

The great book for ESP32forth

version 2.1 - 21 January 2026



Install Forth programming language on ESP32 board. Learn how to
program in Forth language. Use Forth language to drive ESP32 board.
Learn how ESP32 board works.

Author

- Marc PETREMANN petremann@arduino-forth.com

Collaborators

- Bob EDWARDS
- Vaclav POSELT
- Thomas SCHREIN

Contents

Author.....	1
Collaborators.....	1
Introduction.....	12
Translation help.....	12
Why program in Forth language on ESP32?.....	13
Preamble.....	13
Boundaries between language and application.....	13
What is a Forth word?.....	14
A word is a function?.....	14
Forth language compared to C language.....	15
What Forth allows you to do compared to the C language.....	16
But why a stack rather than variables?.....	17
Are you convinced?.....	17
Are there any professional applications written in Forth?.....	17
Discovery of the ESP32 card.....	19
Presentation.....	19
The strong points.....	19
GPIO inputs/outputs on ESP32.....	20
ESP32 Peripherals.....	21
The different ESP32 cards.....	22
Final installation of ESP32forth.....	23
The ESP32 Wroom 32 board.....	23
Connector board.....	24
The ESP32 Wrover board.....	24
Connector board.....	25
The ESP32 S3 board.....	25
Connector board.....	26
Install ESP32Forth.....	27
Download ESP32forth.....	27
Compiling and installing ESP32forth.....	27
Settings for ESP32 WROOM 32.....	29
Start compilation.....	30
Fix Upload Connection Error.....	32
Using numbers with ESP32Forth.....	34
Numbers with the Forth interpreter.....	34
Entering numbers with different numeric bases.....	35
Change of numerical base.....	35
Binary and hexadecimal.....	36
Size of numbers on Forth data stack.....	38
Memory access and logic operations.....	39
A real 32-bit Forth with ESP32Forth.....	42

Values on the data stack.....	42
Values in memory.....	42
Word processing depending on data size or type.....	43
Conclusion.....	44
Comments and debugging.....	46
Write readable Forth code.....	46
Source code indentation.....	47
Comments.....	48
Stack comments.....	48
Meaning of stack parameters in comments.....	49
Word Definition Word Comments.....	49
Textual comments.....	50
Comment at the beginning of the source code.....	50
Diagnostic and tuning tools.....	51
The decompiler.....	51
Memory dump.....	51
Data stack monitor.....	52
Dictionary / Stack / Variables / Constants.....	54
Expand Dictionary.....	54
Dictionary management.....	54
Stacks and reverse Polish notation.....	55
Handling the parameter stack.....	56
The Return Stack and Its Uses.....	56
Memory usage.....	57
Variables.....	57
Constants.....	57
Pseudo-constant values.....	58
Basic tools for memory allocation.....	58
Text colors and display position on terminal.....	60
ANSI coding of terminals.....	60
Text coloring.....	61
Display position.....	62
Local variables with ESP32Forth.....	64
Introduction.....	64
The fake stack comment.....	64
Action on local variables.....	65
Data structures for ESP32forth.....	68
Preamble.....	68
Tables in Forth.....	68
One-dimensional 32-bit data array.....	68
Words for table definitions.....	69
Read and write in a table.....	69
Practical example of managing a virtual screen.....	70
Management of complex structures.....	73
Definition of sprites.....	75

Real numbers with ESP32forth.....	78
The real ones with ESP32forth.....	78
Real number accuracy with ESP32forth.....	78
Real constants and variables.....	79
Arithmetic operators on real numbers.....	79
Mathematical operators on real numbers.....	79
Logical operators on real numbers.....	80
Integer ↔ real transformations.....	80
Displaying numbers and character strings.....	82
Change of numerical base.....	82
Definition of new display formats.....	83
Displaying characters and character strings.....	85
String variables.....	87
Text variable management word code.....	87
Adding character to an alphanumeric variable.....	90
Vocabularies with ESP32forth.....	91
List of vocabularies.....	91
List of vocabulary contents.....	92
Using vocabulary words.....	92
Chaining of vocabularies.....	93
Delayed action words.....	94
Definition and usage of words with defer.....	94
Setting a Forward Reference.....	95
Dependence on the operating context.....	96
A practical case.....	97
Word Creation Words.....	99
Using does>.....	99
Color management example.....	100
Example, writing in pinyin.....	101
Adapt breadboards to ESP32 board.....	103
Breadboards for ESP32.....	103
Build a breadboard suitable for the ESP32 board.....	103
Powering the ESP32 board.....	105
Choice of power source.....	105
Powered by mini-USB connector.....	105
Power supply via 5V pin.....	105
Automatic start of a program.....	107
Install and use the Tera Term terminal on Windows.....	109
Install Tera Term.....	109
Setting up Tera Term.....	110
Using Tera Term.....	112
Compile source code in Forth language.....	113
Access ESP32Forth by TELNET.....	115
Change the DNS name of the ESP32 board.....	115
Connecting to ESP32 boards by their hostname.....	116

Management of source files by blocks.....	119
The blocks.....	119
Open a block file.....	119
Edit the contents of a block.....	120
Compiling block contents.....	121
Practical step-by-step example.....	121
Conclusion.....	122
Editing source files with VISUAL Editor.....	123
Edit a Forth source file.....	123
Editing the Forth code.....	123
Compiling file contents.....	124
The SPIFFS file system.....	125
Access to the SPIFFS file system.....	125
Handling files.....	126
Organize and compile your files on the ESP32 card.....	127
Editing and transmitting source files.....	127
Conclusion.....	127
RECORDFILE and Forth project management.....	129
Save RECORDFILE in autoexec.fs file.....	129
Use modified contents of autoexec.fs file.....	131
Breaking down a project with ESP32forth.....	131
Example project.....	132
The notion of a black box.....	134
Editing and managing source files for ESP32forth.....	136
Text file editors.....	136
Use an IDE.....	137
Storage on GitHub.....	139
Some good practices.....	139
The main.fs file.....	140
Quickly save files to SPIFFS.....	141
Have the word RECORDFILE when starting ESP32forth.....	142
Managing a traffic light with ESP32.....	144
GPIO ports on the ESP32 board.....	144
Mounting the LEDs.....	145
Management of traffic lights.....	146
Conclusion.....	147
Direct access to GPIO registers.....	148
Use of words m! and m@.....	148
The GPIO_OUT_REG register.....	151
Activation and deactivation registers.....	152
Hardware interrupts with ESP32forth.....	156
Interruptions.....	156
Mounting a push button.....	156
Software consolidation of the interrupt.....	157
Further information.....	158

Using the KY-040 rotary encoder.....	160
Encoder Overview.....	160
Mounting the encoder on the breadboard.....	162
Analysis of encoder signals.....	162
Encoder programming.....	163
Testing the encoding.....	164
Increment and decrement a variable with the encoder.....	164
Flashing of an LED per timer.....	166
Getting started with Forth programming.....	166
Flashing by TIMER.....	167
Hardware and software interrupts.....	168
Use the words interval and rerun.....	169
Housekeeper timer.....	171
Preamble.....	171
A solution.....	171
A Forth timer for ESP32Forth.....	172
Management of the light on button.....	173
Conclusion.....	175
Software real-time clock.....	176
The word MS-TICKS.....	176
Managing a software clock.....	176
Measuring the execution time of a Forth word.....	178
Measuring the performance of Forth definitions.....	178
Testing a few loops.....	179
Program a sunshine analyzer.....	180
Preamble.....	180
The miniature solar panel.....	180
Recovery of a miniature solar panel.....	180
Measurement of solar panel voltage.....	181
Solar panel current measurement.....	182
Lowering the solar panel voltage.....	183
Programming the solar analyzer.....	183
Managing activation and deactivation of a device.....	185
Triggered by timer interrupt.....	186
Devices controlled by the sunshine sensor.....	187
Management of N/A (Digital/Analog) outputs.....	189
Digital/analog conversion.....	189
D/A conversion with R2R circuit.....	189
D/A conversion with ESP32.....	189
Possibilities of D/A conversion.....	191
Installing the OLED library for SSD1306.....	192
The I2C interface on ESP32.....	194
Introduction.....	194
Master slave exchange.....	195
Addressing.....	196

Setting GPIO ports for I2C.....	197
I2C bus protocols.....	197
Detecting an I2C device.....	197
The SSD1306 OLED display.....	199
Choosing a display interface.....	199
Online documentation.....	200
Connecting the SSD1306 OLED display.....	200
Memory organization.....	201
Organize the SSD1306 project.....	202
Create the autoexec.fs file.....	202
Creating the main.fs file.....	203
Creating the config.fs file.....	203
Creating the oledTools.fs file.....	204
Test our SSD1306 project.....	204
Use OLED vocabulary.....	205
Initializing the I2C bus for the SSD1306 OLED display.....	206
Initializing the display for SSD1306.....	206
Expand the oled vocabulary.....	209
TEMPVS FVGIT.....	210
Romani non ustulo nulla.....	210
Romani horas et minuta.....	211
Haec omnia integramus pro ESP32forth.....	212
Add the SPI library.....	214
Changes to the ESP32forth.ino file.....	214
First modification.....	214
Second modification.....	215
Third modification.....	215
Fourth modification.....	215
Communicate with the MAX7219 display module.....	216
Locating the SPI port on the ESP32 board.....	217
SPI connectors on the MAX7219 display module.....	217
SPI port software layer.....	218
Installing the HTTP client.....	219
Editing the ESP32forth.ino file.....	219
HTTP Client Testing.....	220
Retrieve the time from a web server.....	222
Transmission and reception of time from a web server.....	222
Understanding transmission by GET to a web server.....	224
Transmission of data to a server by GET.....	224
Parameters in a URL.....	224
Passing multiple parameters.....	224
Managing parameter passing with ESP32forth.....	225
Data transmission to a WEB server.....	227
Data recording on the web server side.....	227
Access protection.....	227

View recorded data.....	228
Add data to transmit.....	229
Conclusion.....	231
Sound synthesis with ESP32Forth.....	232
Simple sound synthesis.....	232
Definition of sound frequency table.....	232
Retrieving the frequency of a musical note.....	233
Managing note duration.....	234
One-note support.....	235
Creating musical notes.....	236
Sound test.....	237
The flight of the bumblebee.....	238
Program in XTENSA assembler.....	240
Preamble.....	240
Compile the XTENSA assembler.....	241
Programming in assembler.....	241
Summary of basic instructions.....	242
A bonus disassembler.....	243
First steps in XTENSA assembler.....	245
Preamble.....	245
Invoking the Xtensa assembler.....	245
Xtensa and the Forth stack.....	245
Writing an Xtensa macro instruction.....	246
Managing the Forth stack in Xtensa assembler.....	248
Efficiency of words written in XTENSA assembler.....	250
Loops and connections in XTENSA assembler.....	252
The LOOP instruction in XTENSA assembler.....	252
Manage a loop in XTENSA assembler with ESP32forth.....	253
Defining loop management macro instructions.....	253
Using the For, and Next macros,.....	254
Connection instructions in XTENSA assembler.....	254
Defining branching macros.....	254
Syntax of branching macro instructions.....	255
Definition and manipulation of registers.....	257
Definition of registers.....	257
Access to register contents.....	258
Handling register bits.....	259
Definition of masks.....	259
Switching from C language to Forth language.....	261
The random number generator.....	263
Characteristic.....	263
Programming procedure.....	264
RND function in XTENSA assembler.....	264
The LoRa transmission system.....	266
Cabling of the REYAX LR890 LoRa transmitter.....	266

The LoRa transmitter for ESP32.....	266
LoRa transmission security.....	267
Review of the REYAX RYLR890 LoRa transmitter.....	270
Required test environment.....	270
Prepare communication with the LoRa transmitter.....	270
Setting up the REYAX RYLR890 LoRa transmitter.....	273
Essential parameters.....	273
ADDRESS Defines the module address.....	274
AT Test LoRa Availability.....	275
BAND Setting the RF frequency.....	275
CPIN Sets the AES128 network password.....	275
CRFOP Selects the output RF power.....	276
FACTORY Sets all current settings to default values.....	276
IPR Sets the UART baud rate.....	276
MODE Selects the working mode.....	277
NETWORKID Selects the network ID.....	277
PARAMETER definition of RF parameters.....	278
Software RESET.....	279
SEND sending data to the designated address.....	279
VER to request firmware version.....	280
Error result codes.....	280
Vectorization of character emissions.....	280
Understanding vectorization in Forth.....	281
Vectorization in ESP32Forth.....	281
Vectorize type to UART2 serial port.....	282
Rewriting a complete listing.....	283
Setting up LoRa transmitters.....	285
Determining the address of LoRa transmitters.....	286
Communication between two REYAX RYLR890 LoRa transmitters.....	287
Transmission from BOSS to SLAV2.....	288
Interfacing a LoRa transmission with ESP32Forth.....	290
The LoRa transmitter side program called BOSS.....	291
Receipt and execution of Forth commands by SLAV1.....	293
Executing a command received by LoRa.....	293
LoRa transmission management loop.....	294
ESP32Forth simple web interface.....	297
ESP-NOW.....	301
Overview.....	301
Security.....	301
Simplicity.....	301
An ESP32 project that uses ESP-NOW.....	302
The prerequisites of our ESP-NOW project.....	302
MAC address management.....	304
How to find the MAC address of an ESP32 board.....	305
Display a MAC address.....	306

Definition of target MAC addresses.....	306
The esp_now_peer_info_t structure.....	307
List of ESP-NOW words.....	307
esp_now_init.....	308
esp_now_register_recv_cb.....	308
esp_now_add_peer.....	309
esp_now_send.....	310
Chronology of actions in transmission and reception.....	310
Prepare the ESP-NOW transmission.....	311
Launch a test sequence.....	313
Application Layer Management.....	315
Application layer control.....	316
The advantages and disadvantages of this control method.....	318
Detailed content of ESP32forth vocabularies.....	320
Version v 7.0.7.21a.....	320
FORTH.....	320
asm.....	321
bluetooth.....	322
editor.....	322
ESP.....	322
espnow.....	322
httpd.....	322
insides.....	322
internals.....	322
interrupts.....	323
ledc.....	323
oled.....	323
recognizers.....	324
registers.....	324
riscv.....	324
rmt.....	324
rtos.....	324
SD.....	324
SD_MMC.....	324
Serial.....	325
sockets.....	325
spi.....	325
SPIFFS.....	325
streams.....	325
structures.....	325
tasks.....	325
telnetd.....	325
timers.....	325
visual.....	326
web-interface.....	326
WiFi.....	326
Wire.....	326

xtensa.....	326
Appendix A – Registers summary.....	328
GPIO registers.....	328
Ressources.....	330
in English.....	330
In french.....	330
GitHub.....	331

Introduction

Since 2019, I manage several websites dedicated to Forth language development for ARDUINO and ESP32 boards, as well as the eForth web version:

- ARDUINO : <https://arduino-forth.com/>
- ESP32 : <https://esp32.arduino-forth.com/>
- eForth web : <https://eforth.arduino-forth.com/>

These sites are available in two languages, French and English. Every year I pay for hosting the main site **arduino-forth.com**.

It will happen sooner or later – and as late as possible – that I will no longer be able to ensure the sustainability of these sites. The consequence will be that the information disseminated by these sites disappears.

This book is the compilation of content from my websites. It is distributed freely from a Github repository. This method of distribution will allow greater sustainability than websites.

Incidentally, if some readers of these pages wish to contribute, they are welcome:

- to suggest chapters ;
- to report errors or suggest changes;
- to help with the translation...

Translation help

Google Translate allows you to translate texts easily, but with errors. So I'm asking for help to correct the translations.

In practice, I provide the chapters already translated in the LibreOffice format. If you want to help with these translations, your role will simply be to correct and return these translations.

Correcting a chapter takes little time, from one to a few hours.

To contact me : petremann@arduino-forth.com


Why program in Forth language on ESP32?

Preamble

I have been programming in Forth since 1983. I stopped programming in Forth in 1996. But I have never stopped monitoring the evolution of this language. I resumed programming in 2019 on ARDUINO with FlashForth then ESP32forth.

I am co-author of several books concerning the Forth langage :

- Introduction au ZX-Forth (ed Eyrolles - 1984 - ASIN:B0014IGOZO)
- Tours de Forth (ed Eyrolles - 1985 - ISBN-13: 978-2212082258)
- Forth pour CP/M et MSDOS (ed Loisitech - 1986)
- TURBO-Forth, manuel d'apprentissage (ed Rem CORP - 1990)
- TURBO-Forth, guide de référence (ed Rem CORP - 1991)



Programming in the Forth language was always a hobby until 1992 when the manager of a company working as a subcontractor for the automobile industry contacted me. They had a concern for software development in C language. They needed to order an industrial automaton.

The two software designers of this company programmed in C language: TURBO-C from Borland to be precise. And their code couldn't be compact and fast enough to fit into the 64 kilobytes of RAM memory. It was 1992 and flash memory type expansions did not exist. In these 64 KB of RAM, we had to fit MS-DOS 3.0 and the application!

For a month, C language developers had been twisting the problem in all directions, even reverse engineering with SOURCER (a disassembler) to eliminate non-essential parts of executable code.

I analyzed the problem that was presented to me. Starting from scratch, I created, alone, in a week, a perfectly operational prototype that met the specifications. For three years, from 1992 to 1995, I created numerous versions of this application which was used on the assembly lines of several automobile manufacturers.

Boundaries between language and application

All programming languages are shared like this :

- an interpreter and executable source code: BASIC, PHP, MySQL, JavaScript, etc... The application is contained in one or more files which will be interpreted whenever necessary. The system must permanently host the interpreter running the source code;
- a compiler and/or assembler: C, Java, etc. Some compilers generate native code, that is to say executable specifically on a system. Others, like Java, compile executable code on a virtual Java machine.

The Forth language is an exception. It integrates :

- an interpreter capable of executing any word in the Forth language
- a compiler capable of extending the dictionary of Forth words

What is a Forth word?

A Forth word designates any dictionary expression composed of ASCII characters and usable in interpretation and/or compilation: words allows you to list all the words in the Forth dictionary.

Certain Forth words can only be used in compilation: **if else then** for example.

With the Forth language, the essential principle is that we do not create an application. In Forth, we extend the dictionary! Each new word you define will be as much a part of the Forth dictionary as all the words pre-defined when Forth starts. Example:

```
: typeToLoRa ( -- )
    0 echo !      \ disable display echo from terminal
    ['] serial2-type is type
;
: typeToTerm ( -- )
    ['] default-type is type
    -1 echo !      \ enable display echo from terminal
;
```

We create two new words: **typeToLoRa** and **typeToTerm** which will complete the dictionary of pre-defined words.

A word is a function?

Yes and no. In fact, a word can be a constant, a variable, a function... Here, in our example, the following sequence :

```
: typeToLoRa ... code... ;
```

would have its equivalent in C langage :

```
void typeToLoRa() { ...code... }
```

In Forth language, there is no limit between language and application.

In Forth, as in C language, you can use any word already defined in the definition of a new word.

Yes, but then why Forth rather than C?

I was expecting this question.

In C language, a function can only be accessed through the main function `main()`. If this function integrates several additional functions, it becomes difficult to find a parameter error in the event of a malfunction of the program.

On the contrary, with Forth it is possible to execute - via the interpreter - any word pre-defined or defined by you, without having to go through the main word of the program.

The Forth interpreter is immediately accessible on the ESP32 card via a terminal type program and a USB link between the ESP32 card and the PC.

The compilation of programs written in Forth language is carried out in the ESP32 card and not on the PC. There is no upload. Example:

```
: >gray ( n -- n' )
    dup 2/ xor      \ n' = n xor ( 1 time right shift logic )
;

```

This definition is transmitted by copy/paste into the terminal. The Forth interpreter/compiler will parse the stream and compile the new word `>gray`.

In the definition of `>gray`, we see the sequence `dup 2/ xor`. To test this sequence, simply type it in the terminal. To execute `>gray`, simply type this word in the terminal, preceded by the number to transform.

Forth language compared to C language

This is my least favorite part. I don't like to compare the Forth language to the C language. But as almost all developers use the C language, I'm going to try the exercise.

Here is a test with `if()` in C language:

```
if(j > 13){          // If all bits are received
    rc5_ok = 1;        // Decoding process is OK
    detachInterrupt(0); // Disable external interrupt (INT0)
    return;
}
```

Test with if in Forth language (code snippet) :

```
var-j @ 13 >          \ If all bits are received
if
    1 rc5_ok ! \ Decoding process is OK
```

```

di          \ Disable external interrupt (INT0)
exit
then

```

Here is the initialization of registers in C langage :

```

void setup() {
    // Timer1 module configuration
    TCCR1A = 0;
    TCCR1B = 0;          // Disable Timer1 module
    TCNT1  = 0;          // Set Timer1 preload value to 0 (reset)
    TIMSK1 = 1;          // enable Timer1 overflow interrupt
}

```

The same definition in Forth langage :

```

: setup
    \ Timer1 module configuration
    0 TCCR1A !
    0 TCCR1B !      \ Disable Timer1 module
    0 TCNT1 !       \ Set Timer1 preload value to 0 (reset)
    1 TIMSK1 !       \ enable Timer1 overflow interrupt
;

```

What Forth allows you to do compared to the C language

We understand that Forth immediately gives access to all the words in the dictionary, but not only that. Via the interpreter, we also access the entire memory of the ESP32 card. Connect to the ESP32 board that has ESP32forth installed, then simply type :

```
hex here 100 dump
```

You should find this on the terminal screen :

3FFEE964	DF DF 29 27 6F 59 2B 42 FA CF 9B 84
3FFEE970	39 4E 35 F7 78 FB D2 2C A0 AD 5A AF 7C 14 E3 52
3FFEE980	77 0C 67 CE 53 DE E9 9F 9A 49 AB F7 BC 64 AE E6
3FFEE990	3A DF 1C BB FE B7 C2 73 18 A6 A5 3F A4 68 B5 69
3FFEE9A0	F9 54 68 D9 4D 7C 96 4D 66 9A 02 BF 33 46 46 45
3FFEE9B0	45 39 33 33 2F 0D 08 18 BF 95 AF 87 AC D0 C7 5D
3FFEE9C0	F2 99 B6 43 DF 19 C9 74 10 BD 8C AE 5A 7F 13 F1
3FFEE9D0	9E 00 3D 6F 7F 74 2A 2B 52 2D F4 01 2D 7D B5 1C
3FFEE9E0	4A 88 88 B5 2D BE B1 38 57 79 B2 66 11 2D A1 76
3FFEE9F0	F6 68 1F 71 37 9E C1 82 43 A6 A4 9A 57 5D AC 9A
3FFEEA00	4C AD 03 F1 F8 AF 2E 1A B4 67 9C 71 25 98 E1 A0
3FFEEA10	E6 29 EE 2D EF 6F C7 06 10 E0 33 4A E1 57 58 60
3FFEEA20	08 74 C6 70 BD 70 FE 01 5D 9D 00 9E F7 B7 E0 CA
3FFEEA30	72 6E 49 16 0E 7C 3F 23 11 8D 66 55 EC F6 18 01
3FFEEA40	20 E7 48 63 D1 FB 56 77 3E 9A 53 7D B6 A7 A5 AB

3FFEEA50	EA 65 F8 21 3D BA 54 10 06 16 E6 9E 23 CA 87 25
3FFEEA60	E7 D7 C4 45

This corresponds to the contents of flash memory.

And the C language couldn't do that?

Yes, but not as simple and interactive as in Forth language.

But why a stack rather than variables?

The stack is a mechanism implemented on almost all microcontrollers and microprocessors. Even the C language leverages a stack, but you don't have access to it.

Only the Forth language gives full access to the data stack. For example, to make an addition, we stack two values, we execute the addition, we display the result: **2 5 + .** displays 7.

It's a little destabilizing, but when you understand the mechanism of the data stack, you greatly appreciate its formidable efficiency.

The data stack allows data to be passed between Forth words much more quickly than by processing variables as in C language or any other language using variables.

Are you convinced?

Personally, I doubt that this single chapter will irremediably convert you to programming in the Forth language. When trying to master ESP32 cards, you have two options :

- program in C language and use the numerous libraries available. But you will remain locked into the capabilities of these libraries. Adapting codes to C language requires real knowledge of programming in C language and mastering the architecture of ESP32 cards. Developing complex programs will always be a problem.
- try the Forth adventure and explore a new and exciting world. Of course, it won't be easy. You will need to understand the architecture of ESP32 cards, the registers, the register flags in depth. In return, you will have access to programming perfectly suited to your projects.

Are there any professional applications written in Forth?

Oh yes! Starting with the HUBBLE space telescope, certain components of which were written in Forth language.

The German TGV ICE (Intercity Express) uses RTX2000 processors to control motors via power semiconductors. The machine language of the RTX2000 processor is the Forth language.



This same RTX2000 processor was used for the Philae probe which attempted to land on a comet.

The choice of the Forth language for professional applications turns out to be interesting if we consider each word as a black box. Each word must be simple, therefore have a fairly short definition and depend on few parameters.

During the debugging phase, it becomes easy to test all the possible values processed by this word. Once made perfectly reliable, this word becomes a black box, that is to say a function in which we have absolute confidence in its proper functioning. From word to word, it is easier to make a complex program reliable in Forth than in any other programming language.

But if we lack rigor, if we build gas plants, it is also very easy to get an application that works poorly, or even to completely crash Forth!

Good programming.

Discovery of the ESP32 card

Presentation

The ESP32 board is not an ARDUINO board. However, development tools leverage certain elements of the ARDUINO eco-system, such as the ARDUINO IDE.

The strong points

In terms of the number of ports available, the ESP32 card is located between an ARDUINO




Figure 1: the base board has 38 connectors

NANO and ARDUINO UNO. The basic model has 38 connectors:

ESP32 devices include :

- 18 analog-to-digital converter (ADC) channels
- 3 SPI interfaces
- 3 UART interfaces
- 2 I2C interfaces
- 16 PWM output channels
- 2 digital-to-analog converters (DAC)
- 2 I2S interfaces
- 10 capacitive sensing GPIOs

The ADC (analog-to-digital converter) and DAC (digital-to-analog converter) functionality are assigned to specific static pins. However, you can decide which pins are UART, I2C,

SPI, PWM, etc. You just need to assign them in the code. This is possible thanks to the multiplexing function of the ESP32 chip.

Most connectors have multiple uses.

But what sets the ESP32 board apart is that it is equipped as standard with WiFi and Bluetooth support, something that ARDUINO boards only offer in the form of extensions.

GPIO inputs/outputs on ESP32

Here, in photo, the ESP32 card from which we will explain the role of the different GPIO inputs/outputs.

The position and number of GPIO I/Os may change depending on the card brand. If this is the case, only the indications appearing on the physical map are authentic. Pictured, bottom row, left to right: CI K, SD0, SD1, G15, G2, G0, G4, G16, ..., G22, G23, GND.




Figure 2: GPIO I/O position

In this diagram, we see that the bottom row begins with 3V3 while in the photo, this I/O is at the end of the top row. It is therefore very important not to rely on the diagram and instead to double check the correct connection of the peripherals and components on the physical ESP32 card.

Development boards based on an ESP32 generally have 33 pins apart from those for the power supply. Some GPIO pins have somewhat particular functions :

GPIO	Possible usage
6	SCK/CLK
7	SCK/CLK
8	SDO/SD0
9	SDI/SD1
10	SHD/SD2
11	CSC/CMD

If your ESP32 card has I/O GPIO6, GPIO7, GPIO8, GPIO9, GPIO10, GPIO11, you should definitely not use them because they are connected to the flash memory of the ESP32. If you use them the ESP32 will not work.

GPIO1(TX0) and GPIO3(RX0) I/O are used to communicate with the computer in UART via USB port. If you use them, you will no longer be able to communicate with the card.

GPIO36(VP), GPIO39(VN), GPIO34, GPIO35 I/O can be used as input only. They also do not have built-in internal pullup and pulldown resistors.

The EN terminal allows you to control the status of the ESP32 via an external wire. It is connected to the EN button on the card. When the ESP32 is turned on, it is at 3.3V. If we connect this pin to ground, the ESP32 is turned off. You can use it when the ESP32 is in a box and you want to be able to turn it on/off with a switch.

ESP32 Peripherals

To interact with modules, sensors or electronic circuits, the ESP32, like any micro-controller, has a multitude of peripherals. There are more of them than on a classic Arduino board.

ESP32 has the following peripherals :

- 3 UART interface
- 2 I2C interfaces
- 3 SPI interfaces
- 16 PWM outputs
- 10 capacitive sensors
- 18 analog inputs (ADC)
- 2 DAC outputs

Some peripherals are already used by ESP32 during its basic operation. There are therefore fewer possible interfaces for each device.

The different ESP32 cards




Figure 3: un grand choix de cartes ESP32 sur AMAZON

If you go to an online sales site to order an ESP32 card, you may end up with a very large choice of cards.

Several questions therefore arise to guide the choice:

- which board can host ESP32forth?
- Which cards are best suited to my projects?
- what is my budget for a given project?

If the price of an ordinary ESP32 card remains affordable, certain variants can see their prices soar. If your goal is to first carry out small experiments, start with a simple ESP32 board. To properly experiment, you will need:

- breadboards, take at least 10. Allow two breadboards per ESP32 card;
- flexible dupont type connectors;
- LEDs, resistors, etc.
- peripherals: OLED display, LCD, relays, synchronous or ordinary motors, servo motors, etc.
- USB cable connecting the PC and the ESP32 card. A USB hub is recommended. In the event of accidental current injection into the USB port, it will be the hub's USB port that will be damaged before the PC's USB port;

For around fifty euros (or US\$), there are ready-to-use kits, including an ESP32 card and peripherals and components.

No kit is complete. If you are carrying out experiments, the ideal is to order a kit, then several ESP32 cards (at least 4) a series of test plates, flat straps, a battery power supply, etc....




Figure 4: kit ESP32

Before embarking on ambitious projects, such as remote control via 3G/4G/5G network, video analysis, etc. start with simple experiments, in C or Forth.

Final installation of ESP32forth

NO ! The installation of ESP32forth on an ESP32 card **is not permanent** ! If you have installed ESP32Forth on one or more ESP32 boards, you can easily download the binary code from any C source, after compiling the contents of an **ino** extension file, to your ESP32 boards and abandon Forth programming .

But, at the risk of advertising ESP32forth, many “makers” have made the definitive choice to program in Forth language. Just one example, **0033mer**’s YouTube channel:

<https://www.youtube.com/@0033mer>

He is one of the most prolific Forth contributors on Youtube. Even though he uses ESP32 very little, the majority of his contributions use the Forth language.

Programming in Forth language requires intellectual effort. This effort is not in vain, because it leads to certain good practices that can be used in other programming languages.

Forth is the only programming language that can be installed on an electronic card and which integrates an interpreter, a compiler, an assembler, a SPIFFS file system, all with significant development space.

The ESP32 card is one of the very rare cards that also has serial communication capabilities (UART0 port via the USB connector), via WiFi or Bluetooth. The majority of cards also have numerous versatile GPIO ports: logic, analog, PWM, UART, SPI, I2C input and output, etc. And all this with one or two processors at nearly 160Mhz, or 10x faster than on an ordinary ARDUINO card. .

And finally, most of the C libraries for ARDUINO can be used on ESP32. Some are accessible from ESP32forth.

The ESP32 Wroom 32 board

ESP32 Wroom is the latest addition to the family of ESP cards from Espressif. This is a particularly fashionable range of development boards, because their low price, low consumption and small size make them an ideal product for carrying out small IoT projects.

- ESP-WROOM-32 processor
- WLAN 802.11 b/g/n
- Bluetooth 4.2 / BLE



Figure 5: carte ESP32 Wroom 32

- Tensilica L108 32-bit 160 MHz processor
 - 512 KB of SRAM and 16 MB of flash memory
 - 32 digital I/O pins ($3.3V$)
 - 6 analog-digital pins
 - 3x UART, 2x SPI, 2x I2C
 - USB to UART CP2102 interface

If you compile ESP32forth for this board, here are the parameters to consider on ARDUINO IDE: TOOLS → BOARD → ESP32 → ESP32 Dev Module:

- **Board:** ESP32 Dev Module

Partition Scheme: No OTA (2M APP, 2M SPIFFS) ← Non-default

Upload Speed: 921600

CPU Frequency: 240MHz

Flash Frequency: 80MHz


Flash Mode: QIO

Flash Size: 4MB (32Mb)

Core Debug Level: None

PSRAM: Disable

Connector board



The ESP32 Wrover board

Espressif Systems ESP32-WROVER MCU Modules are powerful and generic Wi-Fi/BT/BLE MCU modules that target a wide range of applications.




Figure 6: carte ESP32 Wrover


These modules target applications ranging from low-power sensor networks to the most demanding tasks, such as voice coding, music streaming, and MP3 decoding.

The ESP32-WROVER module uses a PCB antenna while the ESP32-WROVER-I uses an IPEX antenna. These modules have an external 4 MB SPI Flash, an external 4 MB PSRAM and a 32 Mbit SPI PSRAM.

If you compile ESP32forth for this board, here are the parameters to consider on ARDUINO IDE: TOOLS → BOARD → ESP32 → ESP32 Dev Module:

- **Board:** ESP32 Dev Module
- Partition Scheme:** No OTA (2M APP, 2M SPIFFS) ← Non-default
- Upload Speed:** 921600
- CPU Frequency:** 240MHz
- Flash Frequency:** 80MHz
- Flash Mode:** QIO
- Flash Size:** 4MB (32Mb)
- Core Debug Level:** None
- PSRAM:** Enabled

Connector board



The ESP32 S3 board

ESP32-S3 is a development board based on an Espressif ESP32-S3-WROOM-2 microcontroller with WiFi and Bluetooth Low Energy interfaces.

- Xtensa LX7 dual-core 32-bit microprocessor
- PSRAM memory: 8 MB
- SRAM memory: 512 KB
- ROM memory: 384 KB
- SRAM memory (RTC): 16 KB SPI



Figure 7: ESP32 S3 board

FLASH memory: 32 MB


WiFi interface: 802.11 b/g/n 2.4 GHz

BLE 5 Mesh interface

If you compile ESP32forth for this board, here are the parameters to consider on ARDUINO IDE: TOOLS → BOARD → ESP32 → ESP32 Dev Module:

- **Board:** ESP32S2 Dev Module
 - **Partition Scheme:** No OTA (2M APP, 2M SPIFFS) ← Non-default
 - **Upload Speed:** 921600
 - **USB CDC On Boot:** Disabled
 - **USB Firmware MSC On Boot:** Disabled
 - **USB DFU On Boot:** Disabled
 - **Upload Mode:** UART0
 - **CPU Frequency:** 240MHz
 - **Flash Frequency:** 80MHz
 - **Flash Mode :** QIO
 - **Flash Size:** 4MB (32Mb)
 - **Core Debug Level:** None
 - **PSRAM:** Enabled

Connector board



Install ESP32Forth

Download ESP32forth

The first step is to get the source code, in C language, of ESP32forth. Preferably use the most recent version:

<https://esp32forth.appspot.com/ESP32forth.html>

Contents of the downloaded file:

```
ESP32forth-7.0.7.2x
  ESP32forth
    readme.txt
    esp32forth.ino
    optional
      SPI-flash.h
      serial-blueooth.h
      ... etc...
```

Compiling and installing ESP32forth

Unzip the **esp32forth.ino** file into a working directory. The **optional** directory contains files for extending ESP32forth. For our first build and upload of ESP32forth, these files are not needed.

To compile ESP32forth you need to have ARDUINO IDE already installed on your computer:

<https://docs.arduino.cc/software/ide-v2>

Once ARDUINO IDE is installed, launch it. ARDUINO IDE is open. Click on *file* and select *Preferences* :




Figure 8: setting preferences

In the window that appears, go to the input box marked *Additional boards manager URLs* : and enter this line:

```
https://dl.espressif.com/dl/package\_esp32\_index.json
```




Figure 9: URL entry Additional Cards Management

Then click on *Tools* and select *Board* .:




Figure 10: ESP32 card selection

This selection should offer you the installation of packages for ESP32. Accept this installation.

You should then be able to access the ESP32 card selection.

ESP32 Dev Module Board Selection :




Figure 11: ESP32 Dev Module card selection

Settings for ESP32 WROOM 32

Here are the other settings needed before compiling ESP32forth. Access the settings by clicking on *Tools* again:

```
-- TOOLS----+-- BOARD      ----+-- ESP32 -----+-- ESP32 Dev Module
      +-- Port: -----+-- COMx
      |
```

```

+-- CPU Frequency -----+-- 240 Mhz
+-- Core Debug Level -----+-- None
+-- Erase All Flash...-----+-- Disabled
+-- Events Run On -----+-- Core 1
+-- Flash Frequency -----+-- 80 Mhz
+-- Flash Mode -----+-- QIO
+-- Flash Size -----+-- 4MB
+-- JTAG Adapter -----+-- FTDI Adapter
+-- Arduino Runs on -----+-- Core 1
+-- PSRAM -----+-- Disabled
+-- Partition Scheme -----+-- Default 4MB with SPIFFS
+-- Upload Speed -----+-- 921600

```

Start compilation

All that remains is to compile ESP32forth. Load the source code using *File* and *Open*.

Assume that your ESP32 board is connected to a USB port. Start the compilation by clicking on *Sketch* and selecting *Upload*:




Figure 12: upload start

If everything goes well, you should transfer the binary code automatically to the ESP32 board. If the compilation goes without errors, but there is a transfer error, recompile the **esp32forth.ino file**. At the time of the transfer, press the button marked **BOOT** on the ESP32 board. This should make the board available for the transfer of the ESP32forth binary code.

Installation and configuration of ARDUINO IDE in video:


- Windows: <https://www.youtube.com/watch?v=2AZQfieHv9g>

- Linux: https://www.youtube.com/watch?v=JeD3nz0_nc

Fix Upload Connection Error

Learn how to fix the fatal error that occurred: "Failed to connect to ESP32: Timed out waiting for packet header" when trying to upload a new code to your ESP32 Card once and for all.

Some ESP32 development boards (read Best ESP32 Boards) do not enter flash/upload mode automatically when downloading new code.




The screenshot shows the Arduino IDE interface with a sketch named "Blink" open. The serial monitor window displays the following error message:


```
it is attached to digital pin 13, on MKR1000 on pin 6, LED BUILTIN is set to
A fatal error occurred: Failed to connect to ESP32: Timed out waiting for packet header
python /home/enfin/.arduino15/packages/esp32/tools/esptool_py/3.0.0/esptool.py --chip esp32 -e
esptool.py v3.0-dev
python /home/enfin/.arduino15/packages/esp32/hardware/esp32/1.0.6/tools/gen_esp32part.py -q /
/home/enfin/.arduino15/packages/esp32/tools/xtensa-esp32-elf-gcc/1.22.0-97-gc752ad5-5.2.0/bin
Le croquis utilise 198842 octets (15%) de l'espace de stockage de programmes. Le maximum est
Les variables globales utilisent 13248 octets (4%) de mémoire dynamique, ce qui laisse 314432
python /home/enfin/.arduino15/packages/esp32/tools/esptool_py/3.0.0/esptool.py --chip esp32 -
esptool.py v3.0-dev
Serial port /dev/ttyUSB0
Connecting.....A fatal error occurred: Failed to connect to ESP32: Timed out waiting for packet header
A fatal error occurred: Failed to connect to ESP32: Timed out waiting for packet header

abled, Default 4MB with spiffs (1.2MB APP/1.5MB SPIFFS), 240MHz (WIFI/BT), QIO, 80MHz, 4MB (32Mb), 921600, None sur /dev/ttyUSB0
```

This means that when you try to upload a new sketch to your ESP32 board, ARDUINO IDE fails to connect to your board and you get the following error message:

To make the ESP32 board automatically switch to flash/download mode, we can connect a 10uF electrolytic capacitor between the EN and GND pin:





This manipulation is only necessary if you are in the uploading phase of ESP32forth from ARDUINO IDE. Once ESP32forth is installed on the ESP32 board, the use of this capacitor is no longer necessary.

Using numbers with ESP32Forth

We started ESP32Forth without problem. We will now delve into some manipulations with numbers to understand how to master the microcontroller in Forth language.

Like many books, we could start with a trivial example program, flashing LEDs for example. Like this for example:

```
define LEDs GPIOs
25 constant ledRED
26 constant ledYELLOW
27 constant ledGREEN
    \ define masks for red yellow and green LEDs
1 ledRED      defMASK: mLED_RED
1 ledYELLOW   defMASK: mLED_YELLOW
1 ledGREEN    defMASK: mLED_GREEN

\ initialisation GPIO G25 G26 and G27 in output mode
: GPIO.init ( -- )
    1 mLED_RED      GPIO_ENABLE_REG regSet
    1 mLED_YELLOW  GPIO_ENABLE_REG regSet
    1 mLED_GREEN   GPIO_ENABLE_REG regSet
;
\ define a ON and OFF sequence
: GPIO.on.off.sequence { position mask delay -- }
    1 position mask GPIO_OUT_W1TS_REG regSet
    delay ms
    1 position mask GPIO_OUT_W1TC_REG regSet ;
```

This code, apparently simple, already requires a knowledge base, such as the notion of memory address, register, binary masks, hexadecimal numbers.

We will therefore start by addressing these basic notions by inviting you to carry out simple manipulations.

Numbers with the Forth interpreter

When ESP32Forth starts, the TERA TERM terminal window (or any other terminal program of your choice) should indicate that ESP32Forth is available. Press the *ENTER* key on the keyboard once or twice . ESP32Forth responds with confirmation of successful execution **ok** . .

We are going to test the entry of two numbers, here **25** and **33** . Type these numbers, then *ENTER* on the keyboard. ESP32Forth always responds with **ok** . . You have just stacked two numbers on the ESP32Forth language stack. Now enter **+** . then press the *ENTER* key . ESP32Forth displays the result:

```
rst:0x1 (POWERON_RESET),boot:0x13 (SPI_FAST_FLASH_BOOT)
configip: 0, SPTMP:0xee
clk_drv:0x00_n_drv:0x00_d_drv:0x00,cs0_drv:0x00,hd_drv:0x00,wp_drv:0x00
mode:DIO, clock div:1
load:0x3fff0030, len:1344
load:0x40078000, len:13836
load:0x40080400, len:3608
entry 0x4008085f0
[...]
L:\151D\1\esp32-hal-cpu.c:2441 setOpFrequencyMhz(): PLL: 480 / 2 = 240
ESP32-D0WDQ6 rev: 0.7.15 rev: 0x5440fc6015454ab3ab
ESP32-D0WDQ6 240 MHz 2 cores 4194304 bytes flash
System Heap: 159972 free + 300288 used = 460260 total (34% free)
86084 bytes max contiguous
3 x Forth stacks: 2048 bytes each
Forth dictionary: 70504 free + 39828 used = 104332 total (67% free)
3 x Forth stacks: 2048 bytes each
ok
--> 25 33
ok
25 33 --> +
58 ok
-->
```

This operation was processed by the Forth interpreter.

ESP32Forth, like all versions of the Forth language, has two states:

- **interpreter** : the state you have just tested by performing a simple sum of two numbers;
- **compiler** : a state which allows new words to be defined. This aspect will be explored further later.

Entering numbers with different numeric bases

In order to fully assimilate the explanations, you are invited to test all the examples via the TERA TERM terminal window.

Numbers can be entered naturally. In decimal, it will ALWAYS be a sequence of numbers, example:

```
-1234 5678 + .
```

The result of this example will show **4444**. Forth numbers and words must be separated by at least one *space character*. The example works perfectly if you type a number or word per line:

```
-1234  
5678  
+  
.
```

Numbers can be prefixed if you want to enter values other than in decimal form:

- **\$** sign to indicate that the number is a hexadecimal value;

Example :

```
255 . \ display 255  
$ff . \ display 255
```

The purpose of these prefixes is to avoid any error of interpretation in the case of similar values:

```
$0305  
0305
```

are not **equal numbers** if the hexadecimal number base is not explicitly defined !

Change of numerical base

ESP32Forth has words allowing you to change the numerical base:

- **hex** to select the hexadecimal numeric base;
- **binary** to select the binary number base;
- **decimal** to select the decimal numeric base.

Any number entered in a numeric base must respect the syntax of numbers in this base:

```
3E7F
```

will cause an error if you are in decimal base.

```
hex 3e7f
```

will work perfectly in hexadecimal base. The new numerical base remains valid as long as another numerical base is not selected:

```
hex  
$0305  
0305
```

are equal numbers !

Once a number is dropped onto the data stack in a numeric base, its value no longer changes. For example, if you drop the value **\$ff** on the data stack, this value which is **255** in decimal, or **11111111** in binary, will not change if we return to decimal:

```
hex ff decimal . \ display: 255
```

At the risk of insisting, **255** in decimal is **the same value** as **\$ff** in hexadecimal!

In the example given at the start of the chapter, we define a constant in hexadecimal:

```
25 constant ledRED
```

If we type:

```
hex ledRED .
```

This will display the contents of this constant in hexadecimal form. The change of base has **no consequences** on the final operation of the Forth program.

Binary and hexadecimal

The modern binary number system, the basis of binary code, was invented by Gottfried Leibniz in 1689 and appears in his article Explanation of Binary Arithmetic in 1703.

In his article, LEIBNITZ uses only the characters **0** and **1** to describe all numbers:

```
: bin0to15 ( -- )  
    binary  
    $10 0 do  
        cr i .  
    loop  
    cr decimal ;  
bin0to15 \ display:  
0  
1  
10  
11  
100  
101
```

```
110  
111  
1000  
1001  
1010  
1011  
1100  
1101  
1110  
1111
```

Is it necessary to understand binary coding? I will say yes and no. **Not** for everyday uses. **Yes** to understand the programming of microcontrollers and mastery of logical operators.

It was Georges Boole who formally described logic. His work was forgotten until the appearance of the first computers. It was Claude Shannon who realized that this algebra could be applied in the design and analysis of electrical circuits.

Boolean algebra deals exclusively with **0** and **1**.

The fundamental components of all our computers and digital memories use binary coding and Boolean algebra.

The smallest unit of storage is the byte. It is a space made up of 8 bits. A bit can only have two states: **0** or **1**. The smallest value that can be stored in a byte is **00000000**, with the largest being **11111111**. If we cut a byte in two, we will have:

- four low-order bits, which can take the values **0000** to **1111** ;
- four most significant bits which can take one of these same values.

If we number all the combinations between 0000 and 1111, starting from 0, we arrive at 15:

```
: bin0to15 ( -- )  
    binary  
    $10 0 do  
        cr i .  
        i hex . binary  
    loop  
    cr decimal ;  
bin0to15 \ display:  
0 0  
1 1  
10 2  
11 3  
100 4  
101 5  
110 6  
111 7
```

```
1000 8
1001 9
1010 A
1011 B
1100 C
1101 D
1110 E
1111 F
```

In the right part of each line, we display the same value as in the left part, but in hexadecimal: **1101** and **D** are the same values!

Hexadecimal representation was chosen to represent numbers in computing for practical reasons. For the high or low order part of a byte, on 4 bits, the only combinations of hexadecimal representation will be between **0** and **F**. Here, the letters A to F **are hexadecimal numbers** !

```
$3E      \ is more readable as 00111110
```

Hexadecimal representation therefore offers the advantage of representing the content of a byte in a fixed format, from **00** to **FF** . In decimal, 0 to 255 should have been used.

Size of numbers on Forth data stack

ESP32forth uses a data stack of 32 bits of memory size, or 4 bytes (8 bits x 4 = 32 bits). The smallest hexadecimal value that can be stacked on the Forth stack will be **00000000** , the largest will be **FFFFFFFF** . Any attempt to stack a larger value results in clipping of that value:

```
hex
abcdefabcdefabcdef .      \ display:      EFABCDE
```

Let's stack the largest possible value in 32-bit (4 byte) hexadecimal format:

```
decimal
$ffffffff .      \ display:      -1
```

I see you surprised, but this result is **normal** ! Word **.** Shows the value that is at the top of the data stack in its signed form. To display the same unsigned value, you must use the word **u.** :

```
$ffffffff u.      \ display:      4294967295
```

This is because of the 32 bits used by Forth to represent an integer, the most significant bit is used as the sign:

- if the most significant bit is **0** , the number is positive;
- if the most significant bit is **1** , the number is negative.

So, if you followed correctly, our decimal values 1 and -1 are represented on the stack, in binary format in this form:

And this is where we will call on our mathematician, Mr LEIBNITZ, to add these two numbers in binary. If we do like in school, starting from the right, you will simply have to respect this rule: $1 + 1 = 10$ in binary. We put the results on a third line:

10

Next step :

At the end, we will have the result:

But since this result has a 33rd most significant bit at 1, knowing that the integer format is strictly limited to 32 bits, the final result is 0 . It is surprising ? Yet this is what every digital clock does. Hide the hours. When you reach 59, add 1, the clock will display 0.

The rules of decimal arithmetic, namely $-1 + 1 = 0$, have been perfectly respected in binary logic!

Memory access and logic operations

The data stack is in no way a data storage space. Its size is also very limited. And the stack is shared by many words. The order of the parameters is fundamental. An error can cause malfunctions. Let's take the case of the word **dump** which displays the contents of a memory space:

```
hex  
0 variable score  
score 10 dump \ display:  
1073670412 00 00 00 00  
1073670416 55 51 54 55 48 51
```

In bold and red we find the four bytes reserved for storing a value in our score variable. Let's store any value in **score** :

decimal

```

1900 score !
hex
score 10 dump \ display :
3FFEE90C
3FFEE910      37 33 36 37 30 33 34 34 79 64 31 30
                                         6C 07 00 00

```

We find the four bytes containing our decimal value **1900**, **0000076C** in hexadecimal. Still surprised? So it is the effect of binary coding and its subtleties that are the cause. In memory, bytes are stored starting with the least significant ones. Upon recovery, the transformation mechanism is transparent:

```
score @ . \ display 1900
```

Let's go back to the code that makes an LED blink. Extract :

```
1 mLED_RED     GPIO_ENABLE_REG regSet
```

This code activates a GPIO output associated with an LED. The word **GPIO_ENABLE_REG** is a constant, the content of which is a mask pointing to this LED. We might as well have written this:

```
1 25 lshift GPIO_ENABLE_REG !
```

Here, the word **lshift** performs a logical shift of 25 bits to the left:

```
\ before shift : %00000000000000000000000000000000 1
\ after shift: %000000 1 00000000000000000000000000000000
```

As a reminder, GPIOs ¹are numbered from 0 to 31. To activate another GPIO, for example GPIO17, we would have executed this:

```
1 17 lshift GPIO_ENABLE_REG !
```

Suppose we want to activate GPIOs 17 and 25 in a single command. We will execute this:

```
1 25 lshift
1 17 lshift or GPIO_ENABLE_REG !
```

What have we done? Here are the details of the operations:

```
\ 1 25 lshift \ %00000010000000000000000000000000
\ 1 17 lshift \ %00000010000000001000000000000000
\ or           \ %00000010000000010000000000000000
```

The word **or** has performed an operation that combines the two offsets into a single binary mask.

Let's return to our **score** variable. We want to isolate the least significant byte. Several solutions are available to us. One solution uses binary masking with the logical operator **and** :

ex

```
score @ . \ display: 76C
```

¹ General Purpose Input/Output = General purpose input-output

```
score @  
$000000FF and . \ display: 6C
```

To isolate the second byte from the right:

```
score @  
$0000FF00 and . \ display: 0700
```

Here we had fun with the content of a variable. To master a microcontroller like the one mounted on the ESP32 card, the mechanisms are hardly different. The hardest part is finding the right registers. This will be the subject of another chapter.

To conclude this chapter, there is still a lot to learn about binary logic and the different possible digital codings. If you have tested the few examples given here, you certainly understand that Forth is an interesting language:

- thanks to its interpreter which allows numerous tests to be carried out interactively without requiring recompilation by uploading code;
- a dictionary most of the words of which are accessible from the interpreter;
- a compiler allowing you to add new words *on the fly*, then test them immediately.

Finally, which doesn't spoil anything, the Forth code, once compiled, is certainly as efficient as its equivalent in C language.

A real 32-bit Forth with ESP32Forth

ESP32Forth is a real 32-bit Forth. What does it mean?

The Forth language favors the manipulation of integer values. These values can be literal values, memory addresses, register contents, etc.

Values on the data stack

When ESP32Forth starts, the Forth interpreter is available. If you enter any number, it will be dropped onto the stack as a 32-bit integer :

```
35
```

If we stack another value, it will also be stacked. The previous value will be pushed down one position :

```
45
```

To add these two values, we use a word, here `+`:

```
+
```

Our two 32-bit integer values are added together and the result is dropped onto the stack.

To display this result, we will use the word `.` :

```
. \ display 80
```

In Forth language, we can concentrate all these operations in a single line :

```
35 45 + . \ display 80
```

Unlike the C language, we do not define an **int8** or **int16** or **int32** type.

With ESP32Forth, an ASCII character will be designated by a 32-bit integer, but whose value will be bounded [32..255]. Example :

```
67 emit \ display C
```

Values in memory

ESP32Forth allows you to define constants and variables. Their content will always be in 32-bit format. But there are situations where that doesn't necessarily suit us. Let's take a simple example, defining a Morse code alphabet. We only need a few bytes :

- one to define number of marks in Morse code character
- one or more bytes for Morse code marks

```
create mA ( -- addr )
2 c,
char . c,    char - c,
```

```

create mB ( -- addr )
  4 c,
  char - c,   char . c,   char . c,   char . c,

create mC ( -- addr )
  4 c,
  char - c,   char . c,   char - c,   char . c,

```

Here we define only 3 words, **mA**, **mB** and **mC**. In each word, several bytes are stored. The question is: how will we retrieve the information in these words ?

The execution of one of these words deposits a 32-bit value, a value which corresponds to the memory address where we stored our Morse code information. It is the word **c@** that we will use to extract the Morse code from each letter :

```

mA c@ . \ display 2
mB c@ . \ display 4

```

The first byte placed on the stack will be used to manage a loop to display the code of a character in Morse code :

```

: .morse ( addr -- )
  dup 1+ swap c@ 0 do
    dup i + c@ emit
  loop
  drop
;
mA .morse \ display .-
mB .morse \ display -...
mC .morse \ display ---.

```

There are plenty of certainly more elegant examples. Here we show a way to manipulate 8-bit values, our bytes, while operating these bytes on a 32-bit stack.

Word processing depending on data size or type

In all other languages, we have a generic word, like **echo** (in PHP) which displays any type of data. Whether integer, real, string, we always use the same word. Example in PHP language:

```

$bread = "Baked bread";
$price = 2.30;
echo $bread . " : " . $price;
// display Baked bread: 2.30

```

For all programmers, this way of doing things is THE STANDARD! So how would Forth do this example in PHP?

```

: bread s" Baked bread" ;

```

```

: price s" 2.30" ;
bread type  s" : " type    price type
\ display   Baked bread: 2.30

```

Here, the word **type** tells us that we have just processed a character string.

Where PHP (or any other language) has a generic function and a parser, Forth compensates with a single data type, but adapted processing methods which inform us about the nature of the data processed.

Here is an absolutely trivial case for Forth, displaying a number of seconds in HH:MM:SS format:

```

: :##           # 6 base !
# decimal
[char] : hold
;
: .hms ( n -- )
<# :## :## # # #> type
;
4225 .hms \ display: 01:10:25

```

I love this example because, to date, **NO OTHER PROGRAMMING LANGUAGE** is capable of achieving this HH:MM:SS conversion so elegantly and concisely.

You have understood, the secret of Forth is in its vocabulary.

Conclusion

Forth has no data typing. All data passes through a data stack. Each position in the stack is **ALWAYS** a 32-bit integer!

That's all there is to know.

Purists of hyper-structured and verbose languages, such as C or Java, will certainly cry heresy. And here, I will allow myself to answer them : why do you need to type your data ?

Because it is in this simplicity that the power of Forth lies : a single stack of data with an untyped format and very simple operations.

And I'm going to show you what many other programming languages can't do, define new definition words :

```

: morse: ( comp: c -- | exec -- )
create
c,
does>
dup 1+ swap c@ 0 do

```

```
    dup i + c@ emit
loop
drop space
;
2 morse: mA      char . c,    char - c,
4 morse: mB      char - c,    char . c,    char . c,    char . c,
4 morse: mC      char - c,    char . c,    char - c,    char . c,
mA mB mC \ display  .. -... -..
```

Here, the word **morse:** has become a definition word, in the same way as constant or variable...

Because Forth is more than a programming language. It is a meta-language, that is to say a language to build your own programming language....

Comments and debugging

There is no IDE² to manage and present code written in Forth language in a structured way. At worst you use an ASCII text editor, at best a real IDE and text files:

- **edit** or **wordpad** on Windows
- **edit** under Linux
- **PsPad** under windows
- **Netbeans** under Windows or Linux...

Here is a code snippet that could be written by a beginner:

```
: cycle.stop -1 +to MAX_LIGHT_TIME MAX_LIGHT_TIME 0 = if  
LOW myLIGHTS pin else 0 rerun then ;
```

This code will be perfectly compiled by ESP32forth. But will it remain understandable in the future if it needs to be modified or reused in another application?

Write readable Forth code

Let's start with the name of the word to be defined, here **cycle.stop**. ESP32forth allows you to write very long word names. The size of the defined words has no influence on the performance of the final application. We therefore have a certain freedom to write these words :

- like object programming in JavaScript: **cycle.stop**
- the Camel wayCoding **cycleStop**
- for programmers wanting very understandable code **cycle-stop-lights**
- programmer who likes concise code: **csl**

There is no rule. The main thing is that you can easily reread your Forth code. However, computer programmers in Forth language have certain habits:

- constants in uppercase characters **MAX_LIGHT_TIME_NORMAL_CYCLE**
- word defining other words **defPin:** , i.e. word followed by a colon;
- address transformation word **>date** , here the address parameter is incremented by a certain value to point to the appropriate data;
- memory storage word **date@** or **date!**
- Data display word **.date**

² Integrated Development Environment = Integrated Development Environment

And what about naming Forth words in a language other than English? Here again, only one rule: **total freedom** ! Be careful though, ESP32forth does not accept names written in alphabets other than the Latin alphabet. However, you can use these alphabets for comments:

```
: .date      \ Плакат сегодняшней даты  
....coded... ;
```

Or

```
: .date      \海報今天的日期  
....coded... ;
```

Source code indentation

Whether the code is two lines, ten lines or more has no effect on the performance of the code once compiled. So, you might as well indent your code in a structured way:

- one line per word of control structure **if else then , begin while repeat...**. For the word if, we can precede it with the logical test that it will process;
- a line by execution of a predefined word, preceded if necessary by the parameters of this word.

Example :

```
60 constant MAX_LIGHT_TIME_NORMAL_CYCLE  
: cycle.stop  
    -1 +to MAX_LIGHT_TIME  
    MAX_LIGHT_TIME 0 =  
    if  
        LOW myLIGHTS pin  
    else  
        0 rerun  
    then  
;
```

If the code processed in a control structure is sparse, the Forth code can be compacted:

```
: cycle.stop  
    -1 +to MAX_LIGHT_TIME  
    MAX_LIGHT_TIME 0 =  
    if        LOW myLIGHTS pin  
    else      0 rerun          then  
;
```

This is often the case with **case of endof endcase** structures ;

```
: socketError ( -- )  
    errno dup  
    case  
        2 of      ." No such file "           endof
```

```

5 of      ." I/O error "
9 of      ." Bad file number "
22 of     ." Invalid argument "
endcase
.quit
;

```

Comments

Like any programming language, the Forth language allows the addition of comments in the source code. Adding comments has no impact on the performance of the application after compiling the source code.

In Forth language, we have two words to delimit comments:

- the word `(` must be followed by at least one space character. This comment is completed by the character `) ;` ;
- the word `\` must be followed by at least one space character. This word is followed by a comment of any size between this word and the end of the line.

The word `(` is widely used for stack comments. Examples:

```

dup  ( n -- nn)
swap ( n1 n2 -- n2 n1)
drop ( n --)
emit ( c -- )

```

Stack comments

As we have just seen, they are marked by `(` and `)`. Their content has no effect on the Forth code during compilation or execution. So we can put anything between `(` and `)`. As for the stack comments, we will remain very concise. The `--` sign symbolizes the action of a Forth word. The indications before `--` correspond to the data placed on the data stack before the execution of the word. The indications after `--` correspond to the data left on the data stack after execution of the word. Examples :

- `words (--)` means that this word does not process any data on the data stack;
- `emit (c --)` means that this word processes data as input and leaves nothing on the data stack ;
- `bl (-- 32)` means that this word does not process any input data and leaves the decimal value 32 on the data stack;

There is no limitation on the amount of data processed before or after execution of the word. As a reminder, the indications between `(` and `)` are only there for information.

Meaning of stack parameters in comments

To begin with, a small but very important clarification is necessary. This is the size of the data on stack. With ESP32Forth, the stack data takes up 4 bytes. So these are integers in 32-bit format. However, some words process data in 8-bit format. So what do we put on the data stack? With ESP32Forth, it will **ALWAYS be 32 BIT DATA !** An example with the **c word!** :

```
create myDelimiter
  0 c,
64 myDelimiter c!  ( c addr -- )
```

Here, the parameter **c** indicates that we stack an integer value in 32-bit format, but whose value will always be included in the interval [0..255].

The standard parameter is always **n** . If there are several integers, we will number them: **n1 n2 n3** , etc.

We could therefore have written the previous example like this :

```
create myDelimiter
  0 c,
64 myDelimiter c!  ( n1 n2 -- )
```

But it is much less explicit than the previous version. Here are some symbols that you will see throughout the source codes:

- **addr** indicates a literal memory address or delivered by a variable;
- **c** indicates an 8-bit value in the interval [0..255]
- **d** indicates a double precision value.
Not used with ESP32Forth which is already in 32-bit format;
- **f1** indicates a Boolean value, 0 or non-zero;
- **n** indicates an integer. 32-bit signed integer for ESP32Forth;
- **str** indicates a character string. Equivalent to **addr len --**
- **u** indicates an unsigned integer

Nothing prevents us from being a little more explicit:

```
: SQUARE ( n -- n-exp2 )
  dup *
;
```

Word Definition Word Comments

Definition words use **create** and **does>** . For these words, it is advisable to write stack comments like this:

```
\ define a command or data stream for SSD1306
```

```

: streamCreate: ( comp: <name> | exec: -- addr len )
    create
        here      \ leave current dictionnary pointer on stack
        0 c,      \ initial lenght data is 0
    does>
        dup 1+ swap c@
        \ send a data array to SSD1306 connected via I2C bus
        sendDatasToSSD1306
;

```

Here, the comment is split into two parts by the character | :

- on the left, the action part when the definition word is executed, prefixed by **comp:**
- on the right the action part of the word that will be defined, prefixed with **exec:**

At the risk of insisting, this is not a standard. These are only recommendations.

Textual comments

They are indicated by the word \ followed by at least one space character and explanatory text:

```

\ store at <WORD> addr length of datas compiled between
\ <WORD> and here
: ;endStream ( addr-var len ---)
    dup 1+ here
    swap -      \ calculate cdata length
    \ store c in first byte of word defined by streamCreate:
    swap c!
;

```

These comments can be written in any alphabet supported by your source code editor:

```

\ 儲存在 <WORD> addr 之間編譯的資料長度
\ <WORD> 和這裡
: ;endStream ( addr-var len ---)
    dup 1+ here
    swap -      \ 計算 cdata 長度
    \ 將 c 儲存在由 StreamCreate 定義的字的第一個位元組中:
    swap c!
;

```

Comment at the beginning of the source code

With intensive programming practice, you quickly find yourself with hundreds or even thousands of source files. To avoid file choice errors, it is strongly recommended to mark the start of each source file with a comment:

```

\ ****
\ Manage commands for OLED SSD1306 128x32 display
\   Filename:      SSD10306commands.fs
\   Date:         21 may 2023

```

```

\ Updated:      21 may 2023
\ File Version: 1.0
\ MCU:          ESP32-WROOM-32
\ Forth:         ESP32forth all versions 7.x++
\ Copyright:    Marc PETREMAN
\ Author:        Marc PETREMAN
\ GNU General Public License
\ ****

```

All this information is at your discretion. They can become very useful when you come back to the contents of a file months or years later.

To conclude, do not hesitate to comment and indent your source files in Forth language.

Diagnostic and tuning tools

The first tool concerns the compilation or interpretation alert:

```

3 5 25 --> : TEST ( ---)
ok
3 5 25 -->      [ HEX ] ASCII A DDUP      \ DDUP don't exist

```

Here, the word **DDUP** does not exist. Any compilation after this error will fail.

The decompiler

In a conventional compiler, the source code is transformed into executable code containing the reference addresses to a library equipping the compiler. To have executable code, you must link the object code. At no time can the programmer have access to the executable code contained in his library with the resources of the compiler alone.

With ESP32Forth, the developer can decompile their definitions. To decompile a word, simply type **see** followed by the word to decompile:

```

: C>F ( °C --- °F) \ Conversion Celsius in Fahrenheit
  9 5 */ 32 +
;
see c>f
\ display:
: C>F
  9 5 */ 32 +
;

```

Many words in ESP32Forth's Forth dictionary can be decompiled.

Decompiling your words allows you to detect possible compilation errors.

Memory dump

Sometimes it is desirable to be able to see the values that are in memory. The word **dump** accepts two parameters: the starting address in memory and the number of bytes to display:

```
create myDATAS 01 c, 02 c, 03 c, 04 c,
```

```

hex
myDATAS 4 dump      \ displays :
3FFEE4EC                      01 02 03 04

```

Data stack monitor

The contents of the data stack can be displayed at any time using the word **.s** . Here is the definition of the word **.DEBUG** which exploits **.s** :

```

variable debugStack

: debugOn ( -- )
    -1 debugStack !
;

: debugOff ( -- )
    0 debugStack !
;

: .DEBUG
    debugStack @
    if
        cr ." STACK: " .s
        key drop
    then
;

```

To use **.DEBUG**, simply insert it in a strategic place in the word to be debugged:

```

\ example of use:
: myTEST
    128 32 do
        i .DEBUG
        emit
    loop
;

```

Here, we will display the contents of the data stack after execution of word **i** in our **do loop** . We activate the focus and run **myTEST** :

```

debugOn
myTest
\ displays:
\ STACK: <1> 32
\ 2
\ STACK: <1> 33
\ 3
\ STACK: <1> 34
\ 4
\ STACK: <1> 35
\ 5
\ STACK: <1> 36
\ 6
\ STACK: <1> 37

```

```
\ 7  
\ STACK: <1> 38
```

When debugging is enabled by **debugOn** , each display of the contents of the datastack pauses our **do loop**. Run **debugOff** so that the **myTEST word** executes normally.

Dictionary / Stack / Variables / Constants

Expand Dictionary

Forth belongs to the class of woven interpretive languages. This means that it can interpret commands typed on the console, as well as compile new subroutines and programs.

The Forth compiler is part of the language and special words are used to create new dictionary entries (i.e. words). The most important are `:` (start a new definition) and `;` (finishes the definition). Let's try this by typing :

```
: *+ * + ;
```

What happened? The action of `:` is to create a new dictionary entry named `*+` and switch from interpretation mode to compilation mode. In compile mode, the interpreter searches for words and, rather than executing them, installs pointers to their code. If the text is a number, instead of pushing it onto the stack, ESP32forth constructs the number in the dictionary space allocated for the new word, following special code that puts the stored number on the stack each time the word is executed. The execution action of `*+` is therefore to sequentially execute the previously defined words `*` and `+`.

Word `;` is special. It is an immediate word and it is always executed, even if the system is in compile mode. Which makes `;` is twofold. First, it installs code that returns control to the next external level of the interpreter, and second, it returns from compilation mode to interpretation mode.

Now let's try this new word :

```
decimal 5 6 7 *+ . \ display 47 ok<#,ram>
```

This example illustrates two main work activities in Forth : adding a new word to the dictionary, and trying it as soon as it has been defined.

Dictionary management

The word `forget` followed by the word to delete will remove all dictionary entries you have made since that word :

```
: test1 ;
: test2 ;
: test3 ;
forget test2 \ delete test2 and test3 in dictionary
```

Stacks and reverse Polish notation

Forth has an explicitly visible stack that is used to pass numbers between words (commands). Using Forth effectively forces you to think in terms of the stack. This can be difficult at first, but as with anything, it gets much easier with practice.

In Forth, The pile is analogous to a pile of cards with numbers written on them. Numbers are always added to the top of the stack and removed from the top of the stack.

ESP32forth integrates two stacks: the parameter stack and the feedback stack, each consisting of a number of cells that can hold 32-bit numbers.

The Forth input line :

```
decimal 2 5 73 -16
```


leaves the parameter stack as it is

Cell	Content	comment
0	-16	(TOS) Top of stack
1	73	(NOS) Next in stack
2	5	
3	2	

We will typically use zero-based relative numbering in Forth data structures such as stacks, arrays, and tables. Note that when a sequence of numbers is entered like this, the rightmost number becomes TOS and the leftmost number is at the bottom of the stack.

Let's continue with this:

```
+ - * .
```



The operations would produce successive stack operations :

After the two lines, the console displays :

```
decimal 2 5 73 -16 \ display: 2 5 73 -16 ok
+ - * . \ display: -104 ok
```

Note that ESP32forth conveniently displays the stack elements when interpreting each line and that the value of **-16** is displayed as a 32-bit unsigned integer. Furthermore, the word

. consumes data value **-104**, leaving the stack empty. If we execute . on the now empty stack, the external interpreter aborts with a stack pointer error STACK UNDERFLOW ERROR.

The programming notation where the operands appear first, followed by the operator(s) is called Reverse Polish Notation (RPN).

Handling the parameter stack

Being a stack-based system, ESP32forth must provide ways to put numbers on the stack, remove them and rearrange their order. We have already seen that we can put numbers on the stack simply by typing them. We can also integrate numbers into the definition of a Forth word.

The word **drop** removes a number from the top of the stack thus putting the next one on top. The word **swap** exchanges the first 2 numbers. **dup** copies the number at the top, pushing all other numbers down. **rot** rotates the first 3 numbers. These actions are

	drop	swap	rot	dup
-16	73	5	2	2
73	5	73	5	2
5	2	2	73	5
2				73

presented below.

The Return Stack and Its Uses

When compiling a new word, ESP32forth establishes links between the calling word and previously defined words that are to be invoked by the execution of the new word. This linking mechanism, at runtime, uses the return stack. The address of the next word to be invoked is placed on the back stack so that when the current word has finished executing, the system knows where to move to the next word. Since words can be nested, there must be a stack of these return addresses.

In addition to serving as a reservoir of return addresses, the user can also store and retrieve from the return stack, but this must be done carefully because the return stack is essential to program execution. If you use the return stack for temporary storage, you must return it to its original state, otherwise you will likely crash the ESP32forth system. Despite the danger, there are times when using return stack as temporary storage can make your code less complex.

To store on the return stack, use `>r` to move the top of the parameter stack to the top of the return stack. To retrieve a value, `r>` moves the top value from the return stack back to the top of the parameter stack. To simply remove a value from the top of the return stack, there is the word `rdrop`. The word `r@` copies the top of the return stack back into the parameter stack.

Memory usage

In ESP32forth, 32-bit numbers are fetched from memory to the stack by the word `@` (fetch) and stored from the top to memory by the word `!` (store). `@` expects an address on the stack and replaces the address with its contents. `!` expects a number and an address to store it. It places the number in the memory location referenced by the address, consuming both parameters in the process.

Unsigned numbers that represent 8-bit (byte) values can be placed in character-sized characters. memory cells using `c@` and `c!`.

```
create testVar
    cell allot
$F7 testVar c!
testVar c@ . \ display 247
```

Variables

A variable is a named location in memory that can store a number, such as the intermediate result of a calculation, off the stack. For example :

```
variable x
```

creates a storage location named `x`, which executes leaving the address of its storage location at the top of the stack :

```
x . \ display address
```

We can then retrieve or store at this address :

```
variable x
3 x !
x @ . \ display: 3
```

Constants

A constant is a number that you would not want to change while a program is running. The result of executing the word associated with a constant is the value of the data remaining on the stack.

```
\ define VSPI pins
19 constant VSPI_MISO
23 constant VSPI_MOSI
```

```

18 constant VSPI_SCLK
05 constant VSPI_CS

\ define SPI frequency port
4000000 constant SPI_FREQ

\ select SPI vocabulary
only FORTH SPI also

\ initialize the SPI port
: init.VSPI ( -- )
    VSPI_CS OUTPUT pinMode
    VSPI_SCLK VSPI_MISO VSPI_MOSI VSPI_CS SPI.begin
    SPI_FREQ SPI.setFrequency
;

```

Pseudo-constant values

A value defined with **value** is a hybrid type of **variable** and **constant**. We set and initialize a value and it is invoked as we would a constant. We can also change a value like we can change a variable.

```

decimal
13 value thirteen
thirteen .      \ display: 13
47 to thirteen
thirteen .      \ display: 47

```

The word **to** also works in word definitions, replacing the value following it with whatever is currently at the top of the stack. You need to be careful that **to** is followed by a value defined by **value** and not something else.

Basic tools for memory allocation

The words **create** and **allot** are the basic tools for reserving memory space and attaching a label to it. For example, the following transcription shows a new dictionary entry **graphic-array** :

```

create graphic-array ( --- addr )
    %00000000 c,
    %00000010 c,
    %00000100 c,
    %00001000 c,
    %00010000 c,
    %00100000 c,
    %01000000 c,
    %10000000 c,

```

When executed, the word **graphic-array** stacks the address of the first entry.


We can now access the memory allocated to **graphic-array** using the fetch and store words explained earlier. To calculate the address of the third byte assigned to **graphic-array** we can write **graphic-array 2 +**, remembering that the indices start at 0.

```
30 graphic-array 2 + c!
graphic-array 2 + c@ .      \ display 30
```

Text colors and display position on terminal

ANSI coding of terminals

If you are using terminal software to communicate with ESP32forth, there is a good chance that this terminal emulates a VT type terminal or equivalent. Here, TeraTerm configured to emulate a VT100 terminal:



These terminals have two interesting features :

- color the page background and the text to display
- position the display cursor

Both of these features are controlled by ESC (escape) sequences. This is how the words **bg** and **fg** are defined in ESP32forth :

```
forth definitions ansi
: fg ( n -- ) esc ." [38;5;" n. ." m" ;
: bg ( n -- ) esc ." [48;5;" n. ." m" ;
: normal   esc ." [0m" ;
: at-xy ( x y -- ) esc ." [" 1+ n. ." ;" 1+ n. ." H" ;
: page     esc ." [2J" esc ." [H" ;
```

The word **normal** overrides the coloring sequences defined by **bg** and **fg**.


The word **page** clears the terminal screen and positions the cursor at the upper left corner of the screen.

Text coloring

Let's see how to color the text first :

```
: testFG ( -- )
page
16 0 do
 16 0 do
    j 16 * i + fg
    ." X"
loop
cr
loop
normal
;
```


Running **testFG** gives this on display :



To test the background colors, we will proceed as follows :

```
: testBG ( -- )
page
16 0 do
 16 0 do
    j 16 * i + bg
    space space
loop
cr
loop
normal
;
```

Running testBG gives this on display :



Display position


The terminal is the simplest solution to communicate with ESP32forth. With ANSI escape sequences it is easy to improve the presentation of data.

```
09 constant red
11 constant yellow
14 constant cyan
15 constant whyte
: box { x0 y0 xn yn color -- }
      color bg
      yn y0 - 1+      \ determine height
      0 do
          x0  y0 i + at-xy
          xn x0 - spaces
      loop
      normal
;

: 3boxes ( -- )
page
2 4 20 6 cyan box
8 6 28 8 red  box
14 8 36 10 yellow box
```

```
0 0 at-xy  
;
```

Running **3boxes** shows this :



You are now equipped to create simple and effective interfaces allowing interaction with Forth definitions compiled by ESP32forth.

Local variables with ESP32Forth

Introduction

The Forth language processes data primarily through the data stack. This very simple mechanism offers unrivaled performance. Conversely, following the flow of data can quickly become complex. Local variables offer an interesting alternative.

The fake stack comment

If you follow the different Forth examples, you will have noticed the stack comments framed by (and) . Example:

```
\ addition two unsigned values, leaves sum and carry on the stack
: um+ ( u1 u2 -- sum carry )
    \ here the definition
;
```

Here, the comment (u1 u2 -- sum carry) has absolutely no action on the rest of the Forth code. This is pure commentary.

When preparing a complex definition, the solution is to use local variables framed by { and } . Example :

```
: 2OVER { a b c d }
    a b c d a b
;
```

We define four local variables a b c and d.

The words { and } are similar to the words (and) but do not have the same effect at all. Codes placed between { and } are local variables. The only constraint: do not use variable names that could be Forth words from the Forth dictionary. We might as well have written our example like this :

```
: 2OVER { varA varB varC varD }
    varA varB varC varD varA varB
;
```

Each variable will take the value of the data stack in the order of their deposit on the data stack. here, 1 goes into varA, 2 into varB, etc.:

```
--> 1 2 3 4
ok
1 2 3 4 --> 2over
ok
1 2 3 4 1 2 -->
```

Our fake stack comment can be completed like this :

```
: 2OVER { varA varB varC varD -- varA varB varC varD varA varB }
```

The characters following `--` have no effect. The only point is to make our fake comment look like a real stack comment.

Action on local variables

Local variables act exactly like pseudo-variables defined by **value**. Example :

```
: 3x+1 { var -- sum }
      var 3 * 1 +
;
```

A le même effet que ceci:

```
0 value var
: 3x+1 ( var -- sum )
  to var
  var 3 * 1 +
;
```

In this example, **var** is defined explicitly by **value**.

We assign a value to a local variable with the word **to** or **+to** to increment the content of a local variable. In this example, we add a local variable **result** initialized to zero in the code of our word:

```
: a+bEXP2 { varA varB -- (a+b)EXP2 }
  0 { result }
  varA varA *      to result
  varB varB *      +to result
  varA varB * 2 * +to result
  result
;
```

Isn't it more readable than this?

```
: a+bEXP2 ( varA varB -- result )
  2dup
  * 2 * >r
  dup *
  swap dup * +
  r> +
;
```

Here is a final example, the definition of the word **um+** which adds two unsigned integers and leaves the sum and the overflow value of this sum on the data stack:

```
\ add two unsigned integers, leaves sum and carry on the stack
```

```

: um+ { u1 u2 -- sum carry }
  0 { sum }
  cell for
    aft
      u1 $100 /mod to u1
      u2 $100 /mod to u2
      +
      cell 1- i - 8 * lshift +to sum
    then
  next
  sum
  u1 u2 + abs
;

```

Here is a more complex example, rewriting **DUMP** using local variables:

```

\ local variables in DUMP:
\ START_ADDR      \ first address for dump
\ END_ADDR        \ last address for dump
\ 0START_ADDR     \ first address for loop in dump
\ LINES            \ number of lines for dump loop
\ myBASE           \ current numerical base
internals
: dump ( start len -- )
  cr cr ." --addr--- "
  ." 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F  -----chars-----"
  2dup + { END_ADDR }          \ store latest address to dump
  swap { START_ADDR }         \ store START address to dump
  START_ADDR 16 / 16 * { 0START_ADDR } \ calc. addr for loop start
  16 / 1+ { LINES }
  base @ { myBASE }          \ save current base
  hex
  \ outer loop
  LINES 0 do
    0START_ADDR i 16 * +      \ calc start address for current line
    cr <# # # # # [char] - hold # # # # #> type
    space space    \ and display address
    \ first inner loop, display bytes
    16 0 do
      \ calculate real address
      0START_ADDR j 16 * i ++
      ca@ <# # # #> type space \ display byte in format: NN
    loop
    space
    \ second inner loop, display chars
    16 0 do

```

```

    \ calculate real address
    0START_ADDR j 16 * i +
    \ display char if code in interval 32-127
    ca@      dup 32 < over 127 > or
    if       drop [char] . emit
    else     emit
    then
    loop
loop
myBASE base !           \ restore current base
cr cr
;
forth

```

The use of local variables greatly simplifies data manipulation on stacks. The code is more readable. Note that it is not necessary to pre-declare these local variables, it is enough to designate them when using them, for example: **base @ { myBASE }**.

WARNING: if you use local variables in a definition, no longer use the words **>r** and **r>**, otherwise you risk disrupting the management of local variables. Just look at the decompilation of this version of **DUMP** to understand the reason for this warning:

```

: dump  cr cr s" --addr--  " type
s" 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F  -----chars-----" type
2dup + >R SWAP >R -4 local@ 16 / 16 * >R 16 / 1+ >R base @ >R
hex -8 local@ 0 (do) -20 local@ R@ 16 * + cr
<# # # # # 45 hold # # # # > type space space
16 0 (do) -28 local@ j 16 * R@ ++ CA@ <# # # # > type space 1 (+loop)
0BRANCH rdrop rdrop space 16 0 (do) -28 local@ j 16 * R@ ++ CA@ DUP 32 < OVER 127 > OR
0BRANCH DROP 46 emit BRANCH emit 1 (+loop) 0BRANCH rdrop rdrop 1 (+loop)
0BRANCH rdrop rdrop -4 local@ base ! cr cr rdrop rdrop rdrop rdrop ;

```

Data structures for ESP32forth

Preamble

ESP32forth is a 32-bit version of the Forth language. Those who have practiced Forth since its beginnings have programmed with 16-bit versions. This data size is determined by the size of the elements deposited on the data stack. To find out the size in bytes of the elements, you must execute the word `cell`. Running this word for ESP32forth :

```
cell . \ display 4
```

The value 4 means that the size of the elements placed on the data stack is 4 bytes, or 4×8 bits = 32 bits.

With a 16-bit Forth version, `cell` will stack the value 2. Likewise, if you use a 64-bit version, `cell` will stack the value 8.

Tables in Forth

Let's start with fairly simple structures : tables. We will only discuss one- or two-dimensional arrays.

One-dimensional 32-bit data array

This is the simplest type of table. To create a table of this type, we use the word `create` followed by the name of the table to create :

```
create temperatures
  34 ,   37 ,   42 ,   36 ,   25 ,   12 ,
temperatures      \ push addr on stack
  0 cell *       \ calculate offset 0
  +
  @ .             \ display 34

temperatures      \ push addr on stack
  1 cell *       \ calculate offset 0
  +
  @ .             \ display 37
```

We can factor the access code to the desired value by defining a word which will calculate this address :

```
: temp@ ( index -- value )
  cell * temperatures + @
;
0 temp@ . \ display 34
```

```
2 temp@ . \ display 42
```

You will notice that for n values stored in this table, here 6 values, the access index must always be in the interval [0..n-1].

Words for table definitions

Here's how to create a word definition of one-dimensional integer arrays :

```
: array ( comp: -- | exec: index -- addr )
    create
    does>
        swap cell * +
    ;
array myTemps
    21 ,    32 ,    45 ,    44 ,    28 ,    12 ,
0 myTemps @ . \ display 21
5 myTemps @ . \ display 12
```

In our example, we store 6 values between 0 and 255. It is easy to create a variant of **array** to manage our data in a more compact way :

```
: arrayC ( comp: -- | exec: index -- addr )
    create
    does>
        +
    ;
arrayC myCTemps
    21 c,    32 c,    45 c,    44 c,    28 c,    12 c,
0 myCTemps c@ . \ display 21
5 myCTemps c@ . \ display 12
```

With this variant, the same values are stored in four times less memory space.

Read and write in a table

It is entirely possible to create an empty array of n elements and write and read values in this array :

```
arrayC myCTemps
    6 allot          \ allocate 6 bytes
    0 myCTemps 6 0 fill \ fill this 6 bytes with value 0
    32 0 myCTemps c! \ store 32 in myCTemps[0]
    25 5 myCTemps c! \ store 25 in myCTemps[5]
    0 myCTemps c@ . \ display 32
```

In our example, the array contains 6 elements. With ESP32forth, there is enough memory space to process much larger arrays, with 1,000 or 10,000 elements for example. It's easy to create multi-dimensional tables. Example of a two-dimensional array :

```

63 constant SCR_WIDTH
16 constant SCR_HEIGHT
create mySCREEN
    SCR_WIDTH SCR_HEIGHT * allot          \ allocate 63 * 16 bytes
    mySCREEN SCR_WIDTH SCR_HEIGHT * bl fill \ fill this memory with 'space'

```

Here, we define a two-dimensional table named **mySCREEN** which will be a virtual screen of 16 rows and 63 columns.

Simply reserve a memory space which is the product of the dimensions X and Y of the table to use. Now let's see how to manage this two-dimensional array :

```

: xySCRaddr { x y -- addr }
    SCR_WIDTH y *
    x + mySCREEN +
;
: SCR@ ( x y -- c )
    xySCRaddr c@
;
: SCR! ( c x y -- )
    xySCRaddr c!
;
char X 15 5 SCR!      \ store char X at col 15 line 5
15 5 SCR@ emit        \ display X


```

Practical example of managing a virtual screen

Before going further in our example of managing a virtual screen, let's see how to modify the character set of the TERA TERM terminal and display it.

Launch TERA TERM terminal :

- in the menu bar, click on *Setup*
- select *Font* and *Font...*
- configure the font below :



Here's how to display the table of available characters :

```
: tableChars ( -- )
  base @ >r  hex
  128 32 do
    16 0 do
      j i + dup . space emit space space
      loop
      cr
    16 +loop
    256 160 do
      16 0 do
        j i + dup . space emit space space
        loop
        cr
      16 +loop
      cr
    r> base !
;
tableChars
```

Here is the result of running **tableChars** :

```
--> tableChars
20   21 ! 22 " 23 # 24 $ 25 % 26 & 27 ' 28 ( 29 ) 2A * 2B + 2C , 2D - 2E . 2F /
30 0 31 1 32 2 33 3 34 4 35 5 36 6 37 7 38 8 39 9 3A : 3B ; 3C < 3D = 3E > 3F ?
40 @ 41 A 42 B 43 C 44 D 45 E 46 F 47 G 48 H 49 I 4A J 4B K 4C L 4D M 4E N 4F O
50 P 51 Q 52 R 53 S 54 T 55 U 56 V 57 W 58 X 59 Y 5A Z 5B [ 5C \ 5D ] 5E ^ 5F o
60 ` 61 a 62 b 63 c 64 d 65 e 66 f 67 g 68 h 69 i 6A j 6B k 6C l 6D m 6E n 6F o
70 p 71 q 72 r 73 s 74 t 75 u 76 v 77 w 78 x 79 y 7A z 7B { 7C } 7D } 7E ~ 7F o
80 á A1 í A2 ó A3 ú A4 ñ A5 Á A6 à A7 ö A8 å A9 ø A0 AA Å B8 Ø B9 Æ BA Ç BB ¶ CC Ç CD =
90 á B1 ï B2 ó B3 ú B4 à B5 Á B6 à C7 Á C8 Ç C9 Æ CA Ç CB ¶ DC Ç DD = CE ¶ CF =
E0 ó E1 B E2 ô E3 ò E4 ö E5 ö E6 µ E7 þ E8 þ E9 ú EA Õ EB Ú EC ß ED ÿ EE ß EF =
F0 - F1 ± F2 = F3 % F4 11 F5 § F6 ÷ F7 - F8 ° F9 .. FA . FB FD FE . FF
```

These characters are those from the MS-DOS ASCII set. Some of these characters are semi-graphic. Here is a very simple insertion of one of these characters into our virtual screen :

```
$db dup 5 2 SCR!      6 2 SCR!
$b2 dup 7 3 SCR!      8 3 SCR!
$b1 dup 9 4 SCR!      10 4 SCR!
```

Now let's see how to display the contents of our virtual screen. If we consider each line of the virtual screen as an alphanumeric string, we just need to define this word to display one of the lines of our virtual screen :

```
: dispLine { numLine -- }
    SCR_WIDTH numLine *
    mySCREEN + SCR_WIDTH type
;
```

Along the way, we will create a definition allowing the same character to be displayed n times :

```
: nEmit ( c n -- )
    for
        aft dup emit then
    next
    drop
;
```

And now, we define the word allowing us to display the content of our virtual screen. To clearly see the content of this virtual screen, we frame it with special characters :

```
: dispScreen
0 0 at-xy
\ display upper border
$da emit    $c4 SCR_WIDTH nEmit    $bf emit    cr
\ display content virtual screen
SCR_HEIGHT 0 do
    $b3 emit    i dispLine          $b3 emit    cr
loop
\ display bottom border
$c0 emit    $c4 SCR_WIDTH nEmit    $d9 emit    cr
;
```



Running our **dispScreen** word displays this :

In our virtual screen example, we show that managing a two-dimensional array has a concrete application. Our virtual screen is accessible for writing and reading. Here we display our virtual screen in the terminal window. This display is far from efficient.

Management of complex structures

ESP32forth has the **structures** vocabulary. The content of this vocabulary makes it possible to define complex data structures.

Here is simple example of structure :

```
structures
struct YMDHMS
    ptr field >year
    ptr field >month
    ptr field >day
    ptr field >hour
    ptr field >min
    ptr field >sec
```

Here, we define the **YMDHMS** structure. This structure manages the **>year** **>month** **>day** **>hour** **>min** and **>sec** pointers.

The sole purpose of the **YMDHMS** word is to initialize and group the pointers in the complex structure. Here is how these pointers are used :

```
create DateTime
YMDHMS allot

2022 DateTime >year !
03 DateTime >month !
21 DateTime >day !
```

```

22 DateTime >hour !
36 DateTime >min !
15 DateTime >sec !

: .date ( date -- )      \ date is address of structure
  >r
  ." YEAR: " r@ >year    @ . cr
  ." MONTH: " r@ >month   @ . cr
  ." DAY: " r@ >day     @ . cr
  ." HH: " r@ >hour    @ . cr
  ." MM: " r@ >min     @ . cr
  ." SS: " r@ >sec     @ . cr
  r> drop
;

```

DateTime .date

We defined word **DateTime** as simple table of 6 consecutive cells each 32 bits. Access to each cell is with specific pointer. We can redefine our structure **YMDHMS** with **i8** pointers to bytes.

```

structures
struct cYMDHMS
  ptr field >year
  i8  field >month
  i8  field >day
  i8  field >hour
  i8  field >min
  i8  field >sec

create cDateTime
  cYMDHMS allot

2022 cDateTime >year !
03 cDateTime >month c!
21 cDateTime >day  c!
22 cDateTime >hour c!
36 cDateTime >min  c!
15 cDateTime >sec  c!

: .cDate ( date -- )
  >r
  ." YEAR: " r@ >year    @ . cr
  ." MONTH: " r@ >month   c@ . cr
  ." DAY: " r@ >day     c@ . cr
  ." HH: " r@ >hour    c@ . cr
  ." MM: " r@ >min     c@ . cr

```

```

."      SS: " r@ >sec      c@ . cr
r> drop
;
cDateTime .cDate    \ displays:
\ YEAR: 2022
\ MONTH: 3
\ DAY: 21
\ HH: 22
\ MM: 36
\ SS: 15

```

In this **cYMDHMS** structure, we kept the year in 32-bit format and reduced all other values to 8-bit integers. We see, in the **.cDate** code, that the use of pointers allows easy access to each element of our complex structure....

Definition of sprites

We previously defined a virtual screen as a two-dimensional array. The dimensions of this array are defined by two constants. Reminder of the definition of this virtual screen :

```

63 constant SCR_WIDTH
16 constant SCR_HEIGHT
create mySCREEN
    SCR_WIDTH SCR_HEIGHT * allot
    mySCREEN SCR_WIDTH SCR_HEIGHT * b1 fill

```

With this programming method, the disadvantage is that the dimensions are defined in constants, therefore outside the table. It would be more interesting to embed the dimensions of the table in the table. To do this, we will define a structure adapted to this case :

```

structures
struct cARRAY
    i8 field >width
    i8 field >height
    i8 field >content

create myVscreen    \ define a screen 8x32 bytes
    32 c,          \ compile width
    08 c,          \ compile height
    myVscreen >width c@
    myVscreen >height c@ * allot

```

To define a software sprite, we will very simply share this definition :

```

: sprite: ( width height -- )
create
    swap c, c, \ compile width et height

```

```

    does>
;
2 1 sprite: blackChars
    $db c, $db c,
2 1 sprite: greyChars
    $b2 c, $b2 c,
blackChars >content 2 type \ display content of sprite blackChars

```

Here's how to define a 5 x 7 byte sprite :

```

5 7 sprite: char3
$20 c, $db c, $db c, $db c, $20 c,
$db c, $20 c, $20 c, $20 c, $db c,
$20 c, $20 c, $20 c, $20 c, $db c,
$20 c, $db c, $db c, $db c, $20 c,
$20 c, $20 c, $20 c, $20 c, $db c,
$db c, $20 c, $20 c, $20 c, $db c,
$20 c, $db c, $db c, $db c, $20 c,

```

To display the sprite, from an x y position in the terminal window, a simple loop is enough :

```


: .sprite { xpos ypos sprAddr -- }
sprAddr >height c@ 0 do
    xpos ypos at-xy
    sprAddr >width c@ i * \ calculate offset in sprite datas
    sprAddr >content + \ calculate real addr for line n in sprite
datas
    sprAddr >width c@ type \ display line
    1 +to ypos \ increment y position
loop
;

0 constant blackColor
1 constant redColor
4 constant blueColor
10 02 char3 .sprite
redColor fg
16 02 char3 .sprite
blueColor fg
22 02 char3 .sprite
blackColor fg
cr cr

```

Result of displaying our sprite :

```
COM3 - Tera Term VT
File Edit Setup Control Window Help
ok
-->
ok
-->
ok
-->
ok
-->
ok
-->
ok
--> blackColor fg
ok
```

A 3x3 grid of colored squares. The first column contains three black squares. The second column contains three red squares. The third column contains three blue squares. The squares are arranged in a 3x3 pattern.

I hope the content of this chapter has given you some interesting ideas that you would like to share...

Real numbers with ESP32forth

If we test the operation **1 3 /** in Forth language, the result will be 0.

It's not surprising. Basically, ESP32forth only uses 32-bit integers via the data stack. Integers offer certain advantages:

- speed of processing;
- result of calculations without risk of drift in the event of iterations;
- suitable for almost all situations.

Even in trigonometric calculations, we can use a table of integers. Simply create a table with 90 values, where each value corresponds to the sine of an angle, multiplied by 1000.

But integers also have limits:

- impossible results for simple division calculations, like our 1/3 example;
- requires complex manipulations to apply physics formulas.

Since version 7.0.6.5, ESP32forth includes operators dealing with real numbers.

Real numbers are also called floating point numbers.

The real ones with ESP32forth

In order to distinguish real numbers, they must end with the letter "e":

```
3          \ push 3 on the normal stack
3e         \ push 3 on the real stack
5.21e f.  \ display 5.210000
```

It's the word **f.** which allows you to display a real number located at the top of the reals stack.

Real number accuracy with ESP32forth

The word **set-precision** allows you to indicate the number of decimal places to display after the decimal point. Let's see this with the constant **pi** :

```
pi f.      \ display 3.141592
4 set-precision
pi f.      \ display 3.1415
```

The limit precision for processing real numbers with ESP32forth is six decimal places :

```
12 set-precision  
1.987654321e f.          \ display 1.987654668777
```

If we reduce the display precision of real numbers below 6, the calculations will still be carried out with a precision to 6 decimal places.

Real constants and variables

A real constant is defined with the word **fconstant** :

```
0.693147e fconstant ln2      \ natural logarithm of 2
```

A real variable is defined with the word **fvariable** :

```
fvariable intensity  
170e 12e F/ intensity SF!    \ I=P/U    ---  P=170W   U=12V  
intensity SF@ f.            \ display 14.166669
```

ATTENTION: all real numbers pass through the **real number stack**. In the case of a real variable, only the address pointing to the real value passes through the data stack.

The word **SF!** stores a real value at the address or variable pointed to by its memory address. Executing a real variable places the memory address on the classic data stack.

The word **SF@** stacks the real value pointed to by its memory address.

Arithmetic operators on real numbers

ESP32Forth has four arithmetic operators **F+ F- F* F/** :

```
1.23e 4.56e F+ f.    \ display 5.790000    1.23-4.56  
1.23e 4.56e F- f.    \ display -3.330000   1.23-4.56  
1.23e 4.56e F* f.    \ display 5.608800    1.23*4.56  
1.23e 4.56e F/ f.    \ display 0.269736    1.23/4.56
```

ESP32forth also has these words:

- **1/F** calculates the inverse of a real number;
- **fsqrt** calculates the square root of a real number.

```
5e 1/F f.          \ display 0.200000    1/5  
5e fsqrt f.        \ display 2.236068    sqrt(5)
```

Mathematical operators on real numbers

ESP32forth has several mathematical operators:

- **F**** raises a real r_val to the power r_exp
- **FATAN2** calculates the angle in radian from the tangent.

- **FCOS** (r1 -- r2) Calculates the cosine of an angle expressed in radians.
- **FEXP** (ln-r -- r) calculates the real corresponding to e EXP r
- **FLN** (r -- ln-r) calculates the natural logarithm of a real number.
- **FSIN** (r1 -- r2) calculates the sine of an angle expressed in radians.
- **FSINCOS** (r1 -- rcos rsin) calculates the cosine and sine of an angle expressed in radians.

Some examples :

```
2e 3e f** f.      \ display 8.000000
2e 4e f** f.      \ display 16.000000
10e 1.5e f** f.  \ display 31.622776

4.605170e FEXP F.    \ display 100.000018

pi 4e f/
FSINCOS f. f.    \ display 0.707106 0.707106
pi 2e f/
FSINCOS f. f.    \ display 0.000000 1.000000
```

Logical operators on real numbers

ESP32forth also allows you to perform logic tests on real data:

- **F0<** (r -- fl) tests if a real number is less than zero.
- **F0=** (r -- fl) indicates true if the real is zero.
- **f<** (r1 r2 -- fl) fl is true if $r_1 < r_2$.
- **f<=** (r1 r2 -- fl) fl is true if $r_1 \leq r_2$.
- **f<>** (r1 r2 -- fl) fl is true if $r_1 \neq r_2$.
- **f=** (r1 r2 -- fl) fl is true if $r_1 = r_2$.
- **f>** (r1 r2 -- fl) fl is true if $r_1 > r_2$.
- **f>=** (r1 r2 -- fl) fl is true if $r_1 \geq r_2$.

Integer ↔ real transformations

ESP32forth has two words to transform integers into reals and vice versa:

- **F>S** (r -- n) converts a real to an integer. Leave the integer part on the data stack if the real has decimal parts.
- **S>F** (n -- r: r) converts an integer to a real number and transfers this real number to the reals stack.

Example :

```
35 S>F  
F. \ display 35.00000  
3.5e F>S . \ display 3
```

Displaying numbers and character strings

Change of numerical base

FORTH does not process just any numbers. The ones you used when trying the previous examples are single-precision signed integers. The definition domain for 32-bit integers is -2147483648 to 2147483647. Example :

```
2147483647 . \ displays 2147483647  
2147483647 1+ . \ displays -2147483648  
-1 u. \ displays 4294967295
```

These numbers can be processed in any number base, with all number bases between 2 and 36 being valid :

```
255 HEX. DECIMAL \displays FF
```

You can choose an even larger numerical base, but the available symbols will fall outside the alpha-numeric set [0..9,A..Z] and risk becoming inconsistent.

The current numerical base is controlled by a variable named **BASE** and whose content can be modified. So, to switch to binary, simply store the value **2** in **BASE**. Example:

```
2 BASE !
```

and type **DECIMAL** to return to the decimal numeric base.

ESP32forth has two pre-defined words allowing you to select different numerical bases:

- **DECIMAL** to select the decimal numeric base. This is the numerical base taken by default when starting ESP32forth;
- **HEX** to select the hexadecimal numeric base.

Upon selection of one of these numerical bases, the literal numbers will be interpreted, displayed or processed in this base. Any number previously entered in a number base other than the current number base is automatically converted to the current number base. Example :

```
DECIMAL \ base to decimal  
255 \ stacks 255  
HEX \ selects hexadecimal base  
1+ \ increments 255 becomes 256  
. \ displays 100
```

One can define one's own numerical base by defining the appropriate word or by storing this base in **BASE**. Example :

```
: BINARY ( ---)          \ selects the binary number base
  2 BASE ! ;
DECIMAL 255 BINARY .   \ displays 11111111
```

The contents of **BASE** can be stacked like the contents of any other variable :

```
VARIABLE RANGE_BASE      \ RANGE-BASE variable definition
BASE @ RANGE_BASE !     \ storage BASE contents in RANGE-BASE
HEX FF 10 + .           \ displays 10F
RANGE_BASE @ BASE !    \ restores BASE with contents of RANGE-BASE
```

In a definition : , the contents of **BASE** can pass through the return stack :

```
: OPERATION ( ---)
  BASE @ >R        \ stores BASE on back stack
  HEX FF 10 + .   \ operation of the previous example
  R> BASE ! ;    \ restores initial BASE value
```

WARNING : the words **>R** and **R>** cannot be used in interpreted mode. You can only use these words in a definition that will be compiled.

Definition of new display formats

Forth has primitives allowing you to adapt the display of a number to any format. With ESP32forth, these primitives deal with integers numbers :

- **<#** begins a format definition sequence;
- **#** inserts a digit into a format definition sequence;
- **#S** is equivalent to a succession of **#** ;
- **HOLD** inserts a character into a format definition;
- **#>** completes a format definition and leaves on the stack the address and length of the string containing the number to display.

These words can only be used within a definition. Example, either to display a number expressing an amount denominated in euros with the comma as a decimal separator :

```
: .EUROS ( n ---)
  <# # # [char] , hold #S #>
  type space ." EUR" ;
1245 .euros
```

Execution examples:

35 .EUROS	\ displays	0,35 EUR
3575 .EUROS	\ displays	35,75 EUR
1015 3575 + .EUROS	\ displays	45,90 EUR

In the EUROS definition, the word `<#` begins the display format definition sequence. The two words `#` place the ones and tens digits in the character string. The word `HOLD` places the character `,` (comma) following the two digits on the right, the word `#S` completes the display format with the non-zero digits following `,`. The word `#>` closes the format definition and places on the stack the address and the length of the string containing the digits of the number to display. The word `TYPE` displays this character string.

At runtime, a display format sequence deals exclusively with signed or unsigned 32-bit integers. The concatenation of the different elements of the string is done from right to left, i.e. starting with the least significant digits.

The processing of a number by a display format sequence is executed based on the current numeric base. The numerical base can be modified between two digits.

Here is a more complex example demonstrating the compactness of FORTH. This involves writing a program converting any number of seconds into HH:MM:SS format:

```
:00 ( ---)
    DECIMAL #
    6 BASE !
    #
    [char] : HOLD
    DECIMAL ;
: HMS ( n ---)           \ displays number seconds format HH:MM:SS
<# :00 :00 #S #> TYPE SPACE ;
```

Execution examples :

```
59 HMS      \ displays      0:00:59
60 HMS      \ displays      0:01:00
4500 HMS    \ displays      1:15:00
```

Explanation: The system for displaying seconds and minutes is called the sexagesimal system. Units are expressed in decimal numerical base, **tens are** expressed in base six. The word `:00` manages the conversion of units and tens in these two bases for formatting the numbers corresponding to seconds and minutes. For times, the numbers are all decimal.

Another example, to define a program converting a single precision decimal integer into binary and displaying it in the format bbbb bbbb bbbb bbbb:

```
: FOUR-DIGITS ( ---)
    # # # # 32 HOLD ;
: AFB ( n ---)           \ format 4 digits and a space
    BASE @ >R              \ Current database backup
```

```

2 BASE !           \ Binary digital base selection
<#
4 0 DO             \ Format Loop
    FOUR-DIGITS
LOOP
#> TYPE SPACE      \ Binary display
R> BASE ! ;        \ Initial digital base restoration

```

Execution example :

```

DECIMAL 12 AFB   \ displays      0000 0000 0000 0110
HEX 3FC5 AFB     \ displays      0011 1111 1100 0101

```

Another example is to create a telephone diary where one or more telephone numbers are associated with a surname. We define a word by surname :

```

: .## ( ---)
  # # [char] . HOLD ;
: .TEL ( d ---)
  CR <# .## .## .## # # #> TYPE CR ;
: WACHOWSKI ( ---)
  0618051254 .TEL ;
WACHOWSKI \ displays: 06.18.05.12.54

```

This calendar, which can be compiled from a source file, is easily editable, and although the names are not classified, the search is extremely fast.

Displaying characters and character strings

A character is displayed using the word **EMIT** :

```
65 EMIT           \ displays A
```

The displayable characters are in the range 32..255. Codes between 0 and 31 will also be displayed, subject to certain characters being executed as control codes. Here is a definition showing the entire character set of the ASCII table:

```

variable #out
: #out+! ( n -- )
  #out +!           \ increment #out
;
: (.) ( n -- a 1 )
  DUP ABS <# #S SWAP SIGN #>
;
: .R ( n 1 -- )
  >R (.) R> OVER - SPACES TYPE
;
: ASCII-SET ( ---)

```

```

cr 0 #out !
128 32
DO
    I 3 .R SPACE      \ displays character code
    4 #out+!
    I EMIT 2 SPACES   \ displays character
    3 #out+!
    #out @ 77 =
    IF
        CR 0 #out !
    THEN
LOOP ;

```

Running ASCII-SET displays the ASCII codes and characters whose code is between 32 and 127. To display the equivalent table with the ASCII codes in hexadecimal, type **HEX ASCII-SET** :

hex	ASCII-SET
20	21 !
2B	2C ,
36	37 7
41	42 B
4C	4D M
57	58 X
62	63 c
6D	6E n
78	79 y
	22 "
	23 #
	24 \$
	25 %
	26 &
	27 '
	28 (
	29)
	2A *
2F	30 0
3A	3B ;
3C	3D <
3D	3E =
3E	3F >
40	41 A
45	46 E
47	48 G
48	49 H
49	4A I
4A	4B J
4B	4C K
50	51 Q
51	52 R
52	53 S
53	54 T
54	55 U
55	56 V
56	57 W
58	59 Y
59	5A Z
5B	5C \
5C	5D]
5D	5E ^
5E	5F _
60	61 a
61	62 b
63	64 d
64	65 e
65	66 f
66	67 g
67	68 h
68	69 i
69	6A j
6A	6B k
6B	6C l
70	71 q
71	72 r
72	73 s
73	74 t
74	75 u
75	76 v
76	77 w
77	78 x
78	79 y
79	7A z
7A	7B {
7B	7C
7C	7D }
7D	7E ~
7E	7F ok

Character strings are displayed in various ways. The first, usable in compilation only, displays a character string delimited by the character " (quote mark):

```

: TITLE ." GENERAL MENU";
    TITLE      \ displays      GENERAL MENU

```

The string is separated from the word **."** by at least one space character.

A character string can also be compiled by the word **s"** and delimited by the character **"** (quotation mark):

```

: LINE1 ( --- adr len)
    S" E..Data logging" ;

```

Executing **LINE1** places the address and length of the string compiled in the definition on the data stack. The display is carried out by the word **TYPE**:

```

LINE1 TYPE      \displays      E..Data logging

```

At the end of displaying a character string, the line break must be triggered if desired:

```

CR TITLE CR CR LINE1 CR TYPE

```

```
\ displays:  
\ GENERAL MENU  
\  
\ E..Data logging
```

One or more spaces can be added at the start or end of the display of an alphanumeric string :

```
SPACE          \ displays a space character  
10 SPACES     \ displays 10 space characters
```

String variables


Alpha-numeric text variables do not exist natively in ESP32forth. Here is the first attempt to define the word **string** :

```
\ define a strvar  
: string  ( comp: n --- names_strvar | exec: --- addr len )  
    create  
    dup  
    c,      \ n is maxlen  
    0 c,    \ 0 is real length  
    allot  
    does>  
        2 +  
        dup 1 - c@  
    ;
```

A character string variable is defined like this:

```
16 string strState
```

Here is how the memory space reserved for this text variable is organized:



Text variable management word code

Here is the complete source code for managing text variables:

```
DEFINED? --str [if] forget --str  [then]  
create --str
```

```

\ compare two strings
: $= ( addr1 len1 addr2 len2 --- f1)
    str=
;

\ define a strvar
: string ( n --- names_strvar )
    create
        dup
        ,
            \ n is maxlen
        0 ,
            \ 0 is real length
        allot
    does>
        cell+ cell+
        dup cell - @@
;

\ get maxlen of a string
: maxlen$ ( strvar --- strvar maxlen )
    over cell - cell - @@
;

\ store str into strvar
: $! ( str strvar --- )
    maxlen$           \ get maxlen of strvar
    nip rot min       \ keep min length
    2dup swap cell - ! \ store real length
    cmove             \ copy string
;

\ Example:
\ : s1
\     s" this is constant string" ;
\ 200 string test
\ s1 test $!

\ set length of a string to zero
: 0$! ( addr len -- )
    drop 0 swap cell - !
;

\ extract n chars right from string
: right$ ( str1 n --- str2 )
    0 max over min >r + r@ - r>

```

```

;

\ extract n chars left frop string
: left$  ( str1 n --- str2 )
    0 max min
;

\ extract n chars from pos in string
: mid$  ( str1 pos len --- str2 )
    >r over swap - right$ r> left$
;

\ append char c to string
: c+$!  ( c str1 -- )
    over >r
    + c!
    r> cell - dup @ 1+ swap !
;

\ work only with strings. Don't use with other arrays
: input$ ( addr len -- )
    over swap maxlen$ nip accept
    swap cell - !
;

```

Creating an alphanumeric character string is very simple :

```
64 string myNewString
```

Here we create an alphanumeric variable **myNewString** which can contain up to 64 characters.

To display the contents of an alphanumeric variable, simply use **type** . Example :

```
s" This is my first example.." myNewString $!
myNewString type \ display: This is my first example..
```

If we try to save a character string longer than the maximum size of our alphanumeric variable, the string will be truncated:

```
s" This is a very long string, with more than 64 characters. It can't store
complete"
myNewString $!
myNewString type
\ displays: This is a very long string, with more than 64 characters. It can
```

Adding character to an alphanumeric variable

Some devices, the LoRa transmitter for example, require processing command lines containing the non-alphanumeric characters. The word **c+\$!** allows this code insertion:

```
32 string AT_BAND
s" AT+BAND=868500000" AT_BAND $! \ set frequency at 865.5 Mhz
$0a AT_BAND c+$!
$0d AT_BAND c+$! \ add CR LF code at end of command
```

The memory dump of the contents of our alphanumeric variable **AT_BAND** confirms the presence of the two control characters at the end of the string:

```
--> AT_BAND dump
--addr--- 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F -----chars-----
3FFF-8620 8C 84 FF 3F 20 00 00 00 13 00 00 00 41 54 2B 42 ...? .....AT+B
3FFF-8630 41 4E 44 3D 38 36 38 35 30 30 30 30 0A 0D BD AND=868500000...
OK
```

Here is a clever way to create an alphanumeric variable allowing you to transmit a carriage return, a **CR+LF** compatible with the end of commands for the LoRa transmitter:

```
2 string $crlf
$0d $crlf c+$!
$0a $crlf c+$!

:crlf ( -- ) \ same action as cr, but adapted for LoRa
$crlf type
;
```

Vocabularies with ESP32forth

In Forth, the notion of procedure and function does not exist. Forth instructions are called WORDS. Like a traditional language, Forth organizes the words that compose it into VOCABULARIES, a set of words with a common trait.

Programming in Forth consists of enriching an existing vocabulary, or defining a new one, relating to the application being developed.

List of vocabularies

A vocabulary is an ordered list of words, searched from most recently created to least recently created. The search order is a stack of vocabularies. Running a vocabulary name replaces the top of the search order stack with that vocabulary.

To see the list of different vocabularies available in ESP32forth, we will use the word **voclist** :

```
--> internals voclist \ displays
registers
ansi
editor
streams
tasks
rtos
sockets
Serial
ledc
SPIFFS
SD_MMC
SD
Wireless
Wire
ESP
structures
internalized
internals
FORTH
```

This list is not limited. Additional vocabularies may appear if we compile certain extensions.

The main vocabulary is called **FORTH**. All other vocabularies are attached to the **FORTH** vocabulary.

List of vocabulary contents

To see the content of a vocabulary, we use the word **vlist** having previously selected the appropriate vocabulary :

```
vlist sockets
```

Select **sockets** vocabulary and displays its contents :

```
--> sockets vlist\displays:  
ip. ip# ->h_addr ->addr! ->addr@ ->port! ->port@ sockaddr l, s, bs, SO_REUSEADDR  
SOL_SOCKET sizeof(sockaddr_in) AF_INET SOCK_RAW SOCK_DGRAM SOCK_STREAM  
socket setsockopt bind listen connect socket accept select poll send sendto  
sendmsg recv recvfrom recvmsg gethostbyname errno sockets-builtins
```

Selecting a vocabulary gives access to the words defined in this vocabulary.

For example, the word **voclist** is not accessible without first invoking the vocabulary **internals**.

The same word can be defined in two different vocabularies and have two different actions : the word **l** is defined in both **asm** and **editor** vocabularies.

This is even more obvious with the word **server**, defined in the **httpd**, **telnetd** and **web-interface** vocabularies.

Using vocabulary words

To compile a word defined in a vocabulary other than Forth, there are two solutions. The first solution is to simply call this vocabulary before defining the word which will use words from this vocabulary.

Here, we define a word **serial2-type** which uses the word **Serial2.write** defined in the **serial** vocabulary :

```
serial \ Selection vocabulary Serial  
: serial2-type (an --)  
Serial2.write drop  
;
```

The second solution allows you to integrate a single word from a specific vocabulary :

```
: serial2-type (an --)  
[ serial ] Serial2.write [ FORTH ] \ compile word from vocabulary serial  
drop  
;
```

The selection of a vocabulary can be carried out implicitly from another word in the Forth vocabulary.

Chaining of vocabularies

The order in which a word is searched in a vocabulary can be very important. In the case of words with the same name, we remove any ambiguity by controlling the search order in the different vocabularies that interest us.

Before creating a chain of vocabularies, we restrict the search order with the word **only** :

```
asm xtensa
order\display: xtensa >> asm >> FORTH
only
order\display: FORTH
```

We then duplicate the chaining of vocabularies with the word **also** :

```
only
order\display: FORTH
asm also
order\display: asm >> FORTH
xtensa
order\display: xtensa >> asm >> FORTH
```

Here is a compact chaining sequence :

```
only asm also xtensa
```

The last vocabulary thus chained will be the first explored when we execute or compile a new word.

```
only
order\display: FORTH
also ledc also serial also SPIFFS
order \ displays: SPIFFS >> FORTH
\ Serial >> FORTH
\ ledc >> FORTH
\ FORTH
```

The search order, here, will start with the **SPIFFS** vocabulary, then **Serial**, then **ledc** and finally the **FORTH** vocabulary :

- if the searched word is not found, there is a compilation error ;
- if the word is found in a vocabulary, it is this word that will be compiled, even if it is defined in the following vocabulary.

Delayed action words

Deferred action words are defined by the definition word **defer**. To understand the mechanisms and the interest in exploiting this type of word, let's look in more detail at the functioning of the internal interpreter of the Forth language.

Any definition compiled by : (colon) contains a sequence of coded addresses corresponding to the code fields of the words previously compiled. At the heart of the Forth system, the word **EXECUTE** accepts as parameters these code field addresses, addresses which we abbreviate by **cfa** for Code Field Address. Every Forth word has a **cfa** and this address is used by the internal Forth interpreter :

```
' <word>
\ drops the cfa of <word> onto the data stack
```

Example:

```
' WORDS
\ stacks the WORDS cfa.
```

From this **cfa** , known as the only literal value, the execution of the word can be carried out with **EXECUTE**:

```
' WORDS EXECUTE
\ executes WORDS
```

Of course, it would have been easier to type **WORDS** directly . From the moment a **cfa** is available as the only literal value, it can be manipulated and notably stored in a variable :

```
variable vector
' WORDS vector !
vector @ .
\ displays cfa of WORDS stored in vector variable
```

You can run **WORDS** indirectly from the contents of **vector**:

```
vector @ EXECUTE
```

This launches the execution of the word whose **cfa** was stored in the **vector** variable then put back on the stack before use by **EXECUTE**.

This is a similar mechanism that is exploited by the execution part of the **defer** definition word. To simplify, **defer** creates a header in the dictionary, like a **variable** or **constant**, but instead of simply dropping an address or value on the stack, it starts execution of the word whose **cfa** was stored in the parametric area of the word defined by **defer** .

Definition and usage of words with **defer**

The initialization of a word defined by **defer** is carried out by **is** :

```
defer vector
' words is vector
```

Executing **vector** causes the word whose **cfa** was previously assigned to be executed:

```
vector      \ execute  words
```

A word created by **defer** is used to execute another word without explicitly calling on that word. The main interest of this type of word lies above all in the possibility of modifying the word to be executed:

```
' page is vector
vector now executes page and no longer words.
```

We essentially use the words defined by **defer** in two situations:

- definition of a forward reference;
- definition of a word depending on the operating context.

In the first case, the definition of a before reference makes it possible to overcome the constraints of the sacrosanct precedence of definitions.

In the second case, the definition of a word depending on the operating context makes it possible to resolve most of the interfacing problems with an evolving software environment, to maintain the portability of applications, to adapt the behavior of a program to situations controlled by various parameters without harming software performance.

Setting a Forward Reference

Unlike other compilers, Forth does not allow a word to be compiled into a definition before it is defined. This is the principle of precedence of definitions:

```
: word1 ( ---)    word2      ;
: word2 ( ---)      ;
```

This generates an error when compiling **word1** , because **word2** is not yet defined. Here's how to get around this constraint with **defer** :

```
defer word2
: word1 ( ---)    word2      ;
: (word2) ( ---)      ;
' (word2) is  word2
```

This time **word2** compiled without errors. It is not necessary to assign a cfa to the vectorized execution word **word2** . It is only after the definition of (**word2**) that the parameter area of **word2** is updated. After assignment of the vectorized execution word **word2** , **word1** will be able to execute the content of its definition without error. The exploitation of words created by **defer** in this situation must remain exceptional.

Dependence on the operating context

ESP32forth natively uses a connection via serial port 1 as input and output stream.

In the ESP32forth source code, we find these lines:

```
defer type  
defer key  
defer key?
```

To pass through the serial port, ESP32forth initializes the word **type** like this:

```
' default-type is type
```

type flow will be redirected as follows:

```
: server ( port -- )  
    server  
    ['] serve-key is key  
    ['] serve-type is type  
    webserver-task start-task  
;
```

type redirection if we use a TELNET flow:

```
: connection ( n -- )  
    dup 0< if drop exit then to clientfd  
    0 echo !  
    ['] telnet-key is key  
    ['] telnet-type is type quit ;
```

And if we wanted to redirect the text display to an OLED display, we would just have to act on **type** in the same way. In the chapter *Setting up the REYAX RYLR890 LoRa transmitter*, we exploit this **type** property as follows:

```
serial \ Select Serial vocabulary  
: serial2-type ( a n -- )  
    Serial2.write drop ;  
: typeToLoRa ( -- )  
    0 echo ! \ disable display echo from terminal  
    ['] serial2-type is type  
;  
: typeToTerm ( -- )  
    ['] default-type is type  
    -1 echo ! \ enable display echo from terminal  
;
```

By doing this, it becomes very easy to transmit a text stream to serial port 2:

```
: optionChoice  
    ." choice option:" ;  
optionChoice      \ display      choice options:  on terminal  
typeToLoRa
```

```

optionChoice      \ display    choice options:  thru serial2
typeToTerm       \ restaure normal display

```

In this specific case, we define lots of words allowing you to control a LoRa transmitter using ordinary words like **emit**, **type**, etc,. If we do not activate transmission to serial port 2, therefore to the LoRa transmitter, the words that communicate with this transmitter can be easily developed:

```

\ Set the ADDRESS of LoRa transmitter:
\ s" <address>" value in interval [0..65535][?] (default 0)
: ATAddress ( addr len -- )
  ." AT+ADDRESS="
  type crlf
;

```

If we execute **ATAddress** , the text stream will be displayed on the terminal. If you followed correctly, you know which word to execute to redirect the flow from **ATAddress** to serial port 2.

In summary, thanks to deferred execution words, we can act on the action of Forth words already defined.

A practical case

You have an application to create, with displays in two languages. Here is a clever way by exploiting a word defined by defer to generate text in French or English. To begin, we will simply create a table of days in English:

```

:noname s" Saterday" ;
:noname s" Friday" ;
:noname s" Thursday" ;
:noname s" Wednesday" ;
:noname s" Tuesday" ;
:noname s" Monday" ;
:noname s" Sunday" ;

create ENdayNames ( --- addr)
, , , , ,

```

Then we create a similar table for the days in French:

```

:noname s" Samedi" ;
:noname s" Vendredi" ;
:noname s" Jeudi" ;
:noname s" Mercredi" ;
:noname s" Mardi" ;
:noname s" Lundi" ;
:noname s" Dimanche" ;

```

```
create FRdayNames ( -- addr
    , , , , , ,
```

Finally we create our deferred action word **dayNames** and how to initialize it:

```
defer dayNames

: in-ENGLISH
    ['] ENdayNames is dayNames ;

: in-FRENCH
    ['] FRdayNames is dayNames ;
```

Here are now the words to manage these two tables:

```
: _getString { array length -- addr len }
    array
    swap cell *
    + @ execute
    length ?dup if
        min
    then
;

10 value dayLength
: getDay ( n -- addr len )      \ n interval [0..6]
    dayNames dayLength _getString
;
```

Here's what running **getDay** does :

```
: .dayList { size -- }
    size to dayLength
    7 0 do
        i getDay type space
    loop
;

in-ENGLISH 3 .dayList cr  \ display : Sun Mon Tue Wed Thu Fri Sat
in-FRENCH  1 .dayList cr  \ display : D L M M J V S
```

In the second line, we only display the first letter of each day of the week.

In this example, we leverage **defer** to simplify programming. In web development, we would use templates *to* manage multilingual sites. In Forth, we simply move a vector in a delayed action word. Here we only manage two languages. This mechanism can easily be extended to other languages, because we have separated the management of text messages from the purely application part.

Word Creation Words

Forth is more than a programming language. It's a meta-language. A meta-language is a language used to describe, specify or manipulate other languages.

With ESP32forth, we can define the syntax and semantics of programming words beyond the formal framework of basic definitions.

We have already seen the words defined by **constant** , **variable** , **value** . These words are used to manage digital data.

In the Data Structures for ESP32forth chapter, we also used the word **create**. This word creates a header allowing access to a data area stored in memory. Example :

```
create temperatures
34, 37, 42, 36, 25, 12,
```

Here, each value is stored in the parameters area of the word **temperatures** with the word **,**.

With ESP32forth, we will see how to customize the execution of words defined by **create**.

Using does>

However, there is a combination of "**CREATE**" and "**DOES>**" keywords, which are often used together to create custom words (vocabulary words) with specific behaviors.

Here's how it generally works in Forth:

- **CREATE** : this keyword is used to create a new data space in the ESP32Forth dictionary. It takes one argument, which is the name you give your new word;
- **DOES>** : this keyword is used to define the behavior of the word you just created with **CREATE** . It is followed by a block of code that specifies what the word should do when encountered during program execution.

Together it looks something like this:

```
forth
CREATE my-new-word
\ code to execute when encountering my-new-word
DOES>
;
```

When the word **my-new-word** is encountered in the Forth program, the code specified in the **does>... ;** will be executed.

```
\ define a register, similar as constant
: defREG:
```

```

create ( addr1 -- <name> )
      ,
does> ( -- regAddr )
      @
;

```

Here, we define the definition word **defREG:** which has exactly the same action as **constant**. But why create a word that recreates the action of a word that already exists?

```
$3FF44004 constant GPIO_OUT_REG
```

or

```
$3FF44004 defREG: GPIO_OUT_REG
```

are similar. However, by creating our registers with **defREG:** we have the following advantages:

- a more readable ESP32forth source code. We easily detect all the constants naming an ESP32 register;
- we leave ourselves the possibility of modifying the **does> part** of **defREG:** without then having to rewrite the lines of code which would not use **defREG:**

Here is a classic case, processing a data table:

```

\ definition word for one dimension arrays
:array (comp: -- <name> | exec: index <name> -- addr)
      create
      does>
          swap cell * +
;
array temperatures
    21 ,    32 ,    45 ,    44 ,    28 ,    12 ,
0 temperatures @ . \ display 21
5 temperatures @ . \ display 12

```

The execution of **temperatures** must be preceded by the position of the value to extract in this table. Here we only get the address containing the value to extract.

Color management example

In this first example, we define the word **color:** which will retrieve the color to select and store it in a variable:

```

0 value currentCOLOR

\ define word as COLOR constant
: color: ( n -- <name> )
      create
      ,
does>

```

```
@ to currentCOLOR  
;  
  
$00 color: setBLACK  
$ff color: setWHITE
```

Running the word **setBLACK** or **setWHITE** greatly simplifies the ESP32forth code. Without this mechanism, one of these lines would have had to be repeated regularly :

```
$00 currentCOLOR !
```

Or

```
$00 constant BLACK  
BLACK currentCOLOR !
```

Example, writing in pinyin

Pinyin is commonly used around the world to teach Mandarin Chinese pronunciation, and it is also used in various official contexts in China, such as street signs, dictionaries, and learning textbooks. It makes learning Chinese easier for people whose native language uses the Latin alphabet.

To write Chinese on a QWERTY keyboard, the Chinese generally use a system called "pinyin input". Pinyin is a system of romanization of Mandarin Chinese, which uses the Latin alphabet to represent the sounds of Mandarin.

On a QWERTY keyboard, users type Mandarin sounds using pinyin romanization. For example, if someone wants to write the character "你" ("ní" meaning "you" in English), they can type "ni".

In this very simplified code, you can program pinyin words to write in Mandarin. The following code only works with the PuTTY terminal:

```
\ Work only with PuTTY terminal  
internals  
: chinese:  
    create ( c1 c2 c3 -- )  
        c, c, c,  
    does>  
        3 serial-type  
    ;  
forth
```

To find the UTF8 code of a Chinese character, copy the Chinese character, from Google Translate for example. Example :

```
Good Morning --> 早安 (Zao an)
```

Copy 早 and go to PuTTY terminal and type :

```
key key key \ followed by key <enter>
```

paste the character 早. ESP32forth should display the following codes:

```
230 151 169
```

For each Chinese character, we will use these three codes as follows:

```
169 151 230 chinese: Zao  
137 174 229 chinese: Year
```

Use :

```
Zao An      \ display 早安
```

Admit that programming like this is something other than what you can do in C language.
No?

Adapt breadboards to ESP32 board

Breadboards for ESP32

You have just received your ESP32 cards. And first bad surprise, this card fits very poorly on the test board :



There is no breadboard specifically suited to ESP32 boards.


Build a breadboard suitable for the ESP32 board

We're going to build our own test plate. For this, two identical test plates must be available.


On one of breadboard, we will remove a power line. To do this, use a cutter and cut from below. You should be able to separate this power line like this :



We can then reassemble the entire breadboard with this board. You have rafters on the sides of the test plates to connect them together :



And there you go! We can now install our ESP32 card :



The I/O ports can now be used without difficulty.

Powering the ESP32 board

Choice of power source


Here we will see how to power an ESP32 card. The goal is to give solutions for running Forth programs compiled by ESP32forth.

Powered by mini-USB connector

This is the simplest solution. We replace the power supply coming from the PC with a different source:

- a mains power supply such as those used to charge a mobile phone;
- a backup battery for a mobile phone (power bank).

Here, we power our ESP32 board with a backup battery for mobile devices.




Power supply via 5V pin


The second option is to connect an external unregulated power supply to the 5V pin and ground. Anything between 5 and 12 volts should work.

But it's best to keep the input voltage at around 6 or 7 Volts to avoid losing too much power as heat on the voltage regulator.

Here are the terminals allowing an external 5-12V power supply:



To use the 5V power supply, you need this equipment:



- two 3.7V lithium batteries
- a battery holder
- two dupont wires

We solder one end of each dupont wire to the terminals of the battery holder. Here our holder accepts three batteries. We will only operate two battery units, so solder correctly only two cells. The batteries are connected in series.

Once the dupont wires are soldered, we install the battery and check that the output polarity is respected:



Now, we can power our ESP32 card via the 5V pin.

WARNING : the battery voltage should be between 5 to 12 Volts.

Automatic start of a program

How can we be sure that the ESP32 card works well once powered by our batteries?

The simplest solution is to install a program and set this program to start automatically when the ESP32 board is powered on. Compile this program:

```
18 constant myLED
0 value LED_STATE
: led.on ( -- )
HIGH dup myLED pin
to LED_STATE
;
: led.off ( -- )
LOW dup myLED pin
to LED_STATE
;
timers also \ select timers vocabulary
: led.toggle ( -- )
LED_STATE if
led.off
else
led.on
then
0 rerun      \ warning, rerun removed in 7.0.7.15!
;
: led.blink ( -- )
myLED output pinMode
['] led.toggle 500000 0 interval
led.toggle
;
startup: led.blink
bye
```

Install LED on the G18 pin.

Turn off the power and reconnect the ESP32 board. If everything went well, the LED should flash after a few seconds. This is a sign that the program is running when the ESP32 board starts up.

Unplug the USB port and plug in the battery. The ESP32 board should boot up and the LED will flash.

The whole secret lies in the **startup** sequence: **led.blink**. This sequence freezes the Forth code compiled by ESP32forth and designates the word **led.blink** as the word to be executed when starting ESP32forth after ESP32 board is powered on.


Install and use the Tera Term terminal on Windows

Install Tera Term

The English page for Tera Term is here:

<https://ttssh2.osdn.jp/index.html.en>

Go to the download page, get the exe or zip file:




Install Tera Term. Installation is quick and easy.

Setting up Tera Term


To communicate with the ESP32 card, you must adjust certain parameters:

- click on Configuration -> serial port




For comfortable viewing:

- click on Configuration -> window



For readable characters:

- click on Configuration -> font



To find all these settings the next time you launch the Tera Term terminal, save the configuration:

- click on *Setup -> Save setup*
- accept the name **TERATERM.INI** .


Using Tera Term

Once configured, close Tera Term.

Connect your ESP32 board to an available USB port on your PC.

Relaunch Tera Term, then click *file -> new connection*

Select the serial port :



If everything went well, you should see this:



The screenshot shows a terminal window titled "COM3 - Tera Term VT". The menu bar includes File, Edit, Setup, Control, Window, and Help. The main text area displays the version information "ESP32forth v7.0.6.10 - rev 17c8b34289028a5c731d" followed by the command "ok" and a prompt "-->".

Compile source code in Forth language

First of all, let's remember that the Forth language is on the ESP32 board! Forth is not on your PC. Therefore, you cannot compile the source code of a program in Forth language on the PC.

To compile a program in Forth language, you must first open a source file on the PC with the editor of your choice.

Then, we copy the source code to compile. Here, open source code with Wordpad:

A screenshot of the Microsoft WordPad application window. The title bar reads "transmitDecode.txt - WordPad". The menu bar includes "Fichier", "Accueil", "Affichage", "Copier", "Coller", "Police", "Paragraphe", "Insertion", "Édition", and "Rechercher". The toolbar includes icons for Cut, Copy, Paste, Font, Paragraph, Insert, and Selection. The main text area contains Forth source code:

```
\ *** decode content of LoRaRX
*****
2 string $crlf
$0d $crlf c+$!
$0a $crlf c+$!

\ delete crlf at end of string
: normalize$ ( addr len -- )
  2dup + 2 - 2
  $crlf $= if      \ if end string = crlf
    2 - swap
    cell - !          \ subtract 2 at length of string
  else
    2drop
  then
;

\ test if string begin with "+RCV="
: RCV? ( addr len -- fl )
  dup 0 > if
    drop 5
    s" +RCV=" $=
```

The source code in Forth language can be composed and edited with any text editor: notepad, PSpad, Wordpad..

Personally I use the Netbeans IDE. This IDE allows you to edit and manage source codes in many programming languages.

Select the source code or portion of code that interests you. Then click copy. The selected code is in the PC edit buffer.

Click on the Tera Term terminal window. Make Paste:

Simply validate by clicking OK and the code will be interpreted and/or compiled.

To run compiled code, simply type the word Forth to launch, from the Tera Term terminal.

Access ESP32Forth by TELNET

Before managing a connection, you must establish a network link. The ESP32 board has a WiFi interface. To establish a WiFi connection, you must:

- have a modem/router that manages WiFi connections
- have the available WiFi port SSID and its access key

Connection to the WiFi network is ensured by the word **login** :

```
\ connection to local WiFi LAN
: myWiFiConnect ( -- )
    z" Mariloo"
    z" 1925144D91DE5373C3XXXXXXXX"
    login
    ;
```

Running **myWiFiConnect** displays :

```
--> myWiFiConnect
192.168.1.8
MDNS started
```

Change the DNS name of the ESP32 board

To connect to an ESP32 board, there are two methods:

- by knowing its IP address on the internal network. In the above case, the IP address is 192.168.1.8. This address may change if it is not locked by the WiFi router;
- by the DNS name declared when connecting to the WiFi network. By default, ESP32forth assigns the name **forth** to the card that connects to the WiFi network.

forth hostname instead of the IP address:




Figure 14: use DNS name with PuTTY

If you want to communicate with several ESP32 cards on the same network, each card must declare a distinct host name. Example code for two ESP32 cards:

```
\ set forthCOM3 for 1st ESP32 card
z" Mariloo"
z" 1925144D91DE5373C3C2D7XXXX"
login
z" forthCOM3" MDNS.begin
cr telnetd 552 server
```

Code for the second ESP32 card:

```
\ set forthCOM6 for 2nd ESP32 card
z" Mariloo"
z" 1925144D91DE5373C3C2D7959F"
login
z" forthCOM6" MDNS.begin
cr telnetd 552 server
```

forthCOM3 and **forthCOM6** hostnames on the internal network.

Connecting to ESP32 boards by their hostname

Launch PuTTY. We enter the host name and the open port to access **forthCOM3** :




Figure 15: PuTTY access to forthCOM3

Then we launch a new PuTTY session and simply change the host name for this session, here **forthCOM6**. Here are two PuTTY sessions allowing you to communicate with these two ESP32 cards:




Figure 16: PuTTY accesses two separate ESP32 boards

To automatically launch the TELNET client on the ESP32 card, we will integrate our connection code into **autoexec.fs**. Here is the code to type from the terminal. First type:

```
visual edit /spiffs/autoexec.fs
```

Then enter these few lines:

```
\ set forthCOM3 for 1st ESP32 card
z" Mariloo"
z" 1925144D91DE5373C3C2DXXXXX"
login
z" forthCOM3" MDNS.begin
cr telnetd 552 server
forth
```

Then do CTRL-X and Y. The code is saved and will be loaded the next time you start ESP32forth. The TELNET client will be restarted automatically when ESP32forth starts. It is no longer necessary to use the terminal to communicate with the ESP32 card declared with the host name **forthCOM3**:

- unplug the ESP32 board;

- reconnect the ESP32 board, but do not open the terminal!
- wait a few seconds...
- launch puTTY and activate a TELNET connection with forthCOM3 on port 552

TELNET access via PuTTY allows the same operations as via the terminal. Only restriction: if you transmit Forth code by copy/paste, limit the size of the transmitted code.


NOTE: ESP32 cards configured in this way can be accessed from the Internet if the configuration of the WiFi router allows it.

Management of source files by blocks

Use **editor** only for block file edits in the SPIFFS file system. Use **RECORDFILE** first. See the *RECORDFILE* chapter.

The blocks

Here a block on an old computer:



BLK# 2 of 23 ; File=Forth Blocks
(Finger Painting Window Definition)

NEW.WINDOW SHEET
" Finger Paint Window" SHEET W.TITLE
60 5 200 300 SHEET W.BOUNDS
SIZE.BOX CLOSE.BOX + SHEET W.ATTRIBUTES
SYS.WINDOW SHEET W.BEHIND

SHEET ADD.WINDOW

A block is a storage space whose unit has 16 lines of 64 characters. The size of a block is therefore $16 \times 64 = 1024$ bytes. It's exactly the size of a kilobyte!

Open a block file

A file is already open by default when ESP32forth starts.

blocks.fb file .

If in doubt, run **default-use** .

To find out what's in this file, use the editor commands by first typing **editor** .

Here are our first commands to know to manage the content of blocks:

- **l** lists the contents of the current block
- **n** selects the next block
- **p** selects the previous block

ATTENTION: a block always has a number between 0 and n. If you end up with a negative block number, it throws an error.

Edit the contents of a block

Now that we know how to select a particular block, let's see how to insert source code in Forth language...

One strategy is to create a source file on your computer using a text editor. You will then just need to copy/paste your source code by line into the block files.

Here are the essential commands for managing the contents of a block:

- **wipe** empties the contents of the current block
- **d** deletes line n. The line number must be in the range 0..14. The following lines move upwards. Example: 3 D erases the contents of line 3 and brings up the contents of lines 4 to 15.
- **e** erases the contents of line n. The line number must be in the range 0..15. The other lines do not go up.
- **a** inserts a line n. The line number must be in the range 0..14. The lines located after the inserted line move back down. Example: 3 A test inserts test in line 3 and moves down the contents of lines 4 to 15.
- **r** replaces the contents of line n. Example: 3 R test replaces the contents of line 3 with test

Here is our block 0 currently being edited:

```
Block 0
| 0
create sintab \ 0...90 Grad, Index in Grad
0000 , 0175 , 0349 , 0523 , 0698 ,
0872 , 1045 , 1219 , 1392 , 1564 ,
1736 , 1908 , 2079 , 2250 , 2419 ,
2588 , 2756 , 2924 , 3090 , 3256 ,
3420 , 3584 , 3746 , 3907 , 4067 ,
4226 , 4384 , 4540 , 4695 , 4848 ,
5000 , 5150 , 5299 , 5446 , 5592 ,
5736 , 5878 , 6018 , 6157 , 6293 ,
| 10
| 11
| 12
| 13
| 14
| 15
ok
--> 10 R 6428 , 6561 , 6691 , 6820 , 6947 ,
CTRL-A Z for help | 115200 8N1 | NOR | Minicom 2.7.1 | VT102 | Déconr
```

At the bottom of the screen, line **10 R 6428, 6561,** is being integrated into our block at line 10.

You notice that line 0 has no content. This generates an error when compiling the Forth code. To fix this, simply type **0 R** followed by two spaces.

With a little practice, in a few minutes, you will have inserted your Forth code into this block.

Do the same for the following blocks if necessary. When moving to the next block, you force the contents of the blocks to be saved by typing **flush** .

Compiling block contents

Before compiling the contents of a block file, we will check that their contents are well saved. For that:

- type **flush** , then unplug the ESP32 board;
- wait a few seconds and reconnect the ESP32 board;
- type **editor** and **l** . You must find your block 0 with the content that you edited.

To compile the content of your blocks, you have two words:

- **load** preceded by the number of the block whose content we want to execute and/or compile. To compile the contents of our block 0, we will execute **0 load** ;
- **thru** preceded by two block numbers will execute and/or compile the contents of the blocks as if we were executing a succession of **load words** . Example: **0 2 thru** executes and/or compiles the contents of blocks 0 to 2.

The speed of execution and/or compilation of block content is almost instantaneous.

Practical step-by-step example

We will see, with a practical example, how to insert source code in block 1. We take a code ready to be integrated into our block:

```
1 list
editor
0 r \ tools for REGISTERS definitions and manipulations
1 r : mclr { mask addr -- }    addr @ mask invert and addr ! ;
2 r : mset { mask addr -- }    addr @ mask or addr ! ;
3 r : mtst { mask addr -- x }  addr @ mask and ;
4 r : defREG: \ define a register, similar as constant
5 r      create ( addr1 -- <name> ) ,
6 r      does> ( -- regAddr )      @ ;
```

```

7 r : .reg ( reg -- ) \ display reg content
8 r      base @ >r binary @ <#
9 r      4 for aft 8 for aft # then next
10 r     bl hold  then next  #
11 r     cr space ." 33222222 22221111 11111100 00000000"
12 r     cr space ." 10987654 32109876 54321098 76543210"
13 r     cr type  r> base ! ;
14 r : defMASK: create ( mask0 position -- )    lshift ,
15 r     does> ( -- mask1 )           @ ;
save-buffers

```

Simply copy/paste parts of the code above and run this code through ESP32 Forth:

- **1 list** to select and see what block 1 contains
- **editor** to select vocabulary **editor**
- copy the lines **n r....** in packs of three and run them
- **save-buffers** hard-saves code in block file

Turn off the ESP32 board. Restart it. If you type **1 list** you should see the code edited and saved.

To compile this code, simply type **1 load** .

Conclusion

The available file space for ESP32forth is close to 1.8MB. You can therefore worry-free manage hundreds of blocks for source files in Forth language. It is recommended to install source codes of stable code parts. Thus, during the program development phase, it will be much easier to integrate into your code in the development phase:

```
2 5 thru \ integrate pwm commands for motors
```

instead of systematically reloading this code via serial line or WiFi.

The other advantage of blocks is to allow the on-site embedding of parameters, data tables, etc. which can then be used by your programs.

Editing source files with VISUAL Editor

Use **visual edit** only for editing source files in the SPIFFS file system. Use **RECORDFILE** first. See the *RECORDFILE* chapter.

Edit a Forth source file

To edit a Forth source file with ESP32forth, we will use the visual editor.

To edit a **dump.fs file**, proceed like this from the terminal connected to an ESP32 card containing ESP32forth:

```
visual edit /spiffs/dump.fs
```

The full **DUMP code** is available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/tools/dumpTool.txt>

The word **edit** is followed by the directory where the source files are stored:

- if the file does not exist, it is created;
- if the file exists, it is retrieved in the editor.

Note the name of the file you created.

fs as the file extension, for **F**orth **S**ource.

Editing the Forth code

In the editor, move the cursor with the left-right-up-down arrows available on the keyboard.



The terminal refreshes the display each time the cursor is moved or the source code is modified.

To exit the editor :

- CTRL-S : saves the contents of the file currently being edited
- CTRL-X : exits editing:
 - N: without saving file changes
 - Y: with saving of changes

Compiling file contents

Compiling the contents of our **dump.fs** file is done like this:

```
include /spiffs/dump.fs
```

Compiling is much faster than via the terminal.

The source files embedded in the ESP32 card with ESP32forth are persistent. After turning off the power and reconnecting the ESP32 card, the saved file remains available immediately.

You can define as many files as necessary.

It is therefore easy to integrate into the ESP32 card a collection of tools and routines from which you can draw as needed.

The SPIFFS file system

ESP32Forth contains a rudimentary file system on internal Flash memory. The files are accessible via a serial interface called SPIFFS for Serial Peripheral Interface Flash File System.

Even though the SPIFFS file system is simple, it considerably increases the flexibility of your developments with ESP32Forth:

- manage configuration files
- integrate software extensions accessible on request
- modularize developments into reusable functional modules

And many other uses that we will let you discover...

Access to the SPIFFS file system

To compile the contents of a source file edited by visual edit, type:

```
include /spiffs/dumpTool.fs
```

The word **include** must always be used from the terminal.

To see the list of SPIFFS files, use the word **ls** :

```
ls /spiffs/  
 \ displays:  
 \ dumpTool.fs
```

Here, the **dumpTool.fs file** has been saved. For SPIFFS, file extensions are irrelevant. File names must not contain space characters or the / character.

Let's edit and save a new **myApp.fs file** with **visual editor**. Let's run **ls again** :

```
ls /spiffs/  
 \ displays:  
 \ dumpTool.fs  
 \ myApp.fs
```

The SPIFFS file system does not manage subfolders like on a Linux computer. To create a pseudo directory, simply indicate it when creating a new file. For example, let's edit the **other/myTest.fs file**. Once edited and saved, let's run **ls** :

```
ls /spiffs/  
 \ displays:  
 \ dumpTool.fs
```

```
\ myApp.fs  
\ other/myTest.fs
```

If you want to view only the files in this **other** pseudo directory , you must follow **/spiffs/** with the name of this pseudo directory :

```
ls /spiffs/other  
\ displays:  
\ myTest.fs
```

There is no option to filter file names or pseudo directories.

Handling files

To completely delete a file, use the word **rm** followed by the name of the file to be deleted :

```
rm /spiffs/other/myTest.fs  
ls /spiffs/  
\ poster:  
\ dumpTool.fs  
\ myApp.fs
```

To rename a file, use the word **mv** :

```
mv /spiffs/myApp.fs /spiffs/main.fs  
ls /spiffs/  
\ displays:  
\ dumpTool.fs  
\ main.fs
```

To copy a file, use the word **cp** :

```
cp /spiffs/main.fs /spiffs/mainTest.fs  
ls /spiffs/  
\ displays:  
\ dumpTool.fs  
\ main.fs  
\ mainTest.fs
```

To see the contents of a file, use the word **cat** :

```
cat /spiffs/dumpTool.fs  
\ displays contents of dumpTool.fs
```

To save the contents of a string to a file, act in two phases :

- create a new file with **touch**
- save string contents with **dump-file**

```
touch /spiffs/mTest.fs \ creates new mTest,fs file  
ls /spiffs/ \ displays:
```

```

\ dumpTool.fs
\ main.fs
\ mainTest.fs
\ mTests.fs

\ save string "Insert my text into mTest" in mTest.fs
r| ." Insert my text into mTest" | s" /spiffs/mTest.fs" dump-file

include /spiffs/mTest.fs \ displays: Insert my text in mTest

```

Organize and compile your files on the ESP32 card

We will see how to manage files for an application being developed on an ESP32 board with ESP32forth installed on it.

It is agreed that all files used are in ASCII text format.

The following explanations are given as advice only. They come from a certain experience and aim to facilitate the development of large applications with ESP32forth.

Editing and transmitting source files

All the source files for your project are on your computer. It is advisable to have a subfolder dedicated to this project. For example, you are working on an SSD1306 OLED display. So you create a directory named SSD1306.

Regarding file name extensions, we recommend using the **fs** extension .

Editing files on a computer is carried out with any text file editor.

In these source files, do not use any characters not included in the ASCII code characters. Some extended codes can disrupt program compilation.

These source files will then be copied or transferred to the ESP32 card via the serial link and a terminal type program :

- by copy/pasted using visual on ESP32forth, to be reserved for small files ;
- with a specific procedure which will be detailed later for important files.

Conclusion

Files saved in the ESP32forth SPIFFS file system are permanently available.

If you take the ESP32 board out of service and then plug it back in, the files will be available immediately.

The content of the files can be modified in situ with **visual edit** .

This convenience will make developments much faster and easier.

RECORDFILE and Forth project management

This chapter is devoted to a single key element: **RECORDFILE** . This word allows fast saving of files in the SPIFFS file system.

I advise you to read it carefully before trying to manipulate source files with **visual** or **editor** .

Here's step-by-step how to save the definition of **RECORDFILE** and then use it effectively.

Save RECORDFILE in autoexec.fs file

When ESP32forth starts, the system tests for the presence of the **autoexec.fs file** . If this file is present, its contents will be interpreted.

Here is the source code for **RECORDFILE** . This definition was developed by Bob EDWARDS. Copy this code and paste it into the terminal window to compile it. This maneuver will only have to be performed once:

```
\ These chars terminate all text lines in a file
create crlf 13 C, 10 C,

\ Records the input stream to a spiffs file until
\ an <EOF> marker is encountered, then close file
: RECORDFILE ( "filename" "filecontents" "<EOF>" -- )
    bl parse          \ read the filename ( a n )
    W/O CREATE-FILE throw >R      \ create the file to record to -
                                \ put file id on R stack
BEGIN
    \ read a line of the file from the input stream
    tib #tib accept
    tib over
    S" <EOF>" startswith? \ does the line start with <EOF> ?
    DUP IF
        \ Yes, so drop the end line of the file containing <EOF>
        swap drop
    ELSE
        swap
        tib swap
        \ No, so write the line to the open file
        R@ WRITE-FILE throw
        \ and terminate line with cr-lf
        crlf 2 R@ WRITE-FILE throw
THEN
```

```

    UNTIL                                \ repeat until <EOF> found
    R> CLOSE-FILE throw                  \ Close the file
;

```

Once this word is compiled, we will see how to proceed so that this word is permanently available from **autoexec.fs**.

On your PC, in your development area dedicated to ESP32Forth, create an **autoexec.fs file**.

Copy the **RECORDFILE** code as given above into this **autoexec.fs file**. Add these two lines of code:

```

RECORDFILE /spiffs/autoexec.fs
\ These chars terminate all text lines in a file
create crlf 13 C, 10 C,

\ Records the input stream to a spiffs file until
\ an <EOF> marker is encountered, then close file
: RECORDFILE ( "filename" "filecontents" "<EOF>" -- )
    bl parse          \ read the filename ( a n )
    W/O CREATE-FILE throw >R      \ create the file to record to -
                                \ put file id on R stack
BEGIN
    \ read a line of the file from the input stream
    tib #tib accept
    tib over
    S" <EOF>" startswith? \ does the line start with <EOF> ?
    DUP IF
        \ Yes, so drop the end line of the file containing <EOF>
        swap drop
    ELSE
        swap
        tib swap
        \ No, so write the line to the open file
        R@ WRITE-FILE throw
        \ and terminate line with cr-lf
        crlf 2 R@ WRITE-FILE throw
    THEN
    UNTIL                                \ repeat until <EOF> found
    R> CLOSE-FILE throw                  \ Close the file
;
<EOF>

```

Copy this source code again, including the lines of code in red. Paste this code into the terminal window again. Transmit this code to the ESP32 board.

Unlike the first manipulation which consists of compiling the code, this time this code is saved in the **/spiffs/autoexec.fs** file.

autoexec.fs file is saved, run **ls** :

```
ls /spiffs/
```

autoexec.fs file should appear in the file list. To check the contents of **autoexec.fs**, type:

```
cat /spiffs/autoexec.fs
```

This should display the contents of **autoexec.fs**.

Use modified contents of autoexec.fs file

Relaunch ESP32forth. If all went well, **RECORDFILE** is now available when ESP32forth starts. Run **words**. You should find **RECORDFILE** in the first words of the Forth dictionary:

```
RECORDFILE crlf FORTH spi oled telnetd registers webui login web-interface
httpd ok LED OUTPUT INPUT HIGH LOW tone freq duty adc pin default-key?
default-key default-type visual set-title page at-xy normal bg fg ansi....
```

Don't clutter **autoexec.fs** with other definitions. We will see how to create a project.

Breaking down a project with ESP32forth

A Forth development project for ESP32forth is created on your PC:

- editing the source code with the text editor of your choice or an IDE (Netbeans for example);
- have a terminal linked by USB to the ESP32 card;
- have ESP32forth enabled on the ESP32 board.

On the PC, work in a structured manner. The following explanations are only recommendations.

Start by defining the general working directory for all ESP32forth developments. For example, a folder named **ESP32forth developments**.

Then, in this folder, create two additional folders:

- **_my Projects** which is intended to accommodate all your projects;
- **_sandbox** which is intended to receive all small programs to be tested and not having a specific use;
- **Tools** which is intended to accommodate all source files of general interest. These are tested files and do not require adaptation;
- **Documentation** which is intended for documents of any kind.

Example project

I will use the TEMPVS FVGIT source code as an example project. The full source codes are available here:

https://github.com/MPETREMANN11/ESP32forth/tree/main/_my%20projects/display/OLED%20SSD1306%20128x32/TEMPVS%20FVGIT

The first file to create is called **main.fs**. This file must be written in a TEMPVS FVGIT folder:

```
ESP32forth developments
    +-----> _my Projects
        +-----> TEMPVS FVGIT
            +-----> main.fs
                config.fs
                strings.fs
```

Once again, these are only recommendations. The main interest is to bring together all the components of a single project. Contents of the **main.fs** file :

```
RECORDFILE /spiffs/main.fs
DEFINED? --tempusFugit [if] forget --tempusFugit [then]
create --tempusFugit

s" /spiffs/strings.fs"           included
s" /spiffs/RTClock.fs"          included

s" /spiffs/clepsydra.fs"        included

s" /spiffs/config.fs"           included
s" /spiffs/oledTools.fs"         included
( part of code removed here )
<EOF>
```

In red, we find our word **RECORDFILE**. To save the code from **main.fs** to the SPIFFS file system on the ESP32 board, simply copy this source code and pass it to ESP32forth with the terminal program.

In blue, in the code above, the content of **main.fs** makes a call to the **strings.fs** file. The source code for this file comes from the **Tools** folder. It is a copy of **strings.fs** which is then modified like this:

```
RECORDFILE /spiffs/strings.fs
structures
struct __STRING
    ptr field >maxLength      \ point to max length of string
    ptr field >realLength     \ real length of string
    ptr field >strContent     \ string content
```

```

forth
( ... removed part of file )
\ work only with strings. Don't use with other arrays
: input$ { addr len -- }
    addr len maxlen$ nip accept
    addr __STRING - cell+ >realLength !
;
<EOF>

```

Copying and passing this source code creates the **strings.fs** file in the SPIFFS file system on the ESP32 board.

At this stage, we start to have several files on the ESP32 card. To compile all the transferred files, we will simply execute:

```
include /spiffs/main.fs
```

There is no limit to the files that can be saved on the ESP32 card, other than a physical space limit. The available space in the SPIFFS file system exceeds 1MB of recording space.

If you need to modify the content of a general-purpose software component, always do so on a copy of the source file for that component. Remember to version and date these modifications.

For each of the files in this project, we integrate **RECORDFILE** and its **<EOF>** terminator




Figure 17: Structuration de projets avec l'IDE Netbeans

In each project, we find the **main.fs** and **config.fs** files. But their content is adapted to each project. For a specific project, all **fs** extension files are loaded onto the ESP32 board in the SPIFFS file system. Compiling their content is incredibly fast. But above all, the content of these files **is preserved** between two restarts of the ESP32 card. At the slightest blockage of Forth it is easy to restart the card and find all the word definitions of the project without requiring a new transfer via the terminal.

The notion of a black box

It's an old concept, which dates from the time when we mainly developed in assembler on micro-controller board. We also find it with classes in object programming. In the concept of "black box", we must consider a subroutine, a function, a method as a black box. We know what we put in there. We know what can come out of it or how this box works, but we don't worry about its internal functioning. We trust those who programmed the "black box".

In Forth language, a word has a definition. When you pass parameters through the stack, in the end, only the designer of the definition must ensure the proper functioning of the definition. And to ensure the proper functioning of a definition, it is strongly recommended not to make definitions that are too long.

My tip for marking verified code is simply to put a comment line just before the definition. Example of unverified code:

```
: fpi* ( fn - fn*pi )
    pi f*
;
```

The code will be tested in a *sandbox* or in a project file. Never mind. Once tested with different values, I modify the source code:

```
\ multiply fn by pi
: fpi* ( fn - fn*pi )
    pi f*
;
```

If you doubt the reliability of your code, you can define a **tests.fs** file.

This file is only intended for carrying out unit test batteries. See the definition of the word **assert(** which carries out these tests, definition visible here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/tools/assert.fs>

And here is an example of tests saved in our **tests.fs** file :

```
assert( 0 >gray 0 = )
assert( 1 >gray 1 = )
assert( 2 >gray 3 = )
assert( 3 >gray 2 = )
assert( 4 >gray 6 = )
assert( 5 >gray 7 = )
assert( 6 >gray 5 = )
assert( 7 >gray 4 = )
```

assert(generates an alert if the tested word does not behave as expected.

For a definition as simple as that of **fpi*** , a battery of tests may be necessary if we have not made the word **f*** reliable . We integrate these tests into the **main.fs** file :

```

RECORDFILE /spiffs/main.fs
DEFINED? --tempusFugit [if] forget --tempusFugit [then]
create --tempusFugit

s" /spiffs/strings.fs"           included
s" /spiffs/RTClock.fs"          included

s" /spiffs/clepsydra.fs"        included
s" /spiffs/config.fs"           included
s" /spiffs/oledTools.fs"         included

s" /spiffs/tests.fs"           included
<EOF>

```

tests.fs file must also be transferred to the ESP32 card.

In this way, the complete compilation cycle also includes a battery of tests. Testing does not guarantee that the code is reliable. They only make it possible to detect possible side effects if we have to modify parts of the application code.

In summary, I advise fragmenting your code by systematically integrating these files:

- **main.fs** which is the main file. Usually, whatever the name of the project, you will compile it with a simple execution of **include /spiffs/main.fs**
- **config.fs** which contains the global configuration parameters, WiFi access passwords for example;
- **tests.fs** which contains a battery of tests. If you are not doing any testing, creating this file is not necessary.

All other files will have the extension **fs**, except for files not processed by ESP32forth.

By following these few guidelines, you will have an easier time managing complex applications. Saving time each time reliable parts of code are compiled and saved in files in the SPIFFS file system is the main argument for adopting **RECORDFILE**.


Editing and managing source files for ESP32forth

As with the vast majority of programming languages, source files written in Forth language are in simple text format. The extension of forth files is free:

- **txt** generic extension for all text files;
- **forth** used by some Forth programmers;
- **fth** compressed form for Forth;
- **4th** other compressed form for Forth;
- **fs** our favorite extension...

Text file editors

On Windows, **edit** file editor is the simplest:



```
\ ****
\ Sinus and Cosinus
\ Filename: integerSinus.fs
\ Date: 17 jul 2021
\ File Version: 1.0
\ MCU: ESP32-WROOM-32
\ Forth: ESP32forth all versions 7.x++
\ Copyright: Marc PETREMANN
\ Author: Marc PETREMANN
\ GNU General Public License
\ *****

\ Sinus and Cosinus
\ Use table calculating integer sin.
\ Get values scaled by 10K.

create sintab \ 0..90 Grad, Index in Grad
0000 , 0175 , 0349 , 0523 , 0698 , 0872 , 1045 ,
1219 , 1392 , 1564 , 1736 , 1908 , 2079 , 2250 ,
2419 , 2588 , 2756 , 2924 , 3090 , 3256 , 3420 ,
3584 , 3746 , 3907 , 4067 , 4226 , 4384 , 4548 ,
4695 , 4848 , 5000 , 5150 , 5299 , 5446 , 5592 ,
5736 , 5878 , 6018 , 6157 , 6293 , 6428 , 6561 ,
6691 , 6828 , 6947 , 7071 , 7193 , 7314 , 7431 ,
7547 , 7660 , 7771 , 7880 , 7986 , 8090 , 8192 ,
8299 . 8387 . 8480 . 8577 . 8660 . 8746 . 8829 .
```

editing with edit under Windows 11


Other editors, such as **WordPad**, are not recommended because you risk saving the Forth language source code in a file format that is not compatible with ESP32forth.

On Linux, the equivalent is called **gEdit**. MacOS also has a simple text editor.

If you use a custom file extension, such as **fs**, for your Forth language source files, you must have this file extension recognized by your system to allow them to be opened by the text editor.

Use an IDE

Nothing stops you from using an IDE³. For my part, I have a preference for **Netbeans** which I also use for PHP, MySQL, Javascript, C, assembler... It is a very powerful IDE and as efficient as **Eclipse**:




```
arduin-forth - Apache Netbeans IDE 12.6
File Edit View Navigate Source Profile Run Debug Tools Window Help
arduin-forth
Projects | Files
charTable.txt
Sources
1  \-----+
2  \-----+
3  \-----+
4  \-----+
5  \-----+
6  \-----+
7  \-----+
8  \-----+
9  \-----+
10 \-----+
11 \-----+
12 \-----+
13 \-----+
14 \-----+
15 \-----+
16 \-----+
17 \-----+
18 \-----+
19 \-----+
20 \-----+
21 \-----+
22 \-----+
23 \-----+
24 \-----+
25 \-----+
26 \-----+
27 \-----+
28 \-----+
29 \-----+
30 \-----+
31 \-----+
32 \-----+
33 \-----+
34 \-----+
35 \-----+
36 \-----+
37 \-----+
38 \-----+
39 \-----+
40 \-----+
41 \-----+
42 \-----+
43 \-----+
44 \-----+
45 \-----+
46 \-----+
47 \-----+
48 \-----+
49 \-----+
50 \-----+
51 \-----+
52 \-----+
53 \-----+
54 \-----+
55 \-----+
56 \-----+
57 \-----+
58 \-----+
59 \-----+
60 \-----+
61 \-----+
62 \-----+
63 \-----+
64 \-----+
65 \-----+
66 \-----+
67 \-----+
68 \-----+
69 \-----+
70 \-----+
71 \-----+
72 \-----+
73 \-----+
74 \-----+
75 \-----+
76 \-----+
77 \-----+
78 \-----+
79 \-----+
80 \-----+
81 \-----+
82 \-----+
83 \-----+
84 \-----+
85 \-----+
86 \-----+
87 \-----+
88 \-----+
89 \-----+
90 \-----+
91 \-----+
92 \-----+
93 \-----+
94 \-----+
95 \-----+
96 \-----+
97 \-----+
98 \-----+
99 \-----+
100 \-----+
```

editing with Netbeans

Netbeans offers several interesting features:


- version management with **GIT** ;
- recovery of previous versions of modified files;
- file comparison with **Diff** ;
- one-click **FTP transmission** to the online hosting of your choice;

GIT option , possibility of sharing files on a repository and managing collaborations on complex projects. Locally or collaboratively, GIT allows you to manage different versions of the same project, then merge the versions. You can create your local GIT repository. Each *time* a file or a complete directory is committed, the developments are kept as is. This allows you to find old versions of the same file or file folder.



GIT commit operation in Netbeans of a folder

With NetBeans, you can define a development branch for a complex project. Here we create a new branch:



creating a branch on a project

Example of a situation that justifies the creation of a branch:


- you have a functional project;
- you plan to optimize it;
- create a branch and do the optimizations in this branch...

Changes to source files in a branch have no influence on files in the *main trunk*.


Incidentally, it is more than advisable to have physical backup media. An SSD hard drive costs around €50 for 300Gb of storage space. The read or write access speed of SSD media is simply astonishing!

Storage on GitHub

GitHub⁴ website is, along with **SourceForge**⁵, one of the best places to store source files. On GitHub, you can share a working folder with other developers and manage complex projects. The Netbeans editor can connect to the project and allows you to pass or retrieve file changes.



On **GitHub**, you can manage project *forks*. You can also make certain parts of your projects confidential. Here the branches in the **flagxor/ueforth** projects:



Some good practices

The first good practice is to name your working files and folders correctly. You are developing for ESP32Forth, so create a folder named **ESP32forth**.

For various tests, create a **sandbox subfolder in this folder**.

4 <https://github.com/>

5 <https://sourceforge.net/>

For well-constructed projects, create a folder per project. For example, you want to control a robot, create a **robot** subfolder .

tools subfolder . If you are using a file from this tools folder in a project, copy and paste that file into that project's folder. This will prevent a modification of a file in **tools** from subsequently disrupting your project.

The second best practice is to distribute the source code of a project into several files:

- **config.fs** to store project settings;
- folder **documentation** to store your preferred file format for project documentation;
- **myApp.fs** for your project definitions. Choose a fairly explicit file name. For example, to manage a robot, take the name **robot-commands.fs** .

..	
LOTTOinterface.jpg	Add files via upload
README.md	Create README.md
euroMillionFR.fs	LOTO wining combinaisons numbers
generalWords.fs	general words for LOTTO program
gridsManage.fs	Manage content of LOTTO grids
interface.fs	text interface for LOTTO program
main.fs	LOTTO game main file
numbersFrequency.fs	stats frequency for LOTTO numbers

Forth source file naming example

It is the content of these files which must be transferred via the terminal to the ESP32 card so that ESP32forth interprets and compiles the Forth code.

The **main.fs** file

ESP32forth manages a SPIFFS file system ⁶. See the chapter *The SPIFFS file system* .

These files are therefore stored in the ESP32 card and can be read by ESP32forth. If you wrote a **config.fs** file in the SPIFFS file system, here is the line of code to write in **main.fs** to access the contents of **config.fs** :

```
s" /spiffs/config.fs" included
```

From this point on, you have two options for interpreting the contents of **config.fs** . From the terminal:

```
include /spiffs/config.fs
```

Or

```
include /spiffs/main.fs
```

⁶ Serial Peripheral Interface Flash File System

The point is that **main.fs** can call other files. Example :

```
\ OLED SSD1306 128x32 dev and display tests
s" /spiffs/config.fs" included
s" /spiffs/SSD10306commands.fs" included
```

Processing many files takes less than a second. This strategy avoids the repeated transmission of the source code by serial link via the terminal.

And when you manage multiple projects on multiple ESP32 boards, it's easier to test each project with a simple **include /spiffs/main.fs** command.

Quickly save files to SPIFFS

This method was developed by Bob EDWARDS. Code to compile:

```
\ These chars terminate all text lines in a file
create crlf 13 C, 10 C,

\ Records the input stream to a spiffs file until
\ an <EOF> marker is encountered, then close file
: RECORDFILE ( "filename" "filecontents" "<EOF>" -- )
    bl parse           \ read the filename ( a n )
    W/O CREATE-FILE throw >R      \ create the file to record to -
                                \ put file id on R stack
BEGIN
    \ read a line of the file from the input stream
    tib #tib accept
    tib over
    S" <EOF>" startswith? \ does the line start with <EOF> ?
    DUP IF
        \ Yes, so drop the end line of the file containing <EOF>
        swap drop
    ELSE
        swap
        tib swap
        \ No, so write the line to the open file
        R@ WRITE-FILE throw
        \ and terminate line with cr-lf
        crlf 2 R@ WRITE-FILE throw
    THEN
    UNTIL
        R> CLOSE-FILE throw          \ repeat until <EOF> found
                                    \ Close the file
;
```

RECORDFILE works is simple. This word is followed by the name of the destination file in the SPIFFS file system.

The following lines contain the code to save.

At the end of the code, simply add the **<EOF>** directive. WARNING: write **<EOF>** and not **<eof>** !!! Example :

```
RECORDFILE /spiffs/main.fs
```

```
\ OLED SSD1306 128x32 dev and display test
s" /spiffs/SSD1306/config.fs" included
s" /spiffs/SSD1306/commands.fs" included
<EOF>
```

Then, we copy all the code, from **RECORDFILE** to **<EOF>** . We paste this code into the terminal window and transmit it to the ESP32 card. This is what the terminal displays while transmitting the source code:

```
--> RECORDFILE /spiffs/main.fs
--> \ OLED SSD1306 128x32 dev and display test
--> s" /spiffs/SSD1306/config.fs" included
--> s" /spiffs/SSD1306/commands.fs" included
--> <EOF>
ok
```

You can check that the new file has been saved:

```
ls /spiffs/ \ displays :
SSD1306/config.fs
SSD1306/commands.fs
SSD1306/dispChars.fs
SSD1306.fs
main.fs
```

main.fs file is successfully saved.

If you modify the contents of a source file and save it with **RECORDFILE** , this modified content will replace the previously saved content.

In our example, we saved a file of a few lines. But it also works for very large source files. Preferably send large, finalized files that no longer require modifications or adjustments.

Have the word **RECORDFILE** when starting **ESP32forth**

Use **visual** to create or edit the **autoexec.fs** file :

```
visual edit /spiffs/autoexec.fs
```

This opens the **visual editor** . Copy the following code:

```
create crlf 13 C, 10 C,
: RECORDFILE
  bl parse
  W/O CREATE-FILE throw >R
  BEGIN
    tib #tib accept
    tib over
    S" <EOF>" startswith?
    DUP IF
      swap drop
    ELSE
      swap
```

```
tib swap
R@ WRITE-FILE throw
crlf 2 R@ WRITE-FILE throw
THEN
UNTIL
R> CLOSE-FILE throw
;
```

visual edit space . At the end of editing, press CTRL-S (Save) then CTRL-X and Y to exit **visual** .

Relaunch ESP32forth. If all went well, **RECORDFILE** is now available when ESP32forth starts:

```
RECORDFILE crlf FORTH spi oled telnetd registers webui login web-interface
httpd ok LED OUTPUT INPUT HIGH LOW tone freq duty adc pin default-key?
default-key default-type visual set-title page at-xy normal bg fg ansi....
```

Managing a traffic light with ESP32

GPIO ports on the ESP32 board

GPIO ports (General Purpose Input/Output) are input-output ports widely used in the world of microcontrollers.

The ESP32 board comes with 48 pins having multiple functions. Not all pins are used on ESP32 development boards, and some pins cannot be used.

There are many questions about how to use ESP32 GPIOs. Which connectors should you use? Which connectors should you avoid using in your projects?

If we look under a magnifying glass at an ESP32 card, we see this:




Each connector is identified by a series of letters and numbers, here from left to right in our photo: G22 TXD RXD G21 GND G19 G18, etc...

The connectors that interest us for this handling are prefixed by the letter G followed by one or two numbers. For example, G2 corresponds to GPIO 2.

Defining and operating a GPIO connector in output mode is quite simple.

Mounting the LEDs

The assembly is quite simple and only one photo is enough:



- Green LED connected to G2 - green wire
- Yellow LED connected to G21 - yellow wire
- Red LED connected to G17 - red wire
- black wire connected to GND

Our code uses the word **include** followed by the file to load.

We define our LEDs with **defPin**:

```
\ Use:  
\ numGPIO defPIN: PD7  ( define portD pin #7)  
: defPIN: ( GPIOx --- <word> | <word> --- GPIOx )  
    value  
;  
  
2 defPIN: ledGREEN  
21 defPIN: ledYELLOW  
17 defPIN: ledRED  
  
: LEDinit  
    ledGREEN    output pinMode  
    ledYELLOW   output pinMode  
    ledRED      output pinMode  
;
```

Many programmers have the bad habit of naming connectors by their number. Example :

```
17 defPin: pin17
```

Or

```
17 defPin: GPIO17.
```

To be effective, you must name the connectors by their function. Here we define the **ledRED** or **ledGREEN** connectors .

For what? Because the day you need to add accessories and release for example the G21 connector, simply redefine **21 defPIN: ledYELLOW** with the new connector number. The rest of the code will be unchanged and usable.

Management of traffic lights

Here is the part of code that controls our LEDs in our traffic light simulation:

```
\ traficLights execute one light cycle
: trafficLights ( ---)
    high ledGREEN    pin      3000 ms    low ledGREEN    pin
    high ledYELLOW   pin      800 ms     low ledYELLOW   pin
    high ledRED      pin      3000 ms    low ledRED      pin
;

\ classic traffic lights loop
: lightsLoop ( ---)
    LEDinit
    begin
        trafficLights
    key? until
;

\ german trafic light style
: Dtraffic ( ---)
    high ledGREEN    pin      3000 ms    low ledGREEN    pin
    high ledYELLOW   pin      800 ms     low ledYELLOW   pin
    high ledRED      pin      3000 ms
    ledYELLOW high    800 ms
    \ simultaneous red and yellow ON
    high ledRED      pin \ simultaneous red and yellow OFF
    high ledYELLOW   pin
;

\ german traffic lights loop
: DlightsLoop ( ---)
    LEDinit
    begin
        Dtraffic
    key? until
;
```

Conclusion

This traffic light management program could perfectly have been written in C language. But the advantage of the Forth language is that it gives control, via the terminal, to analyze, debug and modify functions very quickly (in Forth we say words).

Managing traffic lights is an easy exercise in C language. But when the programs become a little more complex, the compilation and upload process quickly becomes tedious.

Simply act via the terminal and simply copy/paste any fragment of Forth language code for it to be compiled and/or executed.

If you are using a terminal program to communicate with the ESP32 board, simply type **DlightsLoop** or **lightsLoop** to test how the program works. These words use a conditional loop. Simply press a key on the keyboard and the word stops playing at the end of the loop.

Direct access to GPIO registers

In some situations, it is much more beneficial to have direct access to the GPIO registers on the ESP32 board. For example, to manage complex activation or deactivation sequences.

With ESP32forth, access to an ESP32 register is done using words **m!** and **m@**. These words allow in particular direct access to GPIO registers.

The GPIO register managing the first 32 inputs/outputs is at the hexadecimal address 3ff44004:

```
$3ff44004 defREG: GPIO_OUT_REG
```

If we plug an LED into the GPIO2 port, we can turn it on and off like this:

```
0 GPIO_OUT_REG m!      \ turn LED on G2 off  
4 GPIO_OUT_REG m!      \ turn LED on G2 on
```

The disadvantage, in sequence **4 GPIO_OUT_REG m!**, this is what only activates the G2 port. If other ports were active, they will be disabled. The solution coming to mind would therefore be to read the state of the **GPIO_OUT_REG register** using **m@** and perform logical operations on the value before reinjecting it by **m!**.

It turns out that the ESP32 card has dedicated registers which perform these selective activation and deactivation operations without going through these logical operations on this **GPIO_OUT_REG register**.

Use of words **m!** and **m@**

These two words are defined in the **registers** vocabulary. These are also the only words defined in this vocabulary:

- **m!** (val shift mask addr --)
modifies the contents of a register pointed to by **addr**, applies a logical mask with **mask** and shifts **val** by n bits according to **shift** ;
- **m@** (shift mask addr -- val)
reads the contents of a register pointed to by **addr**, applies a logical mask with **mask** and shifts by n bits according to **shift** .

In order to fully understand how these two words work, we will first define a word which displays the 32-bit content of any address :

```
\ display n in bbbbbbbb bbb..... format  
: .binDisp ( n -- )
```

```

base @ >r binary \ select binary base
<#                      \ start num formating
4 for
    aft
        8 for
            aft # then
        next
            bl hold \ add 'space' in number formating
        then
    next
    #>
    cr space ." 33222222 22221111 11111100 00000000"
    cr space ." 10987654 32109876 54321098 76543210"
    cr type          \ display n in binary format
    r> base !       \ restore current numeric base
;

```

Let's define any memory space, initialized with a zero value:

```

create myReg
0 ,

```

Here's how to display the contents of **myReg** with **.binDisp**:

```

myReg @ .binDisp \ display :
\ 33222222 22221111 11111100 00000000
\ 10987654 32109876 54321098 76543210
\ 00000000 00000000 00000000 00000000 ok

```

Bit b22 has been highlighted. Here it is at zero. ATTENTION: the bits are numbered from 0 to 31 in the display rendered by **.binDisp**. Here is how to set this single b22 bit to one :

```

registers
1 22 $ffffffff myReg m!
forth
myReg @ .binDisp \ display :
\ 33222222 22221111 11111100 00000000
\ 10987654 32109876 54321098 76543210
\ 00000000 01000000 00000000 00000000 ok

```

It's done. But we took the 32-bit value **\$ffffffff** as a mask . The choice of this mask unfortunately allows an action on all the bits of the content of **myReg**. If we want to act only on one or more bits, we must choose a mask which limits the action of the word **m!** . For example, to set the single bit b07 to 1, you will need to use a mask. Here is this mask in binary:

00000000 00000000 00000000 **1** 00000000

Which translates in hexadecimal to **\$00000080** . If you have any doubts, you can check with **.binDisp**:

```
$00000080 .binDisp      \ display :  
\ 33222222 22221111 11111100 00000000  
\ 10987654 32109876 54321098 76543210  
\ 00000000 00000000 00000000 10000000 ok
```

This binary mask corresponds to bit b07. We will now set this bit to 1 in **myReg** without modifying the other bit b22 already at one:

```
registers  
1 7 $00000080 myReg m!  
forth  
myReg @ .binDisp      \ display :  
\ 33222222 22221111 11111100 00000000  
\ 10987654 32109876 54321098 76543210  
\ 00000000 01000000 00000000 10000000 ok
```

Among the parameters necessary for **m!** , we have the pair shift addr. To simplify the ESP32forth code, we will create a word **defMASK ::**

\define a mask for registers

```
\ define a mask for registers  
: defMASK: ( comp: mask0 position -- <name> | exec: -- position mask1 )  
    create  
        dup ,  
        lshift ,  
    does>  
        dup @  
        swap cell + @  
    ;
```

To define a mask, you only need two parameters:

- **mask0** : 1 is the minimum mask on one bit, 3 on 2 bits, 7 on 3 bits, etc...
- **position** : indicates the position, in the interval [0..31]

To modify for example the single bit b12, we can define our mask as follows :

```
1 12 defMASK: mB12
```

To set this single bit b12 to 1:

```
registers  
1 mB12 myREG m!  
forth  
myREG @ .binDisp      \ display :  
\ 33222222 22221111 11111100 00000000
```

```
\ 10987654 32109876 54321098 76543210
\ 00000000 01000000 00010000 10000000 ok
```

To reset this bit b12:

```
registers
0 mB12 myREG m!
forth
myREG @ .binDisp      \ display :
\ 33222222 22221111 11111100 00000000
\ 10987654 32109876 54321098 76543210
\ 00000000 01000000 00000000 10000000 ok
```

Masks defined by **defMASK:** are applicable to any register. This is what you will see later. These masks are also very useful for determining the state of a specific bit or bits. Let's test the state of our bit b12 with **m@** :

```
registers
mB12 myREG m@ .          \ display : 0
forth
```

Let's reset this bit b12 to 1 and test it again:

```
registers
1 mB12 myREG m!
1 mb12 myREG m@ .      \ display : 1
forth
```

We now have the keys to act bit by bit on the contents of any register. In this chapter, we will particularly focus on the **GPIO_OUT_REG** register :

```
\ GPIO 0-31 output register R/W
$3FF44004 defREG: GPIO_OUT_REG
```

The **GPIO_OUT_REG** register

This register allows you to control GPIO ports G0 to G31. We will wire three colored diodes on pins G25, G26 and G27. We therefore define three constants attached to these pins:

```
\ definie LEDs GPIOs
25 constant ledRED
26 constant ledYELLOW
27 constant ledGREEN
```

In the chapter *Managing a traffic light with ESP32* , we defined these same words with **defPIN:** . Here we do it with **constant**, which has the same behavior. We will use these constants to define the masks :

```
\ define masks for red yellow and green LEDs
1 ledRED      defMASK: mLED_RED
1 ledYELLOW   defMASK: mLED_YELLOW
```

```
1 ledGREEN    defMASK: mLED_GREEN
```

To activate the red LED on G25, enter:

```
registers
1 mLED_RED GPIO_OUT_REG m!
forth
```

To avoid the recurring selection of the **registers** vocabulary, we define two words which will simplify programming for us:

```
\ set mask in addr
: regSet ( val shift mask addr -- )
  [ registers ] m! [ forth ]
;

\ test mask in addr
: regTst ( shift mask addr -- val )
  [ registers ] m@ [ forth ]
;
```

Activation of the red LED on G25:

```
1 mLED_RED GPIO_OUT_REG regSet
```

This will work, provided the GPIO pins are correctly initialized. This is what we are going to address.

Activation and deactivation registers

The register names are taken from the *ESP32 Technical Reference Manual documentation* (pdf):

https://www.espressif.com/sites/default/files/documentation/esp32_technical_reference_manual_en.pdf

Before turning on and off our LEDs connected to GPIO ports G25 G26 and G27, we will start by initializing these ports. This is done by acting on the **GPIO_ENABLE_REG** register:

```
: GPIO.init ( -- )
  1 mLED_RED    GPIO_ENABLE_REG regSet
  1 mLED_YELLOW GPIO_ENABLE_REG regSet
  1 mLED_GREEN   GPIO_ENABLE_REG regSet
;
```

Executing the word **GPIO.init** initializes the output ports G25 G26 and G27. Let's test LEDs on :

```
GPIO.init
1 mled_red    GPIO_OUT_REG regSet
1 mled_yellow GPIO_OUT_REG regSet
1 mled_green   GPIO_OUT_REG regSet
```

If the LEDs are wired correctly, they should light up. Here the lighting of the LEDs is carried out sequentially by the word **regSet** repeated three times. By acting directly on the contents of the **GPIO_OUT_REG** register, we can light up the three LEDs in a single execution of **regSet** :

```
GPIO.init
7 mled_red
    mled_yellow nip +
    mled_green  nip + GPIO_OUT_REG regSet
```

Let's see how to use the two registers **GPIO_OUT_W1TS_REG** and **GPIO_OUT_W1TC_REG** to act indirectly on the state of one or more GPIO ports :

- **GPIO_OUT_W1TS_REG** : *GPIO Output Write to Set Register* is used to set the bits corresponding to the GPIO pins in "Output" mode to logic high (1). It allows you to activate the specified GPIO outputs, by setting the corresponding bits to 1.
- **GPIO_OUT_W1TC_REG** : *GPIO Output Write to Clear Register* is used to clear (zero) specific bits in the GPIO output register. Each bit of this register is associated with a particular GPIO pin, and **by writing a 1** to a given bit of this register, you can zero (clear) the corresponding output of the GPIO pin.

We can therefore light all our LEDs in a single **regSet sequence** like this :

```
7 mled_red
    mled_yellow nip +
    mled_green  nip + GPIO_OUT_W1TS_REG regSet
```

And to turn off all the LEDs in a single **regSet** sequence, we will execute this :

```
7 mled_red
    mled_yellow nip +
    mled_green  nip + GPIO_OUT_W1TS_REG regSet
```

If we want to manage a timed on and off cycle of an LED, we will create a word that manages an on and off cycle :

```
\ define a ON and OFF sequence for one LED
: GPIO.on.off.sequence { position mask delay -- }
    1 position mask GPIO_OUT_W1TS_REG regSet
    delay ms
    1 position mask GPIO_OUT_W1TC_REG regSet
;
```

We are going to test our word **GPIO.on.off.sequence**:

```
mLED_RED 1000 GPIO.on.off.sequence
```

This sequence should light the red LED for one second. Now let's define a complete cycle simulating a fire at a road intersection:

```

: traffic-light ( -- )
    mLED_GREEN 3000 GPIO.on.off.sequence
    mLED_YELLOW 1000 GPIO.on.off.sequence
    mLED_RED    3000 GPIO.on.off.sequence
;

```

Running **traffic-light** will simulate a classic road light. However, we cannot simulate a road light where the red and orange lights are on at the same time. We will therefore first write the word **TRAFFIC.sequence** which uses the code of **GPIO.on.off.sequence**, with the difference that we will also need to use a value in addition to the other parameters :

```

\ define a ON and OFF sequence
: TRAFFIC.sequence { val position mask delay -- }
    val position mask GPIO_OUT_W1TS_REG regSet
    delay ms
    val position mask GPIO_OUT_W1TC_REG regSet
;

```

Then we define these four words which will make it possible to manage a complex road fire cycle:

```

: TRAFFIC.red ( -- )
    1 mLED_RED    2500 TRAFFIC.sequence ;
: TRAFFIC.yellow ( -- )
    1 mLED_YELLOW 1000 TRAFFIC.sequence ;
: TRAFFIC.green ( -- )
    1 mLED_GREEN  3000 TRAFFIC.sequence ;
: TRAFFIC.red-yellow ( -- )
    3 mLED_RED
    mLED_YELLOW nip + 500 TRAFFIC.sequence ;

```

We now define a road light cycle as we might encounter it in Germany :

```

: TRAFFIC.german.cycle ( -- )
    TRAFFIC.red
    TRAFFIC.red-yellow
    TRAFFIC.green
    TRAFFIC.yellow
;

```

In Germany, traffic lights have four cycles. The particularity of these lights is to simultaneously light the red and yellow lights before switching back to the green light.

And finally, we test our road light cycle in a loop:

```

: TRAFFIC.loop ( -- )

```




Figure 18: four-state light cycle

```
begin
    TRAFFIC.german.cycle
    key? until
;
```

Running **TRAFFIC.loop** will simulate our road light. When the sequence **TRAFFIC.red-yellow** arrives , we can see that the yellow and red lights are lit simultaneously.

We saw in this chapter how to act on several GPIO outputs at the same time by acting directly on the GPIO registers.

However, with ESP32 cards, it is not recommended to manage too many LEDs simultaneously. The ESP32 card should be reserved for managing signals to very energy-efficient accessories. We will therefore avoid assemblies such as 8 to 16 LEDs to create a chase type effect for example, unless we use control elements using a separate power supply.

Hardware interrupts with ESP32forth

Interruptions

When we want to manage external events, a push button for example, we have two solutions :

- test the state of the button as regularly as possible, through a loop. We will act according to the state of this button.
- use an interrupt. We assign the execution code to an interrupt attached to a pin. The button is connected to this pin and the state change will execute this word.

The interrupt solution is the most elegant. It allows you to relieve the main program by avoiding monitoring the button in a loop.

In his ESP32forth documentation, Brad NELSON gives a simple example of interrupt handling :

```
17 input pinMode
: test ." pinvalue: " 17 digitalRead . cr ;
  ' test 17 pinchange
```

Except that this example, as written, has a good chance of not working. We will see why and provide the elements to make it work.

Mounting a push button

The button is connected as an input to the 3.3V power supply of the ESP32 board.

The push button output is connected to the GPIO17 pin. That's all.

For Brad NELSON's example to be functional, you must select the **interrupts vocabulary** before configuring the interrupt using **pinchange** . Along the way, we will define the **button constant** :

```
17 constant button
button input pinMode
: test ." pinvalue: "
  button digitalRead . cr
  ;
interrupts
' test button pinchange
forth
```

It works, but there is an unexpected effect which causes the interrupt to be triggered unexpectedly :

```
pinvalue: 0
pinvalue: 1
pinvalue: 1
pinvalue: 1
pinvalue: 0
pinvalue: 0
pinvalue: 0
pinvalue: 1
pinvalue: 1
pinvalue: 1
pinvalue: 0
pinvalue: 0
pinvalue: 1
```

The hardware solution would consist of putting a high value resistor at the button output and connected to GND.

Software consolidation of the interrupt

In the ESP32 card, you can activate a resistor on any GPIO pin. This activation is carried out by the word **gpio_pulldown_en** . This word accepts as a parameter the GPIO pin number whose resistance must be activated. In return this word returns 0 if the action was successful, an error code otherwise :

```
17 constant button
button input pinMode
: test ." pinvalue: "
    button digitalRead . cr
    ;
interrupts
button gpio_pulldown_en drop
' test button pinchange
forth
```

The result of executing the interrupt is significantly better :

```

ok      button digitalRead . cr
ok      ;
ok  interrupts
ok button gpio_pulldown_en drop
ok ' test button pinchange
ok forth
--> pinvalue: 1
pinvalue: 0
pinvalue: 1
pinvalue: 0
pinvalue: 1
pinvalue: 0

```

At each change of state, we have an interruption. On the screenshot above, each state change displays **pinvalue: 1** then **pinvalue: 0**.

It is possible to take into account an interruption on the rising edge alone. This is possible by indicating :

```
button GPIO_INTR_POSEDGE gpio_set_intr_type drop
```

The word **gpio_set_intr_type** accepts these parameters:


- **GPIO_INTR_ANYEDGE** to manage rising or falling edge interrupts
- **GPIO_INTR_NEGEDGE** to handle interrupts on falling edge only
- **GPIO_INTR_POSEDGE** to manage interrupts on rising edge only
- **GPIO_INTR_DISABLE** to disable interrupts

Complete Forth code with rising edge detection:

```

17 constant button
0 constant GPIO_PULLUP_ONLY
button input pinMode
: test ." pinvalue: "
    button digitalRead . cr
;
interrupts
button gpio_pulldown_en drop
button GPIO_INTR_POSEDGE gpio_set_intr_type
drop
' test button pinchange
forth

```



Further information

For ESP32, all GPIO pins can be used as interrupt except GPIO6 to GPIO11.

Do not use pins colored orange or red. Your program might behave unexpectedly when using these.

Using the KY-040 rotary encoder

Encoder Overview


To vary a signal, we have several solutions:

- a variable resistor in a potentiometer
- two buttons managing the variation by software
- a rotary encoder


The rotary encoder is an interesting solution. It can be made to act as a potentiometer, with the advantage of not having a start and end stop.

Its principle is very simple. Here are the signals emitted by our rotary encoder :

Here is our encoder :



Internal functioning diagram :




According to this diagram, two terminals interest us :

- A (DT) -> switch
- B (CLK) -> switch Y

This encoder can be powered with 5V or 3.3V. This suits us, because the ESP32 card has a 3.3V output.

Mounting the encoder on the breadboard

Wiring our encoder to the ESP32 board only requires 4 wires :




PLEASE NOTE : the position of pins G4 and G15 may vary depending on the version of your ESP32 card.

Analysis of encoder signals

As our encoder is connected, each terminal A or B receives a voltage, here 3.3V, the intensity of which is limited by a resistor of 10Kohms.

Analysis of the signal on terminal G15 clearly shows the presence of the 3.3V voltage :



In this signal capture, the low level on terminal G15 appears when operating the encoder control rod. When idle, the signal on terminal G15 is at high level.

This changes everything, because, at the programming level, we must process the G15 interrupt as a falling edge.

Encoder programming

The encoder will be managed by interrupt. Interrupts trigger the program only if a particular signal reaches a well-defined level.

We will manage a single interrupt on the GPIO G15 terminal:

```
interrupts

\ enable interrupt on GPIO G15
: intG15enable ( -- )
    15 GPIO_INTR_POSEDGE gpio_set_intr_type drop
;

\ disable interrupt on GPIO G15
: intG15disable ( -- )
    15 GPIO_INTR_DISABLE gpio_set_intr_type drop
;

: pinsInit ( -- )
    04 input pinmode          \ G04 as an input
    04 gpio_pulldown_en drop  \ Enable pull-down on GPIO 04
```

```

15 input pinmode          \ G15 as an input
15 gpio_pulldown_en drop \ Enable pull-down on GPIO 15
intG15enable
;

```

In the word **pinsInit** , we initialize the GPIO pins G4 and G15 as input. Then we determine the interrupt mode of G15 on falling edge with **15 GPIO_INTR_POSEDGE** **gpio_set_intr_type drop** .

Testing the encoding

This part of code is not to be used in a final assembly. It is only used to check that the encoder is correctly connected and working properly:

```

: test ( -- )
  cr ." PIN: "
  cr ." - G15: " 15 digitalRead .
  cr ." - G04: " 04 digitalRead .
;

pinsInit \ initialise G4 and G15
' test 15 pinchange

```

It is the '**test 15 pinchange**' sequence which tells ESP32Forth to execute the test code if an interrupt is triggered by action of terminal G15.

Result of the action on our encoder. We only kept the results of actions arriving at the stop, once counterclockwise, then clockwise:

```

PIN:
- G15: 1 \ reverse clockwise turn
- G04: 1
PIN:
- G15: 0 \ clockwise turn
- G04: 1

```

Increment and decrement a variable with the encoder

Now that we have tested the encoder by hardware interrupt, we will be able to manage the content of a variable. To do this, we define our variable **KYvar** and the words allowing us to modify its content:

```

0 value KYvar \ content is incremented or decremented

\ increment content of KYvar
: incKYvar ( n -- )
  1 +to KYvar
;

\ decrement content of KYvar
: decKYvar ( n -- )

```

```
-1 +to KYvar  
;
```

The word **incKYvar** increments the content of **Kyvar** . The word **deckKYvar** decrements the content of **KYvar** .

We test the modification of the content of the variable **KYvar** via this word **testIncDec** defined as follows:

```
\ used by interruption when G15 activated  
: testIncDec ( -- )  
    intG15disable  
    15 digitalRead if  
        04 digitalRead if  
            deckKYvar  
        else  
            incKYvar  
        then  
            cr ." KYvar: " KYvar .  
    then  
    1000 0 do loop \ small wait loop  
    intG15enable  
;  
  
pinsInit  
' testIncDec 15 pinchange
```

Turning the encoder control to the right (clockwise) will increment the contents of the KYvar variable. A rotation to the left decrements the content of the KYvar variable:

```
pinsInit  
' testIncDec 15 pinchange  
-->  
KYvar: 1\rotate Clockwise  
KYvar: 2  
KYvar: 3  
KYvar: 4  
KYvar: 3 \ rotate Contra Clockwise  
KYvar: 2  
KYvar: 1  
KYvar: 0  
KYvar: -1  
KYvar: -2
```

Flashing of an LED per timer

Getting started with Forth programming

Any beginner in programming knows this more than classic example very well: the flashing of an LED. Here is the source code, in C language for ESP32 :

```
/*
 * This ESP32 code is created by esp32io.com
 * This ESP32 code is released in the public domain
 * For more detail (instruction and wiring diagram),
 * visit https://esp32io.com/tutorials/esp32-led-blink
 */

// the code in setup function runs only one time when ESP32 starts
void setup() {
    // initialize digital pin GIOP18 as an output.
    pinMode(18, OUTPUT);
}

// the code in loop function is executed repeatedly infinitely
void loop() {
    digitalWrite(18, HIGH); // turn the LED on
    delay(500);           // wait for 500 milliseconds
    digitalWrite(18, LOW); // turn the LED off
    delay(500);           // wait for 500 milliseconds
}
```

In Forth language, it's not much different :

```
18 constant myLED

: led.blink ( -- )
    myLED output pinMode
    begin
        HIGH myLED pin
        500 ms
        LOW myLED pin
        500 ms
    key? until
;
```

If you compile this Forth code with ESP32forth installed on your ESP32 board and type **led.blink** from the terminal, the LED connected to the GPIO18 port will blink.

To inject code written in C language, it will be necessary to compile it on the PC, then upload it to the ESP32 card, operations which take some time. Whereas with the Forth language, the compiler is already operational on our ESP32 board. The compiler will compile the program written in Forth language in two to three seconds and allow its

immediate execution by simply typing the word containing this code, here **led.blink** for our example.

In Forth language, we can compile hundreds of words and test them immediately, all individually, which is not possible at all in the C language.

We factor our Forth code like this :

```
18 constant myLED

: led.on ( -- )
    HIGH myLED pin
;

: led.off ( -- )
    LOW myLED pin
;

: waiting ( -- )
    500 ms
;

: led.blink ( -- )
    myLED output pinMode
    begin
        led.on      waiting
        led.off     waiting
        key? until
;
```

From the terminal, we can simply turn the LED on by typing **led.on** and turn it off by typing **led.off** . Execution of **led.blink** remains possible.

The aim of factorization is to divide a complex and difficult to read function into a set of simpler and more readable functions. With Forth, factorization is recommended, on the one hand to allow easier debugging, and on the other hand to allow the reuse of factored words.

These explanations may seem trivial to those who know and master the Forth language. This is far from obvious for people programming in C, who are forced to group function calls into the general **loop() function** .

Now that this is explained, we will forget everything ! Because...

Flashing by TIMER

We will forget everything that was explained previously. Because this LED blinking example has a huge downside. Our program does just that and nothing else. In short, it's

a real waste of hardware and software to flash an LED on our ESP32 card. We will see a very different way of producing this flashing, in Forth language exclusively.

ESP32forth has two words that will be very useful to manage this LED flashing: **interval** and **rerun** .

But before discussing how these two words work, let's take a look at the notion of interruption...

Hardware and software interrupts

If you plan to manage microcontrollers without worrying about hardware or software interrupts, then abandon computer development for ESP32 boards!

You have the right to start and not experience interruptions. And we're going to explain interrupts to you and how to use timer interrupts.

Here is a non-computer example of what an interrupt is:

- you are expecting an important package;
- you go down to the gate of your home every minute to see if the postman has arrived.

In this scenario, you actually spend your time going down, looking, back up. In fact, you hardly have time to do anything else...

In reality, this is what should happen :

- you stay in your home ;
- the postman arrives and rings the doorbell ;
- you go down and collect your package...

A microcontroller, which includes the ESP32 card, has two types of interrupts :

- **hardware interrupts** : they are triggered by a physical action on one of the GPIO inputs of the ESP32 card ;
- **software interrupts** : they are triggered if certain registers reach pre-defined values.

This is the case for timer interrupts, which we will define as software interrupts.

Use the words interval and rerun

The word **interval** is defined in the **timers vocabulary** . It accepts three parameters :

- **xt** which executes the code for the word to be thrown when the interrupt is triggered ;
- **usec** is the wait time, in microseconds, before triggering the interrupt ;
- **t** is the number of the timer to trigger. This parameter must be in the range [0..3]

Let's partially take the factored code of our LED flashing :

```
18 constant myLED
0 value LED_STATE
: led.on ( -- )
    HIGH dup myLED pin
    to LED_STATE
;
: led.off ( -- )
    LOW dup myLED pin
    to LED_STATE
;
timers \ select timers vocabulary
: led.toggle ( -- )
    LED_STATE if
        led.off
    else
        led.on
    then
    0 rerun
;
' led.toggle 500000 0 interval
: led.blink
    myLED output pinMode
    led.toggle
;
```

The word **rerun** is preceded by the number of timer activated before the definition of **interval** . The word **rerun** must be used in the definition of the word executed by the timer.

The word **led.blink** initializes the GPIO output used by the LED, then executes **led.toggle**.

In this sequence Forth '`led.toggle 500000 0 interval`' , we initialize timer 0 by recovering the execution code of the word using `rerun` , followed by the time interval, here 500 milliseconds, then the number of the timer to trigger.

The LED flashing starts immediately after execution of the word `led.blink` .

The Forth interpreter of ESP32forth remains accessible while the LED flashes, something impossible in C language!

Housekeeper timer

Preamble

It's 1990. He's a computer programmer who works a lot. So he sometimes leaves his office a little late.

And it was during one of his late exits from the office that he entered the corridor, one of those corridors with a timer button at each end. The light is already on. But out of reflex, our programmer friend presses the switch and pricks his finger. A wooden point is inserted into the switch to block the timer.



It's the cleaning lady who is cleaning the floor who explains to him: "yes. The timer only lasts one minute. And I often find myself in the dark. As I'm tired of pressing again without stop on the timer switch, I block the button with this little wooden point"...

A solution

This anecdote sparked an idea in our programmer's head. As he had some knowledge about microcontrollers, he set out to find a solution for the cleaning lady.

History does not say in what language he programmed his solution. Certainly in assembler.

He derived the control of the lights to his circuit :

- an ordinary press starts the timer for one minute ;
- if the light is on, any brief press of a button reduces the ignition delay to one minute ;
- our programmer's secret is to have planned a long press of 3 seconds or more. This long press starts the timer for 10 minutes of lighting ;
- if the timer is in long circuit, another long press reduces the timer delay to one minute ;

- a short beep acknowledges the activation or deactivation of a long timer cycle.

The cleaning lady really appreciated this improvement in the timer. She no longer needed to block the button in any way.

What about the other workers? Since no one was aware of this feature, they continued to use the timer by briefly pressing the activation switch.

A Forth timer for ESP32Forth

You understand, we are going to use **timers** to manage a timer by integrating the scenario described previously.

```
\ myLIGHTS connecté à GPIO18
18 constant myLIGHTS

\ définit temps max pour cycle normal ou étendu, en secondes
60 constant MAX_LIGHT_TIME_NORMAL_CYCLE
600 constant MAX_LIGHT_TIME_EXXTENDED_CYCLE

\ temps max pour cycle normal ou étendu, en secondes
0 value MAX_LIGHT_TIME

timers
\ coupe éclairage si MAX_LIGHT_TIME égal 0
: cycle.stop ( -- )
    -1 +to MAX_LIGHT_TIME      \ décrémente temps max de 1 seconde
    MAX_LIGHT_TIME 0 = if
        LOW myLIGHTS pin      \ coupe éclairage
    else
        0 rerun
    then
;

\ initialise timer 0
' cycle.stop 1000000 0 interval

\ démarre un cycle d'éclairage, n est délai en secondes
: cycle.start ( n -- )
    1+ to MAX_LIGHT_TIME      \ sélect. Temps max
    myLIGHTS output pinMode
    HIGH myLIGHTS pin         \ active éclairage
    0 rerun
;
```

We can already test our timer :

```
3 cycle.start      \ turns on lights for 3 seconds
10 cycle.start     \ turns on lights for 10 seconds
```

If we restart **cycle.start** while the light is on, we start again for a new lighting cycle of n seconds.

We therefore still have to manage the activation of these cycles from a switch.

Management of the light on button

This is not rocket science. We will manage a push button. As we have an ESP32 card on hand, programmable with ESP32Forth, we will take advantage of it to manage this button by interrupts. The interrupts managing the GPIO terminals on the ESP32 board are hardware interrupts.

Our button is mounted on the GPIO17 (G17) terminal.

We define two words, **intPosEdge** and **intNegEdge**, which determine the type of triggering of the interrupt:

- **intPosEdge** to trigger the interrupt on rising edge;
- **intNegEdge** to trigger the interrupt on falling edge.

```
17 constant button \ mount button on GPIO17

interrupts           \ select interrupts vocabulary

\ interrupt activated for upraising signal
: intPosEdge ( -- )
    button #GPIO_INTR_POSEDGE gpio_set_intr_type drop
;

\ interrupt activated for falldown signal
: intNegEdge ( -- )
    button #GPIO_INTR_NEGEDGE gpio_set_intr_type drop
;
```

We then need to define some variables and constants:

- two constants, **CYCLE_SHORT** and **CYCLE_LONG** which will be used to define the duration of lighting of the lights. Here we chose 3 and 10 seconds to do our tests ;
- the **msTicksPositiveEdge variable** which stores the position of the wait counter delivered by ms-ticks ;
- **DELAY_LIMIT** constant which determines the threshold for determining a short or long press on the push button. Here, it's 3000 milliseconds, or 3 seconds. A normal user will NEVER press the light on button for 3 seconds. Only the cleaning lady knows the maneuver to have a long continuous lighting...

```
03 constant CYCLE_SHORT      \ lighting duration for short press, in seconds
10 constant CYCLE_LONG       \ lighting duration for long press

\ stores value of ms-ticks on rising edge
variable msTicksPositiveEdge
```

```
\ deadline limit: if delay < DELAY_LIMIT, short cycle  
3000 constant DELAY_LIMIT
```

The word **getButton** is launched at each interrupt triggered by pressing the push button connected to GPIO17 (G17) on our ESP32 board.

getButton execution , interrupts on G17 are disabled. This interruption will be reactivated at the end of the definition. This disabling is necessary to prevent interrupt stacking.

Disabling is followed by the **70000 0 do loop** . This loop is used to manage contact bounces. Here we manage the debounce by software.

```
\ word executed by interrupt  
: getButton ( -- )  
    button gpio_intr_disable drop  
    70000 0 do loop  \ anti rebond  
    button digitalRead 1 =  
    if  
        ms-ticks msTicksPositiveEdge !  
        intNegEdge  
    else  
        intPosEdge  
        ms-ticks msTicksPositiveEdge @ -  
        DELAY_LIMIT >  
        if      CYCLE_LONG cr ." BEEP"  
        else    CYCLE_SHORT cr ." ----"  
    then  
    cycle.start  
    button gpio_intr_enable drop  
;
```

On the rising edge, the word **getButton** records the state of the delay counter and positions the interrupts on the falling edge. Then we leave this word by reactivating the interruptions.

At the falling edge, the word **getButton** calculates the time elapsed since the rising edge. If this delay is greater than **DELAY_LIMIT** , a long ignition cycle is initiated. Otherwise, a short ignition cycle is initiated.

The engagement of a long ignition cycle is indicated by the display of “BEEP” on the terminal.

In the original scenario, this is materialized by a short beep.

Finally, we initialize the button and the hardware interrupt on this button:

```
\ initialize button and interrupt vectors
```

```
button input pinMode          \ selects G17 in input mode
button gpio_pulldown_en drop \ activates internal resistance of G17
' getButton button pinchange
intPosEdge

forth
```

Conclusion

Watch the assembly video: https://www.youtube.com/watch?v=OHWMh_bIWz0

This very simple case study shows how to simultaneously manage the timer and a hardware interrupt.

These two mechanisms are very little preemptive. The timer leaves access to the Forth interpreter available. The hardware interrupt is operational even if Forth is running another process.

We don't multitask. It's important to say it!

I only hope that this textbook case will now give you a lot of ideas for your developments...

Software real-time clock

The word MS-TICKS

The word **MS-TICKS** is used in the definition of the word **ms** :

```
DEFINED? ms-ticks [IF]
: ms ( n -- )
    ms-ticks >r
    begin
        pause ms-ticks r@ - over
    >= until
    rdrop drop
;
[THEN]
```

This word **MS-TICKS** is at the heart of our investigations. If we start up the ESP32 card, its execution restores the number of milliseconds elapsed since the ESP32 card was started up. This value is still growing. The saturation value of this count is $2^{32}-1$, or 4294967295 milliseconds, or approximately 49 days...

Each time the ESP32 card is restarted, this value restarts from zero.

Managing a software clock

From the **HH MM SS** (Hours, minutes, seconds) data, it is easy to reconstruct an integer value, in milliseconds, corresponding to the time elapsed since 00:00:00. If we subtract the value of **MS-TICKS** from this time , we have a starting time value to determine the real time . We therefore initialize a basic counter **currentTime** from the word **RTC.set-time** :

```
0 value currentTime

\ store current time
: RTC.set-time { hh mm ss -- }
    hh 3600 *
    mm 60 *
    ss + + 1000 *
    MS-TICKS - to currentTime
;
```

Initialization example: **22 52 00 RTC.set-time** initializes the time base for 22:52:00...

To properly initialize, prepare the three values **HH MM SS** followed by the word **RTC.set-time**, watch your watch. When the expected time arrives, execute the initialization sequence.

HH MM and **SS** values of the current time, using this word :

```
\ get current time in seconds
: RTC.get-time ( -- hh mm ss )
    currentTime MS-TICKS + 1000 /
    3600 /mod swap 60 /mod swap
;
```

Finally, we define the word **RTC.display-time** which allows you to display the current time after initialization of our software clock:

```
\ used for SS and MM part of time display
: :## ( n -- n' )
    # 6 base ! # decimal [char] : hold
;

\ display current time
: RTC.display-time ( -- )
    currentTime MS-TICKS + 1000 /
    <# :## :## 24 MOD #S #> type
;
```

The next step would be to connect to a time server, with the NTP protocol, to automatically initialize our software clock.

Measuring the execution time of a Forth word

Measuring the performance of Forth definitions

Let's start by defining the word **measure:** which will perform these execution time measurements:

```
: measure: ( exec: -- <word> )
    ms-ticks >r
    ' execute
    ms-ticks r> -
    cr ." execution time: "
    <# # # # [char] . hold #s #> type ." sec." cr
;
```

In this word, we retrieve the time by **ms-ticks** , then we retrieve the execution code of the word that follows **measure:**, we execute this word, we retrieve the new time value by **ms-ticks** . We make the difference, which corresponds to the elapsed time, in milliseconds, taken by the word to execute. Example :

```
measure: words
\ displays: execution time: 0.210sec.
```

The word **words** was executed in 0.2 seconds. This time does not take into account transmission delays by the terminal. This time also does not take into account the delay taken by **measure:** to retrieve the execution code of the word to be measured.

If there are parameters to pass to the word to measure, these must be stacked before calling **measure:** followed by the word to measure:

```
: SQUARE ( n -- n-exp2 )
    dup *
;
3 measure: SQUARE
\ poster:
\ execution time: 0.000sec.
```

This result means that our **SQUARE definition** runs in less than a millisecond.

We will repeat this operation a certain number of times:

```
: test-square ( -- )
    1000 for
        3 SQUARE drop
    next
;
3 measure: test-square
```

```
3 measure: test-square
\ poster:
\ execution time: 0.001sec.
```

By executing the word **SQUARE** 1000 times , preceded by a stacking of values and unstacking of the result, we arrive at an execution time of 1 millisecond. We can reasonably deduce that **SQUARE** executes in less than a micro-second!

Testing a few loops

We are going to test a few loops, with 1 million iterations. Let's start with a **do-loop** :

```
: test-loop ( -- )
  1000000 0 do
    loop
  ;
measure: test-loop
\ display:
\ execution time: 1.327sec.
```

Now let's see with a **for-next loop** :

```
: test-for ( -- )
  1000000 for
  next
  ;
measure: test-for
\ displays:
\ execution time: 0.096sec.
```

The **for-next loop** runs almost 14 times faster than the **do-loop**.

Let's see what a **begin-until** loop has in store:

```
: test-begin ( -- )
  1000000 begin
    1- dup 0=
    until
  ;
measure: test-begin
\ displays:
\ execution time: 0.273sec.
```

This is more efficient than the **do-loop** , but still three times slower than the **for-next loop** .

You are now equipped to create even more efficient Forth programs.

Program a sunshine analyzer

Preamble

As part of a solar project using several solar panels and their microinverter, there appear some problems with managing the electrical energy produced.

The main concern is to activate large consumer devices only if the solar panels produce in full sun. One device in particular is concerned, the hot water cumulus:

- activate the device when the panels are in direct sunlight;
- deactivate the device when clouds pass.

Microinverters inject power into the general electrical grid. If a device that consumes a lot of electricity is active when clouds pass, this device will be powered primarily by the general network.

In this article, we present a solution enabling cloud detection using a miniature solar panel and an ESP32 card.

Full code available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/ADC/solarLightAnalyzer.txt>

The miniature solar panel


To create our cloud detector, we will use a very small solar panel, here a 25mm x 25mm panel.

Recovery of a miniature solar panel

This miniature solar panel is recovered from a garden lamp that is out of order :



Here is our mini solar panel taken out of this garden lamp :



We sacrifice two dupont connectors to allow various measurements to be carried out on our prototype plate. These connectors are soldered onto the two red and blue wires coming out of the mini solar panel.

Measurement of solar panel voltage

We start by measuring the no-load voltage of our mini solar panel, here with an oscilloscope. This voltage measurement can also be carried out with a voltmeter :



In bright light, the measured voltage amounts to 14.2 Volts!

Under diffused light, the voltage drops to 5.8 Volts.

By covering the mini solar panel with your hand, the voltage drops to almost 0 Volts.


Solar panel current measurement

The current, i.e. the intensity, must be measured using an ammeter. The ammeter function of a universal controller will be suitable. Short-circuiting the mini solar panel in bright light allows a current of 10 mA to be measured.

Our mini solar panel therefore has an approximate power of 0.2 Watt.

Before connecting our mini solar panel to the ESP32 card, it is essential to lower the output voltage. It is out of the question to inject this voltage of 14.2 Volts into an input of the ESP32 card. Such voltage would destroy the internal circuitry of the ESP32 board.

Lowering the solar panel voltage



The idea is to lower the voltage across our mini solar panel. After a few tests, we choose two resistors, one of 220 Ohms, the other of 1K Ohms. Mounting these resistors:


The voltage measurement is taken between the two resistors and the positive terminal of the solar panel.

The voltmeter now indicates a maximum voltage of 3.2V in bright light, a voltage of 0.35V in diffused light.

Programming the solar analyzer

The ESP32 board has 18 12-bit channels enabling analog-to-digital conversion (ADC). To analyze the voltage of our mini solar panel, a single ADC channel is necessary and sufficient.

Only 15 ADC channels are available:



We will use one, the **ADC1_CH6 channel** which is attached to pin **G34** :

```
34 constant SOLAR_CELL  
  
: init-solar-cell ( -- )  
    SOLAR_CELL input pinMode  
    ;  
  
init-solar-cell
```

To read the voltage at the point between the two resistors, simply run **SOLAR_CELL analogRead** . This sequence drops a value between 0 and 4095. The lowest value corresponds to zero voltage. The highest value corresponds to a voltage of 3.3 Volts.

solar-cell-read definition to recover this voltage:

```
: solar-cell-read ( -- n )  
    SOLAR_CELL analogRead  
    ;
```

Let's test this definition in a loop :

```
: solar-cell-loop ( -- )  
    init-solar-cell  
    begin  
        solar-cell-read cr .  
        200 ms  
        key? until  
    ;
```

When running **solar-cell-loop** , every 200 milliseconds, the ADC voltage conversion value is displayed:

```
...  
322  
331  
290  
172  
39  
0  
0  
0  
0  
19  
79  
86  
...
```

Here the values were obtained by illuminating the mini solar panel with a high power lamp. Zero values correspond to the absence of lighting.

Tests with the real sun show measurements exceeding 300.

Managing activation and deactivation of a device

To begin, we will define two pins, one pin reserved for managing an activation signal, the other for a deactivation signal:

- G17 pin connected to a green LED. This pin is used to activate a device.
- G16 pin connected to a red LED. This pin is used to deactivate a device.

```
17 constant DEVICE_ON      \ green LED
16 constant DEVICE_OFF     \ red LED

: init-device-state ( -- )
    DEVICE_ON  output pinMode
    DEVICE_OFF output pinMode
;
```

We could have used a single pin to manage the remote device. But some devices, like bistable relays, have two coils :

- the first coil is powered so that the contacts switch. The state does not change when the coil is no longer energized ;
- to return to the initial state, the second coil is powered.

For this reason, our programming will take this type of device into account.

```
\ define trigger high state delay
500 value DEVICE_DELAY

\ set HIGH level of trigger
: device-activation { trigger -- }
    trigger HIGH digitalWrite
    DEVICE_DELAY ?dup
    if
        ms
        trigger LOW digitalWrite
    then
;
```

Here, the pseudo-constant **DEVICE_DELAY** is used to indicate the delay during which the control signal should be kept high. After this time, the control signal returns to the low state.

If the value of **DEVICE_DELAY** is zero, the control signal remains high.

It is the word **trigger-activation** which manages the activation of the corresponding pin:

- **TRIGGER_ON trigger-activation** permanently or transiently sets the pin attached to the green LED high;

- **TRIGGER_OFF trigger-activation** sets the pin attached to the red LED permanently or transiently high.

We now define two words, **device-ON** and **device-OFF** , respectively responsible for activating and deactivating the device intended to be controlled by pins G16 and G17:

```
\ define device state: 0=LOW, -1=HIGH
0 value DEVICE_STATE

: enable-device ( -- )
  DEVICE_STATE invert
  if
    DEVICE_OFF LOW digitalWrite
    DEVICE_ON device-activation
    -1 to DEVICE_STATE
  then
;

: disable-device ( -- )
  DEVICE_STATE
  if
    DEVICE_ON LOW digitalWrite
    DEVICE_OFF device-activation
    0 to DEVICE_STATE
  then
;
```

The device state is stored in **DEVICE_STATE** . This state is tested before attempting to change state. If the device is active, it will not be reactivated repeatedly. Same if the device is inactive.

```
\ define trigger value for sunny or cloudy sky
300 value SOLAR_TRIGGER

\ if solar light > SOLAR_TRIGGER, activate action
: action-light-level ( -- )
  solar-cell-read SOLAR_TRIGGER >=
  if
    enable-device
  else
    disable-device
  then
;
```

Triggered by timer interrupt

The most elegant way is to use a timer interrupt. We will use timer 0:

```
0 to DEVICE_DELAY
200 to SOLAR_TRIGGER
init-solar-cell
init-device-state

timers
```

```

: action ( -- )
    action-light-level
    0 rerun
;

' action 1000000 0 interval

```

From now on, the timer will analyze the light flow every second and act accordingly. Link to video: <https://youtu.be/lAjeev2u9fc>

For this video, we act on two parameters:

- **0 to DEVICE_DELAY** lights the LEDs permanently. The red LED indicates that the device is deactivated. The green LED indicates device activation;
- **200 to SOLAR_TRIGGER** determines the threshold for triggering the sunshine state. This parameter is adjustable to adapt to the characteristics of the mini solar panel.

The **action word** works by timer interrupt. It is therefore not necessary to have a general loop for the detector to work.

Devices controlled by the sunshine sensor

In summary, we have two control wires, one wire corresponding to the green LED in the video, the other wire corresponding to the red LED. The program is designed so that both control wires cannot be active at the same time.

To have a continuous signal on either control wire, the **DEVICE_DELAY value only needs to** be zero. Here is how to initialize this scenario :

```

\ start with Constant Command Signal
: start-CCS ( -- )
    0 to DEVICE_DELAY
    200 to SOLAR_TRIGGER
    init-solar-cell
    init-device-state
    disable-device
    [ timers ] ['] action 1000000 0 interval
;

```

And to have timed commands, we will assign to **DEVICE_DELAY** the delay of the level of the activation or deactivation command of the device.

```

\ start with Temporized Command Signal
: start-TCS ( -- )
    300 to DEVICE_DELAY
    200 to SOLAR_TRIGGER
    init-solar-cell
    init-device-state

```

```
disable-device  
[ timers ] ['] action 1000000 0 interval  
;
```

start-TCS scenario is typical of a pulse-operated bistable relay control. The relay activates if it receives an activation command. Even if the activation signal falls, the bistable relay remains active. To deactivate the bistable relay, a deactivation command must be transmitted to it on the deactivation line.

In conclusion, our solar light analyzer can control a wide variety of devices. It is enough to adapt the control interfaces of these devices to the characteristics of the GPIO ports of the ESP32 card.

Management of N/A (Digital/Analog) outputs


Digital/analog conversion

The conversion of a digital quantity into an electrical voltage proportional to this digital quantity is a very interesting functionality on a micro-controller.

When you use the Internet and make a VOIP phone call, your voice is transformed into numerical values. That of your correspondent will be inversely transformed from numbers to sound signals. This process uses analog to digital conversion and vice versa.

D/A conversion with R2R circuit

Here is the basic diagram of a 4-bit digital to analog converter :



The value to convert, on 4 bits, is distributed over 4 pins a_0 to a_3 . The reference voltage is injected at the top left of the circuit. This voltage generates an intensity $2I$ if this current does not pass through any resistance.

Depending on the activated bits, for each bit the voltage is divided and added to that of the other active bits. For example, if a_2 and a_0 are active, the output current I_s will be the sum $I/2$ and $I/8$.

For this 4-bit circuit, the conversion step is $I/16$. With ESP32, the conversion is done on 8 bits. The conversion step will therefore be $I/256$.

D/A conversion with ESP32

No ARDUINO board has a D/A conversion output. To perform a D/A conversion with an ARDUINO board, you must use an external component.

With the ESP32 card, we have two pins, G25 and G26, corresponding to D/A conversion outputs.

For our first D/A conversion experiment with the ESP32 board, we will connect two LEDs to pins G25 and G26:

```
\ define Gx to LEDs
25 constant ledBLUE      \ blue led on G25
26 constant ledWHITE     \ white led on G26
```

Before performing a D/A conversion, we plan to initialize pins G25 and G26:


```
\ init Gx as output
: initLeds ( -- )
    ledBLUE  output pinMode
    ledWHITE output pinMode
;
```

And we define two words allowing us to control the intensity of our two LEDs:

```
\ set intensity for BLUE led
: BLset ( val -- )
    ledBLUE  swap dacWrite
;

\ set intensity for WHITE led
: WHset ( val -- )
    ledWHITE swap dacWrite
;
```

The words **BLset** and **WHset** accept as parameters a numeric value in the range 0..255.



In the photo, after **initLeds**, the sequence **200 BLset** lights the blue LED at reduced power.

To turn it on at full power, we will use the sequence **255 BLset**

To turn it off completely, we will send this sequence **0 BLset**

Possibilities of D/A conversion

Here, with our two LEDs, we have created a simple and uninteresting assembly.

This montage has the merit of showing that the D/A conversion works perfectly. The D/A conversion allows:

- power control through a dedicated circuit, a variator for an electric motor for example;
- generation of signals: sinusoid, square, triangle, etc...
- sound file conversion
- sound synthesis...

Full code available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/DAC/DAoutput.txt>

Installing the OLED library for SSD1306


Since ESP32forth version 7.0.7.15, the options are available in the optional folder :

Téléchargements > ESP32forth-7.0.7.15(1).zip > ESP32forth > optional	
Nom	Type
assemblers.h	Fichier H
camera.h	Fichier H
interrupts.h	Fichier H
oled.h	Fichier H
README-optional.txt	Document texte
rmt.h	Fichier H
serial-bluetooth.h	Fichier H
spi-flash.h	Fichier H

To have the **oled vocabulary**, copy the **oled.h** file to the folder containing the **ESP32forth.ino** file.

Then launch ARDUINO IDE and select the most recent **ESP32forth.ino** file.

If the OLED library has not been installed, in ARDUINO IDE, click *Sketch* and select *Include Library*, then select *Manage Libraries*.



In the left sidebar, look for the **Adafruit SSD1306 by Adafruit** library.

You can now start compiling the sketch by clicking on *Sketch* and selecting *Upload*.

Once the sketch is uploaded to the ESP32 board, launch the TeraTerm terminal. Check that the **OLED** vocabulary is present :

```
oled vlist \ display:  
OledInit SSD1306_SWITCHCAPVCC SSD1306_EXTERNALVCC WHITE BLACK OledReset  
HEIGHT WIDTH OledAddr OledNew OledDelete OledBegin OledHOME OledCLS OledTextc  
OledPrintln OledNumln OledNum OledDisplay OledPrint OledInvert OledTextsize  
OledSetCursor OledPixel OledDrawL OledCirc OledCircF OledRect OledRectF  
OledRectR OledRectRF oled-builtins
```

The I2C interface on ESP32


Introduction

I2C (means: Inter-Integrated Circuit, in English) is a computer bus which emerged from the "war of standards" launched by players in the electronic world. Designed by Philips for home automation and domestic electronics applications, it makes it possible to easily connect a microprocessor and various circuits, notably those of a modern television: remote control receiver, low frequency amplifier settings, tuner, clock, management of scart socket, etc.

There are countless peripherals using this bus, it can even be implemented by software in any microcontroller. The weight of the consumer electronics industry has enabled very low prices thanks to these numerous components.

This bus is sometimes called TWI (Two Wire Interface) or TWSI (Two Wire Serial Interface) by certain manufacturers.

Exchanges always take place between a single master and one (or all) slave(s), always at the initiative of the master (never from master to master or from slave to slave). However, nothing prevents a component from going from master to slave status and vice versa.




The connection is made via two lines:

- SDA (Serial Data Line): bidirectional data line,
- SCL (Serial Clock Line): bidirectional synchronization clock line.

We must not forget the mass which must be common to the equipment.

Both lines are pulled at voltage level VDD through pull-up resistors (RP).




OLED display connected to the I2C bus

Master slave exchange

The message can be broken down into two parts:

- The master is the transmitter, the slave is the receiver:
 - emission of a START condition by the master ("S"),
 - transmission of the address byte or bytes by the master to designate a slave, with the R/W bit at 0 (see the section on addressing below),
 - response from the slave with an ACK acknowledgment bit (or NACK non-acknowledgement bit),
 - after each acknowledgment, the slave can request a pause ("PA").
 - emission of a command byte by the master for the slave,
 - response from the slave with an ACK acknowledgment bit (or NACK non-acknowledgement bit),
 - emission of a RESTART condition by the master ("RS"),
 - transmission of the address byte or bytes by the master to designate the same slave, with the R/W bit at 1.
 - response from the slave with an ACK acknowledgment bit (or NACK non-acknowledgement bit).
- The master becomes a receiver, the slave becomes a transmitter:
 - emission of a data byte by the slave for the master,
 - response from the master with an ACK acknowledgment bit (or NACK non-acknowledgement bit),

- transmission of other data bytes by the slave with acknowledgment from the master,
- for the last byte of data expected by the master, it responds with a NACK to end the dialogue,
- emission of a STOP condition by the master ("P").



Start condition : The SDA line goes from a high voltage level to a low voltage level before the SCL line goes from high to low.

Shutdown condition : The SDA line changes from a low voltage level to a high voltage level after the SCL line changes from low to high.

Address frame : a unique 7 or 10 bit sequence for each slave that identifies the slave when the master wants to talk to it.

Read/Write Bit : A single bit specifying whether the master is sending data to the slave (low voltage level) or requesting data from it (high voltage level).

ACK/NACK bit : each frame of a message is followed by an acknowledgment/non-acknowledgment bit. If an address frame or data frame has been successfully received, an ACK bit is returned to the sender.

Addressing

I2C doesn't have slave select lines like SPI, so it needs another way to let the slave know that data is being sent to it, and not another slave. It does this by addressing. The address frame is always the first frame after the start bit in a new message.

The master sends the address of the slave with which it wants to communicate to each slave connected to it. Each slave then compares the address sent by the master to its own. If the address matches, it returns a low voltage ACK bit to the master. If the address does not match, the slave does nothing and the SDA line remains high.

This is how the **Wire.detect** word detects devices connected to the i2c bus.

You can connect several different devices to the i2c bus. You cannot connect several copies of the same device to the same i2c bus.

Setting GPIO ports for I2C

Setting up the GPIO ports for the I2C bus is very simple:

```
\activate the wire vocabulary
wire
\ start the I2C interface using pin 21 and 22 on ESP32 DEVKIT V1
\ with 21 used as sda and 22 as scl.
21 22 wire.begin
```

I2C bus protocols

The dialogue is only between a master and a slave. This dialogue is always initiated by the master (Start condition): the master sends the address of the slave with whom it wants to communicate on the I2C bus.

The dialogue is always terminated by the master (Stop condition).

The clock signal (SCL) is generated by the master.

Detecting an I2C device

This part is used to detect the presence of a device connected to the I2C bus.

You can compile this code to test for the presence of connected and active modules on the I2C bus.

```
\ activates wire vocabulary
wire
\ starts I2C interface using pin 21 and 22 on ESP32 DEVKIT V1
\ with 21 for sda and 22 for scl.
21 22 wire.begin

: spaces ( n -- )
  for
    space
  next
;

: .## ( n -- )
  <# # # #> type
;

\ not all bitpatterns are valid 7bit i2c addresses
: Wire.7bitaddr? ( a -- f )
  dup $07 >=
  swap $77 <= and
;
```

```

: Wire.detect  ( -- )
    base @ >r hex
    cr
    ."      00 01 02 03 04 05 06 07 08 09 0a 0b 0c 0d 0e 0f"
$80 $00 do
    i $0f and 0=
    if
        cr i .## ." : "
    then
    i Wire.7bitaddr? if
        i Wire.beginTransmission
        -1 Wire.endTransmission 0 =
        if
            i .## space
        else
            ." -- "
        then
    else
        2 spaces
    then
loop
cr r> base !
;

```

Here, running the word **Wire.detect** indicates the presence of the OLED display device at hexadecimal address **3c**:

```

Wire.detect
    00 01 02 03 04 05 06 07 08 09 0a 0b 0c 0d 0e 0f
00 :
10 :
20 :
30 : -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- -- 3c
40 :
50 :
60 :
70 :


```

Here we detected a module at hexadecimal address 3c. This is the address that we will use to address this module..... @TODO: to be completed

The SSD1306 OLED display

The OLED display exists in two definitions:

- 128 x 64 pixels, monochrome or colored screen. If the screen is colored, the pixels remain monochrome.
- 128 x 32 pixels, monochrome screen.



These displays are available in SPI or I2C interface.

Favor the I2C interface which allows the connection of several I2C interfaces on the same I2C device. The **OLED** vocabulary is designed to manage transmission via I2C to these OLED displays.

Choosing a display interface

The choice of a display interface is subject to several conditions:


- his price;
- its electricity consumption;
- its robustness;
- its ease of programming and use.

A display interface is very useful on a stand-alone setup to provide very clear textual or graphic information.

After several researches, the choice fell on an OLED display of this type,

It only costs a few euros.

This display uses OLED technology, therefore without backlighting.



Its display resolution is 128x32 pixels. It can display text and images, but only in monochrome.

In **DISPLAYOFF** mode , power consumption is almost zero.

It is a very widespread and rather well documented product.

Online documentation


- Adafruit: OLED display technical documentation and controls
<https://cdn-shop.adafruit.com/datasheets/SSD1306.pdf>
- Adafruit: SSD1306 C library
https://adafruit.github.io/Adafruit_SSD1306/html/files.html
- Adafruit: SSD1306 microPython
<https://github.com/adafruit/micropython-adafruit-ssd1306>
- Punyforth: SSD1306 SPI Forth
<https://github.com/zeroflag/punyforth/blob/master/arch/esp8266/forth/ssd1306-spi.forth>
- TG9541: Forth Oled Display
<https://github.com/TG9541/forth-oled-display/blob/master/ssd1306.fs>
- Yunfan: SSD1306 128x32 i2c forth
<https://gist.github.com/yunfan/2d3ee14697f3ebd3cb43ae411216d9aa>

Connecting the **SSD1306** OLED display

The SSD1306 128x32 OLED display must be used on the I2C bus of the ESP32 card.

This I2C bus is present on all ESP32 boards.

Connection to an ESP32 card:



connection of the SSD1306 I2C OLED display

As we can see, 4 wires are enough: 2 for powering the SSD1306 OLED display (black and red wires), 2 for the connection to the I2C bus (blue and yellow wires).

The display power supply is taken from the ESP32 card. There is no need to use an auxiliary power supply. The very low power consumption of this display allows this. The SSD1306 OLED display has an integrated circuit bringing back the 5V voltage necessary for its operation.

Memory organization


The SSD1306 128x32 display screen uses the same internal component as the SSD1306 128x64 display. The internal memory is common to both models, the 128x32 display screen only uses part of this memory.

The display's internal memory has 1KB RAM.


In this diagram, here is the organization of the screen for a definition of 128x64 pixels:

Each column contains 8 bits. One line is designated per page:

- the 128x64 display: contains 8 pages, numbered from 0 to 7
- the 128x32 display: contains 4 pages, numbered from 0 to 3



Each page is divided into segments:



Here, in blue in the figure, a segment represents a byte. The least significant bit is on top.

We do not need to go further to manage this display with the **oled** vocabulary.

Organize the SSD1306 project

Before getting to the heart of the matter, let's see how we are going to organize our project. On your computer, create a working folder named **display**. In this folder, create a subfolder **SSD1306**.

We will fully exploit managing SPIFFS files and building our FORTH code in a real project.

Create the autoexec.fs file

In your workspace, therefore on your computer, in the SSD1306 subfolder, create the autoexec.fs file and copy this FORTH code into it:

```
create crlf 13 C, 10 C,
: RECORDFILE ( "filename" "filecontents" "<EOF>" -- )
    bl parse
    W/O CREATE-FILE throw >R
    BEGIN
        tib #tib accept
        tib over
        S" <EOF>" startswith?
        DUP IF
            swap drop
        ELSE
            swap
            tib swap
            R@ WRITE-FILE throw
            crlf 1+ 1 R@ WRITE-FILE throw
        THEN
    UNTIL
    R> CLOSE-FILE throw
;
: MAIN
    s" /spiffs/main.fs" included
;
```

Next, open the terminal that communicates with ESP32forth, then run **visual edit**:

```
visual edit /spiffs/autoexec.fs
```

And copy the contents of **autoexec.fs** described above. At the end of editing with the terminal, type CTRL-S and CTRL-X and Y. Relaunch ESP32forth with **bye**. You must find the words **RECORDFILE** and **MAIN** by typing **words** :

```
OK
--> words
MAIN RECORDFILE crlf FORTH spi oled telnetd registers webui login web-interface
httpd ok LED OUTPUT INPUT HIGH LOW tone freq duty adc pin default-key?
default-key default-type visual set-title page at-xy normal bg fg ansi
```

These manipulations will offer us some facilities which we will now detail.

Creating the main.fs file

SSD1306 subfolder , create the **main.fs** file and copy this FORTH code into it :

```
RECORDFILE /spiffs/main.fs
DEFINED? --oledTest [if] forget --oledTest [then]
create --oledTest
s" /spiffs/config.fs"      included
s" /spiffs/oledTools.fs"   included
<EOF>
```

Copy this code again, launch the terminal which communicates with the ESP32 and ESP32forth board. Copy this code into the terminal. Run it. At the end of execution, you should find your main.fs file on the ESP32 card :

```
--> ls /spiffs/
autoexec.fs
main.fs
```

To verify that the contents of **main.fs** have been saved in the SPIFFS file system :

```
cat /spiffs/main.fs
```

should display the contents of the **/spiffs/main.fs** file .

Creating the config.fs file

SSD1306 subfolder , create the **config.fs** file and copy this FORTH code into it :

```
RECORDFILE /spiffs/config.fs
\ set oled SSD1306 dimensions
oled
128 to WIDTH
32 to HEIGHT
forth
\ set address of OLED SSD1306 display 128x32 pixels
$3c constant I2C_SSD1306_ADDRESS
<EOF>
```

As with **main.fs** , pass this content through the terminal to save the new **config.fs** file .

Creating the oledTools.fs file

For now, here is our final **oledTools.fs file** to create on the computer and transmit to the ESP32 board:

```
RECORDFILE /spiffs/oledTools.fs
oled
: Oled128x32Init
  OledAddr @ 0=
  if
    WIDTH HEIGHT OledReset OledNew
    SSD1306_SWITCHCAPVCC I2C_SSD1306_ADDRESS OledBegin drop
  then
    OledCLS
    1 OledTextsize      \ Draw 2x Scale Text
    WHITE OledTextc     \ Draw white text
    0 0 OledSetCursor   \ Start at top-left corner
    z" *Esp32forth*" OledPrintln OledDisplay
  ;
forth
<EOF>
```


Test our SSD1306 project

Here is the structure of our project on our computer disk.

All fs extension files were passed to the ESP32 board to be saved on the SPIFFS file system.

To compile the contents of the /SPIFFS/config.fs file, you can test this from the terminal window which communicates with ESP32forth:

```
include /spiffs/config.fs
```



If you never modify the contents of the **config.fs file**, it will always be available to ESP32forth as soon as the ESP32 board is powered on.

Remember we also passed in a **main.fs file**? The content of this file must be reserved for calling the different project files. Reminder of the contents of our main.fs file as it is recorded on the ESP32 card in the SPIFFS file system:

```
DEFINED? --oledTest [if] forget --oledTest [then]
create --oledTest

s" /spiffs/config.fs"      included
s" /spiffs/oledTools.fs"   included
```

The first two lines allow you to manage a marker. Each time the content of **main.fs is interpreted**, ESP32forth will test if there is an **--oledTest word**. If this word exists, it will be deleted from the dictionary. All words compiled after **--oledTest** will be removed from the dictionary.

In the second line, we recreate the word `--oledTest` . It's not surprising to remove this word and recreate it. In this way, each time the content of **main.fs is interpreted** , the dictionary of ESP32forth restarts with content that will not disrupt our project.

Finally, in the last two lines of **main.fs** , ESP32forth is asked to process the contents of the **config.fs** and **oledTools.fs** files . So, to launch this global processing, we can type:

```
include /spiffs/main.fs
```

When you have a complex project, you may have to type this treatment dozens, or even hundreds of times. Do you remember that in the **autoexec.fs file** we defined the word **MAIN** ? Reminder of the definition of this word:

```
: MAIN
  s" /spiffs/main.fs" included
;
```

Very good ! So we can just type **MAIN** instead of doing `include /spiffs/main.fs...`

Shall we take the test? OK. Power off the ESP32 board. Turn it back on. Open the terminal that communicates with ESP32forth and type **MAIN**. All project content is executed and compiled almost instantly! Type **words** . All the words in our project are in the **forth** vocabulary.

We check that it works by now typing:

```
Oled128x32Init
```

If the SSD1306 128x32 pixel OLED display is properly connected, it should display ***Esp32forth*** :



result of executing the word Oled128x32Init

From this moment, we can write and realize all the other words of our project.

Use OLED vocabulary

The words that we will use to use our SSD1306 128x32 OLED display are available in the **oled** vocabulary:

```
--> oled vlist
OledInit SSD1306_SWITCHCAPVCC SSD1306_EXTERNALVCC WHITE BLACK OledReset
HEIGHT WIDTH OledAddr OledNew OledDelete OledBegin OledHOME OledCLS OledTextc
OledPrintln OledNumln OledNum OledDisplay OledPrint OledInvert OledTextsize
OledSetCursor OledPixel OledDrawL OledFastHLine OledFastVLine OledCirc
OledCircF OledRect OledRectF OledRectR oled-builtins
```

Initializing the I2C bus for the SSD1306 OLED display

Our OLED display is wired to the I2C bus. It is therefore available at hexadecimal address 3c on this I2C bus. The first thing to do is therefore to define a constant:

```
\ set address of OLED SSD1306 display 128x32 pixels
$3c constant I2C_SSD1306_ADDRESS
```

In the **oled** vocabulary, there is an **OledAddr** variable responsible for memorizing this 3c address. If this address contains a zero value, it is because the I2C bus has not established a connection with our SSD1306 display. It is this null value which conditions the initialization of this connection:

```
OledAddr @ 0=
if
    \ init I2C communication with SSD1306
then
```

Here is the initialization part of our communication via the I2C bus to the SSD1306 OLED display:

```
WIDTH HEIGHT OledReset OledNew
SSD1306_SWITCHCAPVCC I2C_SSD1306_ADDRESS OledBegin drop
```

- **WIDTH HEIGHT OledReset OledNew** sequence instantiates a new OLED session for our SSD1306 display;
- the word **OledBegin** will be preceded by two parameters:
 - **SSD1306_SWITCHCAPVCC** which confirms the 3.3V supply from the ESP32 card, which is the case in our assembly. If we had used an external power supply, we replace this word with **SSD1306_EXTERNALVCC** .
 - **I2C_SSD1306_ADDRESS** which indicates the address of the OLED display on the I2C bus.

This initialization of the I2C bus is only carried out once for our SSD1306 OLED display.

Initializing the display for SSD1306

To start a display, we will initialize the display:

- **OledCLS** which requests deletion of the screen content;
- **1 OledTextsize** which indicates the size of the text to display;

- **WHITE** OledTextc which indicates the color of the text to display. For the SSD1306 display, there are only **WHITE** and **BLACK** colors .

Here is the word **Oled128x32Init** allowing proper initialization:

```

oled
: Oled128x32Init
    OledAddr @ 0=
    if
        WIDTH HEIGHT OledReset OledNew
        SSD1306_SWITCHCAPVCC I2C_SSD1306_ADDRESS OledBegin drop
    then
        OledCLS
        1 OledTextsize      \ Draw 1x Scale Text
        WHITE OledTextc     \ Draw white text
        0 0 OledSetCursor   \ Start at top-left corner
        z" *Esp32forth*" OledPrintln OledDisplay
    ;
forth

```

Running **Oled128x32Init** should display the text ***Esp32forth*** on the OLED screen.


Here is a summary of the text display management commands for the OLED display:

- **OledCLS (--)**
Clears the contents of the OLED screen
- **OledDisplay (--)**
transmits to the OLED display the commands awaiting display
- **OledHOME (--)**
Positions the cursor in row 0, column 0 on the OLED display. This position is pixel-perfect.
- **OledInvert (--)**
Inverts the OLED screen display
- **OledNum (n --)**
Displays the number n as a string on the OLED screen
- **OledNumLn (n --)**
Displays an integer on the OLED display and moves to the next line
- **OledPrint (z-string --)**
Displays z-string text on the OLED screen
- **OledPrintln (z-string --)**
Prints z-string text on the OLED screen and moves to the next line
- **OledTextc (WHITE|BLACK --)**
Sets the color of the text to display


- **OledSetCursor** (xy --)
Sets the cursor position
- **OledTextsize** (size --)
Sets the size of text to display on the OLED screen. The value of n must be in the interval [1..3]. For normal sized text, size=1. If you exceed the value 4, the text will be truncated on a 4-line display.

And here is a summary of the graphic display management commands:

- **OledCirc** (xy radius color --)
Draws a circle centered at xy, with radius radius and color color (0|1)



- **OledCircF** (xy radius color --)
Draws a full circle centered at xy, with radius radius and color color (0|1)
- **OledDrawL** (x0 y0 x1 y1 color --)
Draws a line from x0 y0 to x1 y1 of color color.



- **OledFastHLine** (xy length color --)
Draws a horizontal line from xy of dimension length and color color.
- **OledFastVLine** (xy length color --)
Draws a vertical line from xy of dimension length and color color.
- **OledPixel** (xy color)
Activates a pixel at position x y. The color parameter determines the color of the pixel.
- **OledRect** (xy width height color --)
Draws an empty rectangle from the xy position of size width height and color color.
- **OledRectF** (xy width height color --)
Draws a solid rectangle from the xy position of size width height and color color.
- **OledRectR** (xy width height radius color --)
Draws an empty rectangle with rounded corners, from the xy position, of dimension width height, in the color color, with a radius radius.

- **OledRectRF** (xy width height radius color --)

Traces a solid rectangle with rounded corners, from the xy position, of dimension width heigh, in the color color, with a radius radius.

The OLED display allows text and graphics management words to be executed in the same display mode. In short, you can mix text and graphics.

Expand the oled vocabulary

After a few crashes, I discovered that you cannot define an **OledTriangle** extension like this in C language to extend the definitions in the **oled.h file** :

```
vv(oled, OledTriangle, oled_display->drawTriangle(n5, n4, n3, n2, n1, n0 ); DROFn(6)) \
```

but that's without taking into account that we program in FORTH language and that with this language we can extend our **oled** vocabulary. We will therefore create a new file on our computer, in our project directory, with the file name **extendOledVoc.fs** . Content :

```
RECORDFILE /spiffs/extendOledVoc.fs
oled definitions
: OledTriangle { x0 y0 x1 y1 x2 y2 color -- }
  x0 y0 x1 y1 color OledDrawL
  x1 y1 x2 y2 color OledDrawL
  x2 y2 x0 y0 color OledDrawL
;
forth definitions
<EOF>
```


Then we copy and paste this code and transmit it to ESP32forth via the terminal which communicates with the ESP32 card. We should end up with an **extendOledVoc.fs** file in the SPIFFS memory space. We now modify the contents of the **main.fs** file :

```
s" /spiffs/config.fs"           included
s" /spiffs/extendOledVoc.fs"    included
s" /spiffs/oledTools.fs"        included
```

We unplug and reconnect the ESP32 card. We type **MAIN** in the terminal. If everything went well, you should find the word **OledTriangle** by simply typing **oled vlist** .

Test of our new word **OledTriangle** :

```
oled
OledCLS OledDisplay
5 5 60 8 40 30 WHITE OledTriangle OledDisplay
```



TEMPVS FVGIT⁷

What if the Romans had been able to program the display of time in digital form ?

This is an interesting project that combines several files. In this chapter, we are not going to give all of the code used here. It would be too long.

The source codes for this chapter are in this file:

- **ESP32forth-book.zip** → projects → tempusFugit

Link: https://github.com/MPETREMANN11/ESP32forth/blob/main/_documentation/ESP32forth-book.zip

The entire project contains these files:

- **autoexec.fs** content loaded when ESP32forth starts
- **clepsydra.fs** converting numbers to roman numerals
- **config.fs** global configuration settings
- **main.fs** main file loading the other project files
- **oledTools.fs** completes the **oled vocabulary**
- **RTClock.fs** manages the real-time clock
- **strings.fs** handles processing of alphanumeric strings

The sequence of loading project files is written in **main.fs** :

```
s" /spiffs/strings.fs"           included
s" /spiffs/RTClock.fs"          included
s" /spiffs/clepsydra.fs"        included
s" /spiffs/config.fs"           included
s" /spiffs/oledTools.fs"         included
```

Most files in this project are independent, except for **clepsydra.fs** which is dependent on **strings.fs** .

Romani non ustulo nulla⁸

The Romans did not know the number 0. So how can we display **13:00** or **00:15** in Roman numerals?

To solve the problem of hours after midnight, for example 00:15, the Japanese (residents of JAPAN) will be of great

⁷ Tempus fugit = time passes

⁸ Romani non ustulo nulla = The Romans did not know zero



help to us. If you ever go to this country, you will be amazed to see shops open until **25:00** !

This store is open from 09:00 to 25:00! Oh yes. However, JAPAN's clocks are also 24 hours. We knew the Japanese were hardworking, but to the point of working 25-hour days, we have the right to have some doubts...

In fact, there is a very logical explanation. After 12:00, it is 12:01, etc... And so, after 23:59, it is 24:00, then 24:01. So, if a store closes at 25:00, we must understand that it closes at 01:00 for us.

If we transpose this onto our Roman clock, when it is **00:00**, we will be able to display **XXIV** or better **XXIII:LX** (23:60).

Romani horas et minuta⁹

To resolve the case of hours like 01:00, 02:00... 23:00, logically, after 12:59, we can very well display 12:60, then the minute after 13:01.. 12:60 in Roman numerals: **XXII:LX**.

This is what is achieved in the word tempusTo\$:

```
: tempusTo$ { HH MM -- }
    HH 0 = MM 0= AND if
        60 to MM
        23 to HH
    THEN
    HH 0 > MM 0= AND if
        60 to MM
        -1 +to HH
    then
    HH 0 <= if
        24 to HH
    then
    HH roman tempus $!
    [char] : tempus c+$!           \ add char :
    MM roman tempus append$
    tempus
;
```

In the first if..then test, we test if we are at **00:00**. In this case, we force the hour to **23** and the minutes to **60**.

In the second test, if the hour is greater than 00 and the minutes are greater than 00, we decrement the hour and force the minutes to **60**. The disadvantage is that if the time is 00, we change it to -1.

In the last test, if the hour is zero or negative, we force it to **24**.

⁹ Romani horas et minuta = Roman hours and minutes

We can use the word **.tempus** which was used for the development to check this correct operation :

```
--> 23 59 .tempus
XXIII:LIX ok
--> 0 0 .tempus
XXIII:LX ok
--> 0 1 .tempus
XXIV:I ok
--> 1 0 .tempus
XXIV:LX ok
--> 1 1 .tempus
I:I ok
```

Haec omnia integramus pro ESP32forth¹⁰

In the current state of the project, you must manually enter the initial time:

```
22 19 start
```

Means that we initialize the time at **22:19** . This initialization is carried out very simply:

```
0 RTC.setTime
```

Then we initialize the 128x32 OLED display:

```
Oled128x32Init
1 OledTextsize
WHITE OledTextc
```

And finally, we will retrieve the current time and display it :

```
OledCLS OledDisplay
16 20 OledSetCursor
RTC.getTime drop tempusTo$ s>z OledPrintln OledDisplay
```

I did tests with a larger display. The problem, for the channel **XXIII:XXVIII** , there is not enough space to display this channel.

Here is the final loop to run to start displaying the time in Roman numerals:


```
oled
: start ( HH MM -- )
    0 RTC.setTime      \ define current time
    Oled128x32Init
    1 OledTextsize
    WHITE OledTextc
begin
    OledCLS OledDisplay
    16 20 OledSetCursor
    RTC.getTime drop tempusTo$ s>z OledPrintln OledDisplay
    1000 ms
```

10 Haec omnia integramus pro ESP32forth = We integrate all this for ESP32forth

```
key? until  
;  
forth
```

The program can be improved by retrieving the time from a time server. See chapter *Retrieving the time from a WEB server*.

Using **timers** words to free the interpreter. See chapter *Flashing an LED by timer*.



Finally, assembling the different files for this project and the few tests and adjustments took me an afternoon.

However, I would like to emphasize a few points:

- Always make a copy in your project folder of a general purpose file. For example, for the **strings.fs** file ;
- If you copy a general component, for example **strings.fs** or **RTClock.fs** , only make changes to those files copied to your project folder. Version these modifications and indicate the modification date in the head comment of the modified file.

As you carry out your projects, you may find yourself with the same file copied to different folders and modified. This solution is preferable to a single file full of adjustment parameters.

Add the SPI library

The SPI library is not natively implemented in ESP32forth. To install it, you must first create the **spi.h file** which must be installed in the same folder as that containing the **ESP42forth.ino** file.

Content of the **spi.h file** (in C language) :

```
# include <SPI.h>

#define OPTIONAL_SPI_VOCABULARY V(spi)
#define OPTIONAL_SPI_SUPPORT \
    XV(internals, "spi-source", SPI_SOURCE, \
        PUSH spi_source; PUSH sizeof(spi_source) - 1) \
    XV(spi, "SPI.begin", SPI_BEGIN, SPI.begin((int8_t) n3, (int8_t) n2, (int8_t) n1, (int8_t) n0); DROFn(4)) \
    XV(spi, "SPI.end", SPI_END, SPI.end();) \
    XV(spi, "SPI.setHwCs", SPI_SETHWCS, SPI.setHwCs((boolean) n0); DROP) \
    XV(spi, "SPI.setBitOrder", SPI_SETBITORDER, SPI.setBitOrder((uint8_t) n0); DROP) \
    XV(spi, "SPI.setDataMode", SPI_SETDATAMODE, SPI.setDataMode((uint8_t) n0); DROP) \
    XV(spi, "SPI.setFrequency", SPI_SETFREQUENCY, SPI.setFrequency((uint32_t) n0); DROP) \
    XV(spi, "SPI.setClockDivider", SPI_SETCLOCKDIVIDER, SPI.setClockDivider((uint32_t) n0); DROP) \
    XV(spi, "SPI.getClockDivider", SPI_GETCLOCKDIVIDER, PUSH SPI.getClockDivider();) \
    XV(spi, "SPI.transfer", SPI_TRANSFER, SPI.transfer((uint8_t *) n1, (uint32_t) n0); DROFn(2)) \
    XV(spi, "SPI.transfer8", SPI_TRANSFER_8, PUSH (uint8_t) SPI.transfer((uint8_t) n0); NIP) \
    XV(spi, "SPI.transfer16", SPI_TRANSFER_16, PUSH (uint16_t) SPI.transfer16((uint16_t) n0); NIP) \
    XV(spi, "SPI.transfer32", SPI_TRANSFER_32, PUSH (uint32_t) SPI.transfer32((uint32_t) n0); NIP) \
    XV(spi, "SPI.transferBytes", SPI_TRANSFER_BYTES, SPI.transferBytes((const uint8_t *) n2, (uint8_t *) n1, (uint32_t) n0); DROFn(3)) \
    XV(spi, "SPI.transferBits", SPI_TRANSFER_BITES, SPI.transferBits((uint32_t) n2, (uint32_t *) n1, (uint8_t) n0); DROFn(3)) \
    XV(spi, "SPI.write", SPI_WRITE, SPI.write((uint8_t) n0); DROP) \
    XV(spi, "SPI.write16", SPI_WRITE16, SPI.write16((uint16_t) n0); DROP) \
    XV(spi, "SPI.write32", SPI_WRITE32, SPI.write32((uint32_t) n0); DROP) \
    XV(spi, "SPI.writeBytes", SPI_WRITE_BYTES, SPI.writeBytes((const uint8_t *) n1, (uint32_t) n0); DROFn(2)) \
    XV(spi, "SPI.writePixels", SPI_WRITE_PIXELS, SPI.writePixels((const void *) n1, (uint32_t) n0); DROFn(2)) \
    XV(spi, "SPI.writePattern", SPI_WRITE_PATTERN, SPI.writePattern((const uint8_t *) n2, (uint8_t) n1, (uint32_t) n0); DROFn(3))

const char spi_source[] = R"""\n\
vocabularies spi   spi definitions\n\
transfer spi-builtins\n\
forth definitions\n\
)""";
```

The full file is also available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/optional/spi.h>

Changes to the **ESP32forth.ino** file

spi.h file cannot be integrated into ESP32forth without making some changes to the **ESP32forth.ino** file. Here are the few modifications to be made to this file. These changes were made on version 7.0.7.15, but should be applicable to other recent or future versions.

First modification

Code added in red :

```
#define VOCABULARY_LIST \
V(forth) V(internals) \
```

```

V(rtos) V(SPIFFS) V(serial) V(SD) V(SD_MMC) V(ESP) \
V(ledc) V(Wire) V(WiFi) V(sockets) \
OPTIONAL_CAMERA_VOCABULARY \
OPTIONAL_BLUETOOTH_VOCABULARY \
OPTIONAL_INTERRUPTS_VOCABULARIES \
OPTIONAL_OLED_VOCABULARY \
OPTIONAL_SPI_VOCABULARY \
OPTIONAL_RMT_VOCABULARY \
OPTIONAL_SPI_FLASH_VOCABULARY \
USER_VOCABULARIES

```

Second modification

Addition in red after this code :

```

// Hook to pull in optional Oled support.
# if __has_include("oled.h")
# include "oled.h"
# else
# define OPTIONAL_OLED_VOCABULARY
# define OPTIONAL_OLED_SUPPORT
# endif

// Hook to pull in optional SPI support.
# if __has_include("spi.h")
# include "spi.h"
# else
# define OPTIONAL_SPI_VOCABULARY
# define OPTIONAL_SPI_SUPPORT
# endif

```

Third modification

Added in red :

```

#define EXTERNAL_OPTIONAL_MODULE_SUPPORT \
OPTIONAL_ASSEMBLERS_SUPPORT \
OPTIONAL_CAMERA_SUPPORT \
OPTIONAL_INTERRUPTS_SUPPORT \
OPTIONAL_OLED_SUPPORT \
OPTIONAL_SPI_SUPPORT \
OPTIONAL_RMT_SUPPORT \
OPTIONAL_SERIAL_BLUETOOTH_SUPPORT \
OPTIONAL_SPI_FLASH_SUPPORT

```

Fourth modification

Addition in red :

```

internals DEFINED? oled-source [IF]
  oled-source evaluate
[THEN] forth

internals DEFINED? spi-source [IF]

```

```
spi-source evaluate  
[THEN] forth
```

If you follow these instructions carefully, you will be able to compile ESP32forth with ARDUINO IDE and upload it to the ESP32 board. Once these operations are done, launch the terminal. You need to find the ESP32forth welcome prompt. Type :

```
spinnaker vlist
```


You must find the words defined in this **spi vocabulary** :

```
SPI.begin SPI.end SPI.setHwCs SPI.setBitOrder SPI.setDataMode SPI.setFrequency  
SPI.setClockDivider SPI.getClockDivider SPI.transfer SPI.transfer8 SPI.transfer16  
SPI.transfer32 SPI.transferBytes SPI.transferBits SPI.write SPI.write16  
SPI.write32 SPI.writeBytes SPI.writePixels SPI.writePattern spi-builtins
```


You can now drive extensions via the SPI port, such as the MAX7219 LED displays.

Communicate with the MAX7219 display module

In SPI communication, there is always a master *who* controls the peripherals (also called *slaves*). Data can be sent and received simultaneously. This means that the master can send data to a slave and a slave can send data to the master at the same time.




You can have several slaves. A slave can be a sensor, display, microSD card, etc., or another microcontroller. This means you can have your ESP32 connected to **multiple devices**.



A slave is selected by setting the CS1 or CS2 selector to low level. It will take as many CS selectors as there are slaves to manage.

Locating the SPI port on the ESP32 board

There are two SPI ports on an ESP32 board: HSPI and VSPI. The SPI port that we will manage is the one whose pins are prefixed VSPI:




With ESP32forth, we can therefore define the constants pointing to these VSPI pins:

```
\ define VSPI pins
19 constant VSPI_MISO
23 constant VSPI_MOSI
18 constant VSPI_SCLK
05 constant VSPI_CS
```

To communicate to the MAX7219 display module, we will only need to wire the VSPI_MOSI, VSPI_SCLK and VSPI_CS pins.

SPI connectors on the MAX7219 display module


Here is the SPI port connector map on the MAX7219 module:



Connection between the MAX7219 module and the ESP32 card:

MAX7219		ESP32
DIN	<---->	VSPI_MOSI
CS	<---->	VSPI_CS
CLK	<---->	VSPI_SCLK

The VCC and GND connectors are connected to an external power supply:



The GND part of this external power supply is shared with the GND pin of the ESP32 card.

SPI port software layer

All the words for managing the SPI port are already available in the [spi vocabulary](#) .

The only thing to define is the initialization of the SPI port:

```
\ define SPI port frequency
4000000 constant SPI_FREQ

\ select SPI vocabulary
only FORTH SPI also

\ initialize SPI port
: init.VSPI ( -- )
    VSPI_CS OUTPUT pinMode
    VSPI_SCLK VSPI_MISO VSPI_MOSI VSPI_CS SPI.begin
    SPI_FREQ SPI.setFrequency
;
```

We are now ready to use our MAX7219 display module.

Installing the HTTP client

Editing the ESP32forth.ino file

ESP32Forth is provided as a source file, written in C language. This file must be compiled using ARDUINO IDE or any other C compiler compatible with the ARDUINO development environment.

Here are the portions of code to modify. First portion to modify:

```
#define ENABLE_SD_SUPPORT  
#define ENABLE_SPI_FLASH_SUPPORT  
#define ENABLE_HTTP_SUPPORT  
// #define ENABLE_HTTPS_SUPPORT
```

Second portion to modify:

```
// .....  
#define VOCABULARY_LIST \  
    V(forth) V(internal) \  
    V(rtos) V(SPIFFS) V(serial) V(SD) V(SD_MMC) V(ESP) \  
    V(ledc) V(http) V(Wire) V(WiFi) V(bluetooth) V(sockets) V(oled) \  
    V(rmt) V(interrupts) V(spi_flash) V(camera) V(timers)
```

Third portion to modify :

```
OPTIONAL_RMT_SUPPORT \  
OPTIONAL_OLED_SUPPORT \  
OPTIONAL_SPI_FLASH_SUPPORT \  
OPTIONAL_HTTP_SUPPORT \  
FLOATING_POINT_LIST  
  
#ifndef ENABLE_HTTP_SUPPORT  
# define OPTIONAL_HTTP_SUPPORT  
#else  
  
# include <HTTPClient.h>  
HTTPClient http;  
  
# define OPTIONAL_HTTP_SUPPORT \  
    XV(http, "HTTP.begin", HTTP_BEGIN, tos = http.begin(c0)) \  
    XV(http, "HTTP.doGet", HTTP_DOGET, PUSH http.GET()) \  
    XV(http, "HTTP.getPayload", HTTP_GETPL, String s = http.getString(); \  
        memcpy((void *) n1, (void *) s.c_str(), n0); DROPN(2)) \  
    XV(http, "HTTP.end", HTTP_END, http.end())  
#endif
```

Fourth portion to modify :

```

vocabulary ledc ledc definitions
transfer ledc-builtins
forth definitions

vocabulary http http definitions
transfer http-builtins
forth definitions

vocabulary Serial Serial definitions
transfer Serial-builtins
forth definitions

```

Once the **ESP32forth.ino file** is modified, you compile it and upload it to the ESP32 board. If everything went correctly, you should have a new **http vocabulary** :

```

ttt
vlist \ displays :
HTTP.begin HTTP.doGet HTTP.getPayload HTTP.end http-builtins

```

HTTP Client Testing


To test our HTTP client, we can do it by querying any web server. But for what we are considering later, you need to have a personal web server. On this server, we create a subdomain:

- our server is arduino-forth.com
- **ws** subdomain
- we access this subdomain with the URL <http://ws.arduino-forth.com>

This subdomain being created, it does not contain any script to execute. We create the **index.php** page and put this code there:

It's OK

To check that our subdomain is functional, simply query it from our favorite web browser:



If everything goes as planned, we should have the text **It's OK** displayed in our favorite web browser. Let's now see how to perform this same server query from ESP32Forth...

Here is the Forth code written quickly to perform the HTTP client test:

```
WiFi

\ connection to local WiFi LAN
: myWiFiConnect
    z" mySSID"
    z" myWiFiCode"
    login
;

Forth

create httpBuffer 700 allot
    httpBuffer 700 erase

HTTP

: run
    cr
    z" http://ws.arduino-forth.com/" HTTP.begin
    if
        HTTP.doGet dup ." Get results: " . cr 0 >
        if
            httpBuffer 700 HTTP.getPayload
            httpBuffer z>s dup . cr type
        then
    then
    HTTP.end
;
```

We activate the Wifi connection by executing **myWiFiConnect** then **run** :

```
--> myWiFiConnect
192.168.1.23
MDNS started
ok
--> run

Get results: 200
8
It's OK
ok
```

Our HTTP client queried the web server perfectly, displaying the same text as that retrieved from our web browser.

This small successful test opens the way to enormous possibilities.

Retrieve the time from a web server

Software real-time clock chapter , we looked at how to manage a real-time clock using the properties of the **timer**.

However, the initialization of this real time clock must be done manually. Now that we have a way to communicate with a web server, we will see how to perform this initialization through a web server.

Transmission and reception of time from a web server

For the server part, we create a new **gettime.php** script whose content is as follows:

```
<?php  
echo date('H i s')." RTC.set-time";
```

If we run this script <http://ws.arduino-forth.com/gettime.php> , on a web browser, this is what is displayed:

```
15 25 30 RTC.set-time
```

We prepared the work so that the ESP32Forth interpreter only has this line to execute. Here is the Forth code to retrieve the time:

```
WiFi  
\ connection to local WiFi LAN  
: myWiFiConnect  
  z" mySSID"  
  z" myWiFiCode"  
  login  
;  
  
Forth  
  
0 value currentTime  
  
\ store current time  
: RTC.set-time { hh mm ss -- }  
  hh 3600 *  
  mm 60 *  
  ss ++ 1000 *  
  MS-TICKS - to currentTime  
;  
  
\ used for SS and MM part of time display  
: :## ( n -- n' )  
  # 6 base ! # decimal [char] : hold  
;
```

```

\ display current time
: RTC.display-time ( -- )
    currentTime MS-TICKS + 1000 /
    <# :## :## 24 mod #S #> type
;

700 constant bufferSize
create httpBuffer
    bufferSize allot

0 buffer 700 erase

HTTP

: getTime
    cr
    z" http://ws.arduino-forth.com/gettime.php" HTTP.begin
    if
        HTTP.doGet
        if
            httpBuffer bufferSize HTTP.getPayload
            httpBuffer z>s evaluate
        then
    then
    HTTP.end
;

myWiFiConnect
getTime
RTC.display-time

```

In the word **getTime**, this sequence **httpBuffer z>s evaluate** retrieves the contents of the web transaction buffer and evaluates its contents. This is possible because the web server transmitted a sequence compatible with our Forth interpreter. Running the last three lines of this code displays this :

```

--> myWiFiConnect
192.168.1.23
MDNS started
ok
--> getTime
ok
--> RTC.display-time
15:33:09 ok

```

This initialization can be performed only once, generally when starting ESP32Forth. This technique of querying our own web server avoids negotiating with a time server.

Most time servers deliver information in formats that are difficult to process by Forth: csv, JSON, XML, etc.

Understanding transmission by GET to a web server

Transmission of data to a server by GET

There are two methods of transmitting data from a web page to a web server :

POST which is the method generally used from forms

GET which is the method we are going to study

There are other methods, but these are generally reserved for machine-to-machine transactions through web services.

Parameters in a URL

Let's start by explaining what a URL is: <http://my-website.com/> (URL for example).

We analyze a URL starting from the end:

- **.com** is the TLD (Top-Level Domain)
- **my-website** is the domain name
- **http://** is the communication protocol.

We are not going to do an exhaustive course on these elements. The only thing there is to know comes now.

This URL can be followed by the script or the HTML page, example : <http://my-website.com/index.php>

We can complete this URL with a parameter pass :

`http://my-website.com/index.php?temp=32.7`

Here we pass a **temp parameter** whose value is **32.7**.

Passing parameters using the GET method is marked by the **?** sign.

Passing multiple parameters

Several parameters can be transmitted by separating them with the sign **&** :

`http://my-website.com/index.php?log=myLog&pwd=myPassWd&temp=32.7`

Here we pass three parameters :

- **log** with myLog value
- **pwd** with the value myPassWd
- **temp** with value 32.7

To understand how the server will receive this data, we create a **record.php script** which will provisionally simply contain this:

```
<?php  
var_dump($_GET);
```

and which will display this if we query this script with our favorite web browser:

```
array(3) {  
    ["log"]=>  
    string(7) "myLogin"  
    ["pwd"]=>  
    string(10) "mypassword"  
    ["temp"]=>  
    string(4) "32.7"  
}
```

That's pretty much all we need to get the data and save it to the server. This is what we are going to discover...

Managing parameter passing with ESP32forth

To begin, it is necessary to have words to manage character strings. You will find these words in the *Display of numbers and character strings chapter*, part *Code of words for managing text variables*.

We start by creating a character string :

```
256 string myUrl  
s" http://ws.arduino-forth.com/record.php?log=myLog&pwd=myPassWd&temp=" myUrl $!
```

We have just defined an alphanumeric variable **myUrl**. This variable is almost complete. All that is missing is the value of the **temp** parameter. To add this value, we will execute **append\$** :

```
s" 32.5" myUrl append$  
myUrl type  
\ display: http://ws.arduino-forth.com/record.php?  
log=myLog&pwd=myPassWd&temp=32.5
```

C'est cet URL que nous allons utiliser dans cette définition :

```
: sendData ( str -- )
```

```

s" http://ws.arduino-forth.com/record.php?log=myLog&pwd=myPassWd&temp="  

myUrl $!  

myUrl append$  

\ cr myUrl type  

myUrl s>z HTTP.begin  

if  

    HTTP.doGet dup 200 =  

    if drop  

        httpBuffer bufferSize HTTP.getPayload  

        httpBuffer z>s type  

    else  

        cr ." CNX ERR: " .  

    then  

then  

HTTP.end  

;  

myWiFiConnect  

s" 32.65" sendData

```

The word **sendData** retrieves the contents of the string, here **32.65** , concatenates these contents to **myUrl** , then initiates a web client transaction to the server mentioned in **myUrl** .

You will notice that in the URL there is a log parameter. This parameter can be different for each ESP32 card initiating a transaction to the web server. It is possible for ten, twenty, or even a thousand ESP32 cards to save their data to a single web server.

Data transmission to a WEB server

Data recording on the web server side

In the previous chapter *Understanding transmission by GET to a WEB server*, we explained how ESP32Forth transmits information to a web server.

Now let's see how, on the server side, we will save the data. Here is a first script, in PHP, which performs this recording:

```
<?php
// echo "<pre>"; var_dump($_GET);
$handle = fopen("datasRecords.csv", "a");
$myDatas = array(
    'currentDateTime' => date("Y-m-d H:i:s"),
    'currentLogin'     => $_GET['log'],
    'currentTemp'      => $_GET['temp'],
);
fwrite($handle, implode(';', $myDatas)."
");
fclose($handle);
echo "DATAs recorded";
```

This script is very simple:

- we open a **dataRecords.csv** file with **fopen** .
- we prepare the data to save in a **myDatas table**
- we save this data with **fwrite**
- the data is put in csv format using **implode**
- we close the file with **fclose**

csv format file is easy to retrieve with a spreadsheet or read with a simple text editor.

Access protection

If you have followed our explanations carefully, you will have noticed that we transmit two parameters **log** and **pwd** . These two parameters first serve as access keys to our data recording script.

It is this protection that we put in place to prevent any access to the script by an unauthorized transmitter. Here, we accept two transmitters :

```
<?php
// echo "<pre>"; var_dump($_GET);
```

```

$myAuths = array(
    'pooltemp' => 'pool2022',
    'housetemp' => 'house2022',
);

/**
 * Test authorization access
 * @param array $auths
 * @return boolean
 */
function testAuths($auths){
    if(array_key_exists($_GET['log'], $auths) &&
$auths[$_GET['log']]==$_GET['pwd']) {
        return true;
    }
    return false;
}

// Recording datas in CSV file format
if (testAuths($myAuths)) {
    $handle = fopen("datasRecords.csv","a");
    $myDatas = array(
        'currentDateTime' => date("Y-m-d H:i:s"),
        'currentLogin'     => $_GET['log'],
        'currentTemp'      => $_GET['temp'],
    );
    fwrite($handle, implode(';', $myDatas)."
");
    fclose($handle);
    echo "DATAs recorded";
} else {
    echo "AUTH failed";
}

```

This script serves as an example. It is deliberately simple. On a professional application, the keys and passwords would be saved in the database.

Here is a transaction that will be executed successfully :

```
http://ws.arduino-forth.com/record.php?log=pooltemp&pwd=pool2022&temp=27.5
```


log pwd pair whose values are tested and approved by the data recording script.

View recorded data

To access the recorded data, we use an FTP client (Filezilla):

Nom de fichier	Taille d...	Type de ...	Dernière modification
..			
datasRecords.csv	672	Fichier C...	07/08/2022 16:18:09
record.php	927	Fichier P...	07/08/2022 16:17:22
gettime.php	41	Fichier P...	04/08/2022 15:25:24
index.php	8	Fichier P...	03/08/2022 20:14:10

There we find our **datasRecords.csv** file . Just download it to view its contents with any text editor :



```
*datasRecords.csv - Bloc-notes
Fichier Edition Format Affichage Aide
2022-08-05 16:03:07;myLogin;34.6
2022-08-05 16:03:09;myLogin;33.5
2022-08-05 16:03:12;myLogin;32.7
2022-08-05 17:32:30;myLog;32.65
2022-08-05 17:33:03;myLog;32.6531.65
2022-08-05 17:34:53;myLog;32.65
2022-08-05 17:35:16;myLog;32.65
2022-08-05 17:36:42;myLog;34.7734.77
2022-08-05 17:41:19;myLog;34.7734.77
2022-08-05 17:42:52;myLog;31.23
2022-08-07 15:24:23;myLogin;32.7
2022-08-07 15:33:12;myLogin;32.7
2022-08-07 16:05:47;housetemp;32.7
2022-08-07 16:17:31;housetemp;32.7
2022-08-07 16:18:09;pooltemp;32.7
```

We find, in the last lines, our transmission tests with two different logins. The **record.php** script can process transactions with hundreds of different ESP32 cards, each with a different login.

Add data to transmit

If you manage a DHT11 or DHT22 type sensor (temperature and humidity sensor), you would be tempted to record the temperature and humidity values in a single transaction. To do this, nothing could be easier. Here is the aspect of the transaction allowing this:

```
http://ws.arduino-fourth.com/record.php?log=pooltemp&pwd=pool2022&temp=27.5&hygr=62.2
```

But for it to work, you have to act on the PHP script record.php:

```
<?php
// Recording datas in CSV file format
if (testAuths($myAuths)) {
    $handle = fopen("datasRecords.csv", "a");
    $myDatas = array(
        'currentDateTime' => date("Y-m-d H:i:s"),
        'currentLogin'     => $_GET['log'],
        'currentTemp'      => $_GET['temp'],
        'currentHygr'      => $_GET['hygr'],
    );
    fwrite($handle, implode(';', $myDatas)."
");
    fclose($handle);
    echo "DATAs recorded";
} else {
    echo "AUTH failed";
}
```

Here, we simply add a row to the **\$myDatas** table .

On the Forth side, we will improve URL management :

```
256 string myUrl      \ declare string variable

: addTemp ( strAddrLen -- )
    s" &temp=" myUrl append$
    myUrl append$
;

: addHygr ( strAddrLen -- )
    s" &hygr=" myUrl append$
    myUrl append$
;

: sendData ( strHygr strTemp -- )
    s" http://ws.arduino-forth.com/record.php?log=myLog&pwd=myPassWd"
    myUrl $!
    addTemp
    addHygr
    cr myUrl type
    myUrl s>z HTTP.begin
    if
        HTTP.doGet dup 200 =
        if drop
            httpBuffer bufferSize HTTP.getPayload
            httpBuffer z>s type
        else
            cr ." CNX ERR: " .
        then
    then
    HTTP.end
;

\ for test:
myWiFiConnect
s" 64.2"  \ hygrometry
s" 31.23" \ temperature
sendData
```

We added two words, **addTemp** and **addHygr** . Each of these words concatenates a parameter and its value to the URL which will be used for the web transaction between your ESP32 card and the web server.

There are only two limitations to the number of parameters passed by the GET method:

- the length of our URL as defined in Forth, here 256 characters. If you want to increase this limit, simply set our URL with a longer initial length: **512 string myUrl**
- the maximum length of URLs accepted by the HTTP protocol. This length can reach 8000 characters according to recent standards.

Regarding Forth, we have other limitations. In particular, if we wish to transmit textual data. certain characters, "&" for example, will need to be encoded. You will have to handle this encoding in Forth.

Conclusion

QUESTION: What can all this be used for?

An ESP32 card costs less than €/\$10 each. Even more like €/\$5 if you buy in quantity. If you integrate a temperature sensor and a relay, you can for example take temperature readings and transmit commands from the server to activate/deactivate a relay. Managing the temperature of several rooms becomes very easy. The same goes for managing intelligent watering in a greenhouse.

You can also monitor access and trigger lights or alarms very easily. Let's take the case of a portal. You authorize passages between certain times and you lock this same gate (magnetic suction cup) from the ESP32 card.

We trust your imagination to find practical arrangements exploiting this data transmission between ESP32 cards and a web server.

AND WHY A WEB SERVER?

With a web server, it is easy to query it from anywhere, with a web browser installed on your PC, a digital tablet, a smartphone. And a single web server can integrate an indefinite number of different scripts.

Sound synthesis with ESP32Forth

For your first sound experiments, you need to have a speaker that you connect to a GPIO output. But the impedance of the speakers being very low, it will be necessary to go through a transistor. Here is the recommended diagram for a small speaker.

On this diagram, the GPIO4 pin is mentioned. In fact, this assembly can be used on any GPIO output of the ESP32 card. The two outputs that will particularly interest us are GPIO25 and GPIO26 which are reserved for DAC (Digital to Analog Conversion) outputs.




Figure 19: branchement d'un haut parleur

Simple sound synthesis

We will use PWM signal generation, but on the DAC outputs.

Our speaker is connected to the GPIO25 output, via the PN2222A transistor which serves as an impedance adapter.

```
0 constant CHANNEL0      \ define PWM channel 0
25 constant BUZZER        \ buzzer connected to GPIO25

ledc                      \ select ledc vocabulary
: initTones ( -- )
    BUZZER CHANNEL0 ledcAttachPin
;
```

The word **initTones** connects the GPIO25 output to PWM channel 0. Generating a sound is done like this:

```
CHANNEL0 freq ledcWriteTone drop
```

where freq is the desired frequency, multiplied by 1000. Thus, to generate the note LA (A in English notation), whose frequency is 440 Hz, you will need to use the value 440*1000:

```
CHANNEL0 440000 ledcWriteTone drop
```

Definition of sound frequency table

To find the sound frequencies of musical notes, we went to Wikipedia. We construct a frequency table, where each frequency will be recorded in its form usable by **ledcWriteTone** :

```
\ frequency notes
```

```

\ source: https://fr.wikipedia.org/wiki>Note_de_musique
\ frequency is multiplied by 1000
create NOTES
\ octave -1
15350 , 17330 , 18360 , 19450 , 20600 , 21830 ,
23130 , 24500 , 25960 , 27500 , 29140 , 30870 ,
\ octave 0
32700 , 34650 , 36710 , 38890 , 41200 , 43650 ,
46250 , 49000 , 51910 , 55000 , 58270 , 61740 ,
\ octave 1
65410 , 69300 , 73420 , 77780 , 82410 , 87310 ,
92500 , 98000 , 103830 , 110000 , 116540 , 123470 ,
\ octave 2
130810 , 138590 , 146830 , 155560 , 164810 , 174610 ,
185000 , 196000 , 207650 , 220000 , 233080 , 246940 ,
\ octave 3
261630 , 277180 , 293660 , 311130 , 329630 , 349230 ,
369990 , 392000 , 415300 , 440000 , 466160 , 493880 ,
\ octave 4
523250 , 554370 , 587330 , 622250 , 659260 , 698460 ,
739990 , 783990 , 830610 , 880000 , 932330 , 987770 ,
\ octave 5
1046500 , 1108730 , 1174660 , 1244510 , 1318510 , 1396910 ,
1479980 , 1567980 , 1661220 , 1760000 , 1864660 , 1975530 ,
\ octave 6
2093000 , 2217460 , 2349320 , 2489020 , 2637020 , 2793830 ,
2959960 , 3135960 , 3322440 , 3520000 , 3729310 , 3951070 ,
\ octave 7
4186010 , 4434920 , 4698640 , 4978030 , 5274040 , 5587650 ,
5919910 , 6271930 , 6644880 , 7040000 , 7458620 , 7902130 ,
\ octave 8
8372020 , 8869840 , 9397280 , 9956060 , 10548080 , 11175300 ,
11839820 , 12543860 , 13289760 , 14080000 , 14917240 , 15804260 ,

```

There are twelve notes per octave, hence the definition of 12 values per octave. Here, we record only 10 lines, or 10 octaves. Because after 15Khz, the sounds would no longer be audible.

To find a note, you just need to know its position in an octave. For example, our A note in octave 3 will be: $((\text{octave}+1)*12)+\text{position}$. A being in the 10th position in octave 3, the address to be determined will be $\text{NOTES}+4*((\text{OCTAVE}+1)*12)+\text{position}$

Retrieving the frequency of a musical note

We first create a word **set.octave** which will allow us to select the desired octave. Then, we define **get.note** which retrieves the frequency of the desired note:

```

3 value OCTAVE
\ select octave in interval -1..8
: set.octave ( n[-1..8] )
  to OCTAVE
;

\ select note in interval 1..12
: get.note ( n[1..12] -- )
  1- OCTAVE 1+ 12 * +  cell *      \ calc. offset in NOTES array
  NOTES + @                          \ fetch frequency of selected note
;

3 value OCTAVE
\ select octave in interval -1..8
: set.octave ( n[-1..8] )
  to OCTAVE
;

: OCT6 ( -- )    6 set.octave ;
: OCT5 ( -- )    5 set.octave ;
: OCT4 ( -- )    4 set.octave ;
: OCT3 ( -- )    3 set.octave ;
: OCT2 ( -- )    2 set.octave ;
: OCT1 ( -- )    1 set.octave ;

```

We will see later how to manage notes by calling them from their notation.

Managing note duration

The duration of a note is the time interval between the triggering of two consecutive notes.

A base delay is defined by the **WHOLE-NOTE-DURATION constant** .

The durations are defined in a new **music vocabulary** :

```

1600 constant WHOLE-NOTE-DURATION
WHOLE-NOTE-DURATION value duration

vocabulary music
music definitions
music also

\ set duration of a whole note
: o ( -- )
  WHOLE-NOTE-DURATION to duration
;

```

```

\ set duration of a white note
: o| ( -- )
    WHOLE-NOTE-DURATION 2/ to duration
;

\ set duration of a black note
: .| ( -- )
    WHOLE-NOTE-DURATION 2/ 2/ to duration
;

\ set duration of a half black note
: .|' ( -- )
    WHOLE-NOTE-DURATION 2/ 2/ 2/ to duration
;

\ set duration of a quarter black note
: .|" ( -- )
    WHOLE-NOTE-DURATION 2/ 2/ 2/ 2/ to duration
;

```

We define words which symbolize the desired durations: **o** for a full note, **\o** for a half note, **\.** for a black note, etc...

One-note support

The sustain of a note is the amount of time the note is audible during its playing time. We define a **sustain value** which expresses the percentage of emission sustain of the note during its total duration. If this value is 100, the notes follow each other without any silence between the notes.

```

\ sustain of note, in interval [0..100]
90 value SUSTAIN


ledc
\ sustain note in interval [0..100]
: sustain.note ( -- )
    duration SUSTAIN 100 */ ms
    CHANNEL0 0 ledcWriteTone drop
    duration 100 SUSTAIN - 100 */ ms
;

```

On a synthesizer, the envelope of a sound is determined by four parts **A D S R**:

- **Attack:** marks the start of the sound
- **Decay:** follows attack, marks the sound crushing which precedes the sustain part
- **Sustain:** marks the part during which the sound is sustained

- **Release:** marks the part during which the sound is attenuated



In this chapter, the sounds we generate only have a sustain part.

The word **sustain.note** generates two delays. The first delay corresponds to the duration of the note maintenance. The second delay corresponds to a silence maintenance delay. The sum of these two delays always corresponds to the delay defined in duration.

Creating musical notes

We arrive at the most interesting part, defining the notes by their name:

```
: create-note
  \ compile position in octave
  create      ( position -- )
  ,
  \ get note frequency in current octave
  does>
    @ 1- get.note
    CHANNEL0 swap ledcWriteTone drop
    sustain.note
  ;
  \ notes in english notation
  1 create-note C
  2 create-note C#
  3 create-note D
  4 create-note D#
  5 create-note E
  6 create-note F
  7 create-note F#
  8 create-note G
  9 create-note G#
  10 create-note A
  11 create-note A#
  12 create-note B
  \ notes in french notation
  1 create-note DO
```

```

2 create-note DO#
3 create-note RE
4 create-note RE#
5 create-note MI
6 create-note FA
7 create-note FA#
8 create-note SOL
9 create-note SOL#
10 create-note LA
11 create-note LA#
12 create-note SI

: SIL ( -- )
    CHANNEL0 0 ledcWriteTone drop
    duration ms
;


```

forth definitions

In addition to the twelve notes, from **DO** to **SI**, we define our **SIL** which is a silence.

Sound test

We test all the notes, scale by scale:

```

forth definitions
: music-scale ( -- )
    C C# D D# E F F# G G# A A# B
;

initTones
forth also music also
.|_
80 to SUSTAIN
OCT1 music-scale
OCT2 music-scale
OCT3 music-scale
OCT4 music-scale
OCT5 music-scale
OCT6 music-scale

```

If all goes well, we must unfold all the musical notes, by semitones, from octave 1 to the highest octave, here 6. We do not define an additional octave. It's doable. But the sounds emitted enter a limit zone to be audible.

The flight of the bumblebee

This is a first test of transposing a musical score. To do this, we use a particularly difficult musical piece, THE **FLIGHT OF THE BOURDON** by **Rimsky KORSAKOV**. Here is the first measure of the first line:



Figure 20: first measure - The Flight of the Bumblebee - Rimsky KORSAKOV

Here is how we code this first measure, in French notation:

```
OCT5 MI RE# RE DO#      RE DO# DO OCT4 SI
```

Or in English notation:

```
OCT5 E D# D C#      D C# C OCT4 B
```

Here is the code for the first line of this partition:

```
: 1stLine ( -- )
  .|" ( duration of a quarter black note )
  OCT5 MI RE# RE DO#      RE DO# DO OCT4 SI
  OCT5 DO OCT4 SI LA# LA      SOL# SOL FA# FA
  MI RE# RE DO#      RE DO# DO OCT3 SI
;
```

My apologies if I made any mistakes translating the score. At this stage, it is easy to test this musical line:

```
: flightBumbleBee ( -- )
  initTones
  1stLine
;
flightBumbleBee
```

We code two other lines:

```
: 2ndLine ( -- )
  .|" ( duration of a quarter black note )
  OCT4 DO OCT3 SI FA# FA      SOL# SOL FA# FA
  MI RE# RE DO#      RE DO DO# OCT2 SI OCT3
  MI RE# RE DO#      RE DO DO OCT2 SI OCT3
  MI RE# RE DO#      DO FA FA RE#
;

: 3rdLine ( -- )
  .|" ( duration of a quarter black note )
  MI RE# RE DO#      DO DO# RE RE#
```

```
MI RE# RE DO#      DO FA FA RE#
MI RE# RE DO#      DO DO# RE RE#
MI RE# RE DO#      RE DO DO# OCT2 SI OCT3
;

: flightBumbleBee ( -- )
  initTones
  1stLine
  2ndLine
  3rdLine
;
flightBumbleBee
```

We let you code the other three lines of the score.

Program in XTENSA assembler

Preamble

For those unfamiliar with assembly language, it is the lowest level layer in programming. In assembler, we address the processor directly.

It is also a difficult language, not very readable. But on the other hand, the performance is exceptional.

We program in assembler:

- when there is no other solution to access certain functionalities of a processor;
- to make certain parts of the program faster. Code generated by an assembler is the fastest!
- for fun. Assembler programming is an intellectual challenge;
- because no evolved language can do everything. Sometimes, you can program functions in assembly that are too complex to write in another language.

As an example, here is the Huffman decoding code carried out in XTENSA assembler:

```
/* input in t0, value out in t1, length out in t2 */
    srl t1, t0, 6
    li t3, 3
    beq t3, t4, 2f
    li t2, 2
    andi t3, t0, 0x20
    beq t3, r0, 1f
    li t2, 3
    andi t3, t0, 0x10
    beq t3, r0, 1f
    li t2, 4
    andi t3, t0, 0x08
    beq t3, r0, 1f
    li t2, 5
    andi t3, t0, 0x04
    beq t3, r0, 1f
    li t2, 6
    andi t3, t0, 0x02
    beq t3, r0, 1f
    li t2, 7
    andi t3, t0, 0x01
    beq t3, r0, 1f
    li t2, 8
    b 2f
    li t1, 9
```

```
1: /* length = value */
   move t1, t2
2: /* done */
```

Since version 7.0.7.4, ESP32forth includes a complete XTENSA assembler. This assembler uses infix notation:

```
\ in conventional assembler:
\ andi t3, t0, 0x01

\ in XTENSA assembler with ESP32forth:
a3 a0 $01 ANDI,
```

ESP32forth is the **very first high-level programming language** for ESP32 that integrates an XTENSA assembler.

This feature allows the programmer to define his assembly macros.

Any word written in XTENSA assembly language from ESP32forth is immediately usable in any definition in Forth language.

Compile the XTENSA assembler

Since version 7.0.7.15, ESP32forth offers the XTENSA assembler as an option. To compile this option :

- **optional** folder in the folder where you unzipped the ZIP file of the ESP32forth version;
- **assemblers.h** file to the root folder containing the **ESP32forth.ino file**;
- run ARDUINO IDE, compile **ESP32forth.ino** and upload to ESP32 board;

If everything went well, you access the XTENSA assembler by typing once :

```
xtensa-assembler
```

To check the correct availability of the XTENSA instruction set:

```
assemble xtensa vlist
```

Programming in assembler

In order to clearly understand what was stated previously, here is a definition proposed as an example by Brad NELSON :

```
\ example proposed by Brad NELSON
code my2*
    a1 32 ENTRY,
    a8 a2 0 L32I.N,
```

```
a8 a8 1 SLLI,  
a8 a2 0 S32I.N,  
RETW.N,  
end-code
```

We have just defined the word **my2*** which has exactly the same action as the word **2***. Assembling the code is immediate. We can therefore test our definition of **my2*** from the terminal :

```
--> 3 my2*  
OK  
6 --> 21 my2*  
OK  
6 42 -->
```

This possibility of immediately testing an assembled code allows it to be tested in situ. If we have to write somewhat complex code, it will be easy to cut it into fragments and test each part of this code from the ESP32forth interpreter.

The XTENSA assembly code is placed after the word to be defined. It is the code sequence **my2*** which creates the word **my2***.

The following lines contain the XTENSA assembly code. The assembly definition ends with the execution of **end-code**.

Summary of basic instructions

List of basic instructions included in all versions of the Xtensa architecture. The remainder of this section provides an overview of the basic instructions.

Load / loading

```
L8UI, L16SI, L16UI, L32I, L32R,
```

Store / storage

```
S8I, S16I, S32I,
```

Memory ordering

```
MEMW, EXTW,
```

Jumps

```
CALL0, CALLX0, RET, J, JX,
```

Conditional branching

```
BALL, BNALL, BANY, BNONE, BBC, BBCI, BBS, BBSI, BEQ, BEQI, BEQZ, BNE,  
BNEI, BNEZ, BGE, BGEI, BGEU, BGEUI, BGEZ, BLT, BLTI, BLTU, BLTUI, BLTZ,
```

Shift

```
MOVI, MOVEQZ, MOVGEZ, MOVLTZ, MOVNEZ,
```

Arithmetic

```
ADDMI, ADD, ADDX2, ADDX4, ADDX8, SUB, SUBX2, SUBX4, SUBX8, NEG, ABS,
```

Binary logic

```
AND, OR,
```

Shift

```
EXTUI, SRLI, SRAI, SLLI, SRC, SLL, SRL, SRA, SSL, SSR, SSAI, SSA8B, SSA8L,
```

Processor control

```
RSR, WSR, XSR, RUR, WUR, ISYNC, RSYNC, ESYNC, DSYNC, NOP,
```

A bonus disassembler

An assembler is very good. Easy code to integrate with Forth definitions is wonderful. But having an XTENSA disassembler is royal!

Let's take the definition of **my2*** previously assembled. It is easy to disassemble :

```
' my2* cell+ @ 20 disasm
\ displays:
\ 1074338656 -- a1 32 ENTRY,          -- 004136
\ 1074338659 -- a8 a2 0 L32I.N,       -- 0288
\ 1074338661 -- a8 a8 1 SLLI,         -- 1188F0
\ 1074338664 -- a8 a2 0 S32I.N,       -- 0289
\ 1074338666 -- RETW.N,              -- F01D
\ 1074338668 -- .....
```

The code of our word **my2*** is only accessible by indirection, the address of which is placed in the parameters field.

Each line displays :

- the address of the assembled code
- the disassembled code at this address on 2 or 3 bytes
- the hexadecimal code corresponding to the disassembled code

The disassembler can also act on all code already compiled or assembled. Let's see the code for word **2*** :

```
' 2* @ 20 disasm
\ displays:
```

```
\ 1074606252 -- a12 a3 0 L32I.N,          -- 03C8
\ 1074606254 -- a5 a5 1 SLLI,           -- 1155F0
\ 1074606257 -- a15 a12 0 L32I.N,        -- 0CF8
\ 1074606259 -- a3 a3 4 ADDI.N,         -- 334B
\ 1074606261 -- 1074597318 J,          -- F74346
```

Disassembly indicates that the code leads to an unconditional jump **1074597318 J** . It is easy to continue disassembly to this new address :

```
1074597318 20 disasm
\ display:
\ 1074597318 -- a15 JX,          -- 000FA0
\ 1074597321 -- a10 64672 L32R,    -- FCA0A1
\ 1074597324 -- a5 a7 1 S32I,      -- 016752
\ 1074597327 -- 1074633168 CALL8,   -- 08C025
\ 1074597330 -- a12 a3 0 L32I,      -- 0023C2
\ 1074597333 -- a2 a7 4 ADDI,       -- 04C722
\ 1074597336 ....
```

First steps in XTENSA assembler

Preamble

The assembly code is not portable in another environment, or at the cost of enormous efforts to understand and adapt the assembled code.

A Forth version is not complete if it does not have an assembler.

Assembler programming is not required. But in some cases, creating a definition in assembler can be much easier than a version in C language or in pure Forth language.

But above all, a definition written in assembler will have unrivaled speed of execution.

We will see, using very simple and very short examples, how to master the programming of Forth definitions written in Xtensa assembler.

Invoking the Xtensa assembler

When starting ESP32forth, it is impossible to define words in Xtensa assembly without invoking the word **xtensa-assembler** . This word will load the content of the **xtensa vocabulary**. This word must only be invoked once when starting ESP32forth and before any definition of a word in xtensa code :

```
forth
DEFINED? invert code [IF] xtensa-assembler [THEN]
```

Now, if we type **order** , ESP32forth displays :

```
xtensa >> asm >> FORTH
```

It is this order of vocabularies that must be respected when we want to define a new word in Xtensa assembly using the definition words **code** and **end-code** .

Xtensa and the Forth stack

The Xtensa processor has 16 registers, a0 to a15. In reality, there are 64 registers, but we can only access a window of 16 registers among these 64 registers, accessible in the interval 00..15.

Register a2 contains the Forth stack pointer.

Each time a value is stacked, the stack pointer is incremented by four units:

```

SP@ . \ displays 1073632236
1
SP@ . \ displays 1073632240
2
SP@ . \ displays 1073632244
drop drop
SP@ . \ 1073632236

```

Here is how we could rewrite this word **SP@** in Xtensa assembler :

```

\ get SP Stack Pointer - equivalent to SP@
code mySP@
    a1 32      ENTRY,
    a8 a2      MOV.N,  \ copy contents of a2 into a8
    a2 a2 4    ADDI,   \ increment a2
    a8 a2 0    S32I.N, \ copy a8 into address pointed to by a2+0
                  RETW.N,
end-code

```

Let's test this new word **mySP@** :

```

mySP@.
\ displays 1073632240
SP@.
\ displays 1073632240

```

Writing an Xtensa macro instruction

In our definition of the word **mySP@** , the sequence **a2 a2 4 ADDI**, increments the stack pointer by four units. Without this increment, it is impossible to return a value to the top of the Forth stack. With Forth, we will write a macro that automates this operation.

To start, we'll expand the **asm** vocabulary :

```

asm definitions

: macro:
:
;

```

Our **macro definition:** is redundant with **:** but has the advantage of then making the Forth code a little more readable when we define a macro-instruction which will extend the **xtensa vocabulary** :

```

xtensa definitions

macro: sp++,
      a2 a2 4    ADDI,
      ;

```

sp++ macro instruction , we can rewrite the definition of **mySP@** :

```

forth definitions

```

```

asm xtensa

    \ get Stack Pointer SP - equivalent for SP@
code mySP@  

    a1 32      ENTRY,  

    a8 a2      MOV.N,  \ copy content of a2 in a8  

        sp++,  

    a8 a2 0    S32I.N, \ copy a8 in address pointed by a2+0  

                    RETW.N,  

end-code

```

It is perfectly possible to integrate one macro into another. In the `mySP@` code , the code line `a8 a2 0 S32I.N,` copies the contents of register a8 to the address pointed to by a2. Here is this new macro instruction :

```

xtensa definitions

\ increment Stack Pointer and store content of ar in addr pointed by Stack
Pointer
macro: arPUSH, { ar -- }
    sp++,  

    ar a2 0 S32I.N,
;

```

This macro instruction uses a local variable `ar` . We could have done without it, but the advantage of this variable is that the macro code is more readable.

`mySP@` code with this macro-instruction :

```

forth definitions
asm xtensa

\ get Stack Pointer SP - equivalent to SP@
code mySP@3
    a1 32      ENTRY,  

    a8 a2      MOV.N,  

    a8 arPUSH,  

                    RETW.N,  

end-code

```

Let's complete our list of macro instructions :

```

xtensa definitions

\ décrémente pointeur de pile
macro: sp--,   ( -- )
    a2 a2 -4   ADDI,
;

\ Store content of addr pointed by Stack Pointer in ar and decrement Stack
Pointer
macro: arPOP,   { ar -- }
    ar a2 0    L32I.N,
    sp--,

```

```
;
```

With these new macros, let's rewrite **swap** :

```
forth definitions
asm xtensa


code mySWAP
    a1 32      ENTRY,
    a9  arPOP,
    a8  arPOP,
    a9  arPUSH,
    a8  arPUSH,
                    RETW.N,
end-code

17 24 mySWAP
```

Managing the Forth stack in Xtensa assembler

The position of the Forth stack pointer can be accessed by **SP@** . Stacking a 32-bit integer (default size for ESP32forth) increments this stack pointer by four units.

We discussed how to manage the increment or decrement of this stack pointer through the **sp++** , and **sp--** , macro-instructions. These macro instructions move the stack pointer four units.



Here, we have stacked three values, **1** **4** and **9** . Each time you stack, the stack pointer is incremented automatically. In Xtensa assembler, the stack pointer is found in register a2. We have seen that we can manipulate the contents of this register with the macro-instructions **sp++** , and **sp--** . Manipulating this register has a direct action on the stack pointer managed by ESP32forth.

Here is how we rewrote the word **+** in assembly by manipulating the stack pointer through our **arPOP** , and **arPUSH** , macro-instructions :

```
code my+
```

```

a1 32      ENTRY,
a7 arPOP,
a8 arPOP,
a7 a8 a9    ADD,
a9 arPUSH,
          RETW.N,
end-code

```


There is another way to retrieve data from the stack using the **L32I.N**, instruction . This instruction uses an immediate index :

```

code my+
a1 32 ENTRY,
  sp--,
a7 a2 0   L32I.N,
a8 a2 1   L32I.N,
a7 a8 a9   ADD,
a9 a2 0   S32I.N,
RETW.N,
end-code

```

Before retrieving the data from the stack, we decrement the stack pointer with our macro instruction **sp--** ,. In this way, the pointer moves back 4 units.



But just because the pointer moves back doesn't mean the previously stacked data disappears. Let's see this line of code in detail:

```
a7 a2 0 L32I.N,
```

This instruction loads register a7 with the contents of the address pointed to by (a2)+n*4. Here, n is 0. This instruction will put the value 4 in our register a7.

Let's see the following line:

```
a8 a2 1 L32I.N,
```

Register a8 is loaded with the contents pointed to by (a2)+1*4. This instruction puts the value 9 in our register a8.

```
a9 a2 0 S32I.N,
```

Here, the contents of register a9 are stored at the address pointed by (a2)+1*0. In fact, we overwrite the value 4 with the result of adding the contents of registers a7 and a8.

Let's see one last example where we process two parameters and output two of them onto the data stack. In this example, we rewrite the word **/MOD** :

```
code my/MOD ( n1 n2 -- rem quot )
    a1 32      ENTRY,
    a7 arPOP,      \ divisor in a7
    a8 arPOP,      \ value to divide in a8
    a7 a8 a9  REMS,  \ a9 = a8 MOD a7
    a9 arPUSH,
    a7 a8 a9  QUOS,  \ a9 = a8 / a7
    a9 arPUSH,
                RETW.N,
end-code

5 2 my/MOD . .      \ display 2 1
-5 -2 my/MOD . .    \ display 2 -1
```

In the word **my/MOD** we use the same data n1 and n2 placed respectively in registers a8 and a7. It is then the **REMS** and **QUOT** instructions which allow the results returned by **my/MOD** to be calculated.

Efficiency of words written in XTENSA assembler

In our very last example above, we rewrote the word **/MOD**. The question to ask is: "is the word **my/MOD** really faster in execution than the word **/MOD** ?".

To do this, we will use the word **measure**: whose Forth code is explained in the chapter *Measuring the execution time of a Forth word*.

```
: test1
    1000000 for
        5 2 /MOD
        drop drop
    next
;

: test2
    1000000 for
        5 2 my/MOD
        drop drop
    next
;

measure: test1 \ display: execution time: 0.856sec.
measure: test2 \ display: execution time: 0.600sec.
```

The words **test1** and **test2** are similar, except that **test2** executes **my/MOD**. Over 1 million iterations, the time saving amounts to 0.144 seconds. It's not much, but the ratio still seems significant.

Conversely, we see that the Forth language is very fast in execution time.

Loops and connections in XTENSA assembler

The LOOP instruction in XTENSA assembler

The LOOP loop in XTENSA assembler works by using the **LOOP instruction** to tell the processor to repeat a block of instructions until a specified counter reaches zero. The loop is initialized by setting the initial value of the counter, then executing the **LOOP instruction** with that value as an argument. On each iteration of the loop, the counter is decremented by 1 until it reaches zero, at which point the loop stops. In classic assembler:

```
; Initialization of the counter to 10
MOVI a0, 10

; Beginning of the LOOP loop
loop:
    ; Instruction(s) to repeat
    ...
    ; Decrement the counter and test the stop condition
    LOOP a0, loop
```

Here, the LOOP loop repeats the instructions between **loop:** and **LOOP a0,** looping 10 times, decrementing the counter a0 with each iteration. When the counter reaches zero, the loop stops.

When the XTENSA processor encounters the **LOOP** instruction , it initializes three special registers :

- **LCOUNT** \leftarrow **AR[s] – 1**

The special register LCOUNT is initialized with the contents of the register as, here a0 in our example, decremented by one unit. When the counter reaches the value 0, the LOOP instruction completes the loop;

- **LBEG** \leftarrow **PC + 3**

The LBEG special register contains the start address of the currently executing LOOP loop. This address is defined by the LOOP instruction.

- **LEND** \leftarrow **PC + (024 | imm8) + 4** The LEND special register contains the end address of the currently executing LOOP loop. This address is defined by the LOOP instruction.

In XTENSA assembler, the LOOP instruction admits two parameters :

```
LOOP as, label
```

label corresponds to an 8-bit offset after the **LOOP instruction**. You cannot repeat a code of more than 256 bytes in length.

Here is some disassembled XTENSA code using a LOOP loop :

```
.data:00000000 004136          entry a1, 32
.data:00000003 01a082          movi a8, 1
.data:00000006 04a092          movi a9, 4
.data:00000009 048976          loop a9, 0x00000011
.data:0000000c 04c222          addi a2, a2, 4
.data:0000000f 0289            s32i.n a8, a2, 0
.data:00000011 f01d            retw.n
```



The disassembler indicates a branch address. In reality, the assembled code only contains this offset indicated by the label in the form of a positive 8-bit value.

Manage a loop in XTENSA assembler with ESP32forth

The Forth language cannot resolve a forward reference. Unless you fumble around, it is difficult to use the **LOOP** instruction without finding a trick.

Defining loop management macro instructions

LOOP, instruction , we will define two macro instructions, respectively **For**, and **Next**, of which here is the code in Forth language:

```
: For, { as n -- }
  as n MOVI,
  as 0 LOOP,
  chere 1- to LOOP_OFFSET
;

: Next, ( -- )
  chere LOOP_OFFSET - 2 -
  LOOP_OFFSET [ internals ] ca! [ asm xtensa ]
;
```

The **For** macro instruction accepts the same parameters as the **LOOP instruction** :

```
as n For,
```

- as is the register that contains the number of iterations of the loop;
- n is the number of iterations.

Using the For, and Next macros,

We define a **myLOOP** word to test the LOOP instruction , via the **For**, **Next**, macro instructions :

```
code myLOOP ( n -- n' )
    a1 32          ENTRY,
    a8 1           MOVI,
    a9 4           For,           \ LOOP start here
        a8 a8 1   ADDI,
        a8         arPUSH,       \ push result on stack
    Next,
        RETW.N,
end-code
```

Register a8 is initialized with the value 1. The **For**, **Next loop** increments the contents of a8 and stacks its contents. This is what running **MyLOOP gives** :

```
OK
--> myLoop
OK
2 3 4 5 -->
```

ATTENTION : if the number of iterations is zero, the number of iterations increases to 232.

Connection instructions in XTENSA assembler

The XTENSA assembler in the **xtensa** vocabulary has several types of branch instructions :

- connections using Boolean flags defined in the special register **BR** : **BF**, **BT**,
- the connections carrying out tests on the registers: **BALL**, **BANY**, **BBC**, **BBS**, **BEQ**, **BGE**, **BLT**, **BNE**, **BNONE**,

It is this second category of connections that interests us.

Defining branching macros

The ESP32forth xtensa assembler does not have a label management mechanism as is the case for a classic assembler. To be effective, label management must work in several stages if forward branches need to be resolved. This is incompatible with the operation of the Forth language which compiles or assembles in a single pass.

We overcome this difficulty by defining two macro instructions, **If**, and **Then**, which will manage these forward connections:

```
: If, ( -- BRANCH_OFFSET )
```

```

chere 1-
;

: Then, { BRANCH_OFFSET -- }
    chere BRANCH_OFFSET - 2 -
    BRANCH_OFFSET [ internals ] ca! [ asm xtensa ]
;

```

The macro instruction must be preceded by another macro instruction. For our first test, we define the macro <, which will assemble an unresolved branch:

```

: <, ( as at -- )
    0 BGE,
;

```

Using these macros in our first example:

```

code my< ( n1 n2 -- f1 )      \ f1=1 if n1 < n2
    a1 32          ENTRY,
    a8             arPOP,           \ a8 = n2
    a9             arPOP,           \ a9 = n1
    a7 0            MOVI,           \ a7 = 1
    a8 a9 <, If,
        a7 1        MOVI,           \ a7 = 0
    Then,
        a7        arPUSH,
                    RETW.N,
end-code

```

Syntax of branching macro instructions

In our example, we used the macro instruction <, which is associated with the BGE branch instruction , and whose meaning is: "Branch if Greater Than or Equal". Normally, it would be translated by ">=". Why was "<" used?

This is because our macro instruction **If, Then,** has a logic opposite to that of the branch to be carried out. The code enclosed in **If, Then,** will execute if the required condition is not valid. Here is the table which summarizes this reversed logic explaining the choice of the name of these macro instructions used before **If, Then ,**:

XTENSA branch instruction			Macro
BEQ	Branch if Equal	AR[s] = AR[t]	=,
BGE	Branch if Greater Than or Equal	AR[s] ≥ AR[t]	<,
BLT	Branch if Less Than	AR[s] < AR[t]	>=,
BNE	Branch if Not Equal	AR[s] ≠ AR[t]	≠,

my< assembly example . Here is what the execution of the word **my< gives :**

```

10 20 my< .      \ displays: 1
20 20 my< .      \ displays: 0

```

```

20 10 my< .      \ displays: 0
-5 35 my< .      \ displays: 1
-10 -3 my< .     \ displays: 1
-3 -10 my< .    \ displays: 0

```

We see that this reversed logic is respected.

Once this logic is understood, we can define a new macro-instruction `>=` ,:

```

: >=, ( as at -- )
  0 BLT,
;

```

And test this macro-instruction :

```

code my>= ( n1 n2 -- f1 )   \ f1=1 if n1 < n2
  a1 32          ENTRY,
  a8             arPOP,           \ a8 = n2
  a9             arPOP,           \ a9 = n1
  a7 0            MOVI,           \ a7 = 1
  a8 a9 >=, If,
    a7 1          MOVI,           \ a7 = 0
Then,
  a7             arPUSH,
                    RETW.N,
end-code

10 20 my>= .      \ displays: 0
20 20 my>= .      \ displays: 1
20 10 my>= .      \ displays: 1
-5 35 my>= .     \ displays: 0
-10 -3 my>= .    \ displays: 0
-3 -10 my>= .   \ displays: 1

```

Definition and manipulation of registers

In the ESP32 technical document, we find a very large quantity of registers. These registers allow you to control all peripherals and GPIO ports on the ESP32 board.

SAR ADC2 control registers			
SENS_SAR_READ_CTRL2_REG	SAR ADC2 data and sampling control	0x3FF48890	R/W
SENS_SAR_MEAS_START2_REG	SAR ADC2 conversion control and status	0x3FF48894	RO
ULP coprocessor configuration register			
SENS_ULP_CP_SLEEP_CYC0_REG	Sleep cycles for ULP coprocessor	0x3FF48818	R/W
Pad attenuation configuration registers			
SENS_SAR_ATTEN1_REG	2-bit attenuation for each pad	0x3FF48834	R/W
SENS_SAR_ATTEN2_REG	2-bit attenuation for each pad	0x3FF48838	R/W
DAC control registers			
SENS_SAR_DAC_CTRL1_REG	DAC control	0x3FF48898	R/W
SENS_SAR_DAC_CTRL2_REG	DAC output control	0x3FF4889C	R/W

Figure 21: extrait de Technical Reference manual

In general, the manipulation of these registers is carried out by the application layer offered by ESP32forth. It is therefore not necessary to access it directly.

In some cases, it may be interesting to manage registers directly:

- to access features not offered by ESP32forth
- to run Forth code faster

Definition of registers

Defining a register is very simple:

```
$3FF48898 constant SENS_SAR_DAC_CTRL1_REG \ DAC control
```

The first disadvantage of defining a register as a constant is that when reading the source code, we will not be able to distinguish the register from other constants. We will therefore define a register creation word like this:

```
\ define a register, similar as constant
: defREG:
    create ( addr1 --  )
        ,
    does> ( -- regAddr )
        @@
;
;
```

```
$3FF48898 defREG: SENS_SAR_DAC_CTRL1_REG \ DAC control
```

This way, by rereading our code, we know that the word created is a register. The other advantage is that we can modify **defREG:** to change its behavior: adding control tests, initializing parameters, etc...

Access to register contents

Example, bit values in the **SENS_SAR_DAC_CTRL1_REG register** :




Figure 22: bits in register ENS_SAR_DAC_CTRL1_REG

This register contains bits or blocks of bits having defined functions.

We will first create a word allowing us to visualize the contents of a register:

```
\ display reg content
: .reg ( reg -- )
  base @ >r
  binary
  @ <#
  4 for
    aft
    8 for
      aft # then
    next
    bl hold
    then
  next
  #>
  cr space ." 33222222 22221111 11111100 00000000"
  cr space ." 10987654 32109876 54321098 76543210"
  cr type
  r> base !
;
```

Let's see what the contents of our **SENS_SAR_DAC_CTRL1_REG** register give :

```
SENS_SAR_DAC_CTRL1_REG .reg
\ display:
33222222 22221111 11111100 00000000
10987654 32109876 54321098 76543210
00000000 00000000 00000000 00000000 ok
```

The first two lines allow you to read vertically the rank of a bit in this register, here, in red, 25, whose content is 0. To read this bit, proceed as follows:

```
SENS_SAR_DAC_CTRL1_REG @@
1 25 lshift and
```

To modify this bit and set it to 1:

```

1 25 lshift
SENS_SAR_DAC_CTRL1_REG @
or
SENS_SAR_DAC_CTRL1_REG !

```

Let's check with **.reg** :

```

SENS_SAR_DAC_CTRL1_REG .reg
\ display:
33222222 22221111 11111100 00000000
10987654 32109876 54321098 76543210
00000010 00000000 00000000 00000000 ok

```

If it's just once, it helps. Let's see how to do it more efficiently...

Handling register bits

Let's resume modifying bit 25 of our register **SENS_SAR_DAC_CTRL1_REG** , here is how to set bit b25 to 1:

```

SENS_SAR_DAC_CTRL1_REG .reg      \ display:
\ 33222222 22221111 11111100 00000000
\ 10987654 32109876 54321098 76543210
\ 00000000 10000000 00000000 00000000 ok

registers
1 25 $02000000 SENS_SAR_DAC_CTRL1_REG m!
SENS_SAR_DAC_CTRL1_REG .reg      \ display:
\ 33222222 22221111 11111100 00000000
\ 10987654 32109876 54321098 76543210
\ 00000010 00000000 00000000 00000000 ok

```

We use the **m!** word (val shift mask addr --) which accepts four parameters:

- **val** which is the value to modify, here 1
- **shift** which corresponds to the offset to apply to this value, here 25
- **mask** which corresponds to the logical mask of the register part to modify, here \$02000000
- **addr** which is the address of the register, here SENS_SAR_DAC_CTRL1_REG

Definition of masks

A mask is used to indicate which bits are modifiable. In the previous example, we modified bit b25. In the Espressif documentation, bit b25 is marked with the label **SENS_DAC_CLK_INV** . The simplest solution would be to create a constant like this:

```

1 25 lshift constant SENS_DAC_CLK_INV

```

But this does not adjust the value offset which must be the same as the value of the binary mask.

Let's see a more elegant way to define masks:

```
: defMASK:  
    create ( mask0 position -- )  
        dup ,  
        lshift ,  
    does> ( -- position mask1 )  
        dup @  
        swap cell + @  
    ;  
  
1 25 defMASK: mSENS_DAC_CLK_INV
```

In passing, note that the name of the mask is prefixed with the letter '**m**' (for mask). This is by no means obligatory. But when you have compiled many registers and masks, the prefix '**m**' will allow you to find your way between registers and masks:

```
--> words  
mSENS_SW_FSTEP mSENS_SW_TONE_EN mSENS_DAC_DIG_FORCE mSENS_DAC_CLK_FORCE_LOW  
mSENS_DAC_CLK_FORCE_HIGH mSENS_DAC_CLK_INV defMask: SENS_SAR_DAC_CTRL2_REG  
SENS_SAR_DAC_CTRL1_REG GPIO_ENABLE_W1TC_REG GPIO_ENABLE_W1TS_REG GPIO_ENABLE_REG  
GPIO_OUT_W1TC_REG GPIO_OUT_W1TS_REG GPIO_OUT_REG DR_REG_GPIO_BASE PIN_DAC2  
PIN_DAC1 CONFIG_IDF_TARGET_ESP32S3 CONFIG_IDF_TARGET_ESP32S2 .reg AdcREG:  
mtst mset mclr --DAdirect SENS_DAC_CLK_INV defMASK: input$ c+$! mid$ left$  
right$ 0$! $! maxlen$ string $= FORTH camera-server camera telnetd bterm  
.....
```

The word defined with **defMASK:** places the mask offset on the stack, here 25 for **mSENS_DAC_CLK_INV** and the value of the binary mask to apply.

Let's resume modifying bit b25 with this mask definition:

```
1 mSENS_DAC_CLK_INV SENS_SAR_DAC_CTRL1_REG m!  
SENS_SAR_DAC_CTRL1_REG .reg \ display:  
\ 33222222 22221111 11111100 00000000  
\ 10987654 32109876 54321098 76543210  
\ 00000010 00000000 00000000 00000000  
  
0 mSENS_DAC_CLK_INV SENS_SAR_DAC_CTRL1_REG m!  
SENS_SAR_DAC_CTRL1_REG .reg \ display:  
\ 33222222 22221111 11111100 00000000  
\ 10987654 32109876 54321098 76543210  
\ 00000000 00000000 00000000 00000000
```

Switching from C language to Forth language

With ESP32Forth, there are two solutions for adding primitives to the dictionary:

- rewrite the ESP32Forth source code by adding the desired primitives;
- rewrite these C language words in Forth

The first solution, in C language, is written here:

```
#ifndef ENABLE_DAC_SUPPORT
#define OPTIONAL_DAC_SUPPORT
#else
#include <driver/dac.h>
#include <driver/dac_common.h>
#include <soc/rtc_io_reg.h>
#include <soc/rtc_cntl_reg.h>
#include <soc/sens_reg.h>
#include <soc/rtc.h>
#define OPTIONAL_DAC_SUPPORT \
Y(dac_output_enable, n0 = dac_output_enable( (dac_channel_t) n0 ) ) \
Y(dac_output_disable, n0 = dac_output_disable( (dac_channel_t) n0 ) ) \
Y(dac_output_voltage, n0 = dac_output_voltage((dac_channel_t) n1, (gpio_num_t) n0); NIP ) \
Y(dac_cw_generator_enable, PUSH dac_cw_generator_enable () ) \
Y(dac_cw_generator_disable, PUSH dac_cw_generator_disable () ) \
Y(dac_i2s_enable, PUSH dac_i2s_enable() ) \
Y(dac_i2s_disable, PUSH dac_i2s_disable() ) \
Y(rtc_freq_div_set, REG_SET_FIELD(RTC_CNTL_CLK_CONF_REG, RTC_CNTL_CK8M_DIV_SEL, n0 ); DROP ) \
Y(dac_freq_step_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL1_REG, SENS_SW_FSTEP, n0, SENS_SW_FSTEP_S); DROP ) \
Y(dac1_scale_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_SCALE1, 0, SENS_DAC_SCALE1_S); ) \
Y(dac2_scale_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_SCALE2, 0, SENS_DAC_SCALE2_S); ) \
Y(dac1_offset_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_DC1, n0, SENS_DAC_DC1_S); DROP ) \
Y(dac2_offset_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_DC2, n0, SENS_DAC_DC2_S); DROP ) \
Y(dac1_invert_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_INV1, n0, SENS_DAC_INV1_S); DROP ) \
Y(dac2_invert_set, SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_INV2, n0, SENS_DAC_INV2_S); DROP ) \
Y(dac1_cosine_enable, SET_PERI_REG_MASK(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_CW_EN1_M); ) \
Y(dac2_cosine_enable, SET_PERI_REG_MASK(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_CW_EN2_M); ) \
Y(dacWrite, dacWrite(n1, n0); DROPN(2))
#endif
```

The second solution is to look at the source code of a file written in C language and try to understand how registers are manipulated in this language.

Extract from the **dac-cosine.c** file :

```
/*
 * Enable cosine waveform generator on a DAC channel
 */
void dac_cosine_enable(dac_channel_t channel)
{
    // Enable tone generator common to both channels
    SET_PERI_REG_MASK(SENS_SAR_DAC_CTRL1_REG, SENS_SW_TONE_EN);
    switch(channel) {
        case DAC_CHANNEL_1:
            // Enable / connect tone generator on / to this channel
            SET_PERI_REG_MASK(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_CW_EN1_M);
            // Invert MSB, otherwise part of waveform will have inverted
            SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_INV1, 2, SENS_DAC_INV1_S);
            break;
        case DAC_CHANNEL_2:
```

```

        SET_PERI_REG_MASK(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_CW_EN2_M);
        SET_PERI_REG_BITS(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_INV2, 2, SENS_DAC_INV2_S);
        break;
    default :
        printf("Channel %d\n", channel);
    }
}

```

One of the C functions that comes up often is **SET_PERI_REG_MASK**. This function sets the bits designated by a mask in a register to 1. Example:

```
SET_PERI_REG_MASK(SENS_SAR_DAC_CTRL2_REG, SENS_DAC_CW_EN1_M);
```

The C function that sets bits designated by a mask in a register to 0 is

CLEAR_PERI_REG_MASK.

We will be interested in how we will rewrite **dac_cosine_enable(dac_channel_t channel)** in Forth language. We see that the register SENS_SAR_DAC_CTRL2_REG is mentioned. We will define this register:

```
$3FF4889c defREG: SENS_SAR_DAC_CTRL2_REG \ DAC output control
```

SENS_SAR_DAC_CTRL2_REG register , the two bits that interest us are b24 and b25. Let's define the corresponding masks:

```
1 24 defMASK: mSENS_DAC_CW_EN1 \ selects CW generator as source for PDAC1
1 25 defMASK: mSENS_DAC_CW_EN2 \ selects CW generator as source for PDAC2
```

'Bare-metal' programming acting directly on the ESP32 registers does not require defining in Forth all the registers and register masks as the C language does. Limit yourself to the registers and masks essential for your application.

It is recommended to use the names of registers and register masks as appearing in the Espressif documentation, or failing that the register names used in the C language source codes.

Managing some devices is very complex. The Espressif documentation is stingy with examples on the direct use of registers.

The random number generator


Characteristic

The random number generator generates true random numbers, which means a random number generated from a physical process, rather than by means of an algorithm. No number generated within the specified range is more or less likely to appear than any other number.

Each 32-bit value that the system reads from the **RNG_DATA_REG** register of the random number generator is a true random number. These true random numbers are generated based on thermal noise in the system and asynchronous clock skew.

Thermal noise comes from the high speed ADC or the SAR ADC or both. Whenever the high speed ADC or SAR ADC is activated, the bit streams will be generated and fed into the random number generator via an XOR logic gate as random seeds.

When the **RTC8M_CLK** clock is enabled for the digital core, the random number generator will also sample **RTC8M_CLK** (8 MHz) as a random binary seed. **RTC8M_CLK** is an asynchronous clock source and it increases the RNG entropy by introducing circuit metastability. However, to ensure maximum entropy, it is also recommended to always enable an ADC source.



When there is noise from the SAR ADC, the random number generator is fed with an entropy of 2 bits in a clock cycle of **RTC8M_CLK** (8 MHz), which is generated from an internal RC oscillator (see the Reset and Clock chapter for more details). Thus, it is advisable to read the **RNG_DATA_REG** register at a maximum rate of 500 kHz to obtain the maximum entropy.

When there is noise from the high-speed ADC, the random number generator is fed 2-bit entropy in an APB clock cycle, which is normally 80 MHz. Thus, it is advisable to read the **RNG_DATA_REG register** at a maximum rate of 5 MHz to obtain the maximum entropy.

A 2 GB data sample, which is read from the random number generator at a frequency of 5 MHz with only the high speed ADC being enabled, was tested using the Dieharder Random Number test suite (version 3.31.1). The sample passed all tests.

Programming procedure

When using the random number generator, make sure that at least SAR ADC, High Speed ADC, or RTC8M_CLK is allowed. Otherwise, pseudo-random numbers will be returned.

- SAR ADC can be activated using the DIG ADC controller.
- High-speed ADC is automatically enabled when Wi-Fi or Bluetooth modules are enabled.
- RTC8M_CLK is enabled by setting the RTC_CNTL_DIG_CLK8M_EN bit in the RTC_CNTL_CLK_CONF_REG register.

When using the random number generator, read the **RNG_DATA_REG** register several times until there are enough random numbers generated.

Name	Description	Address	Access
RNG_DATA_REG	Random number data	\$3FF75144	RO

```
\ Random number data
$3FF75144 constant RNG_DATA_REG

\ get 32 bits random b=number
: rnd ( -- x )
    RNG_DATA_REG UL@
;

\ get random number in interval [0..n-1]
: random ( n -- 0..n-1 )
    rnd swap mod
;
```

RND function in XTENSA assembler

Since version 7.0.7.4, ESP32forth has an XTENSA assembler. It is possible to rewrite our **rnd** word in XTENSA assembler:

```
forth definitions
asm xtensa
$3FF75144 constant RNG_DATA_REG


code myRND ( -- [addr] )
    a1 32          ENTRY,
    a8 RNG_DATA_REG L32R,      \ a8 = RNG_DATA_REG
    a9 a8 0        L32I.N,    \ a9 = [a8]
    a9            arPUSH,     \ push a9 on stack
                           RETW.N,
end-code
```


The LoRa transmission system

LoRa is a communications technology that uses a low-power wide area network. LoRa allows you to wirelessly connect devices and gateways.


This standard does not yet require any subscription. It offers **peer-to-peer** communication.

LoRa, WiFi and Bluetooth are complementary and do not overlap. Compared to Wi-Fi and Bluetooth which provide very short range, LoRa benefits from very narrow bandwidth. Gateways or hubs are used barely 1% of the time by connected devices. Which significantly reduces bandwidth. Traffic is slow and unidirectional between the sensors and the gateway. LoRa is the best way to communicate over several kilometers, with very little power and in a very simple way!



Cabling of the REYAX LR890 LoRa transmitter

The transmitter is connected to the ESP32 card like this:



ATTENTION: check the position of pins G16 and G17 on your ESP32 card which may be different depending on your version of ESP32 card.

The LoRa transmitter for ESP32

The REYAX LR890 module costs around €15. It weighs 7 grams.

Its consumption, in transmission, is 43 mA (3.3V). In reception, it is 16.5 mA and can drop to 0.5 mA in SLEEP mode.

To ensure point-to-point transmission, two LoRa modules are required. Each module is a transmitter and receiver.

The ESP32 card communicates via its serial port with the LoRa module. All transmissions between the ESP32 board and the LoRa transmitter are processed through AT commands. Example:

```
AT+SEND=50,5,HELLO
```

This chain is transmitted by the ESP32 card to the LoRa transmission module:

- the LoRa module switches to transmission mode and transmits this character string
- immediately after transmission, the LoRa module returns to receive mode
- the remote LoRa module receives the character string.
- the remote LoRa module can acknowledge this reception by **+OK**

A LoRa module can communicate with a LoRaWan gateway. It is generally a box connected to a router via Ethernet connection. It is therefore possible to have a web application that communicates with one or more LoRa modules.

LoRa transmission security

A single LoRa module can communicate with several remote LoRa modules.

These LoRa modules must be differentiated by their **NETWORKID** . The transmitter and receiver must have the same **NETWORKID** .

Then, each module receives an **ADDRESS** , by default 0. This address is between 0..65535.

Transmission can be encrypted using a 32-character **AES key** . The transmitter and receiver LoRa modules must have the same **AES key** . If a module receives a message encrypted with an unknown **AES key** , **it will ignore the message**.

And finally, each module is assigned a transmission frequency. The transmitter and receiver modules must work on the same frequency.

Example of 868.5 MHz frequency selection.

```
\ select frequency 865.5 Mhz for LoRa transmission
```

```
32 string AT_BAND
s" AT+BAND=868500000" AT_BAND $! \ set frequency at 868.5 MHz
$0a AT_BAND c+$!
$0d AT_BAND c+$! \add CR LF code at end of command
AT_BAND Serial2.write drop
```

On the same frequency, we can manage a fleet of 65,535 LoRa modules, each module having its address. If we transmit with address 0, we will address all LoRa modules.

If we add the AES encryption key, there will be hundreds of thousands of LoRa modules that can coexist within a radius of a few kilometers!


The range of the modules can be increased by changing the transmit power. We can also act on the receiving antenna. With a directional antenna, you can reach **20 to 30 kilometers** of range...

Review of the REYAX RYLR890 LoRa transmitter

Required test environment

To test our REYAX RYLR890 LoRa transmitter, you must:

- use string management words.... @todo: reference in file
- use an ESP32Forth version with access to the UART2 port
- wire the REYAX RYLR890 LoRa transmitter as follows:



Prepare communication with the LoRa transmitter

All programmers who manage the UART2 port define a memory area which will serve as a buffer. For our part, we will directly create an alphanumeric variable:

```
128 string LoRaTX \ buffer ESP32 -> LoRa transmitter
```

Here we will only carry out tests of sending commands to the REYAX RYLR890 LoRa transmitter and see how to recover what this same transmitter sends back. We therefore need another alphanumeric variable for reception:

```
128 string LoRaRX \ buffer LoRa transmitter -> ESP32
```

Let's start by initializing the serial transmission to the LoRa transmitter. Here the speed is 115200 baud. This is the default transmission speed of the LoRa transmitter:

```
Serial \ Select Serial vocabulary
```

```
\ initialise Serial2
: Serial2.init ( -- )
    #SERIAL2_RATE Serial2.begin
;
```

For our test command to the LoRa transmitter, we select the working frequency of the transmitter, here 868.5 Mhz:

```
\ Setup LoRa Frequency
: .band8685 ( -- )
    s" AT+BAND=868500000" LoRaTX $!
    $0d LoRaTX c+$!
    $0a LoRaTX c+$!      \ add CR LF code at end of command
    LoRaTX Serial2.write drop
;
```

Finally, we define a word allowing us to retrieve the response from the LoRa transmitter:

```
\ input from LoRa transmitter
: LoRaInput ( -- n )
    Serial2.available dup if
        LoRaRX maxlen$ nip
        Serial2.readBytes
        LoRaRX drop cell - !
    then
;
```

The **LoRaInput** word tests whether a serial link transmission has been received from the UART2 serial port:

- if there is no reception, returns 0
- if it has characters, stores those characters in the **LoRaRX** alphanumeric string and updates the size of that string.

Example of transmission and reception :

```
Serial2.init
.band8685
LoRaInput
```

Here is what a **LoRaRX** memory dump gives :

```
LoRaRX dump
--addr--- 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F  -----chars-----
3FFF-87A0 05 00 00 00 2B 4F 4B 0D 0A 31 0D 0A 2B 4F 4B 0D ....+OK..1..+OK.
```

If we run **LoRaRX** , we recover the address and especially the length of all the characters received, including the characters CR+LF (\$0d \$0a). To only manage characters strictly included in the interval [0..9A..Za..z], you must reduce the size of the string by two units :

```
: LoRaType ( -- )
    LoRaRX dup 0 > if
        2 - type
    else
        2drop
    then
;
LoRaType \ display: +OK
```

Here, running **LoRaType** displays +OK which is the response to our test command **AT+BAND=868500000** .

Setting up the REYAX RYLR890 LoRa transmitter

Before defining commands for our REYAX RYLR896 LoRa transmitter, we define the word **crlf** :

```
: crlf ( -- )      \ same action as cr, but adapted for LoRa
  $0d emit
  $0a emit
;
```

The purpose of this crlf word is to complete the transmission on the UART2 port from the ESP32 board to the LoRa transmitter. The definition of this word uses emit. Do not be surprised. We will see later how to exploit the vectorized execution of words in Forth language to perform the desired action at emit. This solution will surprise beginner Forth language programmers. It will also show how Forth is much more flexible than many other programming languages.

Essential parameters

Here is the list of essential parameters to configure your LoRa module.

The sequence of using the **AT** command :

- Use **AT+ADDRESS** to set ADDRESS. The ADDRESS is considered to be the identification of the specified transmitter or receiver.
- Use **AT+NETWORKID** to set the Lora network ID. This is a group function. Only by setting the same NETWORKID can modules communicate with each other. If the specified recipient's ADDRESS belongs to a different group, it is not able to communicate with each other. The recommended value: 1 ~ 15
- Use **AT+BAND** to adjust the center frequency of the wireless band. The transmitter and receiver must use the same frequency to communicate with each other.
- Use **AT+PARAMETER** to adjust RF wireless settings. The transmitter and receiver must set the same parameters to communicate with each other. The parameters are to be defined as follows:
 - **<Spreading Factor>** : The larger the SF, the better the sensitivity. But the transmission time will take longer.
 - **<Bandwidth>** : The smaller the bandwidth, the better the sensitivity. But the transmission time will take longer.

- **<Coding Rate>** : The coding rate will be fastest if you set it to 1.
- **<Programmed Preamble>** : Preamble code. If the preamble code is bigger, it will result in less chance of losing data. Preamble code generally can be set above 10 if under the authorization of transmission time.
 - * Communication up to 3 km: recommended setting "AT+PARAMETER=10.7,1.7"
 - * More than 3 km: recommended setting "AT+PARAMETER=12.4,1.7"
- Use **AT+SEND** to send data to the specified ADDRESS. Due to the program used by the module, the payload part will increase by more than 8 bytes compared to the actual data length.

It is necessary to pass **crlf at the end of all AT commands**.

You must wait for the module to respond **+OK** so that you can execute the next **AT command**.

ADDRESS Defines the module address

Each LoRa transmission module must have a personal address.

syntax	response
AT+ADDRESS=<address>	+OK
AT+ADDRESS=?	+ADDRESS=22

```
\ Set the ADDRESS of LoRa transmitter:
\ s" " value in interval [0..65535][?] (default 0)
: ATAddress ( addr len -- )
  ." AT+ADDRESS="
  type crlf
;
<Address>=0~65535(default 0)
```

Example: Set module address to **22**. The settings will be stored in LoRa.

```
s" 22" Ataddress
```



AT Test LoRa Availability

syntax	response
A.T.	+OK

```
\ Test LoRa disponibility
: AT_ ( -- )
    ." AT"
    type crlf
;
```

BAND Setting the RF frequency

syntax	response
AT+BAND=<parameter>	+OK
AT+BAND=?	+BAND=868500000

```
\ Set the BAND of LoRa transmitter:
\ s" " value is RF frequency, unit Hz
: ATband ( addr len -- )
    ." AT+BAND="
    type crlf
;
```

Parameter is RF frequency, unit is Hz: 915000000Hz (default: RYLY89x)

Example: Select the frequency at 868500000Hz:

```
s" 868500000" ATband
```

CPIN Sets the AES128 network password

syntax	response
AT+CPIN=<password>	+OK
AT+CPIN=?	+CPIN=FABC0002EEDCA.....

Password: AES password of 32 characters from 00000000000000000000000000000001 to FFFFFFFFFFFFFFFFFFFF.

The exchange is accepted if both modules have the same password. After reset, the previous password is deleted.

```
\ Set the AES32 password:
\ s" <parameter>" value is an 32 character long AES password
\ from 00000000000000000000000000000001 to FFFFFFFFFFFFFFFFFFFF
: ATcpin ( addr len -- )
    ." AT+CPIN="
    type crlf
;
```

Example: Select this password: FABC0002EEDCAA90FABC0002EEDCAA90

```
s" FABC0002EEDCAA90FABC0002EEDCAA90" ATcpin
```

CRFOP Selects the output RF power

syntax	response
AT+CRFOP=<power>	+OK
AT+CRFOP=?	+CRFOP=10

Power: between 0..15, 15dBm (default)

```
\ Set the CRFOP RF output power:  
\ s" " value is RF output power between 0..15  
: ATcrfop ( addr len -- )  
  ." AT+CRFOP="  
  type crlf  
;  
;
```

Example, select the output power at 10dBm:

```
s" 10" ATcrfop
```

FACTORY Sets all current settings to default values

Sets all current settings to manufacturer defaults.

syntax	response
AT+FACTORY	+FACTORY

```
\ Reset the LoRa transmitter with FACTORY parameters  
: ATfactory ( -- )  
  ." AT+FACTORY"  
  crlf  
;  
;
```

IPR Sets the UART baud rate

syntax	response
AT+IPR=<parameter>	+OK
AT+IPR=?	+IPR=38400

UART baud parameter:

- 300
- 1200
- 4800
- 9600

- 19200
- 28800
- 38400
- 57600
- 115200 (default)

The settings will be stored in the EEPROM.

MODE Selects the working mode

syntax	response
AT+MODE=<parameter>	+OK
AT+MODE=?	+MODE=1

Setting:

- 0: Transmit and Receive mode (default).
- 1: Sleep mode.

```
\ Set work MODE:
\ s" " value is [0,1]
\ 0 (defalt) Transmit and Receive mode
\ 1 Sleep mode
: ATmode ( addr len -- )
    ." AT+MODE"
    type crlf
;
```

NETWORKID Selects the network ID

syntax	response
AT+NETWORKID=<Network ID>	+OK
AT+NETWORKID=?	+NETWORKID=6

```
\ Set NETWORKID:
\ s" " value is [0..16] (0 bay default)
: ATnetworkid ( addr len -- )
    ." AT+NETWORKID"
    type crlf
;
```

Network ID: 0~16(0 by default)

Example: select network ID to 6. The settings will be stored in the EEPROM.

The “0” is the public identifier for LoRa. It is not recommended to use 0 to set the NETWORK ID.

s" 6" Atnetworkid

PARAMETER definition of RF parameters

syntax	response
AT+PARAMETER=<Spreading Factor>, <Bandwidth>, <Coding Rate>, <Programmed Preamble>	+OK
AT+PARAMETER=?	+PARAMETER=7,3,4,5

Parameters:

- Spreading Factor
 - 7~12, (default 12)
- Bandwidth / bandwidth (0~9):
 - 0: 7.8KHz (not recommended, over spec.)
 - 1: 10.4KHz (not recommended, over spec.)
 - 2: 15.6KHz
 - 3: 20.8KHz
 - 4: 31.25KHz
 - 5: 41.7KHz
 - 6: 62.5KHz
 - 7: 125KHz (default).
 - 8: 250KHz
 - 9: 500KHz
- Coding rate
 - 1~4, (default 1)
- Programmed Preamble
 - 4~7 (default 4)

Spreading Factor

Spreading factor	Bitrate / flow
7	5469 bps
8	3125 bps
9	1758bps
10	977 bps

Spreading factor	Bitrate / flow
11	537 bps
12	293 bps

Coding rate

LoRa modulation also adds forward error correction (FEC) into each data transmission. This implementation is done by encoding 4-bit data with redundancies in 5 bits, 6 bits, 7 bits, or even 8 bits. Using this redundancy will allow the LoRa signal to cover interference. The coding rate value should be adjusted according to the conditions of the channel used for data transmission. If there is too much interference in the channel, then it is recommended to increase the coding rate value.

However, increasing the CR value will also increase the transmission duration.

Example: setting parameters as below:

<Spreading Factor> 7,<Bandwidth> 20.8KHz, <Coding Rate> 4,<Programmed Preamble>5,

```
s" 7,3,4,5" Atparameter
```

Software RESET

syntax	response
AT+RESET	+OK

```
\ RESET the LoRa transmitter
: ATreset ( -- )
    ." AT+RESET"
    crlf
;
```

SEND sending data to the designated address

syntax	response
AT+SEND=<Address>,<Payload Length>,<Data>	+OK
AT+SEND=?	+SEND=50.5,HELLO

<Address>0~65535, when <address> is 0, it will send data to all address (from 0 to 65535.)

<Payload Length>Maximum 240 bytes

<Data>ASCII Format

Forth code for ESP32Forth:

```
\ convert a number to a decimal string
: .n ( n --- )
```

```

base @ >r decimal
<# #s #> type
r> base !
;
\ SEND Send data to the appointment address
: ATsend { addr len address -- }
." AT+SEND="
address .n [char] , emit
len .n [char] , emit
addr len type crlf
;

```

Example: sends the string **HELLO** to Address 50:

```
s" HELLO" 50 ATsend \ display: AT+SEND=50;5;HELLO
```

VER to request firmware version

```

\ VER to inquire the firmware version
: ATver ( -- )
." AT+VER"
crlf
;
```

Error result codes

- +ERR=1 there is no “enter” or \$0D \$0A at the end of the AT command
- +ERR=2 the head of the AT command is not an “AT” string
- +ERR=3 there is no “=” symbol in the AT command
- +ERR=4 unknown command
- +ERR=10 TX is on time
- +ERR=11 RX is exceeded
- +ERR=12 CRC error
- +ERR=13 TX data of more than 240 bytes
- +ERR=15 Unknown error

Vectorization of character emissions

If you have followed the development of our words to configure the REYAX RYLR890 LoRa transmitter up to this point, something has certainly surprised you:

```

\ Set the ADDRESS of LoRa transmitter:
\ s" <address>" value in interval [0..65535][?] (default 0)
: ATaddress ( addr len -- )
." AT+ADDRESS="
type crlf
;
```

Because, unless I'm mistaken, this sequence "**AT+ADDRESS=**" sends the character string to our terminal, and not to the transmitter via the UART2 serial port, therefore to the LoRa transmitter!

We understand your surprise. And we will see how to divert the character flow to the LoRa transmitter without changing anything in the definition of our word **ATaddress**.

Understanding vectorization in Forth

The Forth language has certain advantages that are completely non-existent in many other programming languages. Among these assets, let us mention the word **defer**. This word allows you to create a word whose action is deferred:

```
defer myWords
```

defer creates a **myWords** word that does NOTHING!!!

Yes!

It is now up to us to give it action. Let's see this definition:

```
: (myWords) ( -- )
    cr ." I display my words: "
;
```

For **myWords** to execute **(myWords)** we get the action code from **(myWords)** and assign it to **myWords**:

```
' (myWords) is myWords
```

From now on, if we type **myWords**, the action defined in **(myWords)** will be executed.

OK. But here, is it necessary to do such an overhead of code if we can simply execute **(myWords)** ?

And you are absolutely right to ask this question. But you can change the action of **myWords** :

```
' vlist is myWords
```

Now, if we type **myWords** , the word **vlist** is executed.

We will see how to use this mechanism to modify the behavior of ESP32Forth...

Vectorization in ESP32Forth

Let's start with a little reverse engineering. Digging into the ESP32Forth code, we find this for the word **."**:

```
: ."
    postpone s" state @ if postpone type else type then ; immediate
```

Here, the word that interests us is **type** whose definition is:

```
defer type
```

Ahhh.... Are you starting to understand?

What action does **type**? We find this in the source code of ESP32Forth :

```
: serial-type ( a n -- ) Serial.write drop ;
: default-type serial-type ;

' default-type is type
```

If we are interested in the word **emit**, we find this definition in the ESP32Forth source code :

```
: emit ( n -- )
>r rp@ 1 type rdrop ;
```

Here again, we find **type**.

It is therefore on this type word that we will act to divert the emission of characters towards the UART2 serial port.

Vectorize type to UART2 serial port

It is by looking at the definition of **serial-type** that we define our version to transmit to the UART2 serial port :

```
: serial2-type ( a n -- )
  Serial2.write drop ;
```

How far has it gone? Isn't that too difficult?

Now, to divert the character transmission stream from ESP32Forth to the UART2 serial port, simply execute the sequence '**' serial2-type is type**' .

But if you do that, you will have a little difficulty returning to normal behavior of ESP32Forth unless you restore **type** to its initial action with the sequence '**' default-type is type**' .

Let's encapsulate these sequences in these two words:

```
: typeToLoRa ( -- )
  ['] serial2-type is type
;

: typeToTerm ( -- )
  ['] default-type is type
;
```

And now, to execute our **ATaddress** word by making it transmit the characters to the UART2 serial port, just type :

```
typeToLoRa
s" 45" ATaddress \ send AT+ADDRESS=45 to UART2
typeToTerm
```

And there, I await your remark: "but what is the point of going through vectorization?"

In our case, vectorization offers many advantages :

- write simple code with words already known from the Forth dictionary of ESP32Forth ;
- offers the possibility of testing all the configuration words of the LoRa transmitter towards the terminal ;
- possibility to divert the flow to another device, for example I2S or UART1, without having to rewrite these parameter definitions...

Regardless of our management of the LoRa transmitter parameters, we understand that it is enough to exploit this same vectorization mechanism for the characters received from the UART2 serial port to easily take control of ESP32Forth from this serial port!

This is actually what ESP32Forth does if you activate the WiFi or Bluetooth port! I invite you to explore the source code of ESP32Forth. Look at the **server** definition :

```
: server ( port -- )
    server
    ['] serve-key is key
    ['] serve-type is type
    webserver-task start-task
;
```

Rewriting a complete listing

The minimum parameters to communicate between ESP32+LoRa cards are: frequency and address:

```
\ *** defining LoRa Setup words ****
create $crlf
    $0d c, $0a c,
:crlf ( -- )           \ same action as cr, but adapted for LoRa
    $crlf 2 type
;

\ Set the ADDRESS of LoRa transmitter:
\ s" " value in interval [0..65535][?] (default 0)
: ATAddress ( addr len -- )
    ." AT+ADDRESS="
    type crlf ;

\ Set the BAND of LoRa transmitter:
\ s" " value is RF frequency, unit Hz
: ATband ( addr len -- )
    s" AT+BAND=" type
    type crlf ;
```

LoRa transmitters, taken out of their original packaging, theoretically communicate at 115200 bauds with the ESP32 card :

```

\ 115200 speed communication for LoRa REYAX
115200 value #SERIAL2_RATE

\ definition of OUTput and INput buffers
128 string LoRaRX \ buffer LoRa transmitter -> ESP32

Serial \ Select Serial vocabulary

\ initialise Serial2
: Serial2.init ( -- )
    #SERIAL2_RATE Serial2.begin
;

```

We also recover the word **LoRaInput** which reads the messages returned by the LoRa transmitter on the UART2 port. The word **rx.** has been added to facilitate handling :

```

\ input from LoRa transmitter
: LoRaInput ( -- n )
    Serial2.available if
        LoRaRX maxlen$ nip
        Serial2.readBytes
        LoRaRX drop cell - !
    else
        0 LoRaRX drop cell - !
    then
;

: rx.
    LoRaINPUT
    LoRaRX type
;

```

Here, the words **typeToLoRa** and **typeToTerm** are used to transfer the text display from the terminal to the UART2 port :

```

\ *** defining defered words ****
serial \ Select Serial vocabulary

: serial2-type ( a n -- )
    Serial2.write drop ;

: typeToLoRa ( -- )
    0 echo ! \ disable display echo from terminal
    '[' serial2-type is type
;


: typeToTerm ( -- )
    '[' default-type is type
    -1 echo ! \ enable display echo from terminal
;

```

Here we have the necessary and sufficient Forth words to configure our three REYAX RYLR890 LoRa transmitters.

Setting up LoRa transmitters

In this photo you have **BOSS** and **SLAV1** labels. These are simple adhesive post-its stuck to the test plates.



We will create three constants associated with these labels:

```
55 constant LoRaBOSS
39 constant LoRaSLAV1
40 constant LoRaSLAV2
```

To communicate with each other, our LoRa transmitters must be set to use the same frequency. The chosen frequency is 868.5 MHz, or 868500000 Hz:

```
: emptyRX ( -- )
    LoRaINPUT
;

: SETband ( -- )
    emptyRX
    typeToLoRa
    s" 868500000" ATband
    typeToTerm
;
```

Let's start configuring our first LoRa transmitter :

```
serial2.init
SETband
rx. \ display +OK
```

If all goes well, performing **rx.** displays **+OK** .

This is the message returned by the LoRa transmitter. It is possible to get an error message, like **+ERR=1**. Repeat the settings command.

The frequency is indicated in 9 digits, without separators or spaces. The unit is Hz ¹¹.

¹¹ For FRANCE, the free frequency band goes from 863 MHz to 868.6 MHz. Source: ARCEP The "free bands portal"

The REYAX RYLR896 LoRa module can operate frequencies from 862 MHz to 1020 MHz.

ATTENTION : the antenna must be tuned to the frequency used! The antenna fitted to the REYAX LoRa module is tuned for frequencies around 868 MHz. Using a poorly tuned antenna will considerably reduce the transmission efficiency of the LoRa module.

LoRa modules transmit in narrow band. Choose any frequency from the frequencies authorized in your country.

Determining the address of LoRa transmitters

To be operational, all transmitters on a network must be on the same frequency. When you want to transmit a message to a particular transmitter, you must indicate the address of the recipient transmitter. For example, if **BOSS** wants to send a message to **SLAV1**, we will transmit the message to the transmitter which has address 39.

ATTENTION : you cannot have two transmitters with the same address on the same frequency!

Here definition of the word allowing to configure address 55 for the **BOSS transmitter** :

```
: SETaddress ( n -- )
    emptyRX
    typeToLoRa
    str ATband
    typeToTerm
;
LoRaBOSS SETaddress
rx. \ display +OK
```

We can take any value for each transmitter, in the interval [1..65535]. Address 0 is reserved for transmissions to all LoRa transmitters listening on the same frequency.

Now that our **BOSS transmitter** is configured, we can disconnect it from the PC and plug in the one labeled **SLAV1**. We compile the source script and start configuring **SLAV1** :

```
serial2.init
SETband
rx. \ display +OK


LoRaSLAV1 SETaddress
rx. \ display +OK
```

Communication between two REYAX RYLR890 LoRa transmitters

To initialize our REYAX RYLR890 LoRa transmitters, you must:

- have two REYAX RYLR890 LoRa transmitters with the ESP32 card
- connect each ESP32 board to an available USB port on your PC

Here, under Linux, we have opened two Minicom terminals :



Linux commands to open these two terminals:

- from the keyboard, launch the command terminal using CTRL-ALT-T
- in the command window, typing sudo minicom launches the minicom terminal connected to /dev/ttyUSB0
- In the prompt, type the Linux administrator password. The first terminal gives access to your first ESP32 card
- again, from the keyboard, launch the command terminal using CTRL-ALT-T
- in this new command window, type sudo minicom -D /dev/ttyUSB1 which launches another minicom terminal connected to /dev/ttyUSB1
- In the prompt, type the Linux administrator password. This other terminal gives access to the second ESP32 card

Transmission from BOSS to SLAV2

The listing of our Forth code is only slightly different from that of the previous chapter. We simply removed the words ATaddress and ATband. These words are no longer necessary to configure our LoRa **BOSS**, **SLAV1** and **SLAV2 transmitters**.

Once configured, a LoRa transmitter maintains this setting, even when powered off. Powering on an ESP32 card and its LoRa transmitter does not change the settings of the LoRa transmitter.

The words **ATaddress** and **ATband** are replaced by **Atsend** :

```
\ SEND Send data to the appointment address
: Atsend { addr len address -- }
    ." AT+SEND="
    address n. [char] , emit
    len      n. [char] , emit
    addr len type crlf
;
```

Now, we integrate this word into **toSLAV2** :

```
: toSLAV2 ( addr len -- )
emptyRX
typeToLoRa
LoRaSLAV2 ATsend
typeToTerm
;
```

We compile the same program on each ESP32 (**BOSS** and **SLAV2**). It's very easy. Just copy from the listing and paste into the terminal. Each ESP32 card will compile its listing in around ten seconds.



Here, the two windows of our minicom terminal. The left window allows you to control **BOSS**.

The right window controls **SLAV2** :

The screenshot shows two terminal windows. The left window, titled 'LoRaSLAV2 ATsend', contains the following text:

```

ok  LoRaSLAV2 ATsend
ok  typeToTerm
ok ;
ok
ok
ok
ok serial2.init
AT+SEND=39,27,this is a transmission testV1
ok
ok
ok 100 ms
+ERR=1.
ok
ok serial2.init
ok s" this is a transmission test" toSLAV2
ok
ok 100 ms
ok rx.
ok
+OK rx.
ok
--> 

```

The right window, titled 'Serial2.end', contains the following text:

```

lable Serial2.end
Serial.available Serial

```

A cursor is visible in the bottom-left corner of the right window.

Let's test the transmission from **BOSS** to **SLAV2**. To do this, place the mouse pointer in the left window and type directly :

```

serial2.init
s" this is a transmission test" toSLAV2
rx. \ display: +OK

```

Are we sure that **SLAV2** received the message? Nothing easier.

We place the mouse cursor in the right window and simply type:

```

serial2.init
rx. \ display: +RCV=55,27,this is a transmission test,-36,40


```

The result of this transmission by **SLAV2** is stored in the alphanumeric variable **LoRaRX**.

Interfacing a LoRa transmission with ESP32Forth

To demonstrate the incredible flexibility of the Forth language, and more particularly the ESP32Forth version on ESP32, we will use the program used in the chapter *Managing a traffic light with ESP32*.

The problem with the definitions in this chapter is that the control of the LEDs uses the GPIO terminals of the serial link. We therefore move the LED connection like this :




Here is the one and only code adaptation that is made to adapt to the new connection of the LEDs :

```
\ new code
27 constant ledGREEN          \ green LED on GPIO2
26 constant ledYELLOW         \ yellow LED on GPIO21
25 constant ledRED            \ red LED on GPIO17
```

Here are the code sequences in Forth language to selectively turn each LED on or off. These sequences can be executed from the terminal window connected to the ESP32 card :

```
LEDinit
ledGREEN high      \ set GREEN led on
ledRED  high       \ set RED led on
ledGREEN low       \ set GREEN led off
```

And it is these sequences, and these alone, which will be transmitted and received by the LoRa transmitters. Here is the assembly of the LEDs and the LoRa transmitter on our test plate called **SLAV1** :



The LoRa transmitter side program called BOSS

We will complete the busy communications program. We start from what is described in the previous chapter.

The listing includes the essential components allowing LoRa transmission from the ESP32 card marked BOSS.

We simply add a few simple definitions to remotely execute the switching on and off of the LEDs which are on the card marked **SLAV1** :

```
\55 constant LoRaBOSS
39 constant LoRaSLAV1
\ 40 constant LoRaSLAV2

: toSLAV1 (addr len --)
emptyRX
typeToLoRa
LoRaSLAV1 ATsend
typeToTerm
;

: REDhigh ( -- )
s" LEDred high" toSLAV1
;

:REDlow(--)
s" LEDred low" toSLAV1
;

: YELLOWhigh ( -- )
s" ledYELLOW high" toSLAV1
;
```

```

: YELLOW low ( -- )
s" ledYELLOW low" toSLAV1
;

: GREENhigh ( -- )
s" ledGREEN high" toSLAV1
;

: GREENlow ( -- )
s" ledGREEN low" toSLAV1
;\ 55 constant LoRaBOSS
39 constant LoRaSLAV1
\ 40 constant LoRaSLAV2

: toSLAV1 ( addr len -- )
emptyRX
typeToLoRa
LoRaSLAV1 ATsend
typeToTerm
;

: REDhigh ( -- )
s" LEDred high" toSLAV1
;

: REDlow ( -- )
s" LEDred low" toSLAV1
;

: YELLOWhigh ( -- )
s" ledYELLOW high" toSLAV1
;

: YELLOWlow ( -- )
s" ledYELLOW low" toSLAV1
;

: GREENhigh ( -- )
s" ledGREEN high" toSLAV1
;

: GREENlow ( -- )
s" ledGREEN low" toSLAV1
;

```

We create a definition per command, with the aim of making it as simple as possible. You are free to create a more interactive way. This is not the purpose of this chapter. For the moment, once the card marked **BOSS** is connected and the code compiled, if we want to transmit a command to **SLAV1**, we simply type in the terminal:

```


serial2.init
REDhigh

```

LEDred high message to **SLAV1**. The last phase will be to execute this command as if it were typed from a terminal connected to **SLAV1**.

Receipt and execution of Forth commands by SLAV1

Let's summarize: the BOSS transmitter sends a message, for example **LEDred high** to the SLAV1 transmitter. Transmitter SLAV1 receives the message and executes **RXdecode** to store this Forth command in the alphanumeric variable **RCVdata**.



Executing a command received by LoRa

In our diagram, we have two ESP32 boards, each with a LoRa transmitter:

- **BOSS** (address 55) which transmits Forth commands
- **SLAV1** (address 39) which receives these Forth commands

The only difference with Forth commands entered directly on the PC keyboard and transmitted by the terminal program connected to **SLAV1** concerns the commands transmitted by LoRa and stored in **RXdata**.

How to execute these commands stored in **Rxdata**? The answer is shockingly simple :

```
RCVdata evaluate
```

We absolutely do not need anything else TO INTERFACE the LoRa transmission with any program embedded in an ESP32 board!!!

Here is a secure definition of this interfacing:

```
: RXinterface ( -- )
  RCVdata ?dup if
    evaluate
  else
    2drop
  then
;
```

Here are some manipulations in Forth to test this interface:

```
LEDinit          \ initialize GPIOs
s" LEDred high" RCVdata $!
RCVdata RXinterface \ turn RED led on
s" LEDred low" RCVdata $!
RCVdata RXinterface \ turn RED led off
```

We kept our promise : to act from LoRa on any program without changing a single line of code.

In any ESP32 card, you can easily compile and test all the functionalities of your programs with the terminal.

To interface these programs, it will then be enough to add the LoRa transmission layer and its interfacing code.

The remote transmitter will only have to send Forth commands to act on your programs.


Only the Forth language allows such a simple transmission -> application interface!

Now let's see the last point: regularly reading the receive buffer of the LoRa transmitter....

LoRa transmission management loop

The LoRa transmitter is connected to the UART2 serial port. When a transmission is received, the word **Serial2.available** indicates the number of bytes waiting in the UART2 serial buffer. If there is no transmission, the value reported by **Serial2.available** will be zero. Here is the code to test for the presence of characters received by UART2:

```
Serial
\ final loop
: LoRaLoop ( -- )
begin
    Serial2.available \ not 0 if chars available
    if
        100 ms          \ ensures that the entire transmission is received
        LoRaRX maxlen$ nip
        Serial2.readBytes
        LoRaRX drop cell - !
        RXdecode        \ analyse content of LoRa message
        RXinterface     \ interpret content or RCVdata
    then
        pause          \ skip to next task
    again
;
```



LoRaLoop code uses an infinite loop. It is not recommended to execute this word as is. If you do this, you will no longer have control of the Forth interpreter of ESP32Forth.

To use **LoRaLoop** without blocking the Forth interpreter, we will define a new **my-loop** task like this :

```
' LoRaLoop 100 100 task my-loop
my-loop start-task
```

From this moment on, any transmission made from the card marked **BOSS** will be interpreted on this card marked **SLAV1**.

So that all our programming remains persistent in our **SLAV1 card**, we finalize the general initialization :

```
\ 115200 speed communication for LoRa REYAX
115200 value #SERIAL2_RATE

Serial
: mainInit ( -- )
  cr ." Starting SLAV1 LoRa" cr
  LEDinit
  #SERIAL2_RATE Serial2.begin      \ initialise Serial2
  my-loop start-task
;
startup: mainInit
```

From this moment, once the program has been compiled in the ESP32 card marked **SLAV1**, when the card is restarted, the word **mainInit** will be executed, which is confirmed by the Starting **SLAV1** LoRa message which should normally be displayed.

Here is a photo of the actions performed from the **BOSS** card, inset at the bottom left :



On the card marked **SLAV1**, the LEDs react with a latency of one to two seconds. This delay is normal. It results from the LoRa protocol which is certainly slow, but extremely robust. In the photo above, the tests were carried out with a distance of one meter. The **BOSS** and **SLAV1** cards were brought together for the photo.

If you leave **SLAV1** connected to the terminal, you will still have control of the Forth interpreter. This is also normal! The **LoRaLoop word** executes in multi-tasking.

Since its origins, the Forth language has been multitasking. It was already on versions under MS-DOS when MS-DOS was not multitasking.

With ESP32Forth, we continue with the functionalities of Forth, including the possibilities of activating concurrent tasks. In our specific case, the monitor task gives you control over the interpreter while managing the LEDs from the **LoRaLoop** task.

ESP32Forth simple web interface

Author : Vaclav POSELT

I restarted Forth usage after years out of any programming with FlashFORTH on Atmega 328 and Arduino. After creating my first construction it was necessary to build some control panel for electronics, some buttons, display aso. I have thought there is time of IoT and wireless control, so better spare construction work and control all wireless. For this I moved to ESP32 with WiFi and BT, I have found tens of program examples of web interfaces in Arduino C with JavaScript, but nothing in ESP32Forth on ESP32. For me as beginner it was problem.

So next is result of my effort - simple example of web interface, on web server running on ESP32 inForth. Code is based on Peter Forth example peter-webpage-dht11-graphic-example.txt. Whole code is in attached file example_web.fs., line numbers are from this file.

Web server runs on ESP32 board with activated WiFi connection and responds to client (browser on PC, mobile etc.) requests.

So basic web interface is simple:

```
: runpage begin handleClient if serve-page 100 ms then 500 ms again ;
```

where **handleClient** detects if there are client requests, resolves request and gives HTML content to client with word serve-page. This **ms** delays improved wifi connection stability in my home net.

```
: serve-page ( -- ) \ simple parsing and action of client respond
  path s" /" str= if
    htmlpagesend exit \ exit leaves from serve-page
  then
  path s" /26/on" str= if
    cr ." ACTION for /26/on " cr \ here put action word
    0 to GPIO26 htmlpagesend exit
  then
  path s" /26/off" str= if
    cr ." ACTION for /26/off " cr
    1 to GPIO26 htmlpagesend exit
  then
  path s" /27/on" str= if
    cr ." ACTION for /27/on " cr
    0 to GPIO27 htmlpagesend exit
  then
  path s" /27/off" str= if
```

```

cr ." ACTION for /27/off " cr
  1 to GPIO27 htmlpagesend exit
then
path respond      \ actions for html forms
htmlpagesend exit \ resend html page
;

```

The word **serve-page** uses text of client request from word path in form addr len and compares it with possible client responds, each match activates relevant action and refreshes HTML page content with word **htmlpagesend**. Action word(s) can be put instead of substitutes as is **." ACTION for /26/on "** aso.

```

: htmlpagesend \ send whole html page
  s" text/html" ok-response
    htmlpage   \ create html page in webintstream buffer
    webcontent send \ and send it to client
;

```

The word **htmlpagesend** sends back to client (browser) at first status code and type of html data. Next is dynamically created html page code in form of text and finally sent to client to show it in browser.

This is whole process in the nutshell.

Next in more detail.

For practical usage I focused to three types of information generated by ESP32 web interface:

- simple passive text data such as results of some measuring, for example from weather station
- buttons for on/off switching to control something by ESP32 circuit
- HTML forms for adjusting some parameters in program running on ESP32.

For this I created this simple example of web page generated on ESP32 with IP address 92.168.1.6.




Figure 23: web page generated by ESP32forth

So text GPIO 26 status is: is text info and value 0 or 1 is value of forth value GPIO26 included to web page during HTML page generation.

Buttons GPIO26 and GPIO27 can switch appropriate forth value to control something, for ex. relay connected to ESP32 GPIOs.

The rest HTML forms can control more advanced adjusting of forth program parameters.

The last *Click me to display...* is only generating actual date/time info of client browser without any program connection to ESP32forth code.

You will find the **example_web.fs** script in the **ESP32forth-book.zip** file available here :

https://github.com/MPETREMANN11/ESP32forth/blob/main/_documentation/ESP32forth-book.zip

Next only in brief:

Lines 8 to 29 create helping word **mvbar**, used as **mvbar** any multi line text | to create temporary string as addr len across more lines of text.

Buffer for HTML page text is created on line 31 by `stream webintstream` and uses `word >stream` to add text parts together.

Long word `htmlpage` on lines 46 to 135 dynamically creates html text after each activation. Lines 67 to 71 create text with actual values of forth values GPIO26, GPIO27. If it is used to show some measured values continually it is necessary to put code for auto refresh of html page into generated html code.

Lines 72 to 88 create buttons in color red or green depending on GPIO values with client info /26/on or /26/off for detection in serve-page word.

Lines 91 to 127 generate html forms data for adjustment of some values as data, time, text or range. Lines 128 to 131 are only generating actual date/time info with JavaScript code.

In the end of code there is activation of server with wifi and start of webinterface as task on background.

I present this code as base for experiments. I am sure it is possible to improve it, comments are welcome.

ESP-NOW

ESP-NOW is a wireless communication protocol for fast response and low-power control. ESP-NOW uses Wi-Fi transmission.

ESP-NOW is a protocol developed by Espressif that offers short packet transmission capabilities. This protocol allows multiple devices to communicate easily with each other. It is designed for efficient, low-latency communication between devices and is capable of sending up to 250 bytes.

Overview

In ESP-NOW, application data is encapsulated in a vendor-specific action frame and then transmitted from one Wi-Fi device to another without a connection to a router.

ESP-NOW is widely used in smart lighting, remote control, sensors, etc.

Security

ESP-NOW uses the CCMP method, described in IEEE Std. 802.11-2012, to protect the vendor-specific framework. The Wi-Fi device maintains a primary master key (PMK) and several local master keys (LMKs). The lengths of the PMK and LMK are 16 bytes.

- PMK is used to encrypt LMK with the AES-128 algorithm. Call `esp_now_set_pmk` to set PMK. If PMK is not set, a default PMK will be used.
- The LMK of the associated device is used to encrypt the vendor-specific action framework using the CCMP method. The maximum number of different LMKs is six. If the LMK of the coupled device is not defined, the vendor-specific action framework will not be encrypted.

Encryption of the provider-specific action framework in multicast is not supported.

Simplicity

Each card using ESP-NOW is a node. With ESP-NOW, nodes do not need to connect to a WiFi router.

To communicate with ESP-NOW, you need at least two ESP32 boards:

- an issuing card, which we will identify as MASTER;
- a receiver card, which we will identify as SLAVE1.




Figure 24: ESP-NOW nodes

In reality, there is no preferred direction. In the ESP-NOW network, any card can transmit to any other ESP32 card. The only constraint is that each card must know the MAC address of the card receiving the message.

An ESP32 project that uses ESP-NOW

We will explore the use of the ESP-NOW communication protocol through a project. The source code for this project is available here:

<https://github.com/MPETREMANN11/ESP32forth/tree/main/espnow>

The prerequisites of our ESP-NOW project

To use ESP-NOW, you need to compile an ESP32Forth version incorporating the contents of this file:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/ESP32forth70721a/optional/espnow-voc.h>

RECORDFILE also needs to be integrated. This is what is done in this adapted version of ESP32Forth:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/ESP32forth70721a/ESP32forth70721a.ino>

Here is the content of the **config.fs** file :

```
RECORDFILE /spiffs/config.fs

$EC $62 $60 $9C $76 $30 define-mac-address: MASTER
$08 $3A $F2 $6E $1E $F8 define-mac-address: SLAVE1
wifi-init 100 ms
getMyMac

espnnowInit

myMac 6 MASTER 6 str= [IF]      \ peers for MASTER
    SLAVE1 add-peer
    SLAVE2 add-peer
    s" MASTER" set-title
[THEN]

myMac 6 SLAVE1 6 str= [IF]      \ peers for SLAV1
    MASTER add-peer
    SLAVE2 add-peer
    s" SLAVE1" set-title
[THEN]

<EOF>
```

Let's take a look at the first line RECORDFILE /spiffs/config.fs .

When ESP32Forth interprets this line of code, it creates the **config.fs file** in the **spiffs** filespace on the ESP32 board.

Then RECORDFILE saves each line of code in this **/spiffs/config.fs file** up to the line containing <EOF>. The creation and saving of the **/spiffs/config.fs file** is done by simply copying and pasting the source code from your FORTH code editor on your PC to the terminal connected to the ESP32 board and running ESP32Forth.

When ESP32Forth starts, to execute the contents of the **/spiffs/config.fs file**, you will need to run **needs /spiffs/config.fs**. This is what is done from the **/spiffs/main.fs file** :

```
RECORDFILE /spiffs/main.fs

DEFINED? --espnow [if] forget --espnow [then]
create --espnow

needs /spiffs/oled128x32-display.fs
needs /spiffs/datas-structs.fs
needs /spiffs/mac-address.fs
needs /spiffs/espnow.fs
needs /spiffs/config.fs
myMac 6 MASTER 6 str= [IF]      \ peers for MASTER
    needs tests-MASTER.fs
[THEN]

myMac 6 SLAVE1 6 str= [IF]      \ peers for SLAV1
    needs tests-SLAVE1.fs
[THEN]
<EOF>
```

By executing include /spiffs/main.fs, the contents of all other files listed in **main.fs** are loaded . This loading is extremely fast. The only requirement is that all files listed in **main.fs** must first be written to the **/spiffs/ directory** by copying and pasting the contents of each source file. For the contents of **mac-address.fs**, the contents of the **mac-address.fs** source file must be copied and pasted into the terminal window associated with the ESP32 board.

Order of operations:

- open access to the first ESP32 card from the terminal open on the PC;
- open a source file of the project with the text editor of your choice;
- From this editor, copy the source code between RECORDILE and <EOF> inclusive;
- Go to the terminal window and copy this code;

- From the terminal, verify that the file has been correctly saved in the **/spiffs/space** with ls /spiffs/ ;
- Proceed in the same way for the second ESP32 board...

WARNING: the project's source files, as well as their content, are subject to change without notice.

Similarly, some parameters need to be adapted to your hardware, such as MAC addresses, for example, described in **config.fs** .

MAC address management

A MAC address, or Media Access Control address, is a unique identifier assigned to a network device. It is encoded using 6 bytes, or 48 bits, and consists of alphanumeric characters. It can be thought of as the device's license plate on the network.

Functioning :

- The MAC address is stored in the network card or a similar network interface of the device;
- It is used to uniquely identify a device on a local area network (LAN);
- It allows devices to communicate with each other and share resources;
- It is also used by routers and switches to control network traffic.

Properties :

- Unique: Each MAC address is supposed to be globally unique. This means that no other device should have the same MAC address.
- physical: The MAC address is tied to the device hardware and does not change, even if you reinstall the operating system or change your Internet service provider (ISP).
- Fixed: The MAC address cannot usually be changed. However, there are methods to spoof it, but this is generally discouraged for security reasons.

Uses:

- Access control: MAC addresses can be used to control access to a network. For example, a network administrator can configure a router to allow only devices with specific MAC addresses to connect to the network.
- conflict resolution: In the event of an IP conflict on a network, MAC addresses can be used to identify the device causing the conflict.
- Device tracking: The MAC address can be used to track the movement of devices across a network.

To operate ESP-NOW, you need to know the MAC address of each ESP32 card integrated into the network.

The source code for managing MAC addresses is available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/espnow/mac-address.fs>

How to find the MAC address of an ESP32 board

The first step is to activate the WiFi port in station mode. In this mode, no SSID is defined for this station mode:

```
also wifi
also espnow

\ Configure WiFi in station mode
wifi-init ( -- )
  Start Wi-Fi in station mode
  WIFI_MODE_STA Wifi.mode
;
```

WIFI_MODE_STA is a Wi-Fi mode parameter commonly used in embedded systems and microcontrollers, especially those based on the ESP32 family. It stands for Station Mode, meaning the device is operating as a Wi-Fi client and intends to connect to an existing Wi-Fi network provided by an access point (AP).

In WIFI_MODE_STA mode, the device actively searches for available Wi-Fi networks, selects the desired network based on its SSID (Service Set Identifier) and password, and establishes a connection with the access point (AP). Once connected, the device can communicate with other devices on the network and access the internet.

A MAC address is defined as 6 bytes. We will retrieve the MAC address of our ESP32 board by reserving 6 bytes in the word myMAC :

```
\ store current mac Address
create myMac 6 allot

\ get mac address for current ESP32 card and store result in myMac
: getMyMac ( -- )
  myMac WiFi.macAddress
;
```

We also take this opportunity to define the word .mac which will display the retrieved MAC address.

Now, we retrieve the MAC address of our ESP32 board using WiFi.macAddress :

```
wifi-init 100 ms
getMyMac
```

Display a MAC address

A MAC address is usually written as six groups of two hexadecimal characters, separated by colons. For example:

00:11:22:33:44:55

This is the form we're interested in. Here's how the .mac extension , which displays a MAC address, is defined:

```
\display MAC address in hex format
: .mac { mac-addr -- }
    base @ hex
    6 0 do
        mac-addr i + c@ <# # # #> type
        i 5 < if
            [char]: emit
        then
    loop
    base !
;
```

Viewing the contents of myMAC with .mac :

```
myMAC .mac
display: EC:62:60:9C:76:30
```

Definition of target MAC addresses

As you've probably gathered, for two ESP32 boards to communicate, the board sending the message must know the MAC address of the target board. If there are multiple target boards, the address of each target board must be stored.

Here's how to set a MAC address for a target ESP32 board:

```
: define-mac-address: ( $n1..$n6 comp: -- | exec: -- addr )
    create
        \ store mac address in 6 first bytes
        >r >r >r >r
        c, r> c, r> c, r> c, r> c, r> c,
        store 0 in other bytes
        esp_now_peer_info_t 6 - 1- for
            0 c,
        next
    does>
;
only forth
```

Here's how to memorize the MAC address of a target card whose MAC address you have retrieved:

```
$EC $62 $60 $9C $76 $30 define-mac-address: MASTER
$08 $3A $F2 $6E $1E $F8 define-mac-address: SLAVE1
$EC $62 $60 $9C $A9 $50 define-mac-address: SLAVE2
```

Here, we define MASTER and SLAVE1 . Executing one of these words places the memory area containing the MAC address of the target ESP32 board onto the data stack. We verify that this MAC address has been correctly compiled:

```
SLAVE1 .mac \ display: 08:3A:F2:6E:1E:F8
```

If you run this code with your own ESP32 boards, you will get very different MAC addresses. It is your own MAC addresses that you need to consider.

In addition to storing a MAC address, the `define-mac-address:` keyword reserves additional bytes for specific uses. These additional bytes depend on the `esp_now_peer_info_t` structure.

The `esp_now_peer_info_t` structure

`esp_now_peer_info_t` is a structure containing information about the target card:

- **peer_address** : The MAC address of the device;
- **lmk** : encryption key. Leave blank if you do not want to encrypt the transmission;
- **channel** : Wi-Fi channel on which to communicate with the twin device. The channel value can be between 0 and 14.

Here is how the `esp_now_peer_info_t` structure is defined:

```
forth
also structures
also espnow
struct esp_now_peer_info_t
    ESP_NOW_ETH_ALEN field ->peer_addr
        16 field ->lmk
        u8 field ->channel
        u8 field ->ifidx
        u8 field ->encrypt
        u32 field ->priv
```

The default WiFi transmission channel is channel 0. Here's how the set-channel command, which allows you to change a transmission channel, is defined:

```
\ set channel for peer
: set-channel ( chan addr -- )
    !field ->channel
;
```

To change the channel of the MASTER and SLAVE1 cards:

```
3 MASTER set-channel
3 SLAVE1 set-channel
```

List of ESP-NOW words

The terms governing the ESP-NOW network are defined in the espnow vocabulary :

```
espnow vlist \ displays:
esp_now_init esp_now_deinit esp_now_get_version esp_now_send
esp_now_register_recv_cb
esp_now_unregister_recv_cb esp_now_register_send_cb esp_now_unregister_send_cb
esp_now_add_peer esp_now_del_peer esp_now_mod_peer esp_now_is_peer_exist
esp_now_get_peer_num esp_now_set_pmk espnow-builtins ESP_NOW_MAX_DATA_LEN
```

```
ESP_NOW_MAX_ENCRYPT_PEER_NUM ESP_NOW_MAX_TOTAL_PEER_NUM ESP_NOW_KEY_LEN  
ESP_NOW_ETH_ALEN
```

In the following, we will only discuss the essential words enabling the management of ESP-NOW transmissions between two or more ESP32 boards.

The source code for this section is available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/espnow/espnow.fs>

esp_now_init

The word `esp_now_init` is used to initialize the ESP-NOW communication protocol with ESP32Forth.

The word `esp_now_init` must be called before using other ESP-NOW words, such as `esp_now_add_peer` or `esp_now_send`.

The `esp_now_init` function takes no arguments and returns an integer value. If execution is successful, it returns 0. If the function fails, it returns an error code.

To simplify access to `esp_now_init`, it is integrated into the word `espnowInit`, which will remain accessible from the FORTH vocabulary:

```
also espnow

0 constant ESP_OK
-1 constant ESP_FAIL

\ Initialize ESPNOW
: espnowInit ( -- )
  wifi-init
  esp_now_init ESP_OK <>      \ 0 for success
  if
    ." ESP-NOW init failed" cr
    -1 throw
  then
    ." ESP-NOW init success" cr
;
```

esp_now_register_recv_cb

The word `esp_now_register_recv_cb` is essential in the ESP-NOW protocol. It is used to register a callback function that will execute automatically each time your module receives a data packet.

It is initialized with the execution code of a FORTH word.

To simplify access to `esp_now_register_recv_cb`, it is integrated into the word `espnowRegisterRecv`, which will remain accessible from the FORTH vocabulary:

```
\ Register receive callback
```

```

: espnowRegisterRecv ( callback -- )
    esp_now_register_recv_cb ESP_OK <>      \ 0 for success
    if cr ." ESP-NOW esp_now_register_recv_cb failed"
        -1 throw
    then ;

```

Exemple :

```

: testReception ( -- )
    cr ." ESP6NOW receiving"
;
` testReception espnowRegisterRecv

```

Each time a frame is received on the receiving card, the word testReception will be executed.

esp_now_add_peer

The word esp_now_add_peer is the essential step to establish targeted communication with another ESP32 board.

This word allows you to register a remote device (a "peer") in your module's memory. Without this step, your module cannot send data to a specific MAC address.

To simplify access to esp_now_add_peer, it is integrated into the word espnowAddPeer, which will remain accessible from the FORTH vocabulary:

```

\ add peer to peer list
\ ex: SLAVE1 add-peer
: espnowAddPeer ( peer-addr -- )
    peer-exist?
    esp_now_add_peer ESP_OK <>      \ 0 for success
    if cr ." ESP-NOW add_peer failed"
        -1 throw
    then ;

```

Example :

```

$EC $62 $60 $9C $76 $30 define-mac-address: MASTER
$08 $3A $F2 $6E $1E $F8 define-mac-address: SLAVE1
$EC $62 $60 $9C $A9 $50 define-mac-address: SLAVE2

```

```

myMac 6 MASTER 6 str= [IF]      \ peers for MASTER
    SLAVE1 espnowAddPeer
    SLAVE2 espnowAddPeer
    s" MASTER" set-title
[THEN]

```

esp_now_send

The term esp_now_send refers to the engine that allows you to send your data to one or more recipients via the ESP-NOW protocol. Here's what you need to know about how it works:

- **Role :** it transmits a data packet (up to a maximum of 250 bytes) to a remote module;
- **Recipients :**
 - **Unicast** : by specifying the MAC address of a specific peer;
 - **Broadcast** : by passing the NULL address (or the address FF:FF:FF:FF:FF:FF), the message is sent to all devices within range.
- **Asynchronous** : the word simply places the message in the sending queue. It returns immediate confirmation that sending has been initiated, but does not guarantee receipt.

```
\ send datas to peer
: espnowSend ( peerAddr myData len -- )
    esp_now_send ?dup ESP_OK <>      \ 0 for success
    if cr ." ESP-NOW esp_now_send failed"
        ." error: " .
        -1 throw
    then ;
```

Example:

```
: ttrans
    SLAVE1 s" TEST TRANSMISSION VERS SLAVE1" espnowSend
;

: tvl
    SLAVE1 s" PAGE VLIST" espnowSend
;
```

Chronology of actions in transmission and reception

We have just seen the essential words for managing actions in sending and receiving.

The order of operations to be performed, on both the sender's and receiver's sides, is as follows:

- definition of each peer in the network with the word define-mac-address: ;
- Initialize WiFi on each card of the ESP-NOW network by running wifi-init 100 ms . The 100 ms delay is necessary to allow the WiFi interface to initialize correctly;
- initialize ESP-NOW by running espnowInit

- peer registration with the word espnowAddPeer .

On the receiver side, the word FORTH to be executed if a signal is received is indicated with the word espnowRegisterRecv , preceded by the execution code for this word FORTH to be executed.

If you wish to manage a two-way dialogue, you will need to initialize the reception on each card with the appropriate code.

In our project, we find this file loading sequence:

```

needs /spiffs/oled128x32-display.fs
needs /spiffs/datas-structs.fs
needs /spiffs/mac-address.fs
needs /spiffs/espnow.fs
needs /spiffs/config.fs
\ needs /spiffs/tests.fs
myMac 6 MASTER 6 str= [IF]      \ peers for MASTER
    needs tests-MASTER.fs
[THEN]

myMac 6 SLAVE1 6 str= [IF]      \ peers for SLAV1
    needs tests-SLAVE1.fs
[THEN]

myMac 6 SLAVE2 6 str= [IF]      \ peers for SLAV2
    needs tests-SLAVE2.fs
[THEN]

```

The first line, needs /spiffs/oled128x32-display.fs , is only necessary if you want to use an OLED display. It can be deleted or commented out.

The second line loads the data structure required for our ESP-NOW project.

The third line handles MAC addresses.

With the line needs /spiffs/espnow.fs , we load the words essential for the initialization of ESP-NOW and the management of the transmission.

The following line, needs / **spiffs/config.fs**, manages the configuration specific to our project. The necessary peers are described in the **config.fs** file.

Prepare the ESP-NOW transmission

We've reached the crucial point in our development. To test the ESP-NOW transmission, you need to:

- two ESP32 boards, each connected to a USB port on our PC;
- all the files for our project have been installed in the **/spiffs/ space** of each ESP32 board;

- Each card is accessible through its terminal window.

Here is our first open terminal window:

```

VT COM3 - Tera Term VT
File Edit Setup Control Window Help
ets Jun 8 2016 00:22:57

rst:0x1 (POWERON_RESET), boot:0x13 (SPI_FAST_FLASH_BOOT)
configsip: 0, SPIWP:0xee
clk_drv:0x00, q_drv:0x00, d_drv:0x00, cs0_drv:0x00, hd_drv:0x00, wp_drv:0x00
mode:DIO, clock div:1
load:0x3fff0030, len:4980
load:0x40078000, len:16612
Load:0x40080400, len:3500
entry 0x400805b4
ESP32forth v7.0.7.21a - rev 9ae74fa18335b0378a98bd8c693b468cc1265ee5
ESP32-D0WDQ6 240 MHz 2 cores 4194304 bytes flash
    System Heap: 156120 free + 314968 used = 471088 total (33% free)
        77812 bytes max contiguous
Forth dictionary: 70988 free + 33340 used = 104328 total (68% free)
3 x Forth stacks: 2048 bytes each
ok
--> []

```

Figure 25: Opening the first terminal window.

Here, our window is connected to the COM3 serial port.

To open the other window:

- Click on File in the menu bar
- Select new connection
- The pop-up window offers to open another terminal window via a different serial port. Click OK.

From there, to load the code contained in the **/spiffs/main.fs file**, type MAIN . This word is pre-implemented in the latest version of ESP32forth v7.0.7.21a. Otherwise, here is its definition:

```

: MAIN
    s" /spiffs/main.fs" included
;

```

The execution of MAIN loads the code contained in our main.fs file, which should be considered the main trunk of our project.

In our project, the **main.fs file** contains the calls to the other project components. To understand the project, simply look at the contents of each of these files. If we turn off our ESP32 board, then turn it back on and reconnect, all the files in the **/spiffs/** directory

will still be available. By typing MAIN again , all the files are loaded and their code executed or compiled in less than a second.


To modify the contents of a file in the **/spiffs/** space :

- open the source code, for example **config.fs** , with the text editor of your choice;
- edit the source code and save the file to your PC
- Copy the code between RECORDFILE and <EOF> and paste this section into the terminal window. For example, if you copy the code including RECORDFILE /spiffs/config.fs up to and including <EOF> , this new code will replace the code previously installed in the **/spiffs/** directory .

The storage space **/ spiffs/** can contain a huge number of files, up to a limit of 2MB of storage space!

Launch a test sequence

Typing MAIN in our project loads FORTH source code, the necessary parameters, and the test file. In our terminal window, we execute the sequence FORTH page main . The window now displays this:



The screenshot shows a terminal window titled "COM3 - MASTER VT". The window has a menu bar with File, Edit, Setup, Control, Window, and Help. The main text area displays the following FORTH output:

```
ESP-NOW init success
EC:62:60:9C:76:30
-----
MASTER: EC:62:60:9C:76:30
SLAVE1: 08:3A:F2:6E:1E:F8
SLAVE2: EC:62:60:9C:A9:50
ok
--> []
```

This results from the execution of the code contained in the **tests-MASTER.fs** file :

```
myMac .mac cr
." - - - " cr
MASTER: MASTER .mac cr
SLAVE1: SLAVE1 .mac cr
SLAVE2: SLAVE2 .mac cr
```

The terminal window is now named **MASTER** . This is a result of the following code snippet in **config.fs** :

```

myMac 6 MASTER 6 str= [IF]      \ peers for MASTER
    SLAVE1 espnowAddPeer
    SLAVE2 espnowAddPeer
    s" MASTER" set-title
[THEN]

```

It's the sequence FORTH `s" MASTER" set-title` that renames our terminal window.
Let's do the same for the second terminal window, by typing page main :




Figure 26: Opening the second terminal window for **SLAVE1**

We have two terminal windows in front of us, each linked to its ESP32 board. The right-hand window is titled **SLAVE1** .

Place the cursor in the **MASTER** window . Then type testTrans, the code for which is as follows:

```

: testTrans
    SLAVE1 s" TEST TRANSMISSION TO SLAVE1" espnowSend
;

```

The word `testTrans` sends the string *TEST TRANSMISSION TO SLAVE1* via ESP-NOW to the card titled **SLAVE1** :




Figure 27: Our first ESP-NOW transmission test, from **MASTER** to **SLAVE1**

Here is the FORTH code processed as a callback by the interrupt managed by ESP-NOW on this **SLAVE1** card :

```
: my-cb-espnow { len dataAddr macAddr -- }  
    cr ." --- ESP-NOW RECEIVED ---"  
    cr ." Length      : "  len .  
    cr ." Datas       : "  dataAddr .  
    cr ." Source MAC : "  macAddr .mac  
    cr ." -----"  
    cr ." Msg : "  dataAddr len type  
;  
  
\ set callback  
' my-cb-espnow espnowRegisterRecv
```

It is the ESP-NOW interrupt that passes the parameters len dataAddr macAddr to the word my-cb-espnow .

These parameters are treated as local variables, which simplifies understanding the Forth code.

Application Layer Management

In our ESP-NOW project, we will refer to the application layer as the set of words acting on the receiver side. For our article, an example of the application layer is available here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/espnow/ledsCommand.fs>

In this **ledsCommand.fs** file , let's look at the last lines of code:

```
: redOn  ( -- )      LED_RED ledOn  ;  
: redOff ( -- )     LED_RED ledOff ;  
: yellowOn ( -- )   LED_YELLOW ledOn ;
```

```

: yellowOff ( -- ) LED_YELLOW ledOff ;
: greenOn  ( -- ) LED_GREEN ledOn  ;
: greenOff ( -- ) LED_GREEN ledOff ;

```

These six lines, in FORTH code, allow you to turn these three LEDs on or off:




Figure 28: The three LEDs to be controlled from ESP-NOW

Application layer control

On our board labeled **SLAVE1**, we mounted three LEDs. To test these three LEDs, we need to load the contents of the **ledsCommand.fs** file into the **/spiffs/** space of the ESP32 board. Then we run include **/spiffs/ledsCommand.fs**.

To automatically incorporate the contents of the **ledsCommand.fs** file, we modify this part of the code in **main.fs**:

```

myMac 6 SLAVE1 6 str= [IF]
    needs /spiffs/ledsCommand.fs      initLeds
    needs /spiffs/tests-SLAVE1.fs
[THEN]

```

The big question is therefore: "how do we control this application layer from ESP-NOW?".

In ESP32forth, we have the **evaluate** keyword. This keyword takes a string of characters and attempts to interpret its contents. Example:

s" redOn" evaluate

In the parameters processed by callback upon receiving a message via ESP-NOW, we have the address and length of the message transmitted to our **SLAVE1 board**.

Here's how we create the interface between ESP-NOW and our application part:

```

\ interpret ESP-NOW transmitted string
: evaluate-cb-espnnow { len dataAddr macAddr -- }
    dataAddr len evaluate

```

```

;
\ register this callback
' evaluate-cb-espnow espnowRegisterRecv

```

Except... it doesn't work!

It doesn't work, for a very simple reason, there is a collision between the Forth interpretation buffer and our evaluate-cb-espnow word which is executed in an ESP-NOW interrupt.

To remedy this malfunction, the data processed by the ESP-NOW interrupt must be transferred. This can be done using the string manipulation available in the **strings.fs** file , which can be found here:

<https://github.com/MPETREMANN11/ESP32forth/blob/main/espnow/strings.fs>

We begin by defining two string variables:

```

also espnow
\ define two strings
ESP_NOW_ETH_ALEN      string RecvMacAddr
ESP_NOW_MAX_DATA_LEN  string RecvDataStr

```

Role of each string of characters:

- RecvMacAddr : receives the 6 bytes that identify the MAC address of the sender of the ESP-NOW transmission;
- RecvDataStr : receives the text string transmitted by the ESP-NOW sender.

Next, a new word is defined for the ESP-NOW callback:

```

\ interpret ESP-NOW transmitted string
: store-cb-espnow { len dataAddr macAddr -- }
  macAddr ESP_NOW_ETH_ALEN RecvMacAddr $!
  dataAddr len RecvDataStr $!
;
only FORTH

\ register this callback
' store-cb-espnow espnowRegisterRecv

```

Now, each time an ESP-NOW transmission is received, the message will be stored in the RecvDataStr string . The contents of this string still need to be processed:

```

: espnow-loop ( -- )
begin
  RecvDataStr nip ?dup if
    RecvDataStr evaluate
    RecvDataStr 0$!
  then
again

```

```
;  
espnow-loop
```

The word espnow-loop is a closed loop. In this loop, we test if the string RecvDataStr is not of zero length. If it is, we interpret its content. Then we reset the length of this string to zero with the sequence RecvDataStr 0\$! .

The advantages and disadvantages of this control method

Let's start with the disadvantages:

- FORTH command transmission only. This can be circumvented by developing more sophisticated processing of the transmitted data;
- Text transmission only. Same answer as above.

Here are the advantages:

- FORTH command transmission related to the application layer. The application layer can be tested intensively and in detail before interfacing with ESP-NOW transmission;
- Extremely simplified interfacing. The application layer completely ignores ESP-NOW transmission, as application commands pass through the FORTH interpreter.
- A very simplified application integration on the issuing card side.

Here, the application layer, on the **SLAVE1 board**, simply involves turning on three separate LEDs. On the **MASTER board**, these on/off commands are simply cloned in this way, here for the red LED:

```
: redOn ( -- )  
    SLAVE1 s" redOn " espnowSend  
;  
: redOff ( -- )  
    SLAVE1 s" redOff " espnowSend  
;
```

On the **MASTER board**, typing and executing redOn will turn on the red LED on the **SLAVE1** board .

This raises another question: "how can we be certain that the **SLAVE1** board has correctly received and executed the command transmitted by **MASTER** via ESP-NOW?"

The only way to ensure proper execution is to establish a bidirectional protocol between the transmitting and receiving cards. But that's beyond the scope of this chapter.

The main advantage of executing FORTH words upon receiving an ESP-NOW transmission is that it simplifies the interface between an application and the ESP-NOW transmission. The code described earlier in this article will remain unchanged regardless of the

application component being handled. In this respect, the FORTH language proves to be much more flexible than the C language.

To conclude this article, here's how, on the **SLAVE1** board , we automatically launch the ESP-NOW reception and the application part:

```
RECORDFILE /spiffs/autoexec.fs
```

```
main
```

```
<EOF>
```

This saves the file **/spiffs/autoexec.fs** . Upon power-up, the ESP32forth checks if the **autoexec.fs** file is present. If it is, the ESP32forth automatically loads its contents. In this case, it only contains the word "main," which will load the entire project contained in the **/spiffs/main.fs** file .

MASTER card side , it remains connected to the PC via the terminal. But there's nothing preventing its interface from being modified, for example by managing switches, whose management will transmit frames to **SLAVE1** .

In our project, we only programmed two cards. In reality, you can program up to 20 target cards for one master card.

It is also possible to use a target broadcast MAC address: FF:FF:FF:FF:FF:FF. With this target address, you can control all the cards at the same time.

In summary, mastering the ESP-NOW network opens up very interesting prospects for any IoT (Internet of Things) or home automation application.

Video demonstration:

<https://www.youtube.com/watch?v=qHtbdMbuZPY>

In this video, the **SLAVE1** board was placed 15 meters from the **MASTER** board . It is powered by a self-contained unit via a lithium battery. Upon power-up, the three LEDs light up rapidly to confirm program initialization.



All LED switching commands are transmitted by the MASTER board, as an overlay in the video.

ESP-NOW communication is handled peer-to-peer, meaning without the need for any WiFi router.

Detailed content of ESP32forth vocabularies

ESP32forth provides numerous vocabularies:

- **FORTH** is the main vocabulary;
- certain vocabularies are used for internal mechanics for ESP32Forth, such as **internals** , **asm...**
- many vocabularies allow the management of specific ports or accessories, such as **bluetooth** , **oled** , **spi** , **wifi** , **wire...**

Here you will find the list of all the words defined in these different vocabularies. Some words are presented with a colored link:

[align](#) is an ordinary Forth word;

[CONSTANT](#) is definition word;

[begin](#) marks a control structure;

[key](#) is a deferred execution word;

[LED](#) is a word defined by **constant** , **variable** or **value** ;

[--registers](#) marks a vocabulary.

FORTH vocabulary words are displayed in alphabetical order. For other vocabularies, the words are presented in their display order.

Version v 7.0.7.21a

FORTH

-	-rot	?	?dup	:	:"	:noname	!
?	?do	[[']	.	[ELSE]	.s	'
(local)	[[[[char]	[IF]	[THEN]	[
1	{	{	{	}transfer	@	*	*/
*/MOD	/	/mod	/mod	#	#!	#>	#fs
#s	#tib	+	+!	+loop	+to	≤	≤
<a href;"="">≤#	<a href;"="">≤=	<a href;"="">≤>	<a href;"="">≤>	<a href;"="">≥	<a href;"="">≥	<a href;"="">≥=	<a href;"="">≥=
<a href;"="">>flags	<a href;"="">>flags&	>in	>link	>link&	>name	<a href;"="">>BODY	<a href;"="">>params
<a href;"="">>R	<a href;"="">>size	0<	0<>	0=	1-	1/F	1/F
1+	2!	2@	2*	2/	2drop	2dup	2dup
4*	4/	abort	abort"	abs	accept	adc	adc
afliteral	aft	again	ahead	align	aligned	allocate	allocate
allot	also	analogRead	AND	ansi	ARSHIFT	asm	asm
assert	at-xy	base	begin	bg	BIN	binary	binary
bl	blank	block	block-fid	block-id	buffer	bye	bye
c,	C!	C@	CASE	cat	catch	CELL	CELL

cell/	cell+	cells	char	CLOSE-DIR	CLOSE-FILE	cmove
cmove>	CONSTANT	context	copy	cp	cr	CREATE
CREATE-FILE	current	dacWrite	decimal	default-key	default-key?	
default-type		default-use	defer	DEFINED?	definitions	DELETE-FILE
depth	digitalRead	digitalWrite		do	DOES>	DROP
dump	dump-file	DUP	duty	echo	editor	else
emit	empty-buffers		ENDCASE	ENDOF	erase	ESP
ESP32-C3?	ESP32-S2?	ESP32-S3?	ESP32?	evaluate	EXECUTE	exit
extract	F-	f.	f.s	F*	F**	F/
F+	F<	F<=	F<>	F=	F>	F>=
F>S	F0<	F0=	FABS	FATAN2	fconstant	FCOS
fdepth	FDROP	FDUP	FEXP	fg	file-exists?	
FILE-POSITION		FILE-SIZE	fill	FIND	fliteral	FLN
FLOOR	flush	FLUSH-FILE	FMAX	FMIN	FNEGATE	FNIP
for	forget	FORTH	forth-builtins	FOVER		FP!
FP@	fp0	free	freq	FROT	FSIN	FSINCOS
FSQRT	FSWAP	fvariable	handler	here	hex	HIGH
hld	hold	httpd	I	if	IMMEDIATE	include
included	included?	INPUT	internals	invert	is	J
K	key	key?	L!	latesttxt	leave	LED
ledc	list	literal	load	login	loop	LOW
ls	LSHIFT	max	MDNS.begin	min	mod	ms
MS-TICKS	mv	n.	needs	negate	nest-depth	next
nip	nl	NON-BLOCK	normal	octal	OF	ok
only	open-blocks	OPEN-DIR	OPEN-FILE	OR	order	OUTPUT
OVER	pad	page	PARSE	pause	PI	pin
pinMode	postpone	precision	previous	prompt	PSRAM?	pulseIn
quit	r"	R@	R/O	R/W	R>	r
r~	rdrop	read-dir	READ-FILE	recurse	refill	registers
remaining	remember	RENAME-FILE	repeat	REPOSITION-FILE		required
reset	resize	RESIZE-FILE	restore	revive	RISC-V?	rm
rot	RP!	RP@	rp0	RSHIFT	rtos	s"
S>F	s>z	save	save-buffers	scr		SD
SD MMC	sealed	see	Serial	set-precision		set-title
sf,	SF!	SF@	SFLOAT	SFLOAT+	SFLOATS	sign
SL@	sockets	SP!	SP@	sp0	space	spaces
SPIFFS	start-task	startswith?	startup:	state	str	str=
streams	structures	SW@	SWAP	task	tasks	telnetd
terminate	then	throw	thru	tib	to	tone
touch	transfer	transfer	type	u.	U/MOD	UL@
UNLOOP	until	update	use	used	UW@	value
VARIABLE interface	visual	vlist	vocabulary	W!	W/O	web-
webui	while	WiFi	Wire	words	WRITE-FILE	XOR
Xtensa?	z"	z>s				

asm

```
xtensa disasm disasm1 matchit address istep sextend m. m@ for-ops op >operands
>mask >pattern >length >xt op-snap opcodes coden, names operand l o bits
bit skip advance advance-operand reset reset-operand for-operands operands
>printop >inop >next >opmask& bit! mask pattern length demask enmask >>1
```

```
odd? high-bit end-code code, code4, code3, code2, code1, callot chere reserve  
code-at code-start
```

bluetooth

```
SerialBT.new SerialBT.delete SerialBT.begin SerialBT.end SerialBT.available  
SerialBT.readBytes SerialBT.write SerialBT.flush SerialBT.hasClient  
SerialBT.enableSSP SerialBT.setPin SerialBT.unpairDevice SerialBT.connect  
SerialBT.connectAddr SerialBT.disconnect SerialBT.connected  
SerialBT.isReady bluetooth-builtins
```

editor

```
a r d e w i p e p n l
```

ESP

```
getHeapSize getFreeHeap getMaxAllocHeap getChipModel getChipCores getFlashChipSize  
getCpuFreqMHz getSketchSize deepSleep getEfuseMac esp_log_level_set ESP-builtins
```

espnow

```
esp_now_init esp_now_deinit esp_now_get_version esp_now_send  
esp_now_register_recv_cb  
esp_now_unregister_recv_cb esp_now_register_send_cb esp_now_unregister_send_cb  
esp_now_add_peer esp_now_del_peer esp_now_mod_peer esp_now_is_peer_exist  
esp_now_get_peer_num esp_now_set_pmk ESP NOW MAX DATA LEN ESP NOW ETH ALEN  
espnow-builtins
```

httpd

```
notfound-response bad-response ok-response response send path method hasHeader  
handleClient read-headers completed? body content-length header crnl= eat  
skipover skipto in@<> end< goal# goal strcase= upper server client-cr client-emit  
client-read client-type client-len client httpd-port clientfd sockfd body-read  
body-1st-read body-chunk body-chunk-size chunk-filled chunk chunk-size  
max-connections
```

insides

```
run normal-mode raw-mode step ground handle-key quit-edit save load backspace  
delete handle-esc insert update crtype cremit ndown down nup up caret length  
capacity text start-size fileh filename# filename max-path
```

internals

```
assembler-source xtensa-assembler-source MALLOC SYSFREE REALLOC heap_caps_malloc  
heap_caps_free heap_caps_realloc heap_caps_get_total_size heap_caps_get_free_size  
heap_caps_get_minimum_free_size heap_caps_get_largest_free_block RAW-YIELD  
RAW-TERMINATE REaddir CALLCODE CALL0 CALL1 CALL2 CALL3 CALL4 CALL5 CALL6  
CALL7 CALL8 CALL9 CALL10 CALL11 CALL12 CALL13 CALL14 CALL15 DOFLIT S>FLOAT?
```

```

fill32 'heap 'context 'latesttxt 'notfound 'heap-start 'heap-size 'stack-cells
'boot 'boot-size 'tib 'argc 'argv 'runner 'throw-handler NOP BRANCH OBRANCH
DONEXT DOLIT DOSET DOCOL DOCON DOVAR DOCREATE DODOES ALITERAL LONG-SIZE
S>NUMBER? 'SYS YIELD EVALUATE1 'builtins internals-builtins autoexec
arduino-remember-filename
arduino-default-use esp32-stats serial-key? serial-key serial-type yield-task
yield-step e' @line grow-blocks use?! common-default-use block-data block-dirty
clobber clobber-line include+ path-join included-files raw-included include-file
sourcedirname sourcefilename! sourcefilename sourcefilename# sourcefilename&
starts../ starts./ dirname ends/ default-remember-filename remember-filename
restore-name save-name forth-wordlist setup-saving-base 'cold park-forth
park-heap saving-base crtype cremit cases (+to) (to) --? }? ?room scope-create
do-local scope-clear scope-exit local-op scope-depth local! local! local@
<>locals locals-here locals-area locals-gap locals-capacity ?ins. ins.
vins. onlines line-pos line-width size-all size-vocabulary vocs. voc. voclist
voclist-from see-all >vocnext see-vocabulary nonvoc? see-xt ?see-flags
see-loop see-one indent+! icr see. indent mem= ARGS_MARK -TAB +TAB NONAMED
BUILTIN_FORK SMUDGE IMMEDIATE_MARK relinquish dump-line ca@ cell-shift
cell-base cell-mask MALLOC_CAP_RTCRAM MALLOC_CAP_RETENTION MALLOC_CAP_IRAM_8BIT
MALLOC_CAP_DEFAULT MALLOC_CAP_INTERNAL MALLOC_CAP_SPIRAM MALLOC CAP DMA
MALLOC CAP 8BIT MALLOC CAP 32BIT MALLOC CAP EXEC #f+s internalized BUILTIN_MARK
zplace $place free. boot-prompt raw-ok [SKIP] ' [SKIP] ?stack sp-limit input-limit
tib-setup raw.s $@ digit parse-quote leaving, leaving )leaving leaving(
value-bind evaluate&fill evaluate-buffer arrow ?arrow. ?echo input-buffer
immediate? eat-till-cr wascr *emit *key notfound last-vocabulary voc-stack-end
xt-transfer xt-hide xt-find& scope

```

interrupts

```

pinchange #GPIO_INTR_HIGH_LEVEL #GPIO_INTR_LOW_LEVEL #GPIO_INTR_ANYEDGE
#GPIO_INTR_NEGEDGE #GPIO_INTR_POSEDGE #GPIO_INTR_DISABLE ESP_INTR_FLAG_INTRDISABLED
ESP_INTR_FLAG_IRAM ESP_INTR_FLAG_EDGE ESP_INTR_FLAG_SHARED ESP_INTR_FLAG_NMI
ESP_INTR_FLAG_LEVELn ESP_INTR_FLAG_DEFAULT gpio config gpio reset pin gpio set intr type
gpio intr enable gpio intr disable gpio set level gpio get level gpio set direction
gpio set pull mode gpio wakeup enable gpio wakeup disable gpio pullup en
gpio pulldown en gpio pulldown dis gpio hold en gpio hold dis
gpio deep sleep hold en gpio deep sleep hold dis gpio install isr service
gpio isr handler add gpio isr handler remove
gpio set drive capability gpio get drive capability esp_intr_alloc esp_intr_free
interrupts-builtins

```

ledc

```

ledcSetup ledcAttachPin ledcDetachPin ledcRead ledcReadFreq ledcWrite ledcWriteTone
ledcWriteNote ledc-builtins

```

oled

```

OledInit SSD1306_SWITCHCAPVCC SSD1306_EXTERNALVCC WHITE BLACK
OledReset HEIGHT WIDTH OledAddr OledNew OledDelete OledBegin OledHOME OledCLS
OledTextc OledPrintln OledNumln OledNum OledDisplay OledPrint OledInvert
OledTextsize OledSetCursor OledPixel OledDrawL OledFastHLine OledFastVLine
OledCirc OledCircF OledRect OledRectF OledRectR OledRectRF oled-builtins

```

recognizers

```
REC-FNUM RECOGNIZE RECSTACK +RECOGNIZER -RECOGNIZER GET-RECOGNIZERS SET-RECOGNIZERS  
RECTYPE-NUM RECTYPE-IMM RECTYPE-WORD RECTYPE-NONE RECTYPE: REC-NUM REC-FIND
```

registers

```
m@ m!
```

riscv

```
C.FSWSP, C.SWSP, C.FSDSP, C.ADD, C.JALR, C.EBREAK, C.MV, C.JR, C.FLWSP,  
C.LWSP, C.FLDSP, C.SLLI, BNEZ, BEQZ, C.J, C.ADDW, C.SUBW, C.AND, C.OR,  
C.XOR, C.SUB, C.ANDI, C.SRAI, C.SRLI, C.LUI, C.LI, C.JAL, C.ADDI, C.NOP,  
C.FSW, C.SW, C.FSD, C.FLW, C.LW, C.FLD, C.ADDI4SP, C.ILL, EBREAK, ECALL,  
AND, OR, SRA, SRL, XOR, SLTU, SLT, SLL, SUB, ADD, SRAI, SRLI, SLLI, ANDI,  
ORI, XORI, SLTIU, SLTI, ADDI, SW, SH, SB, LHU, LBU, LW, LH, LB, BGEU, BLTU,  
BGE, BLT, BNE, BEQ, JALR, JAL, AUIPC, LUI, J-TYPE U-TYPE B-TYPE S-TYPE  
I-TYPE R-TYPE rs2' rs2#' rs2 rs2# rs1' rs1# rd' rd#' rd rd# offset  
ofs ofs. >ofs iiii i numeric register' reg'. reg>reg' register reg. nop  
x31 x30 x29 x28 x27 x26 x25 x24 x23 x22 x21 x20 x19 x18 x17 x16 x15 x14  
x13 x12 x11 x10 x9 x8 x7 x6 x5 x4 x3 x2 x1 zero
```

rmt

```
rmt_set_clk_div rmt_get_clk_div rmt_set_rx_idle_thresh rmt_get_rx_idle_thresh  
rmt_set_mem_block_num rmt_get_mem_block_num rmt_set_tx_carrier rmt_set_mem_pd  
rmt_get_mem_pd rmt_tx_start rmt_tx_stop rmt_rx_start rmt_rx_stop  
rmt_tx_memory_reset  
rmt_rx_memory_reset rmt_set_memory_owner rmt_get_memory_owner rmt_set_tx_loop_mode  
rmt_get_tx_loop_mode rmt_set_rx_filter rmt_set_source_clk rmt_get_source_clk  
rmt_set_idle_level rmt_get_idle_level rmt_get_status rmt_set_rx_intr_en  
rmt_set_err_intr_en rmt_set_tx_intr_en rmt_set_tx_thr_intr_en rmt_set_gpio  
rmt_config rmt_isr_register rmt_isr_deregister rmt_fill_tx_items rmt_driver_install  
rmt_driver_uninstall rmt_get_channel_status rmt_get_counter_clock rmt_write_items  
rmt_wait_tx_done rmt_get_ringbuf_handle rmt_translator_init  
rmt_translator_set_context  
rmt_translator_get_context rmt_write_sample rmt-builtins
```

rtos

```
vTaskDelete xTaskCreatePinnedToCore xPortGetCoreID rtos-builtins
```

SD

```
SD.begin SD.beginFull SD.beginDefaults SD.end SD.cardType SD.totalBytes  
SD.usedBytes SD-builtins
```

SD_MMC

```
SD_MMC.begin SD_MMC.beginFull SD_MMC.beginDefaults SD_MMC.end  
SD_MMC.cardType  
SD_MMC.totalBytes SD_MMC.usedBytes SD_MMC-builtins
```

Serial

```
Serial.begin Serial.end Serial.available Serial.readBytes Serial.write
Serial.flush Serial.setDebugOutput Serial2.begin Serial2.end Serial2.available
Serial2.readBytes Serial2.write Serial2.flush Serial2.setDebugOutput serial-
builtins
```

sockets

```
ip. ip# ->h_addr ->addr! ->addr@ ->port! ->port@ sockaddr l, s, bs, SO_REUSEADDR
SOL_SOCKET sizeof(sockaddr in) AF_INET SOCK_RAW SOCK_DGRAM SOCK_STREAM
socket setsockopt bind listen connect sockaccept select poll send
sendto
sendmsg recv recvfrom recvmsg gethostbyname errno sockets-builtins
```

spi

```
SPI.begin SPI.end SPI.setHwCs SPI.setBitOrder SPI.setDataMode SPI.setFrequency
SPI.setClockDivider SPI.getClockDivider SPI.transfer SPI.transfer8
SPI.transfer16 SPI.transfer32 SPI.transferBytes SPI.transferBits
SPI.write SPI.write16 SPI.write32 SPI.writeBytes SPI.writePixels
SPI.writePattern
SPI-builtins
```

SPIFFS

```
SPIFFS.begin SPIFFS.end SPIFFS.format SPIFFS.totalBytes SPIFFS.usedBytes
SPIFFS-builtins
```

streams

```
stream> >stream stream>ch ch>stream wait-read wait-write empty? full? stream#
>offset >read >write stream
```

structures

```
@field !field field-op field struct-align max! align-by struct last-struct
long ptr i64 u32 i32 u16 i16 u8 i8 sc@ typer last-typer last-align
```

tasks

```
.tasks main-task task-list
```

telnetd

```
server broker-connection wait-for-connection connection telnet-key telnet-type
telnet-emit broker client-len client telnet-port clientfd sockfd
```

timers

```
interval onalarm int-enable! alarm-enable! divider! autoreload! increase!
enable! alarm! alarm@ timer! timer@ tmp t>nx timer_isr_callback_add timer_init_null
```

```

timer_get_counter_value timer_set_counter_value timer_start timer_pause
timer_set_counter_mode timer_set_auto_reload timer_set_divider
timer_set_alarm_value
timer_get_alarm_value timer_set_alarm timer_group_intr_enable
timer_group_intr_disable
timer_enable_intr timer_disable_intr timers-builtins

```

visual

[edit](#) [insides](#)

web-interface

```

server webserver-task do-serve handle1 serve-key serve-type handle-input
handle-index out-string output-stream input-stream out-size webserver index-html
index-html#

```

WiFi

```

WIFI MODE APSTA WIFI MODE AP WIFI MODE STA WIFI MODE NULL WiFi.config WiFi.begin
WiFi.disconnect WiFi.status WiFi.macAddress WiFi.localIP WiFi.mode WiFi.setTxPower
WiFi.getTxPower WiFi.softAP WiFi.softAPIP WiFi.softAPBroadcastIP
WiFi.softAPNetworkID
WiFi.softAPConfig WiFi.softAPdisconnect WiFi.softAPgetStationNum WiFi-builtins

```

Wire

```

Wire.begin Wire.setClock Wire.getClock Wire.setTimeout Wire.getTimeout
Wire.beginTransmission Wire.endTransmission Wire.requestFrom Wire.write
Wire.available Wire.read Wire.peek Wire.flush Wire-builtins

```

xtensa

```

WUR, WSR, WITLB, WER, WDTLB, WAITI, SSXU, SSX, SSR, SSL, SSIU, SSI, SSAI,
SSA8L, SSA8B, SRLI, SRL, SRC, SRAI, SRA, SLLI, SLL, SICW, SICK, SEXT, SDCT,
RUR, RSR, RSIL, RFI, ROTW, RITLB1, RITLB0, RER, RDTLB1, RDTLB0, PITLB,
PDTLB, NSAU, NSA, MULA.DD.HH, MULA.DD.LH, MULA.DD.HL, MULA.DD.LL, MULS.DD
MULA.DA.HH, MULA.DA.LH, MULA.DA.HL, MULA.DA.LL, MULS.DA MULA.AD.HH, MULA.AD.LH,
MULA.AD.HL, MULA.AD.LL, MULS.AD MULA.AA.HH, MULA.AA.LH, MULA.AA.HL, MULA.AA.LL,
MULS.AA MULA.DD.HH.LDINC, MULA.DD.LH.LDINC, MULA.DD.HL.LDINC, MULA.DD.LL.LDINC,
MULA.DD.LDINC MULA.DD.HH.LDDEC, MULA.DD.LH.LDDEC, MULA.DD.HL.LDDEC,
MULA.DD.LL.LDDEC, MULA.DD.HH, MULA.DD.LH, MULA.DD.HL, MULA.DD.LL, MULA.DD
MULA.DA.HH.LDINC,
MULA.DA.LH.LDINC, MULA.DA.HL.LDINC, MULA.DA.LL.LDINC, MULA.DA.LDINC
MULA.DA.HH.LDDEC,
MULA.DA.LH.LDDEC, MULA.DA.HL.LDDEC, MULA.DA.LL.LDDEC, MULA.DA.LDDEC MULA.DA.HH,
MULA.DA.LH, MULA.DA.HL, MULA.DA.LL, MULA.DA MULA.AD.HH, MULA.AD.LH, MULA.AD.HL,
MULA.AD.LL, MULA.AD MULA.AA.HH, MULA.AA.LH, MULA.AA.HL, MULA.AA.LL, MULA.AA
MUL16U, MUL16S, MUL.DD.HH, MUL.DD.LH, MUL.DD.HL, MUL.DD.LL, MUL.DD MUL.DA.HH,
MUL.DA.LH, MUL.DA.HL, MUL.DA.LL, MUL.DA MUL.AD.HH, MUL.AD.LH, MUL.AD.HL,
MUL.AD.LL, MUL.AD MUL.AA.HH, MUL.AA.LH, MUL.AA.HL, MUL.AA.LL, MUL.AA
MOVSP, MOVT.S, MOVF.S, MOVGEZ.S, MOVLTZ.S, MOVNEZ.S, MOVEQZ.S, ULE.S, OLE.S,
ULT.S, OLT.S, UEQ.S, OEQ.S, UN.S, CMPSOP NEG.S, WFR, RFR, ABS.S, MOV.S

```

```

ALU2.S UTRUNC.S, UFLOAT.S, FLOAT.S, CEIL.S, FLOOR.S, TRUNC.S, ROUND.S,
MSUB.S, MADD.S, MUL.S, SUB.S, ADD.S, ALU.S MOVF, MOVGEZ, MOVLTZ, MOVNEZ,
MOVEQZ, MAXU, MINU, MAX, MIN, CONDOP MOV, LSXU, LSX, L32E, LICW, LICL,
LDCT, JX, IITLB, IDTLB, LSIU, LSI, LDINC, LDDEC, L32R, EXTUI, S32E, S32RI,
S32C1I, ADDMI, ADDI, L32AI, L16SI, S32I, S16I, S8I, L32I, L16UI, L8UI,
LDSTORE MOVI, IIU, IHU, IPFL, DIWBI, DIWB, DIU, DHU, DPFL, CACHING2 III,
IHI, IPF, DII, DHI, DHWBI, DHWB, DPFWO, DPFRD, DPFW, DPFR, CACHING1 CLAMPS,
BREAK, CALLX12, CALLX8, CALLX4, CALLX0, CALLXOP CALL12, CALL8, CALL4, CALL0,
CALLOP LOOPGTZ, LOOPNEZ, LOOP, BT, BF, BRANCH2b J, BGEUI, BGEI, BGEZ, BLTUI,
BLTI, BLTZ, BNEI, BNEZ, ENTRY, BEQI, BEQZ, BRANCH2e BRANCH2a BRANCH2 BBSI,
BBS, BNALL, BGEU, BGE, BNE, BANY, BBCI, BBC, BALL, BLTU, BLT, BEQ, BNONE,
BRANCH1 REMS, REMU, QUOS, QUOU, MULSH, MULUH, MULL, XORB, ORBC, ORB, ANDBC,
ANDB, ALU2 ALL8, ANY8, ALL4, ANY4, ANYALL SUBX8, SUBX4, SUBX2, SUB, ADDX8,
ADDX4, ADDX2, ADD, XOR, OR, AND, ALU XSR, ABS, NEG, RFDO, RFDD, SIMCALL,
SYSCALL, RFWU, RFWO, RFDE, RFUE, RFME, RFE, NOP, EXTW, MEMW, EXCW, DSYNC,
ESYNC, RSYNC, ISYNC, RETW, RET, ILL, ILL.N, NOP.N, RETW.N, RET.N, BREAK.N,
MOV.N, MOVI.N, BNEZ.N, BEQZ.N, ADDI.N, ADD.N, S32I.N, L32I.N, tttt t ssss
s rrrr r bbbb b y w iiiii i xxxx x sa sa. >sa entry12 entry12' entry12.
>entry12 coffset18 cofs cofs. >cofs offset18 offset12 offset8 ofs18 ofs12
ofs8 ofs18. ofs12. ofs8. >ofs sr imm16 imm8 imm4 im numeric register reg.
nop a15 a14 a13 a12 a11 a10 a9 a8 a7 a6 a5 a4 a3 a2 a1 a0

```

Appendix A – Registers summary

GPIO registers

Name	Description	Address	Access
GPIO_OUT_REG	GPIO 0-31 output register	\$3FF44004	R/W
GPIO_OUT_W1TS_REG	GPIO 0-31 output register_W1TS	\$3FF44008	WO
GPIO_OUT_W1TC_REG	GPIO 0-31 output register_W1TC	\$3FF4400C	WO
GPIO_OUT1_REG GPIO	GPIO 32-39 output register	\$3FF44010	R/W
GPIO_OUT1_W1TS_REG	GPIO 32-39 output bit set register	\$3FF44014	WO
GPIO_OUT1_W1TC_REG	GPIO 32-39 output bit clear register	\$3FF44018	WO
GPIO_ENABLE_REG	GPIO 0-31 output enable register	\$3FF44020	R/W
GPIO_ENABLE_W1TS_REG	GPIO 0-31 output enable register_W1TS	\$3FF44024	WO
GPIO_ENABLE_W1TC_REG	GPIO 0-31 output enable register_W1TC	\$3FF44028	WO
GPIO_ENABLE1_REG	GPIO 32-39 output enable register	\$3FF4402C	R/W
GPIO_ENABLE1_W1TS_REG	GPIO 32-39 output enable bit set register	\$3FF44030	WO
GPIO_ENABLE1_W1TC_REG	GPIO 32-39 output enable bit clear register	\$3FF44034	WO
GPIO_STRAP_REG	Bootstrap pin value register	\$3FF44038	RO
GPIO_IN_REG	GPIO 0-31 input register	\$3FF4403C	RO
GPIO_IN1_REG	GPIO 32-39 input register	\$3FF44040	RO
GPIO_STATUS_REG	GPIO 0-31 interrupt status register	\$3FF44044	R/W
GPIO_STATUS_W1TS_REG	GPIO 0-31 interrupt status register_W1TS	\$3FF44048	WO
GPIO_STATUS_W1TC_REG	GPIO 0-31 interrupt status register_W1TC	\$3FF4404C	WO
GPIO_STATUS1_REG	GPIO 32-39 interrupt status register1	\$3FF44050	R/W
GPIO_STATUS1_W1TS_REG	GPIO 32-39 interrupt status bit set register	\$3FF44054	WO
GPIO_STATUS1_W1TC_REG	GPIO 32-39 interrupt status bit clear register	\$3FF44058	WO
GPIO_ACPU_INT_REG	GPIO 0-31 APP_CPU interrupt status	\$3FF44060	RO
GPIO_ACPU_NMI_INT_REG	GPIO 0-31 APP_CPU non-maskable interrupt status	\$3FF44064	RO
GPIO_PCPU_INT_REG	GPIO 0-31 PRO_CPU interrupt status	\$3FF44068	RO
GPIO_PCPU_NMI_INT_REG	GPIO 0-31 PRO_CPU non-maskable interrupt status	\$3FF4406C	RO
GPIO_ACPU_INT1_REG	GPIO 32-39 APP_CPU interrupt status	\$3FF44074	RO
GPIO_ACPU_NMI_INT1_REG	GPIO 32-39 APP_CPU non-maskable interrupt status	\$3FF44078	RO
GPIO_PCPU_INT1_REG	GPIO 32-39 PRO_CPU interrupt status	\$3FF4407C	RO
GPIO_PCPU_NMI_INT1_REG	GPIO 32-39 PRO_CPU non-maskable interrupt status	\$3FF44080	RO
GPIO_PIN0_REG	Configuration for GPIO pin 0	\$3FF44088	R/W
GPIO_PIN1_REG	Configuration for GPIO pin 1	\$3FF4408C	R/W
GPIO_PIN2_REG	Configuration for GPIO pin 2	\$3FF44090	R/W
GPIO_PIN38_REG	Configuration for GPIO pin 38	\$3FF44120	R/W
GPIO_PIN39_REG	Configuration for GPIO pin 39	\$3FF44124	R/W
GPIO_FUNC0_IN_SEL_CFG_REG	Peripheral function 0 input selection register	\$3FF44130	R/W
GPIO_FUNC1_IN_SEL_CFG_REG	Peripheral function 1 input selection register	\$3FF44134	R/W
GPIO_FUNC254_IN_SEL_CFG_REG	Peripheral function 254 input selection register	\$3FF44528	R/W
GPIO_FUNC255_IN_SEL_CFG_REG	Peripheral function 255 input selection register	\$3FF4452C	R/W
GPIO_FUNC0_OUT_SEL_CFG_REG	Peripheral output selection for GPIO 0	\$3FF44530	R/W
GPIO_FUNC1_OUT_SEL_CFG_REG	Peripheral output selection for GPIO 1	\$3FF44534	R/W
GPIO_FUNC38_OUT_SEL_CFG_REG	Peripheral output selection for GPIO 38	\$3FF445C8	R/W

Name	Description	Address	Access
GPIO_FUNC39_OUT_SEL_CFG_REG	Peripheral output selection for GPIO 39	\$3FF445CC	R/W
IO_MUX_PIN_CTRL	Clock output configuration register	\$3FF49000	R/W
IO_MUX_GPIO36_REG	Configuration register for pad GPIO36	\$3FF49004	R/W
IO_MUX_GPIO37_REG	Configuration register for pad GPIO37	\$3FF49008	R/W
IO_MUX_GPIO38_REG	Configuration register for pad GPIO38	\$3FF4900C	R/W
IO_MUX_GPIO39_REG	Configuration register for pad GPIO39	\$3FF49010	R/W
IO_MUX_GPIO34_REG	Configuration register for pad GPIO34	\$3FF49014	R/W
IO_MUX_GPIO35_REG	Configuration register for pad GPIO35	\$3FF49018	R/W
IO_MUX_GPIO32_REG	Configuration register for pad GPIO32	\$3FF4901C	R/W
IO_MUX_GPIO33_REG	Configuration register for pad GPIO33	\$3FF49020	R/W
IO_MUX_GPIO25_REG	Configuration register for pad GPIO25	\$3FF49024	R/W
IO_MUX_GPIO26_REG	Configuration register for pad GPIO26	\$3FF49028	R/W
IO_MUX_GPIO27_REG	Configuration register for pad GPIO27	\$3FF4902C	R/W
IO_MUX_MTMS_REG	Configuration register for pad MTMS	\$3FF49030	R/W
IO_MUX_MTDI_REG	Configuration register for pad MTDI	\$3FF49034	R/W
IO_MUX_MTCK_REG	Configuration register for pad MTCK	\$3FF49038	R/W
IO_MUX_MTDO_REG	Configuration register for pad MTDO	\$3FF4903C	R/W
IO_MUX_GPIO2_REG	Configuration register for pad GPIO2	\$3FF49040	R/W
IO_MUX_GPIO0_REG	Configuration register for pad GPIO0	\$3FF49044	R/W
IO_MUX_GPIO4_REG	Configuration register for pad GPIO4	\$3FF49048	R/W
IO_MUX_GPIO16_REG	Configuration register for pad GPIO16	\$3FF4904C	R/W
IO_MUX_GPIO17_REG	Configuration register for pad GPIO17	\$3FF49050	R/W
IO_MUX_SD_DATA2_REG	Configuration register for pad SD_DATA2	\$3FF49054	R/W
IO_MUX_SD_DATA3_REG	Configuration register for pad SD_DATA3	\$3FF49058	R/W
IO_MUX_SD_CMD_REG	Configuration register for pad SD_CMD	\$3FF4905C	R/W
IO_MUX_SD_CLK_REG	Configuration register for pad SD_CLK	\$3FF49060	R/W
IO_MUX_SD_DATA0_REG	Configuration register for pad SD_DATA0	\$3FF49064	R/W
IO_MUX_SD_DATA1_REG	Configuration register for pad SD_DATA1	\$3FF49068	R/W
IO_MUX_GPIO5_REG	Configuration register for pad GPIO5	\$3FF4906C	R/W
IO_MUX_GPIO18_REG	Configuration register for pad GPIO18	\$3FF49070	R/W
IO_MUX_GPIO19_REG	Configuration register for pad GPIO19	\$3FF49074	R/W
IO_MUX_GPIO20_REG	Configuration register for pad GPIO20	\$3FF49078	R/W
IO_MUX_GPIO21_REG	Configuration register for pad GPIO21	\$3FF4907C	R/W
IO_MUX_GPIO22_REG	Configuration register for pad GPIO22	\$3FF49080	R/W
IO_MUX_U0RXD_REG	Configuration register for pad U0RXD	\$3FF49084	R/W
IO_MUX_U0TXD_REG	Configuration register for pad U0TXD	\$3FF49088	R/W
IO_MUX_GPIO23_REG	Configuration register for pad GPIO23	\$3FF4908C	R/W
IO_MUX_GPIO24_REG	Configuration register for pad GPIO24	\$3FF49090	R/W

GPIO configuration / data registers

RTCIO_RTC_GPIO_OUT_REG	RTC GPIO output register	0x3FF48400	R/W
RTCIO_RTC_GPIO_OUT_W1TS_REG	RTC GPIO output bit set register	0x3FF48404	WO
RTCIO_RTC_GPIO_OUT_W1TC_REG	RTC GPIO output bit clear register	0x3FF48408	WO
RTCIO_RTC_GPIO_ENABLE_REG	RTC GPIO output enable register	0x3FF4840C	R/W
RTCIO_RTC_GPIO_ENABLE_W1TS_REG	RTC GPIO output enable bit set register	0x3FF48410	WO
RTCIO_RTC_GPIO_ENABLE_W1TC_REG	RTC GPIO output enable bit clear register	0x3FF48414	WO
RTCIO_RTC_GPIO_STATUS_REG	RTC GPIO interrupt status register	0x3FF48418	WO
RTCIO_RTC_GPIO_STATUS_W1TS_REG	RTC GPIO interrupt status bit set register	0x3FF4841C	WO
RTCIO_RTC_GPIO_STATUS_W1TC_REG	RTC GPIO interrupt status bit clear register	0x3FF48420	WO
RTCIO_RTC_GPIO_IN_REG	RTC GPIO input register	0x3FF48424	RO
RTCIO_RTC_GPIO_PIN0_REG	RTC configuration for pin 0	0x3FF48428	R/W
RTCIO_RTC_GPIO_PIN1_REG	RTC configuration for pin 1	0x3FF4842C	R/W

Name	Description	Address	Access
RTCIO_RTC_GPIO_PIN2_REG	RTC configuration for pin 2	0x3FF48430	R/W
RTCIO_RTC_GPIO_PIN3_REG	RTC configuration for pin 3	0x3FF48434	R/W
RTCIO_RTC_GPIO_PIN4_REG	RTC configuration for pin 4	0x3FF48438	R/W
RTCIO_RTC_GPIO_PIN5_REG	RTC configuration for pin 5	0x3FF4843C	R/W
RTCIO_RTC_GPIO_PIN6_REG	RTC configuration for pin 6	0x3FF48440	R/W
RTCIO_RTC_GPIO_PIN7_REG	RTC configuration for pin 7	0x3FF48444	R/W
RTCIO_RTC_GPIO_PIN8_REG	RTC configuration for pin 8	0x3FF48448	R/W
RTCIO_RTC_GPIO_PIN9_REG	RTC configuration for pin 9	0x3FF4844C	R/W
RTCIO_RTC_GPIO_PIN10_REG	RTC configuration for pin 10	0x3FF48450	R/W
RTCIO_RTC_GPIO_PIN11_REG	RTC configuration for pin 11	0x3FF48454	R/W
RTCIO_RTC_GPIO_PIN12_REG	RTC configuration for pin 12	0x3FF48458	R/W
RTCIO_RTC_GPIO_PIN13_REG	RTC configuration for pin 13	0x3FF4845C	R/W
RTCIO_RTC_GPIO_PIN14_REG	RTC configuration for pin 14	0x3FF48460	R/W
RTCIO_RTC_GPIO_PIN15_REG	RTC configuration for pin 15	0x3FF48464	R/W
RTCIO_RTC_GPIO_PIN16_REG	RTC configuration for pin 16	0x3FF48468	R/W
RTCIO_RTC_GPIO_PIN17_REG	RTC configuration for pin 17	0x3FF4846C	R/W
RTCIO_DIG_PAD_HOLD_REG	RTC GPIO hold register	0x3FF48474	R/W
GPIO RTC function configuration registers			
RTCIO_HALL_SENS_REG	Hall sensor configuration	0x3FF48478	R/W
RTCIO_SENSOR_PADS_REG	Sensor pads configuration register	0x3FF4847C	R/W
RTCIO_ADC_PAD_REG	ADC configuration register	0x3FF48480	R/W
RTCIO_PAD_DAC1_REG	DAC1 configuration register	0x3FF48484	R/W
RTCIO_PAD_DAC2_REG	DAC2 configuration register	0x3FF48488	R/W
RTCIO_XTAL_32K_PAD_REG	32KHz crystal pads configuration register	0x3FF4848C	R/W
RTCIO_TOUCH_CFG_REG	Touch sensor configuration register	0x3FF48490	R/W
RTCIO_TOUCH_PAD0_REG	Touch pad configuration register	0x3FF48494	R/W
,,,	,,,		
RTCIO_TOUCH_PAD9_REG	Touch pad configuration register	0x3FF484B8	R/W
RTCIO_EXT_WAKEUP0_REG	External wake up configuration register	0x3FF484BC	R/W
RTCIO_XTL_EXT_CTR_REG	Crystal power down enable GPIO source	0x3FF484C0	R/W
RTCIO_SAR_I2C_IO_REG	RTC I2C pad selection	0x3FF484C4	R/W

Ressources

in English

- **ESP32forth** page maintained by Brad NELSON, the creator of ESP32forth. You will find all versions there (ESP32, Windows, Web, Linux...)

<https://esp32forth.appspot.com/ESP32forth.html>
- **ESP32forth** (eforth for ESP32) page maintained by Peter FORTH

<https://www.forth2020.org/esp32forth>

In french

- **ESP32 Forth** site in two languages (French, English) with lots of examples

<https://esp32.arduino-forth.com/>

GitHub

- **Ueforth** resources maintained by Brad NELSON. Contains all Forth and C language source files for ESP32forth
<https://github.com/flagxor/ueforth>
- **ESP32forth** source codes and documentation for ESP32forth. Resources maintained by Marc PETREMANN
<https://github.com/MPETREMANN11/ESP32forth>
- **ESP32forthStation** resources maintained by Ulrich HOFFMAN. Stand alone Forth computer with LillyGo TTGO VGA32 single board computer and ESP32forth.
<https://github.com/uho/ESP32forthStation>
- **ESP32Forth** resources maintained by F. J. RUSSO
<https://github.com/FJRusso53/ESP32Forth>
- **esp32forth-addons** resources maintained by Peter FORTH
<https://github.com/PeterForth/esp32forth-addons>
- **Esp32forth-org** Code repository for members of the Forth2020 and ESp32forth groups
<https://github.com/Esp32forth-org>

Index

ADC channels.....	183	fvariable.....	79	see.....	51
allot.....	58	GIT.....	137	Serial.....	325
analogRead.....	184	handleClient.....	297	server.....	116
and.....	41	hex.....	35	SF!.....	79
asm.....	246, 321	HEX.....	82	SF@.....	79
autoexec.fs.....	142	HOLD.....	83	shift.....	40
BASE.....	82	httpd.....	322	sockets.....	325
binary.....	35	include.....	125	spi.....	325
bluetooth.....	322	insides.....	322	SPIFFS.....	125, 325
breadboard.....	103	internals.....	322	streams.....	325
c!.....	57	interrupts.....	323	struct.....	73
c@.....	57	interval.....	168	structures.....	73, 325
code.....	241	is.....	95	tasks.....	325
commande AT.....		ledc.....	323	telnetd.....	116, 325
	273	list.....	122	Tera Term.....	109
constant.....	57	load.....	121	thru.....	121
create.....	58, 99	login.....	115	timers.....	325
decimal.....	35	m!.....	148, 259	type.....	44
DECIMAL.....	82	m@.....	151	u.....	38
default-use.....	119	ms-ticks.....	176	value.....	58
defer.....	95, 281	Netbeans.....	137	variable.....	57
defPin:.....	145	normal.....	60	visual.....	326
démarrage automatique.....		oled.....	192, 323	web-interface.....	326
	107	page.....	60	WiFi.....	326
DOES>.....	99	r@.....	57	wipe.....	120
dump.....	52	r>.....	57	Wire.....	326
editor.....	119, 121, 322	rdrop.....	57	xtensa.....	326
end-code.....	242	recognizers.....	324	xtensa-assembler.....	241
ESP.....	322	RECORDFILE.....	129, 141		245
esp_now_add_peer.....	309	registers.....	324	;	54
esp_now_init.....	308	rerun.....	168	:	54
esp_now_register_recv_cb..	308	ressources.....	330	:noname.....	97
esp_now_send.....	310	riscv.....	324	.s.....	
espnnow.....	322	rmt.....	324		52
fconstant.....	79	rtos.....	324	#.....	83
flush.....	121	S".....	86	#>.....	83
forget.....	54	save-buffers.....	122	#S.....	83
FORTH.....	320	SD.....	324	<#.....	83
Forth word.....	14	SD_MMC.....	324	>r.....	57